TCAD
Tutorial and Examples
Volume III

SILVACO International
4701 Patrick Henry Drive, Bldg. 1
Santa Clara, CA 94054
Telephone: (408) 567-1000
FAX: (408) 496-6080
E-Mail: support@silvaco.com
Internet: http://www.SILVACO.com

Edition Two
January 1999

TCAD
Tutorial and Examples Manual
Volume III

SILVACO International
4701 Patrick Henry Drive, Bldg. 1
Santa Clara, CA 95054

Phone:	(408) 567-1000
FAX:	(408) 496-6080
E-Mail:	support@silvaco.com
Internet:	http://www.SILVACO.com

Notice

The information contained in this document is subject to change without notice.

SILVACO International MAKES NO WARRANTY OF ANY KIND WITH REGARD TO THIS MATERIAL, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED WARRANTY OF FITNESS FOR A PARTICULAR PURPOSE.

SILVACO International Inc. shall not be liable for errors contained herein or for incidental or consequential damages in connection with the furnishing, performance, or use of this material.

SIMULATION STANDARD, TCADDRIVENCAD, VIRTUAL WAFER FAB, ANALOG ALLIANCE, LEGACY, ATHENA, ATLAS, FAST ATLAS, ODIN, VYPER, CRUSADE, RESILIENCE, DISCOVERY, CELEBRITY, PRODUCTION TOOLS, AUTOMATION TOOLS, INTERACTIVE TOOLS, TONYPLOT, DECKBUILD, DEVEDIT, INTERPRETER, ATHENA INTERPRETER, ATLAS INTERPRETER, CIRCUIT OPTIMIZER, MASKVIEWS, PSTATS, SSUPREM3, SSUPREM4, ELITE, OPTOLITH, FLASH, SILICIDES, SPDB, CMP, MC DEPOSIT, MC IMPLANT, PROCESS ADAPTIVE MESHING, S-PISCES, BLAZE, DEVICE 3D, INTERCONNECT3D, BLAZE3D, GIGA3D, MIZEDMODE3D, TFT, LUMINOUS, GIGA, MIZEDMODE, ESD, LASER, FASTBLAZE, FASTMIXEDMODE, FASTGIGA, FASTNOISE, MOCASIM, UTMOST, UTOMST II, UTMOST III, UTMOST IV, PROMOST, SPAYN, SMARTSPICE, MIXSIM, TWISTER, FASTSPICE, SMARTLIB, SDDL, EXACT, CLEVER, STELLAR, HIPEX, LISA, SCHOLAR, SIREN, ESCORT, STARLET, EXPERT, SAVAGE, SCOUT, GUARDIAN, and ENVOY are trademarks of SILVACO INTERNATIONAL.

All other trademarks mentioned in this manual are the property of their respective owners.

Reader Comment Sheet

We welcome your evaluation of this manual. Your comments and suggestions help us to improve our publications. If you have any responses to the questions below, please let us know. Write your observations down and send complaints, bug reports, suggestions or comments to the e-mail address listed below.

- Is this manual technically accurate?
- Are the concepts and wording easy to understand?
- Is the size of this manual convenient for you?
- Is the manual's arrangement convenient for you?
- Do you consider this manual to be easily readable?

Please add any additional relevant comments.

Please FAX your comments to

SILVACO International
Attention Technical Publications
1701 Patrick Henry Drive, Building 1
Santa Clara, CA 95054
at (408) 496-6080.
or
Send e-mail to us at: *support@silvaco.com*
or
Browse our Web Page at *http://www.SILVACO.com*

Intended Audience

The information in this manual is based on the following assumptions:

- The reader is familiar with the basic terminology of semiconductor processing and semiconductor device operation,

  and

- The reader understands the basic operations of the computer hardware and operation systems being used.

Introduction

This manual is intended as an additional guide to the use of Silvaco's process and device simulators. It contains descriptions of all the standard examples that demonstrate the use of SSUPREM3, ATHENA, ATLAS, and the VWF INTERACTIVE TOOLS manuals. Users should consult the relevant "User's Manual" for a full description of the models and syntax of each program.

Included on your distribution media are more than five hundred (500) Standard Examples that demonstrate the way that the simulators are used to model many different technologies. The examples are instructional and it is strongly recommended that new users apply these examples as a starting point for creating their own simulations. One of the first things you should learn is how to access, load, and run these examples.

Accessing the Examples

The examples are accessed from the menu system in DECKBUILD. To select and load an example:

1. Start DECKBUILD as described in the VWF INTERACTIVE TOOLS MANUAL.
2. Pull down the **MainControl** menu using the right hand mouse button. There are options on this menu for **MainControl**, **Optimizer**, **Examples**, **Help**, etc.
3. Select **Examples**. An index will appear in a **DeckBuild: Examples** window (see below). The examples are divided by technology or technology group. The most common technologies are clear (e.g., MOS, BJT) while others are grouped with similar devices (e.g., IGBT and LDMOS are under POWER, and solar cell and photodiode are under OPTOELECTRONICS).



The Examples Index in DeckBuild

4. Choose the technology you are interested in by double-clicking the left mouse button over that item in the examples index.
5. A list of examples for that technology will appear. These examples typically illustrate different devices, applications, or types of simulation.
6. Choose a particular example by double-clicking the left mouse button over that item in the list.
7. A text description of the example will appear in the window. This online text is the same as in this manual. It describes the important physyical mechanisms in the simulation, as well as giving details of the simulator syntax used. You should read this information before proceeding.
8. Press the **Load Example** button. The **Input Command** file for the example will be copied into your current working directory, together with any associated files. A copy of the command file will be loaded into DECKBUILD. (Note that the **Load Example** button remains faded until **Step 6** is performed correctly.
9. To run the example, press the **Run** button in the middle frame of the DECKBUILD application window.
10. Alternatively, most examples are supplied with results that can be copied into the current working directory, along with the input file. To view the results, select (highlight) the name of the **Results File** and select the DECKBUILD menu option, **Tools-Plot**. Details on the use of TONYPLOT can be found in the VWF INTERACTIVE TOOLS manual.

Examples Index:

23.1. ATHENA_IMPLANT: Implant Process Simulation

23.1.1. aniiex01.in: Comparison of Gaussian, Pearson and SVDP methods

Requires: SSUPREM4

This example compares implant analytical models: Gaussian (symmetrical) profile, single Pearson (amorphous implant), and SIMS Verified Dual Pearson (**SVDP**) method. The parameter `Print.mom` specifies that moments used in each simulation should be printed out. The `Moments` statement is used to select the tables.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.1: SVDP Model shows crystalline channeling tail when compared to single Pearson or Gaussian implant profiles

Input File athena_implant/aniiex01.in:

```
go athena

#TITLE: Comparison of Gauss, Pearson and SVDP method


line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing = 0.01
line y loc = 0.50  spacing = 0.01
```

```


#initialize the mesh
init silicon

#Gauss (symmetrical) implant (parameters are in std_tables)
moments std_tables
implant phos dose=1e14 energy=40 gauss

struct outf=aniiex01_0.str


line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing = 0.01
line y loc = 0.50  spacing = 0.01


#initialize the mesh
init silicon

#Use single Pearson (parameters are in std_tables)
moments std_tables
implant phos dose=1e14 energy=40 pearson print.mom

struct outf=aniiex01_1.str


line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing = 0.01
line y loc = 0.50  spacing = 0.01


#initialize the mesh
init silicon

#Use SVDP method (default)
moments svdp_tables
implant phos dose=1e14 energy=40 print.mom

struct outf=aniiex01_2.str

```

```
tonyplot -overlay aniiex01_*.str -set aniiex01.set

quit


```

23.1.2. aniiex02.in: Tilt Angle Dependence using the SVDP Model

Requires: SSUPREM4

This example shows how 35 keV boron implant profile depends on the tilt angle. SIMS Verified Dual Pearson (**SVDP**) method is used. Parameter, `Print.mom`, specifies that moments used in each simulation should be printed out. As can be seen, the boron distribution is very sensitive even for a small variation in the tilt angle.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.2: Tilt Angle dependence of SVDP Models. At zero degrees the channeling probability is extremely high

Input File athena_implant/aniiex02.in :

```
go athena
# Tilt angle dependence using SVDP model
line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 0.7   spac=0.01
init

```

```
implant boron energy=35 dose=1.e13 tilt=0 rotation=0 print.mom
struct outfile=aniiex02_00.str

line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 0.7   spac=0.01
init one.d
implant boron energy=35 dose=1.e13 tilt=1 rotation=0 print.mom
struct outfile=aniiex02_01.str

line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 0.7   spac=0.01
init
implant boron energy=35 dose=1.e13 tilt=2 rotation=0 print.mom
struct outfile=aniiex02_02.str

line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 0.7   spac=0.01
init
implant boron energy=35 dose=1.e13 tilt=7 rotation=0 print.mom
struct outfile=aniiex02_07.str

line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 0.7   spac=0.01
init
implant boron energy=35 dose=1.e13 tilt=10 rotation=0 print.mom
struct outfile=aniiex02_10.str
tonyplot -overlay  aniiex02_*.str -set aniiex02.set

```

23.1.3. aniiex03.in: Screen Oxide Thickness Dependence using the SVDP Model

Requires: SSUPREM4

This example shows how 35 keV boron implant profile depends on the thickness of the surface oxide. Oxide is grown in DRY ambient at a temperature of 900 C. Oxidation time varies from 0 to 160 minutes which results in oxide thicknesses between approx. 50 and 400 Angstroms. The values of oxide thickness are extracted and then substituted into the `s.oxide` parameter of the `implant` statement. The surface screen oxide partially randomizes the ion flux which leads to less channeling with increasing oxide thickness. The SIMS Verified Dual Pearson (**SVDP**) method allows to properly predict this effect. Parameter, `print.mom`, specifies that moments used in each simulation should be printed out.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.3: Effect of screen oxide thickness on implant profiles

Input File athena_implant/aniiex03.in :

```
go athena
# Oxide Thickness Dependence of B Implant Profiles
line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 0.6   spac=0.01
init
# Implant into "bare" silicon (native oxide = 0.001 micron)
implant boron energy=35 dose=1.e13 tilt=0 rotation=0 print.mom
#
extract name="SIMS" curve(depth,impurity="Boron" material="Silicon" \
        mat.occno=1 x.val=0.0) outfile="aniiex03_00.dat"

struct outfile=aniiex03_00.str

line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
```

```
line y loc = 0    spac=0.01
line y loc = 0.6   spac=0.01
init
diffuse time=8 temp=900 dry
#
extract name="tox8" thickness material="SiO~2" mat.occno=1 x.val=0.0
implant boron energy=35 dose=1.e13 tilt=0 rotation=0 \
        s.oxide=1.0e-04*$tox8 print.mom
extract name="SIMS" curve(depth,impurity="Boron" material="Silicon" \
        mat.occno=1 x.val=0.0) outfile="aniiex03_01.dat"


line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 0.7   spac=0.01
init
diffuse time=20 temp=900 dry
#
extract name="tox20" thickness material="SiO~2" mat.occno=1 x.val=0.0
implant boron energy=35 dose=1.e13 tilt=0 rotation=0 \
              s.oxide=1.0e-04*$tox20 print.mom
extract name="SIMS" curve(depth,impurity="Boron" material="Silicon" \
        mat.occno=1 x.val=0.0) outfile="aniiex03_02.dat"


line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 0.6   spac=0.01
init
diffuse time=60 temp=900 dry
#
extract name="tox60" thickness material="SiO~2" mat.occno=1 x.val=0.0
implant boron energy=35 dose=1.e13 tilt=0 rotation=0 \
              s.oxide=1.0e-04*$tox60 print.mom
extract name="SIMS" curve(depth,impurity="Boron" material="Silicon" \
        mat.occno=1 x.val=0.0) outfile="aniiex03_03.dat"


line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
```

```
line y loc = 0.6   spac=0.01
init
diffuse time=160 temp=900 dry
#
extract name="tox160" thickness material="SiO~2" mat.occno=1 x.val=0.0
implant boron energy=35 dose=1.e13 tilt=0 rotation=0 \
              s.oxide=1.0e-04*$tox160 print.mom
extract name="SIMS" curve(depth,impurity="Boron" material="Silicon" \
        mat.occno=1 x.val=0.0) outfile="aniiex03_04.dat"

tonyplot -overlay  aniiex03_*.dat  -set aniiex03.set

```

23.1.4. aniiex04.in: Experimental Verification of Channeling Profiles

Requires: SSUPREM4

This example compares zero-tilt boron implant profiles obtained using the SIMS Verified Dual Pearson (**SVDP**) method with experimental results (R.J. Shreutelkamp, et.al., Nucl.Instr. & Meth., B55, p. 615, 1991) saved in *.exp files.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.4: Comparison of SVDP profiles with measured data for channeling implants

Input File athena_implant/aniiex04.in:

```
go athena
# Comparison of Well-Channeled Profiles with Experiments
```

```
line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 1.   spac=0.01
init
implant boron energy=40 dose=1.e13 tilt=0 rotation=0 print.mom
struct outfile=athena_40.str
extract name="b1" curve(depth,impurity="Boron" material="Silicon" mat.oc-
  cno=1 \
        x.val=-10) outfile="aniiex04_1.dat"
line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 1.   spac=0.01
init one.d
implant boron energy=80 dose=1.e13 tilt=0 rotation=0 print.mom
struct outfile=athena_80.str
extract name="b2" curve(depth,impurity="Boron" material="Silicon" mat.oc-
  cno=1 \
        x.val=-10) outfile="aniiex04_2.dat"
tonyplot -overlay aniiex04_1.* aniiex04_2.* -set aniiex04.set
quit
```

23.1.5. aniiex05.in: Dose Dependent Channeling of Phosphorus

Requires: SSUPREM4

This example shows the dose dependence of zero-tilt P implant profiles obtained using the SIMS Verified Dual Pearson (**SVDP**) method and compares them with experimental results (R.J. Shreutelkamp, et.al., Nucl.Instr. & Meth., B55, p.615, 1991) saved in *.exp files. As the implant dose increases more damage is created which results in additional dechanneling of phosphorus ions. Therefore, the profile tail shortens with increasing dose.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.5: Comparison of SVDP profiles with measured data for dose dependent channeling

Input File athena_implant/aniiex05.in:

```
go athena
# Dose Dependence of P Implants
line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 1.2  spac=0.01
init
implant phos energy=100 dose=1.e13 tilt=0 rotation=0

#
extract name="p1" curve(depth,impurity="Phosphorus" material="Silicon" \
        mat.occno=1 x.val=-10) outfile="aniiex05_01.dat"

line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 1.2  spac=0.01
init
implant phos energy=100 dose=5.e13 tilt=0 rotation=0

#
```

```
extract name="p1" curve(depth,impurity="Phosphorus" material="Silicon" \
        mat.occno=1 x.val=-10) outfile="aniiex05_02.dat"


line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 1.2  spac=0.01
init
implant phos energy=100 dose=2.e14 tilt=0 rotation=0

#
extract name="p1" curve(depth,impurity="Phosphorus" material="Silicon" \
        mat.occno=1 x.val=-10) outfile="aniiex05_03.dat"


line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 1.2  spac=0.01
init
implant phos energy=100 dose=5.e14 tilt=0 rotation=0

#
extract name="p1" curve(depth,impurity="Phosphorus" material="Silicon" \
        mat.occno=1 x.val=-10) outfile="aniiex05_04.dat"


line x loc = 0.0  spac=0.1
line x loc = 1.0  spac=0.1
line y loc = 0    spac=0.01
line y loc = 1.2  spac=0.01
init
implant phos energy=100 dose=1.e15 tilt=0 rotation=0

#
extract name="p1" curve(depth,impurity="Phosphorus" material="Silicon" \
        mat.occno=1 x.val=-10) outfile="aniiex05_05.dat"
tonyplot -overlay aniiex05_*.dat aniiex05_*.exp -set aniiex05.set
quit
```

23.1.6. aniiex06.in: Retrograde Well Formation Using High Energy Implants

This example shows high energy implant capability. The look-up tables for main implants are extended up to 8 MeV which allows simulation of modern technologies involving MEV implants. In this example, the three microns deep retrograde well is formed using two P implants(2 MeV and 0.75 MeV) with subsequent 1 hour, 1100 C anneal.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.6: High Energy Implantation for Well Formation

Input File athena_implant/aniiex06.in:

```
go athena
# Retrograde Well Formation Using High Energy Phosphorus Implants
#the x dimension definition
line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
#the vertical definition
line y loc = 0     spacing = 0.02
line y loc = 5  spacing = 0.05
#initialize the mesh
init silicon c.boron=1e15
```

```
deposit oxide thick=0.01

#perform hogh energy phosphorus implants

implant phos dose=5e13 energy=2000 print.mom
implant phos dose=5e12 energy=750 print.mom

# subsequent anneal
diffuse time=60 temp=1100

struct outf=aniiex06_0.str

tonyplot aniiex06_0.str -set aniiex06.set
quit
```

23.1.7. aniiex07.in: User-Defined MOMENTs with Parameters from MC Simulation

Requires: SSUPREM4/MCIMPLANT

This example can be used as a template which allow to estimate implant moments from Monte Carlo simulation and then use these moments in subsequent analytical implants. This approach is useful for those ion/material/energy combinations which are not included into look-up tables.

In the first part of the example, the Monte Carlo method is used to calculate moments for 70 keV boron in titanium silicide. The parameter `print.mom` specified in the `implant` statement directs ATHENA to save the calculated moments in the standard structure file `aniiex07_01.str`.

In the second part of the example first the `moments` parameter for single Pearson distribution are extracted from the file and then the parameter substitution capability of DECKBUILD is used to path these parameters to the `moments` statement. After that, the moments set in the `moments` statement are used in the analytical implant. Parameter `ANY.PEARSON` is set to FALSE. This ensures that Pearson type-IV distribution will be used. Profiles obtained in Monte Carlo and analytical simulations are compared in TONYPLOT.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.7: Analytical Models with user-defined moments matches the original MC implant data. This technique allows users to quickly build analytical model tables for combinations of ions and materials not supported as defaults in ATHENA

Input File athena_implant/aniiex07.in:

```
go athena

# Using Moments Statement with Parameters Extracted from MC Simulation

#the x dimension definition
line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25

#the vertical definition
line y loc = 0     spacing = 0.01
line y loc = 0.4  spacing = 0.01


#initialize the mesh
set dose=1e13
set energy=70
set ion=boron
set mat=tisi

init $mat
```

```

# First use MC method and save the structure and moments in it
implant $ion dose=$dose energy=$energy monte n.ion=20000  print.mom
struct outf=aniiex07_01.str


#the x dimension definition
line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25

#the vertical definition
line y loc = 0     spacing = 0.01
line y loc = 0.4  spacing = 0.01


#initialize the mesh
init $mat
# Now extract the moments from the saved structure
extract init infile="aniiex07_01.str"
   extract name="rp" param="RP"
   extract name="drp" param="DRP"
   extract name="skew" param="SKEW"
   extract name="kurt" param="KURT"

# Use them in the moments statement
moments $mat i.$ion dose=$dose energy=$energy rp=$rp drp=$drp skewn=$skew
  kurtos=$kurt

# Now analytical implant could be used
implant $ion dose=$dose energy=$energy print.mom any.pearson=f
struct outf=aniiex07_02.str

tonyplot -overlay aniiex07_*.str -set aniiex07.set
quit


```

23.1.8. aniiex08.in: Phosphorus Implant through Thick Nitride Layer

Requires: SSUPREM4

This example compares Monte Carlo and analytical methods for implantation into a multilayered structure. It shows that analytical methods are reasonably accurate even for the case when the profile peak is close to interface between two materials with quite different stopping powers. In most cases the default `dose.match` method appears to be fairly adequate, but sometimes range scaling methods (`max.scale`) for example) may give slightly better profiles in the deeper layers.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.8: Comparison of different analytical techniques for multi-layer implants with MC results. When the range of an implant is close to a material boundary there can be a loss of integrated dopant with some methods

Input File athena_implant/aniiex08.in:

```
go athena
# Phosphorus Implant through Thick Nitride Layer
#the x dimension definition
line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 0.3  spacing = 0.005
init silicon
deposit nitride thick=0.08 div=20
struct outf=aniiex08.str
```

```

# Use Monte Carlo
implant phos energy=100 dose=1e14 amorph monte n.ion=40000
struct outf=aniiex08_01.str

init infile=aniiex08.str
# Use default DOSE.MATCH method
implant phos energy=100 dose=1e14 amorph
struct outf=aniiex08_02.str
#
init infile=aniiex08.str
# Use MAX.SCALE method
implant phos energy=100 dose=1e14 amorph max.scale
struct outf=aniiex08_03.str
tonyplot -overlay aniiex08_*.str -set aniiex08.set
quit
```

23.1.9. aniiex09.in: Lateral Implant Straggle

This example compares analytical methods for simulation of 2D implantation. It also shows how to use Monte Carlo simulation for estimation of spatial moments. In the first part, Monte Carlo simulation is used to save a structure with spatial moments. Only 20000 trajectories is enough to estimate the moments. In the second part, all eight spatial moments are extracted. Then, the default and `full.lat` model are used to simulate 2D implant into the 0.2 microns mask window. The default method uses only five spatial moments, and lateral distribution is gaussian with standard deviation independent from the vertical coordinate. `Full.lat` model uses eight spatial moments and lateral distribution is a symmetrical Pearson function with standard deviation varying as a parabolic function of the vertical coordinate. In general case lateral kurtosis should also depend on the vertical coordinate, however the simplified approach with constant kurtosis slightly smaller than the averaged fourth lateral moment KURTT is used. Obtained 2D profiles are overlaid in TONYPLOT.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.9: Demonstration of FULL.LAT for accurate lateral straggle from implants

Input File athena_implant/aniiex09.in:

```
go athena

# 2D Implant into Narrow Window

#the x dimension definition
line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25

#the vertical definition
line y loc = 0     spacing = 0.01
line y loc = 0.8  spacing = 0.01


#initialize the mesh
set dose=1e13
set energy=100
set ion=boron
set mat=silicon

init $mat

```

```
# First use MC method and save the structure and moments in it
implant $ion dose=$dose energy=$energy monte n.ion=20000 tilt=0 amorph
  print.mom
struct outf=aniiex09_01.str


#the x dimension definition
line x loc = 0.0   spacing=0.01
line x loc = 0.1  spacing=0.01
line x loc = 0.5  spacing=0.04

#the vertical definition
line y loc = 0     spacing = 0.01
line y loc = 0.8   spacing = 0.04


#initialize the mesh
init $mat c.boron=3e13
deposit barrier thick=0.05 div=2
etch barrier left p1.x=0.1
structure mirror left
struct outf=aniiex09_02.str
# Now extract the moments from the saved structure
extract init infile="aniiex09_01.str"
   extract name="rp" param="RP"
   extract name="drp" param="DRP"
   extract name="skew" param="SKEW"
   extract name="kurt" param="KURT"
   extract name="ldrp" param="LDRP"
   extract name="skewxy" param="SKEWXY"
   extract name="kurtxy" param="KURTXY"
   extract name="kurtt" param="KURTT"


# Use them in the moments statement
moments $mat i.$ion dose=$dose energy=$energy rp=$rp drp=$drp \
        skewn=$skew kurtos=$kurt ldrp=$ldrp

# default 2d implant

implant $ion dose=$dose energy=$energy any.pearson=f print.mom
struct outf=aniiex09_03.str
```

```
init infile=aniiex09_02.str

# Full.lat 2d implant
moments $mat i.$ion dose=$dose energy=$energy rp=$rp drp=$drp \
        skewn=$skew kurtos=$kurt ldrp=$ldrp \
        skewxy=$skewxy kurtxy=$kurtxy kurtt=$"kurtt"-0.2
implant $ion dose=$dose energy=$energy full.lat any.pearson=f print.mom

struct outf=aniiex09_04.str
tonyplot -overlay aniiex09_03.str aniiex09_04.str -set aniiex09.set

quit
```

23.1.10. aniiex10.in LDD Formation using High Tilt Angle Implant (LATID)

Requires: SSUPREM4

This example demonstrates the use of the analytical angled implant model. A MOS structure is created with angled full rotation LDD implants. Full rotation is used to provide symmetry of 2D profiles. LDD implant uses off-axes direction (48 degrees) to exclude very deep penetration along 45 degrees direction. A very detailed Monte Carlo simulations would be needed to obtain a reliable set of moments for 45 degrees implants. Source and drain As implant is done at zero degrees.

Note: The parameter, `Std_table,` is used in the moments statement. Otherwise, the SVDP model would overestimate the implant tail because it does not have surface oxide dependence for arsenic.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.10: A high angle implant is used to form an LDD MOSFET with no spacers

Input File athena_implant/aniiex10.in:

```
go athena

#LDD Formation using LATID Full Rotation Implant
#
line x loc = 0.0  spac=0.1
line x loc = 0.4  spac=0.02
line x loc = 0.7  spac=0.05
#
line y loc = 0    spac=0.02
line y loc = 0.50 spac=0.04
line y loc = 1.0  spac=0.1
#
#calculate the mesh
init c.boron=1.0e15
#
#grow gate oxide
diffuse temp=950 time=15 dry
#
#implant threshold adjust
```

```
implant boron dose=4e12 energy=20
#
#deposit gate poly and pattern
deposit poly thick=0.4 div=8
etch poly left p1.x=0.4
#
# remove excess grid in substrate
relax dir.y=f y.min=.4
#
#perform gate reoxidation
diffuse temp=950 time=10 dry
#
# mirror to form complete device
structure mirror right
# implant LDD using angled implant and full rotation
# 48 degrees is selected to avoid channelling
implant phos  dose=5e13 energy=80 tilt=48 fullrotat
#
# implant source/drain vertically
# Use standard tables because SVDP tables are available only for
# native oxide and would overestimate channelling for in this case
moments std_tab
implant arsen dose=5e15 energy=80 tilt=0
#
# activation diffusion
diffuse time=10 temp=900
#
#
structure outfile=aniiex10.str
tonyplot -st aniiex10.str -set aniiex10.set
quit
```

23.1.11. aniiex11.in: Visualization of Collision Cascades

Requires: SSUPREM4/MCIMPLANT

This example demonstrates a capability to visualize trajectories of implanted ions as well as of recoiled target atoms. Only 50 trajectories started in one point (see parameter `Impct.point`) are calculated. Each of them are stored in the special trajectory file (parameter `Traj.file`). The parameter, `damage` specifies that secondary (recoil) trajectories should be also calculated. The parameter, `REC.FR=1` specifies that all recoils should be followed. In order to visualize trajectories, select the LINE (the last icon in the `Display` menu) button. By default, only primary trajectories will be shown (red color). To visualize trajectories of recoils, select **Lines** in the **Display** menu and in the **Lines** menu, change the default color number five to any other number (e.g. 3).

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.11: Monte Carlo Ion Tracks. Tracks after collisions can be seen on-screen as different colors

Input File athena_implant/aniiex11.in:

```
go athena

#Visualization of Collision Cascade

#the x dimension definition
line x loc = 0.0   spacing=0.005
line x loc = 0.2  spacing=0.005

#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 0.20  spacing = 0.005


#initialize the mesh
init silicon two.d

implant phos dose=1e14 energy=50 monte n.ion=50 amorph impct.point=0.501 \
             damage tilt=0 rec.fr=1 traject=1 traj.file=aniiex11.str

tonyplot -st aniiex11.str
```

```
quit
```

23.1.12. aniiex12.in: Angled Monte Carlo Implant into a Trench

This example employs a Monte Carlo implant model to implant boron into the sidewall of a trench. The `Relax` statement is also used to adapt the mesh after trench etching so that a fine mesh can be created at the surface where it is needed to resolve the implantation.

For this example, ions are incident at an angle of 15 degrees from the direction of the y-axis. Thus, the ions are incident directly only on the right side of the trench and at the top exposed surface. The doping on the left side and bottom of the trench is due to reflected ions. There is a visible reduction in the depth of the profile about one micron from the bottom of the trench. This corresponds to the point in the trench where ions are directly incident above this point and shadowed below it. The first TONYPLOT shows the ion trajectories (including those of reflected ions). To facilitate the visulization, select the LINE (the last icon in the **Display** menu) button.

Following the implant calculation, a short diffusion is performed and the results are plotted using TONYPLOT**.** The output shows 1D profiles from the surface of the unetched silicon and also from the implanted sidewall of the trench.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.12: Monte Carlo ion tracks during an angled implant into a trench



Figure 23.13: Final as-implanted profile for the trench sidewall implant

Input File athena_implant/aniiex12.in:

```
go athena

#TITLE: Angled Monte Carlo Implant into a Trench
#
line x loc=-1    spac=0.25
line x loc=-0.8  spac=0.04
line x loc=-0.5  spac=0.04
line x loc=-0.4  spac=0.1
line x loc=0.00  spac=0.5
#
line y loc=0.00  spac=0.04
line y loc=0.20  spac=0.04
line y loc=0.30  spac=0.1
line y loc=4.50  spac=0.5
line y loc=5.00  spac=0.04
line y loc=5.20  spac=0.04
line y loc=6.00  spac=0.5
init c.boron=1.0e16 orientation=100
#
# Etch half of the trench
etch silicon start x=-0.50 y=0.00
etch cont x=-0.50 y=4.50
```

```
etch cont x=-0.48 y=4.90
etch cont x=-0.45 y=4.93
etch cont x=0.00 y=5.00
etch cont x=1 y=5.00
etch done x=1 y=0.00
#
#Perform a series of relax operations to loosen the mesh
relax y.min=.5   x.max=-.8
relax y.min=5.15
relax y.min=5.15
#
#mirror to form complete trench
structure mirror right
#
#Perform Monte Carlo implant with trajectory visualization
implant boron dose=1e16 ener=50 monte n.ion=3000 amorph tilt=15 \
         traject=5 traj.file=aniiex12_01.str
#Plot the structure with trajectories
tonyplot -st aniiex12_01.str -set aniiex12_01.set
#
# Diffuse the implant slightly
diffuse time=10 temp=900
#
structure outfile=aniiex12_02.str
tonyplot -st aniiex12_02.str -set aniiex12_02.set
#
quit
```

23.1.13. aniiex13.in: Comparison of Analytical and MC Implant Models

Requires: SSUPREM4/MCIMPLANT

This example shows that the channeling effect of ion implantation can be simulated by using the Monte Carlo crystal model. It compares boron implants at seven degrees obtained by amorphous and crystalline Monte Carlo models with amorphous and SVDP analytical models. Within the range of the SVDP model validity, the analytical approach is preferred. However outside this range the crystalline MC implant can be used for more accutate profiles.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.14: Crystalline and amorphous MC implant models are compared with the analytical equivalents. SVDP is a good match to the crystal MC implant not possible with previous analytical approaches

Input File athena_implant/aniiex13.in:

```
go athena

#TITLE: Crystal and Amorphous MC Implantation Models

#the x dimension definition
line x loc = 0.0
line x loc = 0.25  spacing=0.25

#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 0.25  spacing = 0.005

#initialize the mesh
init c.boron=1e14

# Amorphous model
implant boron energy=10 dose=1e13 monte n.ion=3000 tilt=0 rotation=0 crys-
  tal=f
structure outfile=aniiex13_1.str


```

```
#the x dimension definition
line x loc = 0.0
line x loc = 0.25  spacing=0.25

#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 0.25  spacing = 0.005

#initialize the mesh
init c.boron=1e14

# Crystal model
implant boron energy=10 dose=1e13 n.ion=3000 monte tilt=7 rotation=0
structure outfile=aniiex13_2.str

#the x dimension definition
line x loc = 0.0
line x loc = 0.25  spacing=0.25

#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 0.25  spacing = 0.005

#initialize the mesh
init c.boron=1e14

# Amorphous analytical model
implant boron energy=10 dose=1.0e13  amorph tilt=7  print.mom
structure outfile=aniiex13_3.str
#the x dimension definition
line x loc = 0.0
line x loc = 0.25  spacing=0.25

#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 0.25  spacing = 0.005

#initialize the mesh
init c.boron=1e14

# Crystalline analytical SVDP model
implant boron energy=10 dose=1e13 tilt=7 rotation=0
structure outfile=aniiex13_4.str
```

```
#plot the current structure along with previous structures
tonyplot  -overlay aniiex13_*.str -set aniiex13.set

quit
```

23.1.14. aniiex14.in : Comparison of Analytical and BCA Implant Models for Well-Channeled Boron Implantation.

Requires: SSUPREM4/MCIMPLANT

This example demonstrates that 0-tilt implantation can be simulated using Binary Collision Approximation (BCA) Model. In the BCA model the deflection of the moving particles is calculated in a strict binary way – between the moving ion and the closest atom in the lattice. The difference between the BCA model and earlier implemented Monte Carlo crystlline model is in much more accurate determination of the closest atom and more accurate calculation of impact dependent inelastic energy losses. Therefore, BCA is capable to much more accurately predict trajectories of well-channeled particles and therefore usually gives much better agreement with experiments in those cases where channeling process is dominant, e.g. for 0 degrees implants.

The simulated profile is compared with experiment (R.J. Shreutelkamp, et.al., Nucl.Instr. & Meth., B55, p.615, 1991) and SVDP simulation which is also based on SIMS experiments. It should be stressed here that experimental results for well channeled implants are quite sensitive to many factors including surface conditions (thickness and uniformity of oxide surface layer) and precision of the ion beam orientation and beamwidth. Even 0.5 degree deviation in these parameters could result in considerable changes of measured implanted profiles. Therefore, agreement between these simulations and experiment should be considered as quite reasonable.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.15: Binary Collision Approximation implant model is compared with experiment and SVDP. BCA is a good match to the experiment which was not possible with older Monte Carlo approaches.

Input Deck athena_implant/aniiex14.in :

```
go athena

#TITLE: BCA and SVDP Modelsmv b10100.dat  and Experiment

#the x dimension definition
line x loc = 0.0
line x loc = 0.25  spacing=0.25

#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 0.3  spacing = 0.005

#initialize the mesh
init silicon

#
deposit oxide thick=0.001 div=2
# First run BCA model
```

```
implant boron energy=10 dose=9.27e12 tilt=0  bca beamw=0.1 n.ion=5000
structure outfile=aniiex14_1.str remove.gas
#
extract name="aa" curve(depth,impurity="Boron" material="Silicon" \
        mat.occno=1 x.val=0) outfile="aniiex14_1.dat"
#the x dimension definition
line x loc = 0.0
line x loc = 0.25  spacing=0.25
#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 0.3  spacing = 0.005
#initialize the mesh
init silicon
#
deposit oxide thick=0.001 div=2
# Run SVDP for comparison
implant boron energy=10 dose=9.27e12 tilt=0  pears s.ox=0.001
structure outfile=aniiex14_2.str
#
extract name="aa" curve(depth,impurity="Boron" material="Silicon" \
        mat.occno=1 x.val=0) outfile="aniiex14_2.dat"
tonyplot -overlay aniiex14_1.dat aniiex14_2.dat aniiex14.exp -set
  aniiex14.set
```

23.1.15. aniiex15.in : Preamorphization Effect on Shallow Junction Profile.

Requires: SSUPREM4/MCIMPLANT

This example demonstrates that the BCA implant model can predict reduction of the channeling tail by using the preamorphization technique. Preamorphization of surface layers using Silicon ion bombardment is often used to avoid deep channeling tails when shallow symmetric doping profiles are required. BCA simulations of three separate 70 keV, 1e13 ion/cm^2, 0-tilt Arsenic implants are performed:

- without preamorphization
- with partial preamorphization (1e14 Si ions/cm^2)
- with full preamorphization (1e15 Si ions/cm^2).

The profile comparison shows that even partial preamorphization significantly reduces the channeling tail, while “full” preamorphization completely eliminates the tail. To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.16: Effect of Silicon pre-amorphising implant to reduce channeling in a low energy arsenic implant

23.1.16. aniiex16.in : Use of C-Interpreter for Control of Damage Models.

Requires: SSUPREM4/C-INTERPRETER

This example shows how `C-Interpreter` can be used for control of `PLUS.ONE` (or `UNIT.DAMAGE`) and <311> Cluster formation models. Both models estimates distributions of point defect and <311> clusters by scaling implant profile using factors `DAM.F` and `CLUSTER.F`, correspondingly. Also, in case of <311> user specifies the concentration limits within which the clusters can be formated.

These methods of damage model control are very limited and sometimes could not represent physical reality. For example, it is observed that clusters are not formed near silicon surface, but it cannot be specified within model limitations. The first part of this example and corresponding Figure on the left show that two <311> cluster bands are formed. This situation is even more pronounced in 2D since the <311> cluster band could propagate laterally and end up at the surface. It means for example that if the model is applied to S/D or LDD implant of MOS device the <311> would "appear" under the gate which is obviously wrong.

Use of `C-interpreter` allows to modify the models In this case the `C-interpreter` file aniiex16.in allows the cluster to be formated only at depths > 0.15 microns. Also, `C-interpreter` allows to modify PLUS.ONE model. In this case it makes the model to be applied to concentrations only above 3e15.

The `C-interpreter` file is specified by `DAMAGEMOD` parameter in the `Moments` statement.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 23.17: Boron, Interstitials, and <311> cluster profiles. Left figure: default models; right figure: with modifications using C-Interpreter.

Input Deck athena_implant/aniiex16.in :

```
go athena

#the x dimension definition
line x loc = 0.0  spacing=0.01
line x loc = 0.1  spacing=0.01

#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 1.0   spacing = 0.020


#initialize the mesh
init silicon

#deposit screen oxide
deposit   oxide thickness=0.005 div=5
```

```

#perform boron  implant with +1 and 311-cluster models

method  cluster.dam
cluster min.clust=1e16 max.clust=3e18 clust.fact=2.0 boron
moments ignore_mom
implant boron dose=1e14 energy=40  unit.dam dam.f=0.1

structure outfile=aniiex16_1.str

#the x dimension definition
line x loc = 0.0  spacing=0.01
line x loc = 0.1  spacing=0.01

#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 1.0   spacing = 0.020


#initialize the mesh
init silicon

#deposit screen oxide
deposit   oxide thickness=0.005 div=5

# The same implant with C-interpreter control of +1 and 311-cluster models
moments damagemod=aniiex16.lib
method  cluster.dam
implant boron dose=1e14 energy=40  unit.dam
structure outfile=aniiex16_2.str

tonyplot aniiex16_1.str aniiex16_2.str -set aniiex16.set
```

This page is intentionally left blank

24.1. ATHENA_DIFFUSION: Diffusion Process Simulation

24.1.1 andfex01.in: Boron Implant and Anneal

Requires: SSUPREM4

This example gives a simple demonstration of implantation and annealing. Boron is implanted and the profile diffused. Because the implant is not masked, ATHENA/SSUPREM4 remains in 1D mode.

For this anneal of a medium dose implant in an inert ambient the default `fermi` diffusion model is used.

After completing the diffusion calculation, TONYPLOT is used to plot a 1D cutline through the structure.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.1: Boron Implant and Anneal using FERMI Model

Input File athena_diffusion/andfex01.in:

```
go athena

#TITLE: Simple Boron Anneal

#the x dimension definition
line x loc = 0.0  spacing=0.1
line x loc = 0.1  spacing=0.1

```

```
#the vertical definition
line y loc = 0     spacing = 0.02
line y loc = 2.0   spacing = 0.20

#initialize the mesh
init silicon c.phos=1.0e14

#perform uniform boron implant
implant boron dose=1e13 energy=70

#perform diffusion
diffuse time=30 temperature=1000
#
extract name="xj" xj silicon mat.occno=1 x.val=0.0 junc.occno=1

#plot the final profile
tonyplot

#save the structure
structure outfile=andfex01.str

quit
```

24.1.2 andfex02.in: Oxidation Enhanced Diffusion of Boron

Requires: SSUPREM4

This example demonstrates the diffusion models for Oxidation Enhanced Diffusion **(OED)**. When diffusion is performed in an oxidizing ambient, point defects are injected into the silicon with a rate that is a function of the rate at which the silicon is oxidizing. This point defect injection gives rise to an increased diffusion that is commonly referred to as OED.

To model OED the `two.dim` diffusion model must be specified on the `method` statement. When this model is specified, the interstitial and vacancy concentration will be calculated along with the impurities. This model loosely couples the point defects, generated by the oxidizing Si-SiO2 interface, with the diffusing boron.

When modeling point defects, it is necessary to extend the substrate of the simulation space to provide an adequate sink for the point defects. Simulations show that a depth of 20 to 50 microns is required in most cases.

This example re-runs the simulation using the `fermi` model to compare results. The overlay plot at the end of the simulation demonstrates the extra oxidation enhancement to the diffusion depth.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.2: Effect of Oxidation Enhanced Diffusion with TWO.DIM Model compared to the default FERMI Model

Input File athena_diffusion/andfex02.in:

```
go athena

# OED of Boron

#the x dimension definition
line x loc = 0.0  spacing=0.1
line x loc = 0.1  spacing=0.1

#the vertical definition
line y loc = 0     spacing = 0.02
line y loc = 2.0   spacing = 0.20
line y loc = 25.0  spacing = 2.5

#initialize the mesh
init silicon c.boron=1.0e14

#perform uniform boron implant
implant boron dose=1e13 energy=70

#set diffusion model for OED
method two.dim
```

```
#perform diffusion
diffuse time=30 temperature=1000 dryo2
#
extract name="xj_two.dim" xj silicon mat.occno=1 x.val=0.0 junc.occno=1

#save the structure
structure outfile=andfex02_0.str

# repeat the simulation with default FERMI model
go athena

#TITLE: Simple Boron Anneal

#the x dimension definition
line x loc = 0.0  spacing=0.1
line x loc = 0.1  spacing=0.1

#the vertical definition
line y loc = 0     spacing = 0.02
line y loc = 2.0   spacing = 0.20
line y loc = 25.0  spacing = 2.5

#initialize the mesh
init silicon c.phos=1.0e14

#perform uniform boron implant
implant boron dose=1e13 energy=70

#select diffusion model
method fermi

#perform diffusion
diffuse time=30 temperature=1000 dryo2
#
extract name="xj_fermi" xj silicon mat.occno=1 x.val=0.0 junc.occno=1


#save the structure
structure outfile=andfex02_1.str

# compare diffusion models
tonyplot -overlay andfex02_0.str andfex02_1.str -set andfex02.set
```

24.1.3 andfex03.in: Damage Enhanced Diffusion of Arsenic

Requires: SSUPREM4

This example demonstrates the damage enhanced diffusion effect in a heavy arsenic implant typical of MOS source/drain or bipolar emitter processing.

Diffusion enhancement due to defects introduced during heavy implants are responsible for the **Transient Enhanced Diffusion (TED)** or **Rapid Thermal Annealing (RTA)** effect.

One of the key parameters in TED modeling using the `FULL.CPL` model is the amount of implant damage generated by the implant. This is controlled using the `unit.damage` model. The parameter, `dam.fact` scales the interstitial concentration relative to the doping profile.

A 15-second heat cycle at 1000 Celcius is performed and the resulting dopant profile saved.

When modeling point defects, it is necessary to extend the substrate of the simulation space to provide an adequate sink for the point defects. Simulations show that a depth of 20 to 50 microns is required in most cases.

This example re-runs the simulation using the `fermi` model to compare results. The overlay plot at the end of the simulation demonstrates the extra implant damage enhancement to the diffusion depth.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.3: Effect of Implant Damage Enhanced Diffusion with FULL.CPL Model compared to the default FERMI Model. Implant Damage is the major cause of TED effects seen with RTA.

Input File athena_diffusion/andfex03.in :

```
go athena

#the x dimension definition
line x loc = 0.0  spacing=0.1
```

```
line x loc = 0.1  spacing=0.1

#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 2.0   spacing = 0.20
line y loc = 25.0  spacing = 2.5

#initialize the mesh
init silicon c.boron=1.0e17

#deposit screen oxide
deposit oxide thickness=0.005 div=2

#perform arsenic implant with damage
implant arsenic dose=1.0e15 energy=40 tilt=7 unit.damage dam.factor=0.1

#set diffusion model for TED
method full.cpl

#perform diffusion
diffuse time=15/60 temperature=1000
#
extract name="xj_fullcpl" xj silicon mat.occno=1 x.val=0.0 junc.occno=1

#save the structure
structure outfile=andfex03_0.str

# repeat the simulation with FERMI model

#the x dimension definition
line x loc = 0.0  spacing=0.1
line x loc = 0.1  spacing=0.1

#the vertical definition
line y loc = 0     spacing = 0.005
line y loc = 2.0   spacing = 0.20
line y loc = 25.0  spacing = 2.5

#initialize the mesh
init silicon c.boron=1.0e17

#deposit screen oxide
deposit oxide thickness=0.005 div=2
```

```

#perform arsenic implant with damage
implant arsenic dose=1.0e15 energy=40 tilt=7 unit.damage dam.factor=0.1

#set default model
method fermi

#perform diffusion
diffuse time=15/60 temperature=1000
#
extract name="xj_fermi" xj silicon mat.occno=1 x.val=0.0 junc.occno=1

#save the structure
structure outfile=andfex03_1.str

# compare diffusion models
tonyplot -overlay andfex03_0.str andfex03_1.str -set andfex03.set
```

24.1.4 andfex04.in: Transient Enhanced Diffusion using <311> Clusters

Requires: SSUPREM4

This example demonstrates the damage enhanced diffusion effect in a phosphorus implant typical of MOS LDD processing.

Diffusion enhancement due to defects introduced during heavy implants are responsible for the **Transient Enhanced Diffusion (TED)** or **Rapid Thermal Annealing (RTA)** effect.

This example shows the Stanford 311 cluster diffusion model. To enable this model the syntax used is:

```
method full.cpl cluster.dam i.loop.sink high.conc.
```

This model is an extension to the fully coupled model that describes the dissolution of 311 interstitial clusters.

The implant damage is defined in the <311> model by using the `cluster` statement. The parameters, `min.cluster` and `max.cluster`, describe the doping levels at which clusters form.

Tuning of the TED effect is done using the release time of the clusters. The syntax is `cluster tau.311.0=<val> tau.311.E=<val>`. The default for these parameters are taken from the Stanford work.

When modeling point defects, it is necessary to extend the substrate of the simulation space to provide an adequate sink for the point defects. Simulations show that a depth of 20 to 50 microns is required in most cases.

The plot shows the evolution of the transient enhanced diffusion. At 800 degress, the transient effect lasts for about five minutes. The overlay plot shows how the diffusion in the first five minutes is greater the subsequent 55 minutes.

A second plot shows the location and concentration of the interstitial clusters and the free interstitials. As the clusters decay, they release free interstitials that provide the diffusion mechanism for the phosphorus.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.4: Time Evolution of Diffusion Profile. Enhanced Diffusion from Implant Damage lasts around five minutes at this temperature.



Figure 24.5: Diffusion and Recombination of Interstitials generated by Decaying Interstitial <311> Clusters

Input File athena_diffusion/andfex04.in:

1 `go athena`

```
line x loc=0.00 spac=0.10
line x loc=0.1 spac=0.10
#
line y loc=0.00 spac=0.01
line y loc = 0.50  spacing = 0.02
line y loc = 1.0   spacing = 0.25
line y loc=20 spac=5
#
init silicon c.boron=1.0e14 orientation=100

# set RTA models
method full.cpl cluster.dam i.loop.sink high.conc pdinit.time=1e-5

# set max doping level at which clusters form
# at higher doping levels substrate is amorphised so no defects.
cluster  min.cluster=1e15 max.cluster=1e19 clust.fac=1 phos
interstitial damalpha=1 silicon

# using moments statement to fit the implantation to the as-implanted curve.
# fit parameters were range, std.dev and dratio.
moments silicon i.phosphor dose=5.0e13 energy=100 range=0.165 std.dev=0.0998 \
gamma=3.561 kurtosis=49.2516 srange=0.1339 sstd.dev=0.0506 \
sgamma=0.115 skurtosis=3.04804 dratio=0.393

implant phosphor dose=5.0e13 energy=100 tilt=7    print.mom

struct outfile=andfex04.a.str


#
diffus time=15/60 temp=800 nitro press=1.00
struct outfile=andfex04.b.str

diffus time=45/60 temp=800  nitro press=1.00
struct outfile=andfex04.c.str

diffus time=4 temp=800 nitro press=1.00
struct outfile=andfex04.d.str

diffus time=55 temp=800 nitro press=1.00
struct outfile=andfex04.f.str
```

```
#

tonyplot -overlay andfex04*str -set andfex04_ov.set
tonyplot -overlay andfex04*str -set andfex04_i.set
```

24.1.5 andfex05.in: CNET Models: Phos. Predeposition & Percolation

Requires: SSUPREM4

ATHENA includes advanced diffusion models of CNET. This work under the direction of Dr Daniel Mathiot has resulted in several enhanced physical model options available to the user.

This example demonstrates the use of the percolation model.

Percolation is an effect seen at high concentrations, (Cperc=2.5e20 cm-3) where the mechanisms of diffusion change when considering vacancy interactions. Here dopant atoms in the fifth neighboring position of the atomic lattice interact with each other. In this case, the diffusity is affected as if an effective vacancy concentration is increased by the factor:

```
impurity silicon i.phos fperc.0=1.0 fperc.E=1.0
```

The concentration where this effect comes into play is controlled with

```
impurity silicon i.phos cperc=2.5e20
```

The two profiles, with and without this model switched on, are compared after a 900 degrees predep process step.

All of the models may be switched on with the command:

```
method cnet.model
```

Related parameters added to adjust above models are documented in the User's Manual.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.6: Phosphorus Predeposition Simulated using the CNET Model.



Figure 24.7: Lower dose Phosphorus Predeposition Simulated using the CNET Model.

Input File athena_diffusion/andfex05.in:

```
1   go athena
2   # test of ATHENA CNET models
```

```
#
# deep substrate to avoid boundary problems for I/V
line x loc=0.00 spac=0.1
line x loc=0.1 spac=0.1
#
line y loc=0.00 spac=0.02
line y loc=2 spac=0.05
line y loc=50 spac=10
#
init silicon orientation=100

# set diffusion model
method full.cpl cnet

# with CNET model user must source the cnetmod file
# usually found in <install>/lib/athena/<version>/common
source andfex05.cnetmod


diffus time=240 temp=900 nitro press=1.00 c.phos=3.3e20

struct outf=andfex05_1.str

extract name="profile1" curve(depth,log10(impurity="Phosphorus" \
        material="Silicon"  mat.occno=1 x.val=0.05)) out-
  file="andfex05_1.dat"

tonyplot -overlay andfex05_1.dat andfex05_1.exp

# repeat simulation with lower phosphorus dose

line x loc=0.00 spac=0.1
line x loc=0.1 spac=0.1
#
line y loc=0.00 spac=0.02
line y loc=2 spac=0.05
line y loc=50 spac=10
#
init silicon orientation=100

# set diffusion model
method full.cpl cnet

```

```
# with CNET model user must source the cnetmod file
# usually found in <install>/lib/athena/<version>/common
source andfex05.cnetmod


diffus time=240 temp=900 nitro press=1.00 c.phos=9.0e19

struct outf=andfex05_1.str

extract name="profile2" curve(depth,log10(impurity="Phosphorus" \
        material="Silicon" mat.occno=1 x.val=0.05)) out-
  file="andfex05_2.dat"

tonyplot -overlay andfex05_2.dat andfex05_2.exp

tonyplot -overlay andfex05_1.dat andfex05_1.exp andfex05_2.dat
  andfex05_2.exp -set andfex05.set
```

24.1.6 andfex06.in: Stanford High Dose Model

This example demonstrates the high dose extensions to the fully coupled diffusion model from Stanford.

This model results from the work of Scott Crowder and Peter Griffin at Stanford and represents an extended fully coupled diffusion model accounting for dopant-defect coupling, seen at high concentrations.

The model is involked by selecting the method flag: – `method high.conc.`

This model extends the previous fully coupled model (FULL.CPL) to include extra reactions terms related to the pairing of dopant and point defect states. Further, it includes extra reactions describing coupled recombination terms in both the bulk and at the surface.

The parameter, `ivfactor,` controls the relative rate of PD recombination based upon vacancy-dopant pairs recombining with free interstitials with respect to straight interstitial – vacancy recombination. This parameter may be physically considered as a capture cross section ratio.

Similarly, `iifactor` controls the relative rate of PD recombination based upon interstial-dopant pairs recombining with free vacancies with respect to straight interstitial – vacancy recombination. This parameter may be physically considered as a capture cross section ratio.

Also related only to the surface, `isurfact` is the ratio of the surface capture cross sections of V-PI surface recombination and I-V surface recombination.

This model is thought to be significant when dealing with unclustered concentrations above 1e19 cm^-3.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.8: Effect of HIGH.CONC Model on Diffusion of Dopants above 1.0e20

Input File athena_diffusion/andfex06.in:

```
go athena

#Stanford full coupled model for high concentration P diffusion

#the x dimension definition
line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25

#the vertical definition
line y loc = 0     spacing = 0.02
line y loc = 0.50  spacing = 0.02
line y loc = 2.0   spacing = 0.25
line y loc = 25.0  spacing = 2.5

#initialize the mesh
init silicon  c.phos=1.0e14

#set implant damage parameters
clust max.clust=1e19 clust.fact=1.0

#perform high dose phos implant
implant phos dose=5e15 energy=50 amorp
```

```
deposit oxide thick=0.01
struct outf=andfex06_0.str

# diffusion without the high concentration model switched on...
method full.cpl cluster.dam i.loop

trap silicon total=0 enable=f

diffuse time=30 temp=850

#save the structure
structure outfile=andfex06_1.str

# diffusion with the high concentration model switched on ....

init inf=andfex06_0.str

method full.cpl cluster.dam i.loop high.conc

trap silicon total=0 enable=f

#high concentration model parameters
interstitial iifactor=0.3 ivfactor=0.3
interstitial silicon phos neu.0=1.0e-10 neu.E=1.0

diffuse time=30 temp=850

#save the structure
structure outfile=andfex06_2.str

tonyplot -overlay andfex06_1.str andfex06_2.str -set andfex06.set
```

24.1.7 andfex07.in: Emitter Push Effect

Requires: SSUPREM4

This example demonstrates the use of the `FULL.CPL` model to simulate the emitter push effect on boron diffusion.

The emitter push effect is observed experimentally when a high concentration phosphorus doped emitter region is diffused within a boron doped base region. The boron profile beneath the emitter region is seen to diffuse more rapidly than the boron profile in the base region. This is the result of point defects that are coupled to the diffusing phosphorus that are swept outward into the substrate where they enhance the boron diffusion. Such effects were the driving force behind the development of the so-called fully coupled model in ATHENA/SSUPREM4**.** This model includes terms that describe the flux of point defects based on gradients in impurity profiles.

The fully coupled model is invoked for the co-diffusion of boron and phosphorus in this example by specifying the `full.cpl` parameter on the `method` statement. This model is the most time consuming of the ATHENA/SSUPREM4 diffusion models and should be used with caution. The fully coupled model, however, is the only model that can accurately simulate the emitter push effect.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.9: Diffusion of Boron Doping in the Base of this Bipolar Transistor is enhanced by the presence of a highly doped emitter region

Input File athena_diffusion/andfex07.in:

```
go athena

#TITLE: Emitter push effect example
#
line x loc=0.0   spac=0.2
line x loc=2.5   spac=0.8
line x loc=3.0   spac=0.2
```

```
#
line y loc=0.00 spac=0.04
line y loc=0.3  spac=0.06
line y loc=2.0  spac=0.8
line y loc=10.0 spac=2.0
#
init c.phos=1e15
#
implant boron dose=1e13 energy=40
#
deposit nitride thick=.2 div=4
#
etch right nitride p1.x=2.5
relax y.min=1.5
#
implant phosphor dose=1e16 energy=30
#
etch nitride all
#
method compress full.cpl
diffuse time=30 temp=1000
#
structure outfile=andfex07.str
#
tonyplot -st andfex07.str -set andfex07.set

quit
```

24.1.8 andfex08.in: Grain-based Polysilicon Diffusion Model

Requires: SSUPREM4

This example demonstrates the diffusion of impurities in polysilicon in both grain and grain boundary components. A comparison is run between the default diffusion model and the grain-based model.

The grain based model is enabled by the syntax: `method poly.diff`

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.10: The Grain-based Model for diffusion in Polysilicon (POLY.DIFF) correctly predicts enhanced diffusion through polysilicon layers for this emitter compared with the FERMI model

Input File athena_diffusion/andfex08.in:

```
go athena
#TITLE: arsenic poly diffusion
line x loc = 0.0  spacing=0.1
line x loc = 0.5  spacing=0.1
line x loc = 2.5  spacing=0.1
line x loc = 3.0  spacing=0.1
#the vertical definition
line y loc = 0     spacing = 0.02
line y loc = 0.20  spacing = 0.04
line y loc = 1.0   spacing = 0.5
#initialize the mesh
init silicon
deposit nitride thick=.5 div=10
etch nitride start x=.5  y=-10
```

```
etch nitride cont  x=.5  y=10
etch nitride cont  x=2.5 y=10
etch nitride done  x=2.5 y=-10

deposit poly thick=0.2 div=10

#perform uniform arsenic implant
implant arsenic dose=2e16 energy=20 tilt=10

structure outfile=andfex08_0.str

#perform diffusion using the default model
diffuse time=10.0 temperature=900

structure outfile=andfex08_1.str


line x loc = 0.0  spacing=0.1
line x loc = 0.5  spacing=0.1
line x loc = 2.5  spacing=0.1
line x loc = 3.0  spacing=0.1

#the vertical definition
line y loc = 0     spacing = 0.02
line y loc = 0.20  spacing = 0.04
line y loc = 1.0   spacing = 0.5

#initialize the mesh
init silicon

#arsenic poly diffusion coefficients
arsenic poly Dix.0=8.0 Dix.E=4.05
arsenic poly Dim.0=12.8 Dim.E=4.05
arsenic poly gb.dix.0=6.6e+2 gb.dix.E=3.44
arsenic poly gb.seg.0=1.0e+2 gb.seg.E=0.0
arsenic poly gb.tau=1.0e+2
material poly gb.vol.rat=0.1 gb.seg=2.64e15 density=2.328 grain.size=0.2
material poly gb.energ=1.0 gb.dix.0=1.0e-12 gb.dix.E=0.0

deposit nitride thick=.5 div=10
etch nitride start x=.5  y=-10
etch nitride cont  x=.5  y=10
etch nitride cont  x=2.5 y=10
```

```
etch nitride done  x=2.5 y=-10

method poly.diff
deposit poly thick=0.2 div=10

#perform uniform arsenic implant
implant arsenic dose=2e16 energy=20 tilt=10

#perform diffusion with the advanced polysilicon diffusion model
diffuse time=10.0 temperature=900

structure outfile=andfex08_2.str

tonyplot -st andfex08_*.str -set andfex08.set

quit

```

24.1.9 andfex09.in: Arsenic Activation

Requires: SSUPREM4

This simple example demonstrates the activation model for Arsenic. This model is based on clustering of arsenic dopant. The default model for activation of other species is based on solid solubility.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.11: Arsenic Activation at low and high temperatures

Input File athena_diffusion/andfex09.in:

```
go athena

# This Input File illustrates the activation model of ATHENA
line x loc=0.00 spac=0.10
line x loc=0.10 spac=0.10
#
line y loc=0.00 spac=0.01
line y loc=2    spac=0.2
#
init silicon orientation=100
implant arsenic dose=3.0e15 energy=40

diffus time=30 temp=800
struct outfile=andfex09_0.str

diffus time=10/60 temp=1000
struct outfile=andfex09_1.str
tonyplot -overlay -st andfex09_0.str andfex09_1.str -set andfex09.set
```

24.1.10 andfex10.in: Boron Clustering

Requires: SSUPREM4

This example demonstrates the Clustering model. Originally implemented for Arsenic, the model has been extended to Boron, Antimony and Phosphorus.

To switch on this model, use: `method cluster.s4`

This model will override the default solid solubility models. To unset this model, it is possible to specify `method cluster.s4 = false.` In this case, Arsenic will still use a clustering model, but all other dopants will obey the Solid Solubility model. cluster.s4 is false by default.

The model coefficients are controlled with the parameters:

```
impurity i.boron silicon Ctn.0=1e-24 Ctn.E=0.60
```

Refer to the manual section describing 'Impurity Activation and Clustering' for a description of the clustering model used.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.12: Boron Clustering Model predicts boron activation at high doping levels

Input File athena_diffusion/andfex10.in:

```
go athena
# Example of SSUPREM4 static clustering model
# Extended to include B, Ph and Sb. Originaly it included As only.

line x loc=0.00 spac=0.05
line x loc=0.1 spac=0.05
line y loc=0.00 spac=0.002
line y loc=1 spac=0.1
```

```

init silicon orientation=100 c.boron=1e16

# Implant BF2 with a high dose
implant bf2 energy=35 dose=2e15
struct outf=andfex10_0.str

#Firstly without the  clustering model switched on
method full.cpl compress
diffus time=15 temp=900
struct outf=andfex10_1.str

#Switch on the clustering model for all dopants
init inf=andfex10_0.str
method full.cpl compress cluster.s4

# Define the static clustering coefficients
impurity i.boron silicon Ctn.0=1e-24 Ctn.E=0.60

diffus time=15 temp=900
struct outf=andfex10_2.str


# Plot the profiles to compare...
tonyplot -overlay andfex10_1.str andfex10_2.str -set andfex10.set

```

24.1.11 andfex11.in: Phosphorus Predeposition using POCl

Requires: SSUPREM4

This example demonstrates simultaneous phosphorus predeposition and oxide growth. The oxygen content of the gas is specified by the parameters, `dryo2 press=0.8.` The phosphorus concentration is controlled by the `c.phos` parameter.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.13: Simulation of POCl allows simultaneous oxidation and phosphorus doping

Input File athena_diffusion/andfex11.in:

```
go athena

#TITLE: POCl example

line x loc=-1.00 spac=2.00
line x loc=1.00 spac=2.00
#
line y loc=0.00 spac=.01
line y loc=0.05 spac=.02
#
init orientation=100
#
deposit oxide thickness=0.02 div=3
deposit poly  thickness=0.2 div=20

method fermi compress grid.ox=.01
diff time=40 temp=950 dryo2 press=0.8 c.phos=1e23

structure outfile=andfex11.str
```

```
tonyplot andfex11.str

```

24.1.12 andfex12.in: Indium Implant and Anneal

Requires: SSUPREM4/MCIMPLANT

This example demonstrates the implantation and diffusion of Indium in Silicon. Models and coefficients for Indium are available in `fermi` and `CNET` diffusion models. This example uses the `fermi` model to anneal a medium dose implant at high temperature in an inert ambient.

For implantation the `monte` parameter is used to activate the Monte Carlo implantation model. Monte Carlo implants will generally be more accurate than the analytical tables for non-standard species. Analytical tables for Indium do exist.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.14: Indium Modeling. X is as-implanted, O is after anneal

Input File athena_diffusion/andfex12.in:

```
go athena

#TITLE: Indium Anneal

#the x dimension definition
line x loc = 0.0  spacing=0.1
line x loc = 0.1  spacing=0.1

#the vertical definition
```

```
line y loc = 0     spacing = 0.02
line y loc = 2.0   spacing = 0.20

#initialize the mesh
init silicon c.phos=1.0e14

#perform uniform boron implant
implant indium dose=1e13 energy=70 monte
structure outfile=andfex12_0.str

#perform diffusion
diffuse time=30 temperature=1000


extract name="xj" xj silicon mat.occno=1 x.val=0.0 junc.occno=1

#save the structure
structure outfile=andfex12_1.str

tonyplot -overlay andfex12_0.str andfex12_1.str -set andfex12.set

quit
```

24.1.13 andfex13.in: <311> Cluster RTA in a MOSFET

Requires: SSUPREM4

This example demonstrates the use of the new 311 TED model derived from the work of Peter Griffin at Stanford. This model employs a bulk injection of Point Defects from a source of as-implanted <311> Clusters. The clusters may be thought of a groups of interstitials that are released into the bulk during the initial diffusion cycle.

This model is applicable for medium dose situations in the 1e13 ~ 1e14 range.

The focus of the work is driven from the need to produce accurate LDD profiles.

The damage, in the form of Point Defects, <311> Clusters and Dislocation Loops are derived from the implanted dopant profile scaled and clipped to specific dopant concentration 'bands'.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 24.15: 0.4um MOSFET simulated using TED Models

Input File athena_diffusion/andfex13.in:

```
go athena

#
line x loc=-0.5 spac= 0.3
line x loc=0.00 spac=0.01
line x loc=0.2 spac=0.15
#
line y loc=0.00 spac=0.01
line y loc=0.2 spac=0.02
line y loc=20 spac=6

init silicon c.boron=5e16

# deposit gate oxide
deposit oxide thickness=0.010 div=3

#threshold adjust implant
implant bf2 dose=3e12 energy=45
```

```
#deposit gate
deposit poly thickness=0.2 div=6

# poly heat cycle
diff time=10 temp=900

etch poly left p1.x=0.0
structure mirror right
relax y.min=0.75
relax y.min=1.5 dir.y=f
relax y.min=3.0 dir.y=f

# set gate length through stretch command
set drawn_l=0.4
stretch length=$drawn_l poly snap div=15

# extract poly length
extract name="poly_len" thickness poly y.val=-0.1
extract name="poly_len_um" $poly_len*0.0001

#set cluster models implant damage
method full.cpl cluster.dam i.loop.sink high.conc
cluster min.clust=1e16 max.clust=1e19 clust.fact=1.0 arsenic
disloc  min.loop=1e15  max.loop=1e16  loop.fact=1    arsenic

#LDD implant
implant arsenic dose=1e14 energy=50 amorph

#spacer definition
deposit oxide thickness=0.1 div=5
etch oxide thickness=0.1

# source/drain implant
implant arsenic dose=1e15 energy=50 amorph

# source/drain anneal
diff time=30 temp=800

#this is an alternative command to create a diffusion movie
#diff time=30 temp=800 dump=1 dump.pref=andfex13m

etch below p1.y=2.0
structure outf=andfex13_0.str
```

```

# Extract the two sd junction depths....
extract name="n++ xj" xj silicon mat.occno=1 x.val=-0.5 junc.occno=1
extract name="lat1" xj silicon mat.occno=1 y.val=0.01 junc.occno=1
extract name="lat2" xj silicon mat.occno=1 y.val=0.01 junc.occno=2
extract name="met_leff" $lat2 - $lat1
extract name="peak_B" max.conc impurity="Active Boron" material="Silicon" \
        mat.occno=1 y.val=0.01
extract name="surf_B" surf.conc impurity="Active Boron" material="Sili-
  con" \
        mat.occno=1 x.val=$drawn_l/2

#extract VT at different locations in the channel
extract name="1dnvt_center" 1dvt ntype x.val=$drawn_l/2
extract name="1dnvt_edge"   1dvt ntype x.val=0.05

set xgate=$drawn_l/2.0
electrode name=gate x=$xgate
electrode name=substrate backside

tonyplot  andfex13_0.str -set andfex13.set

```

This page is intentionally left blank

25.1. ATHENA_OXIDATION: Oxidation Process Simulation

25.1.1. anoxex01.in: Evolution of a Bird's Beak in Recessed SILO

This example demonstrates the bird's beak profile evolution for a Recessed SILO process. A nitride mask is deposited and patterned. A recess into the silicon is then performed using a sequence of etch statements.

SSUPREM4's capability to dump a structure after a specified number of time steps and TONYPLOT's movie function are used to show the formation of the growing bird's beak. This is accomplished by specifying `dump` and `dump.prefix` parameters in the `diffuse` statement.

To watch the movie, first choose **Select all plots** from the **View** popup, and then select **Movie...** from the **Tools** popup. Evolution of the vertical oxidation velocities is also shown in the movie. This was achieved by saving a set file with the Y velocity contours. This file is called **anoxex02.set**.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.1: Final profile of local oxidation in a shallow recess using simple models

Input File athena_oxidation/anoxex01.in:

```
go athena

#TITLE: Oxide Profile Evolution Example

# Substrate mesh definition
```

```
line y loc=0    spac=0.05
line y loc=0.6  spac=0.2
line y loc=1

line x loc=-1   spac=0.2
line x loc=-0.2 spac=0.05
line x loc=0    spac=0.05
line x loc=1    spac=0.2

init orient=100

# Anisotropic silicon etch
etch    silicon left p1.x=-0.218 p1.y=0.3 p2.x=0 p2.y=0

# Pad oxide and nitride mask
deposit oxide thick=0.02
deposit nitride thick=0.1
etch    nitride left p1.x=0
etch    oxide   left p1.x=0

# Field oxidation with structure file output for movie
diffuse tim=90 tem=1000 weto2 dump=1 dump.prefix=anoxex01m

tonyplot -st anoxex01m*.str

structure outfile=anoxex01_0.str

quit
```

25.1.2. anoxex02.in Mixed Ambient Oxidation

This example demonstrates SSUPREM4 capability to perform oxidation in mixed gas ambients. The gas flows can be specified in relative units for each flow of oxygen, water, hydrogen, and nitrogen by the parameters, `F.O2`, `F.H2O`, `F.H2`, and `F.N2` on the `diffuse` statement.

For this example, the oxidation step is performed several times using SSUPREM4 internal looping capability to specify a different gas mixture each time the loop is executed.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.2: 1D results of oxide thickness as a function of gas flow rates defined in SSUPREM4

Input File athena_oxidation/anoxex02.in:

```
go athena

#TITLE: Mixed Ambient Oxidation Example

# This example demonstrates mixed ambient oxidation in 1D
#
foreach gas (2. to 8. step 2.)
#
line x loc=0.0 sp=1.0
line x loc=1.0 sp=1.0
line y loc=0.0 sp=0.05
line y loc=1.0 sp=0.05

initialize
#
diffuse time=60 temperature=1000 f.o2=gas f.h2=20.
structure outfile=anoxex02_gas.str
#
end

```

```
tonyplot -st anoxex02*.str

quit
```

25.1.3. anoxex03.in: Doping Dependent Oxidation Rate

Requires: SSUPREM4

This example shows the effect of dopant in silicon on oxidation rate. High levels of boron (>1e20) accelerate the oxidation rate on the substrate.

No special model commands are required to activate dopant dependent oxidation. The parameter, `grid.ox` , controls the gridding in the grown oxide.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.3: Enhanced Oxidation Rate due to heavy substrate doping

Input File athena_oxidation/anoxex03.in :

```
go athena

#TITLE: Doping Dependent Oxidation Example
#
# Create the initial mesh
line x loc=-0.30 spac=0.03
line x loc=0.30  spac=0.03
```

```
#
line y loc=0.00  spac=0.02
line y loc=0.50  spac=0.1
#
init orientation=100 c.boron=1e15
#
# Mask part of the surface with nitride
deposit nitride th=.3 div=4
#
etch right nitride p1.x=0.0
#
# Implant with very high dose to dope part of the silicon
implant boron dose=3e15 energy=20
#
etch nitride all
#
# Oxidize and observe thicker oxidation on doped material
# Specify grid.ox to enhance the mesh in the growing oxide
method fermi compress grid.ox=0.03
diff time=5 temp=1000 weto2
extract name="toxlow" thickness material="SiO~2" mat.occno=1 x.val=-0.25
extract name="toxhigh" thickness material="SiO~2" mat.occno=1 x.val=0.25
# save it and plot it
structure outfile=anoxex03.str
tonyplot anoxex03.str -set anoxex03.set
quit
```

25.1.4. anoxex04.in: Polysilicon Oxidation

This example demonstrates the simultaneous oxidation of polysilicon and silicon. This example also shows the lifting of polysilicon as a small bird's beak is formed.

The example begins by forming a structure that is a sandwich of polysilicon, oxide, and silicon with a recessed area in the exposed silicon. Prior to oxidization, the coefficients for polysilicon oxidation are modified using the `oxide` statement.

The structure is then oxidized to show the polysilicon oxidation model. The `dump` parameter of the diffuse statement causes SSUPREM4 to save a structure file after each time step. The `DUMP.PRE` parameter specifies that the name of these structure files will be of the form

`anoxex09mxxx.xxx.str`

where xxx.xxx is the current total time of the diffusion step. This allows the making of a movie using the MOVIE capability of TONYPLOT.

The second TONYPLOT statement loads intermediate files for a movie. To watch the movie, first select “Select all plots” from the “View” popup and then select “Movie..” from the “Tools” popup.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.4: Oxidation of both Silicon and Polysilicon. Oxidation rates between Si and poly are separately controlled in SSUPREM4

Input File athena_oxidation/anoxex04.in:

```
go athena

#TITLE: Polysilicon Oxidation Example

# Polysilicon oxidation and consumption can be observed in the example.

# Substrate mesh definition
line y loc=0  spac=0.05
line y loc=0.6 spac=0.2
line y loc=1
```

```

line x loc=-1   spac=0.2
line x loc=-0.2 spac=0.05
line x loc=0.2  spac=0.05
line x loc=1  spac=0.2

init orient=100

# Anisotropic silicon etch
etch    silicon left p1.x=-0.218 p1.y=0.3 p2.x=0 p2.y=0

# Pad oxide and polysilicon gate
deposit oxide thick=0.02
deposit poly thick=0.5 div=5
etch    poly left p1.x=-0.1 p1.y=0.51 p2.x=0.1 p2.y=-0.55
etch    oxide   left p1.x=0
#
# modify polysilicon oxidation rate relative to silicon
#
# The original coefficients (same as silicon)
#oxide poly wet par.l.0 =283.333     par.l.e =1.17
#oxide poly wet par.h.0 =7.00        par.h.e =0.78    p.break=950.
#oxide poly wet lin.l.0=3.45e4       lin.l.e=1.60
#oxide poly wet lin.h.0=2.95e6       lin.h.e=2.05     l.break=900.

# Modified
oxide poly wet par.l.0=600       par.l.e =1.17
oxide poly wet par.h.0=14.       par.h.e =0.78    p.break=950.
oxide poly wet lin.l.0=7e4       lin.l.e=1.60
oxide poly wet lin.h.0=6e6       lin.h.e=2.05     l.break=900.

# Field oxidation
diffuse tim=60 tem=1000 wet dump=3 dump.pre=anoxex04m
#
tonyplot -st anoxex04m*.str
# Save the structure
structure    outfile=anoxex04.str

quit
```

25.1.5. anoxex05.in: Poly Buffered LOCOS (PBL)

This example demonstrates the use of SSUPREM4 in modeling a Poly Buffer Locos (PBL) isolation structure.

The structure is formed by depositing layers to create a sandwich of silicon, oxide, polysilicon and nitride. The nitride is patterned to expose the area that will be oxidized. At this point, TONYPLOT is invoked to display the material structure prior to oxidation.

Next, the structure is oxidized in wet O2 for a time of 120 minutes at a temperature of 1000 degrees Celsius.

The oxidation completely consumes the polysilicon in the exposed portion of the simulation and lifts the remaining polysilicon and nitride.

The final structure is plotted in TONYPLOT**.**

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.5: Poly Buffered LOCOS shows simultaneous lifting and oxidation of the floating polysilicon region.

Input File athena_oxidation/anoxex05.in:

```
go athena

#TITLE: Poly Buffered LOCOS Example
#
line x loc=-1.5 spac=0.5
line x loc=-0.4 spac=0.03
line x loc=0.0  spac=0.03
line x loc=0.7  spac=0.3
#
line y loc=0
```

```
line y loc=0.2  spac=0.08
line y loc=1
initialize orient=100
oxide ori.dep=f
#
# Pad oxide and nitride mask
deposit oxide   thick=0.02   div=1
deposit poly    thick=0.1    div=4
deposit nitride thick=0.15   div=3
etch    nitride left p1.x=-0.4
#
tonyplot
#
# Field oxidation
method fermi compress
diffuse time=120 temp=1000 weto2
#
tonyplot
#
structure outfile=anoxex05.str

quit
```

25.1.6. anoxex06.in: Stress Dependent Oxidation

This example demonstrates the use of SSUPREM4 stress dependent Visco-Elastic Oxidation model. Visco-Elastic, Stress Dependent Models take longer to run than the default (`COMPRESS`) oxidation model and therefore should only be used when specifically looking at oxides grown at a temperature of greater than 965C (viscous flow temperature), and when the oxide profile is critical. Oxidations in this example are performed with and without the stress dependence. The results of oxidation using the two models are plotted together using TONYPLOT to compare the structures predicted by the two models.

To enable the stress dependent viscous model the syntax required is:

```
method viscous oxide stress.dep=t
```

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.6: Comparison of viscous and stress dependent oxidation models. Stress in the oxide reduces the oxidation rate in the bird's beak region.

Input File athena_oxidation/anoxex06.in:

```
go athena

#TITLE: Stress Dependent Oxidation Example

#  LOCOS cross section with stress effects

# Define a very simple mesh in the substrate
line x loc=-1 spac=0.2
line x loc=0 spac=0.05
line x loc=1 spac=0.2
line y loc=0
line y loc=1
init orient=100
#
# Pad oxide and nitride mask
deposit oxide thick=0.02  div=1
deposit nitride thick=0.15 div=1
etch nitride left p1.x=0
#
```

```
#set some material parameters and models for the oxidation
oxide stress.dep=f
material nitride visc.0=1.8e15
#
# Perform field oxidation
method   viscous
diffuse tim=90 tem=1000 wet
#
structure outfile=anoxex06_1.str
#
# Now perform the same calculation with stress dependence
line x loc=-1 spac=0.2
line x loc=0 spac=0.05
line x loc=1 spac=0.2
line y loc=0
line y loc=1
init orient=100
#
# Pad oxide and nitride mask
deposit oxide thick=0.02  div=1
deposit nitride thick=0.15 div=1
etch nitride left p1.x=0
#
#set some models and material parameters and models for the oxidation
oxide stress.dep=t
material nitride visc.0=1.8e15
method   viscous
#
# Perform field oxidation
diffuse tim=90 tem=1000 wet
structure outfile=anoxex06_2.str
#
tonyplot -st anoxex06_*.str -set anoxex06.set
#
quit
```

25.1.7. anoxex07.in: Effect of Young's Modulus on Bird's Beak

Requires: SSUPREM4

This example illustrates one method for tuning the shape of bird's beak structuresthe user defined material capability. This example also shows how user defined materials can be created in ATHENA.

The example creates a comparison between the default coefficients for nitride and a user defined material that is similar to nitride but with a modified Young's modulus.

The simulation begins with the initial mesh definition in silicon.

Following the inialize statement, a new material called `nitride1` is defined. A set of coefficient statements follows that make the coefficients for `nitride1` equivalent to the default coefficients for nitride.

Following the coefficient definition, a `material` statement redefines the value of Young's modulus for the new material. All other properties of the material `nitride1` will be the same as those of nitride.

The new material can now be deposited and patterned just like any other default material in ATHENA**.** A deposition and an etch of **nitride1** is performed.

Following the patterning of the new material, nitride is deposited and patterned so that the structure has the right side with a nitride barrier to oxidation and the left side with a barrier made of the user defined material `nitride1` that is equivalent to nitride except that it has a different Young's modulus.

The structure is then oxidized and the final structure is plotted. The final structure shows the results of the modified Young's modulus that produce a different shaped bird's beak on each side.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.7: **Demonstration of the effect of Young's Modulus of the nitride of the shape of the oxidation. A stiffer nitride leads to less nitride lifting.**

Input File athena_oxidation/anoxex07.in:

```
go athena

#TITLE: User defined material example

line x loc = 0.0 spacing=.05
line x loc = 2.0 spacing=.05

line y loc=0     spacing = 0.04
line y loc=0.8   spacing = 0.3

init

# define the parameters for the "new" nitride named nitride1
oxide    material=nitride1 wet diff.0=0.0 diff.e=0;
oxide    material=nitride1 dry diff.0=0.0 diff.e=0;
material    material=nitride1 visc.0=5e12   visc.E=0  visc.x=0.499
material    material=nitride1 Young.m=1.0e14  Poiss.r=0.3
antimony donor   material=nitride1Dix.0=0.0Dix.E=0.0
antimony donor   material=nitride1Dim.0=0.0Dim.E=0.0
arsenicdonor    material=nitride1Dix.0=0.0Dix.E=0.0
arsenicdonor    material=nitride1Dim.0=0.0Dim.E=0.0
boron acceptor   material=nitride1Dix.0=0.0Dix.E=0.0
boron acceptor   material=nitride1Dip.0=0.0Dip.E=0.0
phosphorus donor material=nitride1Dix.0=0.0Dix.E=0.0 \
     Dvx.0=0.0Dvx.E=0.0
phosphorus donor material=nitride1Dim.0=0.0Dim.E=0.0 \
Dvm.0=0.0Dvm.E=0.0
phosphorus donor material=nitride1 Dimm.0=0.0Dimm.E=0.0 \
Dvmm.0=0.0Dvmm.E=0.0
interst material=nitride1D.0=0.0 D.E=0.0 Kr.0=0.0 Kr.E=0.0 \
Cstar.0=1.0 Cstar.E=0.0
interst material=nitride1neu.0=1.0 neg.0=0.0 dneg.0=0.0 \
pos.0=0.0 dpos.0=0.0
interst material=nitride1neu.E=0.0 neg.E=0.0 dneg.E=0.0 \
pos.E=0.0 dpos.E=0.0
interst material=nitride1ktrap.0=0.0 ktrap.E=0.0
vacancy material=nitride1D.0=0.0 D.E=0.0 Kr.0=0.0 Kr.E=0.0 \
Cstar.0=1.0  Cstar.E=0.0
vacancy material=nitride1neu.0=1.0 neg.0=0.0 dneg.0=0.0 \
pos.0=0.0 dpos.0=0.0
vacancy material=nitride1neu.E=1.0 neg.E=0.0 dneg.E=0.0 \
pos.E=0.0 dpos.E=0.0
```

```
trap material=nitride1total=1.0 frac.0=0.5 frac.E=0.0
material  material=nitride1 Ni.0=1.0    Ni.Pow=0.0 \
Ni.E=0.0   eps=7.5

# modify the Youngs modulus for the user defined material
material   material=nitride1 Young.m=1.0e15  Poiss.r=0.3

deposit oxide thick=.01
deposit  material=nitride1 thick=0.15 div=2
etch     material=nitride1 right p1.x=0.5

deposit  nitride thick=0.15 div=2
etch     nitride left  p1.x=1.5

#the diffusion card
method fermi compress grid.ox=0.05
diffuse time=30 temp=1100 wet

#plot the resulting structure
tonyplot
#
#save the structure
structure outfile=anoxex07.str

quit
```

25.1.8. anoxex08.in: Effect of Nitride Viscosity on Bird's Beak

Requires: SSUPREM4

This example is a comparison of the effect on oxide growth when varying the nitride thickness. This effect requires the use of `method viscous`

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.8: Demonstration of the effect of nitride viscosity of local oxidation shapes. Nitride viscosity controls the lifting of the nitride mask

Input File athena_oxidation/anoxex08.in:

```
go athena

#TITLE: Viscosity Example
#
line y loc=0    spac=0.1
line y loc=2.0  spac=1.0
#
line x loc=1.0  spac=0.25
line x loc=2    spac=0.05
line x loc=2.4  spac=0.05
init   orient=100
#
deposit oxide   thick=0.020 div=2
deposit poly    thick=0.050 div=4

structure outfile=anoxex08.str

deposit nitride thick=0.200 div=4
#
etch nitride p1.x=2.0   right
#
```

```
method fermi viscous
diffuse time=60 temp=1000 dryo2
diffuse time=98 temp=1000 f.h2=9.0 f.o2=5.0
#
# save the structure
structure outfile=anoxex08_1.str
#
initialize infile=anoxex08.str

deposit nitride thick=0.20 div=4

# set higher nitride viscocity
material nitride visc.0=1.8e16

#
etch nitride p1.x=2.0  right
#
method fermi viscous
diffuse time=60 temp=1000 dryo2
diffuse time=98 temp=1000 f.h2=9.0 f.o2=5.0
#    #
# structure mirror left
#
# save the structure
structure outfile=anoxex08_2.str

tonyplot -overlay anoxex08_1.str anoxex08_2.str -set anoxex08.set
```

25.1.9. anoxex09.in: Trench Sidewall Oxidation with Orientation Dependence

This example shows that orientation of the wafer will affect the growth rate of an oxide on trench sidewalls. The orientation of the substrate is specified by `init orientation=xyz`. Xyz will be 100, 110 or 111. The rotation of the substrate is specified by the parameter, `rot.sub`

The method statement parameters, `grid.ox` and `gridinit.ox` are used to improve the grid within the grown oxide layer. The thicknesses of sidewall oxide are extracted for all cases.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.9: Orientation Dependence in oxidation. On-screen the green line shows the oxide shape in at a trench corner assuming correct orientation dependence. The Red Line shows the effect of neglecting the orientation dependent model

Input File athena_oxidation/anoxex09.in:

```
go athena

#TITLE: Orientation dependent Oxidation Example

line y loc=0   spac=0.1
line y loc=4   spac=0.1

line x loc=-1 spac=0.1
line x loc=1   spac=0.1
#
#initialize with orientation of sidewalls along 100 direction
init orient=100 rot.sub=0

method gridinit.ox=0.02 grid.ox=0.02

# Anisotropic silicon etch
etch silicon right p1.x=0.0 p1.y=2.0 p2.x=0.0 p2.y=0.0

# Field oxidation
diffuse time=20 tem=1000 wet
```

```
#
#
extract name="toxx" thickness oxide mat.occno=1 x.val=0.50
extract name="toxy" thickness oxide mat.occno=1 y.val=1.05

# Save the structure
structure outfile=anoxex09_1.str


line y loc=0  spac=0.1
line y loc=4  spac=0.1

line x loc=-1 spac=0.1
line x loc=1  spac=0.1
#
#initialize with orientation of sidewalls along 110 direction
init orient=100 rot.sub=45

# Anisotropic silicon etch
etch silicon right p1.x=0.0 p1.y=2.0 p2.x=0.0 p2.y=0.0

# Field oxidation
diffuse time=20 tem=1000 wet


extract name="toxx" thickness oxide mat.occno=1 x.val=0.50
extract name="toxy" thickness oxide mat.occno=1 y.val=1.05

# Save the structure
structure outfile=anoxex09_2.str
#
#plot each of the saved structures
tonyplot -overlay anoxex09_1.str anoxex09_2.str -set anoxex09.set
```

25.1.10. anoxex10.in: Void Formation During Trench Oxidation

This example demonstrates SSUPREM4's ability to oxidize when colliding oxide fronts cause a void. A narrow trench is oxidized until the opposing oxidation fronts touch. During oxidation, SSUPREM4 saves structure after each time step. These saved structures, as well as the final structure, are used to monitor the moving boundary progress.

To watch the movie, first select **Select all plots** from the **View** popup and then select **Movie...** from the **Tools** popup.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.10: Voids can be formed during oxidation in SSUPREM4

Input File athena_oxidation/anoxex10.in:

```
go athena

#TITLE: Oxidation Overlap in Narrow Trench Example
#
# This example simulates oxidation in a narrow trench
#
line x loc=0.0  spac=0.3
line x loc=1.0  spac=0.05
#
line y loc=0.0  spac=0.05
line y loc=0.7  spac=0.25
line y loc=1.4  spac=0.05
line y loc=2    spac=0.25
initialize
#
etch silicon start x=0.95 y=0.0
etch continue x=1.0 y=0.0
etch continue x=1.0 y=1.5
etch done x=0.8 y=1.5
```

```
#
structure mirror right
#
# Specifying fill on method statement causes voids in oxide to be filled
method compress fermi
diffuse time=8 temp=1150 wet dump=1 dump.prefix=anoxex10m
structure outfile=anoxex10.str
#
tonyplot -st anoxex10m*.str anoxex10.str

```

25.1.11. anoxex11.in: SWAMI Isolation Using Stress Dependent Oxidation

This example demonstrates the use of the stress dependent oxidation model in **SSUPREM4** to simulate an isolation oxidation using a SWAMI process.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.11: SWAMI Oxidation Process using a nitride mask and spacer. Contours of stress within the oxide are shown



Figure 25.12: SWAMI Oxidation Process using a nitride mask and spacer

Input File athena_oxidation/anoxex11.in:

```
go athena

#TITLE: SWAMI Example

line y loc=0    spac=0.06
line y loc=0.3    spac=0.06
line y loc=0.6  spac=0.25
line y loc=1

line x loc=-1   spac=0.2
line x loc=-0.5 spac=0.06
line x loc=-0.2 spac=0.06
line x loc=0    spac=0.05
line x loc=1    spac=0.2
init orient=100

# Anisotropic silicon etch
deposit oxide thick=0.04
etch oxide p1.x=0.1 left
etch    silicon left p1.x=-0.2 p1.y=0.3 p2.x=0 p2.y=0

```

```
diffuse time=4 temp=1000 wet

# Pad oxide and nitride mask
deposit nitride thick=0.16 div=4
# deposit nitride thick=0.15 div=3

etch    nitride left p1.x=0
deposit nitride thick=0.04
etch    nitride   left p1.x=-0.5
etch    oxide     left p1.x=-0.5

tonyplot

# Field oxidation
oxide stress.dep=t
method viscous oxide.rel=1e-2
diffuse tim=500 tem=1000 weto2

structure   outfile=anoxex11.str

tonyplot  -st anoxex11.str -set anoxex11.set

quit
```

25.1.12. anoxex12.in: Controlling Encroachment using Oxidation Threshold

This example specifies that oxidation only occurs when oxidant concentration is above some certain value.

Two oxidation threshold values are used in this example along with a run using the default model. The threshold model is enabled by: `method ox.thresh`

Parameters for the model are set by material and separately for wet and dry oxidations using the statement:

```
oxide  min.oxidant=<value>
```

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 25.13: Demonstration of controlling oxide encroachment using various values of minimum oxidant concentration

Input File athena_oxidation/anoxex12.in:

```
go athena

line x loc=-0.5000  spacing=0.100
line x loc=0.0000  spacing=0.0400
line x loc=0.1000  spacing=0.0150
line x loc=0.2500  spacing=0.0200
line x loc=0.5500  spacing=0.0400

line y loc=0.0000  spacing=0.0100
line y loc=0.5000  spacing=0.100


init no.imp

deposit   oxide thickness=0.005 divisions=3

deposit   poly   thickness=0.2 temp=550  divisions=6

etch poly right p1.x=0.1000

etch dry oxide thickness=0.0020

```

```
structure outf=anoxex12_0.str

# oxidation threshold model. This will stop oxidation happening
# when the value of weto2 drops below 1e19 at the oxidising interface
method ox.thresh

set min.oxi=1e19
oxide poly wet min.oxidant=$min.oxi
oxide silicon wet min.oxidant=$min.oxi


# grow the oxide that will encroach under the gate....
diffuse   time=20  temp=900  wet

structure outfile=anoxex12_1.str

# Extract some design parameters for calibration...
extract name="tox_at_0.05um_min.oxid_$min.oxi" thickness \
material="SiO~2" mat.occno=2 x.val=0.05

extract name="tox_at_0.um_min.oxid_$min.oxi" thickness \
material="SiO~2" mat.occno=2 x.val=0.0
#
extract name="weto2_conc_0.0_min.oxid_$min.oxi" max.conc \
impurity="Wet O~2" material="SiO~2" mat.occno=2 x.val=0.0

#-----------------------------------------------------------------------

init inf=anoxex12_0.str
# redefine oxidant threshold
set min.oxi=1e18
oxide poly wet min.oxidant=$min.oxi
oxide silicon wet min.oxidant=$min.oxi


# grow the oxide that will encroach under the gate....
diffuse   time=20  temp=900  wet

structure outfile=anoxex12_2.str

# Extract some design parameters for calibration...
extract name="tox_at_0.05um_min.oxid_$min.oxi" thickness \
material="SiO~2" mat.occno=2 x.val=0.05
```

```
extract name="tox_at_0.um_min.oxid_$min.oxi" thickness \
material="SiO~2" mat.occno=2 x.val=0.0
#
extract name="weto2_conc_0.0_min.oxid_$min.oxi" max.conc \
impurity="Wet O~2" material="SiO~2" mat.occno=2 x.val=0.0


#-----------------------------------------------------------------------

init inf=anoxex12_0.str
method ox.thresh=f

# grow the oxide that will encroach under the gate....
diffuse   time=20  temp=900  wet

structure outfile=anoxex12_3.str

# Extract some design parameters for calibration...
extract name="tox_at_0.05um_min.oxid_def" thickness \
material="SiO~2" mat.occno=2 x.val=0.05

extract name="tox_at_0.um_min.oxid_def" thickness \
material="SiO~2" mat.occno=2 x.val=0.0
#
extract name="weto2_conc_0.0_min.oxid_def" max.conc \
impurity="Wet O~2" material="SiO~2" mat.occno=2 x.val=0.0

# now plot the 3 structures and overlay them....
tonyplot -overlay anoxex12_[1-3].str -set anoxex12.set
```

This page is intentionally left blank

26.1. ATHENA_MISC: Miscellaneous Process Simulation Examples

26.1.1. anmiex01.in: Multiple Impurities in a Material (BPSG)

This example demonstrates the deposition of material such as BPSG that contains multiple impurities in ATHENA.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 26.1: Out diffusion of boron and phosphorus from BPSG into undoped oxide

Input File athena_misc/anmiex01.in:

```
go athena

#TITLE: BPSG example

line x loc = 0.0 spacing=.05
line x loc = 2.0 spacing=.05

line y loc=0     spacing = 0.01
line y loc=1.0   spacing = 0.1

init    c.boron=1e15

# define the parameters for BPSG
```

```
impurity i.antimony   donor    material=BPSG Dix.0=1.31e16 Dix.E=8.75
impurity i.antimony   donor    material=BPSG Dim.0=0.0 Dim.E=0.0
impurity i.arsenic    donor    material=BPSG Dix.0=1.75 Dix.E=4.89
impurity i.arsenic    donor    material=BPSG Dim.0=0.0 Dim.E=0.0
impurity i.boron      acceptor material=BPSG Dix.0=3.16e-4 Dix.E=3.53
impurity i.boron      acceptor material=BPSG /oxide Seg.0=1126.0
  Seg.E=0.91 Trn.0=1.66e-7 Trn.E=0.0
impurity i.boron      acceptor material=BPSG Dip.0=0.0 Dip.E=0.0
impurity i.phosphor   donor    material=BPSG /oxide Seg.0=30.0 Seg.E=0.0
  Trn.0=1.66e-7 Trn.E=0.0
impurity i.phosphor   donor    material=BPSG Dix.0=7.6e-3 Dix.E=3.5
  Dvx.0=0.0 Dvx.E=0.0
impurity i.phosphor   donor    material=BPSG Dim.0=0.0 Dim.E=0.0 Dvm.0=0.0
  Dvm.E=0.0
impurity i.phosphor   donor    material=BPSG Dimm.0=0.0 Dimm.E=0.0
  Dvmm.0=0.0 Dvmm.E=0.0

deposit oxide thick=0.1 c.phos=1e12 c.bor=1e12 div=20
deposit  material=BPSG thick=0.5 div=20 c.phos=2e21 c.boron=5e21

#perform diffusion
method fermi compress
diffuse time=150 temp=1050

#plot the resulting structure
tonyplot
structure outfile=anmiex01.str

quit
```

26.1.2. anmiex02.in : Epitaxial Growth

This example makes use of the `epitaxy` statement to create an epitaxial layer with a specific mesh density and boron concentration.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 26.2: Grid control during EPITAXY simulations

Input File athena_misc/anmiex02.in:

```
go athena

#TITLE: Epitaxy Example

line x loc=0.0   sp=0.1
line x loc=1.0   sp=0.1

line y loc=0.    sp=0.04
line y loc=1.0   sp=0.1

init c.boron=5e12

epitaxy boron conc=5.e16 thick=8 time=100 temp=927 dy=.5 ydy=4. div=30

structure outfile=anmiex02.str
tonyplot

quit
```

26.1.3. anmiex03.in: Phosphorus Segregation Modeling

Requires: SSUPREM4

This example shows the use of improved discretization to improve spurious phosphorus segregation through gate oxides.

The problem is clearly seen in the left plot. The original gridding has no mesh points wholly within the oxide layer. It results in too much diffusion through the oxide. The only way to decrease the transport is to increase the number of grid points in the thin oxide.

To increase mesh in thermally grown oxides, the parameter, `method grid.ox`, is used. The default Lowther discretization method means the diffusion into silicon is almost completely eliminated (right plot).

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 26.3: Importance of using grid points within a MOS gate oxide to avoid diffusion from poly gate to substrate

Input File athena_misc/anmiex03.in:

```
go athena

# TITLE: Use improved gridding to avoid excess segregation through gate
  oxide
#
line x loc=0.00 spac=0.10
line x loc=0.1 spac=0.10
#
line y loc=0.00 spac=0.01
line y loc=1.0 spac=0.1
#
init orientation=100
```

```
#
method fermi compress
diffus time=20 temp=900 dryo2 press=1.00 hcl.pc=0
#
deposit poly thick=0.50 c.phosphor=1.0e19 divisions=10

#
method fermi compress
diffus time=20 temp=900 nitro press=1.00
#
structure outfile=anmiex03_1.str

#
line x loc=0.00 spac=0.10
line x loc=0.1 spac=0.10
#
line y loc=0.00 spac=0.02
line y loc=1.0 spac=0.1
#
init orientation=100
#
method fermi compress grid.ox=.003 gridinit.ox=.003
diffus time=20 temp=900 dryo2 press=1.00 hcl.pc=0
#
extract name="tox" thickness oxide mat.occno=1 x.val=0.0
#
deposit poly thick=0.50 c.phosphor=1.0e19 divisions=10
#
method fermi compress
diffus time=20 temp=900 nitro press=1.00
#
structure outfile=anmiex03_2.str

tonyplot  -st anmiex03_[1-2].str -set anmiex03.set

```

26.1.4. anmiex04.in: Substrate Shear Stress

Requires: SSUPREM4

This example demonstrates the capability to calculate shear stress as a result of an overlying nitride layer. The results of the calculation are plotted using TONYPLOT to display contours of each of the three components of stress produced by the calculation.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 26.4: Demonstration of the Thermal Stress Model in ATHENA

Input File athena_misc/anmiex04.in:

```
go athena

#TITLE: Substrate Shear Stress Example
#
#  This example shows a calculation of shear stress
#  for nitride deposited on silicon.
#initialize the simulation regime with varying width
line x loc=-2.5
line x loc=-2 spac=0.3
line x loc= 0 spac=0.1
line x loc= 2 spac=0.3
line x loc=2.5

line y loc=0 spac=0.1
line y loc=2 spac=0.3
line y loc=5

initial orient=111

```

```
# deposit and pattern nitride stripe
deposit    nitride thick=0.1 div=2
etch       nitride left      p1.x=0

# set the value of the intrinsic stress...
material   nitride intrin.sig=1.4e10

# perform the stress analysis
stress  temp1=1000 temp2=1000
structure outfile=anmiex04.str
tonyplot -st anmiex04.str -set anmiex04.set

quit
```

26.1.5. anmiex05.in: MOSFET Scaling with the STRETCH command

Requires: ATHENA

The command,

```
stretch poly length=<val>
```

will stretch the defined material to the indicated final length, while maintaining the absolute coordinate system of the left of the structure, thus making it simpler to predict the absolute position of a future stretched region. eg. it makes placing the position for an electrode's name simpler.

This can be used in short channel effect scaling experiments under the VWF AUTOMATION TOOLS manual.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 26.5: Using the STRETCH feature to scale MOS gate Length

Input File athena_misc/anmiex05.in:

```
go athena

#
line x loc=0.0 spac=0.1
line x loc=0.2 spac=0.006
line x loc=0.4 spac=0.006
line x loc=0.5 spac=0.01
#
line y loc=0.0 spac=0.002
line y loc=0.2 spac=0.005
line y loc=0.5 spac=0.05
line y loc=0.8 spac=0.15
#
init orientation=100 space.mul=3

deposit oxide thickness=0.01 div=2

depo poly thick=0.2 divi=10

etch poly left p1.x=0.35
```

```

structure mirror right

# stretch the structure to the right maintaining the previous coordinate system
# this new command makes it simpler to predict a stretched regions coordinates
# for electrode name placement

structure outf=anmiex05_0.str

stretch poly div=10 length=1.0

structure outf=anmiex05_1.str

tonyplot -st anmiex05_[0-1].str
```

26.1.6. anmiex06.in: Salicide Example Using Titanium Silicide

Requires: SSUPREM4/SILICIDES

This example uses models a self aligned silicide (silicide) process. The example is run without impurities for speed.

No special commands are required to enable silicidation. Any interface between a refractory metal and silicon or polysilicon will cause growth of a silicide material.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 26.6: Self-Aligned Silicide Process on a MOS transistor

Input File athena_misc/anmiex06.in:

```
go athena

#TITLE: Salicide Example

line x loc=0.0 spac=0.2
line x loc=0.9 spac=0.03
line x loc=1.4 spac=0.02
line x loc=2.0 spac=0.2

line y loc=0 spac=0.03
line y loc=1.0 spac=0.5

#10ohm-cm 100 n-type substrate
init c.phos=4.5e13 no.imp space.m=1

deposit oxide thick=0.01 div=1
deposit poly thick=0.25 div=10
etch poly left p1.x=1.4
etch oxide left p1.x=1.4
implant phos dose=5e13 energy=35
deposit oxide thick=.6 div=15
```

```
etch oxide thick=.6
implant arsenic dose=5e15 energy=50

deposit titanium thick=0.1 div=8

tonyplot

meth compress fermi grid.sil=0.02
diffuse tim=5 tem=700

tonyplot

etch titanium all

structure outfile=anmiex06.str

tonyplot

quit
```

26.1.7. anmiex07.in: N+/P+ Poly Strapping with Ti and W Silicide

Requires: SSUPREM4/SILICIDES

This example shows the dopant diffusion strapping for Ti and W silicides. Fast dopant diffusion in silicides is a primary concern for developing a successful dual poly (N+/P+) process.

Boron diffuses fast in WSi but forms immobile TiB complexes in TiSi.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 26.7: Comparison of TiSi and WSi for strapping N+/P+ polysilicon. The dopants from the polysilicon generally diffuse rapidly in silicide materials.

Input File athena_misc/anmiex07.in:

```
go athena

#TITLE: Dopant diffusion strapping effect for Ti and W silicides

line x loc=-5.0 spac=0.1
line x loc=-3.5 spac=0.1
line x loc=0.0  spac=0.1
line x loc=3.5  spac=0.1
line x loc=5.0  spac=0.1
line y loc=0    spac=0.05
line y loc=1.   spac=0.5
init c.boron=4.5e13 space.m=3
#method grid.ox=.05
method grid.ox=.1

deposit oxide thick=0.015 div=1
deposit nitride thick=0.2 div=1
etch start nitride x=-3.5 y=-10
```

```
etch cont x=-3.5 y=10
etch cont x=3.5 y=10
etch done x=3.5 y=-10
diff time=100 temp=1000 wet
etch nitride all
deposit poly thick=0.25 div=4

deposit barrier thick=.1
etch barrier left p1.x=0.0
implant boron dose=1e15 energy=35
etch barrier all

deposit barrier thick=.1
etch barrier right p1.x=0.0
implant arsen dose=1e15 energy=60
etch barrier all

structure outfile=anmiex07_0.str

deposit titanium thick=0.05 div=4

meth compress grid.sil=0.02 lift.poly=f
diffuse tim=1 tem=700

etch titanium all
structure outfile=anmiex07_1.str

diff time=20 temp=800

structure outfile=anmiex07_2.str

init infile=anmiex07_0.str

deposit tungsten thick=0.05 div=4

meth compress grid.sil=0.02 lift.poly=f
diffuse tim=1 tem=700

etch tungsten all
structure outfile=anmiex07_3.str

diff time=20 temp=800

```

```
structure outfile=anmiex07_4.str

tonyplot -st anmiex07_*.str -set anmiex07.set




quit

```

26.1.8. anmiex08.in: Adjusting the Silicide Growth Rate

Requires: SSUPREM4/SILICIDES

This example demonstrates the silicide model applied to a one-dimensional structure including doping. The comments in the input file give tuning hints for adjusting the model.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 26.8: Demonstration of the DSV.0 parameter to adjust the silicide growth rate

Input File athena_misc/anmiex08.in:

```
go athena

#TITLE: 1D Silicide Example
```

```

line x loc=0.0 spac=0.5
line x loc=0.5 spac=0.5

line y loc=0 spac=0.01
line y loc=0.5 spac=0.05

init boron conc=6.5e14 space.m=1

implant arsenic dose=1e15 energy=30
diff temp=900 time=90

deposit titanium thick=0.1 div=8

structure outfile=anmiex08_0.str

# Use Ksurf.0 to modify reaction rate on the metal side
# Increase Ksurf.0 to reduce reaction rate on the metal side
interst tisix /titanium Ksurf.0=.0002 Ksurf.E=0.06

# Use ks.0 to modify reaction rate on the silicon side
# reaction is directly proportional to ks.0
silicide tisix /silicon ks.0=2.9e-07 ks.E=0.06

# use seg.0 to control snow plough effect, value for this parameter un-
  certain
arsenic silicon /tisix Seg.0=1.0 Seg.E=0.0 Trn.0=1.66e-7 Trn.E=0.0

meth compress fermi grid.sil=0.1
diffuse time=.75 tem=700
#

structure outfile=anmiex08_1.str
tonyplot -st anmiex08_*.str

quit
```

26.1.9. anmiex09.in: Mesh Control at Si/SiO2 Interface

Requires: SSUPREM4/PROCESS_ADAPTMESH

This example demonstrates the ability to control the mesh at or around the Si/SiO2 interface. A MOS gate is oxidized to create a non-planar structure. After this grid lines at a fixed spacing form the Si/SiO2 interface are added. A fine grid at the interface is key for many mobility model in device simulation. This input file is run in no impurity mode for speed.

First the:

```
method grid.oxide=0.025
```

command is used, this will limit the size of the mesh, added to the oxide, as the oxide evolves down into the Silicon. After oxide, this value is reverted back the the default 0.1, so as to not effect subsequent oxidations.

This example demonstrates the addition of mesh points, mid-simulation, adjacent to the Si/SiO2 interface, in the Silicon bulk. This function is useful for adding mesh lines at the time you would like to study dopant segregation accurately or at the point you would like to interface from Athena to ATLAS.

The command:

```
adapt.mesh add.i.line=0.002 silicon /oxide
```

will add a mesh point, parallel to the Si/SiO2 interface and in the Silicon, a distance of 2nm (20A) from the interface. It will only add a mesh line if there are no existing lines within the defined distance to the interface. If several lines are required, they should be added in a sequence that orders the addition in steps of decreasing distance to the interface.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 26.9: Addition of surface grid lines at the end of a process simulation. This is important for subsequent device simulations.

Input File athena_misc/anmiex09.in:

```
go athena
#
line x loc=0 spac=0.05
line x loc=0.3 spac=0.02
line x loc=0.6 spac=0.05
```

```
#
line y loc=0.00 spac=0.01
line y loc=0.8 spac=0.1
#
init orientation=100 space.mul=1 no.imp
deposit oxide thickness=0.01 div=2
deposit poly thickness=0.2 div=8
etch poly left p1.x=0.3
diffuse time=20 temp=850 weto2
structure outf=anmiex09_0.str
# Add extra mesh lines, gradually comming CLOSER to the interface
# with every addition....
adapt.mesh add.i.line=0.02 silicon /oxide
adapt.mesh add.i.line=0.015 silicon /oxide
adapt.mesh add.i.line=0.01 silicon /oxide
adapt.mesh add.i.line=0.005 silicon /oxide
adapt.mesh add.i.line=0.002 silicon /oxide
adapt.mesh add.i.line=0.001 silicon /oxide
# Now save the structure with the extra interface mesh lines added
structure outf=anmiex09_1.str
tonyplot anmiex09_0.str anmiex09_1.str -set anmiex09.set
```

This page is intentionally left blank

27.1. ATHENA_ELITE: 2-D Topography Simulation

27.1.1. anelex01.in: Deposit Machine Comparison

This example shows the use of six different deposit machines: unidirectional, planetary, dual directional, hemispherical, CVD and conical. The resulting structures are plotted for comparison.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.1: Comparison of ELITE Deposition Machine Models

Input File athena_elite/anelex01.in:

```
go athena

# Deposit machine comparison
#
line x loc=0.00 spac=0.20
line x loc=1.1 spac=0.05
line x loc=1.9 spac=0.05
line x loc=3.0 spac=0.20
#
line y loc=0.00 spac=0.05
line y loc=1.0  spac=0.5
```

```
#
initialize
#
deposit nitride      thick=.4 divis=10
#
etch nitride         start x=1.1 y=-10
etch                 cont  x=1.1 y=10
etch                 cont  x=1.9 y=10
etch                 done  x=1.9 y=-10
#
#
structure outfile=anelex01_0.str

init infile=anelex01_0.str

rate.depo machine=uni oxide a.m  sigma.dep=0.20 \
        uni dep.rate=1000 angle1=45.0
#
deposit machine=uni time=1 minute divis=5
structure outfile=anelex01_1.str

init infile=anelex01_0.str
rate.depo machine=planet1 oxide a.m sigma.dep=0.20 \
      planetar dep.rate=1000 angle1=45.0 angle2=60.0 \
      angle3=6.0 c.axis=20.0 p.axis=8.75
deposit machine=planet1 time=1 minute divis=5
structure outfile=anelex01_2.str
#
init infile=anelex01_0.str
rate.depo machine=dual oxide a.m  sigma.dep=0.20 \
        dualdirec dep.rate=1000 angle1=45.00 angle2=-45.00
deposit machine=dual time=1 minute divis=5
structure outfile=anelex01_3.str
#
init infile=anelex01_0.str
rate.depo machine=hemi oxide a.m  sigma.dep=0.20 \
          hemisphe dep.rate=1000 angle1=90.00 angle2=-90.00
deposit machine=hemi time=1 minute divis=5
structure outfile=anelex01_4.str

init infile=anelex01_0.str
rate.depo machine=cvd1 oxide a.m  \
         cvd dep.rate=1000 step.cov=0.80
```

```
deposit machine=cvd1 time=1 minute divis=5
structure outfile=anelex01_5.str
#
init infile=anelex01_0.str
rate.depo machine=conic oxide a.m  sigma.dep=0.20 \
      conical dep.rate=1000 angle1=60.00 c.axis=45.00 p.axis=20.00
deposit machine=conic time=1 minute divis=5
structure outfile=anelex01_6.str
#
init infile=anelex01_0.str
rate.depo oxide custom machine=custom infile=anelex01.dat a.m
  dep.rate=10.0
deposit machine=custom time=1 minute divis=5
structure outfile=anelex01_7.str
tonyplot -st anelex01_*.str
quit
```

27.1.2. anelex02.in: Unidirectional Deposit Example

This example uses the unidirectional deposit model with incident angles of 0, 45, -45, and 80 degrees from the surface normal. The unidirectional model uses a cosine law deposit that is applicable to line of sight deposition.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.2: Effects of varying the angle of incidence for a unidirectional deposition

Input File athena_elite/anelex02.in:

```
go athena

line x loc=0.00 spac=0.20
line x loc=1.5 spac=0.05
line x loc=3.0 spac=0.20
#
line y loc=0.00 spac=0.05
line y loc=1.0  spac=0.5
#
initialize
deposit nitride      thick=1 divis=20
etch nitride start x=1.0 y=-10
etch         cont  x=1.0 y=10
etch         cont  x=2.0 y=10
etch         done  x=2.0 y=-10
structure outfile=anelex02_0.str

rate.depo machine=uni oxide a.m  sigma.dep=0.20 \
        uni dep.rate=1000 angle1=0
```

```
deposit machine=uni time=1 minute divis=4
structure outfile=anelex02_1.str

init infile=anelex02_0.str
rate.depo machine=uni oxide a.m  sigma.dep=0.20 \
        uni dep.rate=1000 angle1=45
deposit machine=uni time=1 minute  divis=4
structure outfile=anelex02_2.str

init infile=anelex02_0.str
rate.depo machine=uni oxide a.m  sigma.dep=0.20 \
        uni dep.rate=1000 angle1=-45
deposit machine=uni time=1 minute  divis=4
structure outfile=anelex02_3.str

init infile=anelex02_0.str
rate.depo machine=uni oxide a.m  sigma.dep=0.20 \
        uni dep.rate=1000 angle1=70
deposit machine=uni time=1 minute  divis=4
structure outfile=anelex02_4.str

tonyplot -st anelex02_*.str -ttitle anelex02.in

quit
```

27.1.3. anelex03.in: Trench Etching: A Simple Approach

Simple directional etching can be modeled by combining two etch rate components: isotropical and directional. By selecting positive or negative values for the isotropic component, you can model trench etch with undercuts or conical sections as demonstrated in this example.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.3: Demonstration of the simplest etching technique in ELITE

Input File athena_elite/anelex03.in:

```
go athena

#TITLE: Simple trench etching
#
line x loc=0.00 spac=0.10
line x loc=3.0 spac=0.10
#
line y loc=0.00 spac=0.05
line y loc=2.0  spac=0.05
#
initialize
#
deposit material=SPR500      thick=1.16 divis=11
#
etch material=SPR500 start x=1.1 y=-10
etch                 cont  x=1.1 y=10
etch                 cont  x=1.9 y=10
etch                 done  x=1.9 y=-10
#
```

```
structure outfile=anelex03_0.str
#
rate.etch machine=m1 rie silicon \
iso=0.0 dir=1.0 u.m
rate.etch machine=m1 rie material=SPR500 \
iso=0.0 dir=0.05 u.m
#
etch machine=m1 time=60.0 second dx.mult=0.5
#
structure outfile=anelex03_1.str

tonyplot -ttitle anelex03.in -st anelex03_*.str
#
quit
```

27.1.4. anelex04.in: Trench Etching Using the RIE Model

This example demonstrates simple trench etching using a combination of isotropic and directional etch components. The initial structure was created in three steps. The initialize statement defines silicon as the substrate material and specifies the structure dimensions. The second step is a simple vertical deposition. The third step models lithography in a very simple manner. An opening with critical dimensions of 0.8 um and sidewall a slope of 89 degrees is created with the `etch` statement.

The RIE was performed in two steps. The first step has a high isotropic component. This produces a rounded profile with undercutting. The second portion has a high directional component. This produces a quick and dirty simulation of a strawberry shaped profile that is observed experimentally.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.4: Isotropic and Anisotropic RIE Etching

Input File athena_elite/anelex04.in:

```
go athena

#TITLE: Trench etching using RIE model
#
line x loc=0.00 spac=0.20
line x loc=0.5 spac=0.05
line x loc=1 spac=0.05
line x loc=1.5 spac=0.05
line x loc=2 spac=0.05
line x loc=2.5 spac=0.05
line x loc=3.0 spac=0.20
#
line y loc=0.00 spac=0.02
line y loc=2.5 spac=0.1
initialize
#
deposit material=SPR500      thick=1. div=5
#
etch material=SPR500 start x=1.1 y=-10
etch                 cont  x=1.1 y=10
etch                 cont  x=1.9 y=10
```

```
etch                 done  x=1.9 y=-10
#
structure outfile=anelex04_0.str
rate.etch machine=m1 rie silicon iso=1.0 dir=0.0 u.m
#
etch machine=m1 time=40.0 second dx.mult=0.5
structure outfile=anelex04_1.str
rate.etch machine=m2 rie silicon iso=0.0 dir=1.0 u.m
#
etch machine=m2 time=50.0 second dx.mult=0.5
#
structure outfile=anelex04_2.str
tonyplot -st anelex04_*.str -ttitle anelex04.in
#
quit
```

27.1.5. anelex05.in: Microloading Effects

The angular distributions obtained from the Monte Carlo simulation may be approximated by a Gaussian distribution, overlaid by a strong peak at zero degrees. Using such an approximation or using the original distribution obtained from the Monte-Carlo program for the calculation of local ion flux, results in strawberry-shaped profiles. They have rounded bottoms and a barrel-like underetching in the upper part of the trenches. A microloading or proximity effect can be seen. Small feature sizes have a lesser etch rate than wide feature sizes. The maximum etch rate can only be reached in a distance from the mask feature that is large compared to the etch depth.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.5: Microloading Effect. The depth of the trench depends on the size of the mask opening.

Input File athena_elite/anelex05.in :

```
go athena

#TITLE: Trench etching  with proximity (microloading) effect

line x loc=0.00 spac=0.05
line x loc=5.00 spac=0.05

line y loc=0.00 spac=0.1
line y loc=2.00 spac=0.5

init orientation=100

deposit oxide thick=0.10 div=2

etch oxide start x=.5 y=-10
etch       cont  x=.5 y=10
etch       cont  x=.8 y=10
etch       done  x=.8 y=-10

```

```
etch oxide start x=1.3 y=-10
etch       cont  x=1.3 y=10
etch       cont  x=1.9 y=10
etch       done  x=1.9 y=-10

etch oxide start x=3.0 y=-10
etch       cont  x=3.0 y=10
etch       cont  x=5.0 y=10
etch       done  x=5.0 y=-10

# define the etch rates for silicon (zero for other materials)
rate.etch machine=test rie  n.m silicon   \
isotropic=10.0 directional=20.0 chemical=30.0 divergence=20.

# perform the etch
etch mach=test time=20. minutes dx.mult=0.5


# save the structure and plot
structure outfile=anelex05_0.str
tonyplot anelex05_0.str -ttitle "Microloading Effect"

quit
```

27.1.6. anelex06.in: Comparison of RIE Model Components

This example demonstrates simple trench etching using the `RIE` etching model. The input file consists of four parts. The first part produces the initial multilayered structure using the ATHENA `Initialize`, `Deposit`, and `Etch` statements. In the first step, the `Initialize` statement defines silicon as a substrate 1.0 micron thick and 2 microns wide. The second step is a simple vertical deposition of photoresist. The third step models lithography in a very simple manner. An opening with critical dimensions of 0.6um and a sidewall slope of 87 degrees is created with the `Etch` statement. The second part demonstrates an example of `WET` or isotropical etching, which is modeled using the parameter `isotropic` e.g. `rate.etch machine=ad1 silicon a.s rie isotropic=50.0`. The third part shows anisotropical (directional) etching using the parameter `directional` e.g. `rate.etch machine=ad1 silicon a.s rie directional=50.` The last part illustrates chemical etching using the parameters, `chemical` and `divergence`, e.g. `rate.etch machine=ad1 silicon a.s rie chemical=50.0 divergence=5` where `chemical` is the rate and `divergence` is the standard deviation in degrees of a Gaussian distribution of ions with angle incident upon the surface. The simulation results appear on the joint plot at the very end of simulation.

To load and run this example, select the `ad example` button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.6: Demonstration of RIE Models. Isotropic, 100% Anisotropic and Anisotropic with angular distribution of ions

Input File athena_elite/anelex06.in:

```
go athena
#
line x loc=0.00 spac=0.05
line x loc=2.00 spac=0.05
#
line y loc=0.00 spac=0.05
line y loc=1.00 spac=0.05
#
init orientation=100
#
#
deposit photoresist  thick=0.5 div=8
etch photo start x=.65 y=-10
etch       cont  x=.75 y=10
etch       cont  x=1.25 y=10
etch       done  x=1.35 y=-10
#
structure outfile=anelex06_0.str
#
```

```
# Isotropic only
#
rate.etch machine=ad1 silicon a.s rie isotropic=50.0
#
etch machine=ad1 time=2 minute dx.mult=0.5
#
structure outfile=anelex06_1.str
#
init infile=anelex06_0.str
#
# Directional only
#
rate.etch machine=ad1 silicon a.s rie directional=50.0
#
etch machine=ad1 time=2 minute dx.mult=0.5
#
structure outfile=anelex06_2.str
#
init infile=anelex06_0.str
#
# Chemical only
#
rate.etch machine=ad1 silicon a.s rie chemical=50.0 divergence=5
#
etch machine=ad1 time=2 minute dx.mult=0.5
#
structure outfile=anelex06_3.str
#
tonyplot -st anelex06_*.str
#
quit
```

27.1.7. anelex07.in: Effect of Beam Divergence on Etch Rates

This example demonstrates simple trench etching using the `RIE` etching model. The example illustrates the influence of the `divergence` parameter on the final profile obtained by the `chemical` etch mode. The input file consists of four parts. The first part produces the initial multilayered structure using the `athena initialize`, `deposit`, and `etch` statements. The `initialize` statement defines silicon as a substrate 1.0 micron thick and 2 microns wide. The second step is a simple vertical deposition of photoresist. The third step models lithography in a very simple manner. An opening with critical dimensions of 0.6um and a sidewall slope of 87 degree is created with the `etch` statement.

The second part demonstrates an example of chemical etching, modeled using the parameter, `chemical` , e.g. `rate.etch machine=ad2 silicon a.s rie chemical=50.0 diver-`

gence=0.1 The third part shows the same etching mode, except the parameter divergence=10. The last part illustrates the same etching mode with divergence=45

The simulation results appear on the joint plot at the very end of the simulation.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.7: Demonstration of etching using different angular distribution of ions in RIE

Input File athena_elite/anelex07.in:

```
go athena

# Etching simulation using "RIE" model
#
line x loc=0.00 spac=0.05
line x loc=2.00 spac=0.05
#
line y loc=0.00 spac=0.05
line y loc=1.00 spac=0.05
#
init orientation=100
#
deposit photoresist  thick=0.5 div=8
etch photo start x=.65 y=-10
```

```
etch       cont  x=.75 y=10
etch       cont  x=1.25 y=10
etch       done  x=1.35 y=-10
#
structure outfile=anelex07_0.str
#
rate.etch machine=ad2 silicon a.s rie chemical=50.0 divergence=0.1
#
etch machine=ad2 time=2 minute dx.mult=0.5
#
#
structure outfile=anelex07_1.str
#
init infile=anelex07_0.str
#
rate.etch machine=ad2 silicon a.s rie chemical=50.0 divergence=10
#
etch machine=ad2 time=2 minute dx.mult=0.5
#
#
structure outfile=anelex07_2.str
#
init infile=anelex07_0.str
#
#
rate.etch machine=ad2 silicon a.s rie chemical=50.0 divergence=45
#
etch machine=ad2 time=2 minute dx.mult=0.5
#
#
structure outfile=anelex07_3.str
#
tonyplot -st anelex07_*.str
#
quit
```

27.1.8. anelex08.in: Trench Etch Tuning using RIE Model

This example demonstrates etching using the `RIE` etching model. The example illustrates the influence of the `divergence` of the ion beam on the final profile obtained by the `directional` etch mode. The parameter, `divergence` sets the standard deviation of the distribution function (in degrees). The input file consist of two major parts. The first one creates the initial multilayer structure using the `athena`, `initialize`, `deposit`, and `etch` statements. The `initialize` statement defines silicon as a substrate 1.0 micron thick and 2 microns wide. The second step is a simple vertical deposition of photoresist. The third step models lithography in a very simple manner. An opening with critical dimensions of 0.6um and a sidewall slope of 87 degree is created with the `etch` statement.

The second part demonstrates an example of `RIE` etching simulation using the parameters `directional, isotropical, chemical` and `divergence. Divergence` determines the standard deviation of the ion's Gaussian angular distribution function for the `chemical` component of the `RIE` etch model. This example shows a number of simulations with a variation of the of the three etching components: `directional, isotropical` and `chemical` etch rates.

The simulation results appear on the joint plot at the very end of the simulation.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.8: Tuning of Trench Etch Shapes using RIE Models

Input File athena_elite/anelex08.in:

```
go athena
#
line x loc=0.00 spac=0.05
line x loc=2.00 spac=0.05
#
line y loc=0.00 spac=0.05
line y loc=1.00 spac=0.05
#
init orientation=100
#
```

```
deposit photoresist  thick=0.5 div=8
etch photo start x=.65 y=-10
etch       cont  x=.75 y=10
etch       cont  x=1.25 y=10
etch       done  x=1.35 y=-10
#
structure outfile=anelex08_0.str

set DIVERG = 10
#
rate.etch machine=ad3 silicon a.s rie chemical=50 diverg=$DIVERG \
isotropic=0 directional=0
#
etch machine=ad3 time=2 minute dx.mult=0.5
#
structure outfile=anelex08_1.str
#
init infile=anelex08_0.str
#
rate.etch machine=ad3 silicon a.s rie chemical=30 diverg=$DIVERG \
isotropic=0 directional=20
#
etch machine=ad3 time=2 minute dx.mult=0.5
#
structure outfile=anelex08_2.str
#
init infile=anelex08_0.str
#
rate.etch machine=ad3 silicon a.s rie chemical=30 diverg=$DIVERG \
isotropic=20 directional=0
#
etch machine=ad3 time=2 minute dx.mult=0.5
#
structure outfile=anelex08_3.str
#
init infile=anelex08_0.str
#
rate.etch machine=ad3 silicon a.s rie chemical=30 diverg=$DIVERG \
isotropic=10 directional=10
#
etch machine=ad3 time=2 minute dx.mult=0.5
#
structure outfile=anelex08_4.str
```

```
#
tonyplot -st anelex08_*.str
#
quit

```

27.1.9. anelex09.in: Trench Filling and Planarization

A simple trench filling, planarization, and etchback is demonstrated in this example. The initial trench profile was defined using the `etch` statement with a bottom critical dimension of 0.6 um and 87 degree sidewall slopes.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.9: Complete simulation of trench etch, refill and planarization

Input File athena_elite/anelex09.in:

```
go athena

#TITLE: Trench filling and etch-back planarization
#
line x loc=0.00 spac=0.05
line x loc=1.2   spac=0.05
line x loc=1.8   spac=0.05
```

```
line x loc=3.00 spac=0.05
#
line y loc=0.00 spac=.2
line y loc=3.00 spac=.2
line y loc=5.0   spac=.2
#
init orientation=100
#
# geometrical trench etching
#
etch silicon   start x=1.1  y=0.0
etch  cont  x=1.2 y=3.0
etch  cont  x=1.8 y=3.0
etch  done  x=1.9 y=0.0
#
structure outfile=anelex09_1.str
# oxide deposition (step coverage 90%)
rate.depo machine=cvd1 cvd oxide  a.m dep.rate=200. step.cov=.9
#
deposit mach=cvd1 time=5.0 minutes div=2
#
structure outfile=anelex09_2.str
#
deposit polysilicon thick=.2  divis=5
deposit polysilicon thick=.15 divis=10
deposit polysilicon thick=.1  divis=4
structure outfile=anelex09_3.str
#
# planarization step
#
rate.depo machine=d3 hemis photoresist a.m \
 angle1=90. angle2=-90. dep.rate=1000. sigma.dep=0.2 \
 smooth.win=.25 smooth.step=3
#
deposit mach=d3 time=15.0 minutes divis=6
#
structure outfile=anelex09_4.str
#
# Define machine for etching of photoresist and poly
rate.etch mach=ws rie silicon iso=1400.0 dir=0.0 a.m
rate.etch mach=ws rie photoresist iso=1400 dir=0.0 a.m
```

```
rate.etch mach=ws rie polysilicon iso=1400 dir=0.0 a.m
rate.etch mach=ws rie oxide iso=0 dir=14.0 a.m
#
etch mach=ws time=10 minutes
#
structure outfile=anelex09_5.str
etch mach=ws time=5 minutes
#
structure outfile=anelex09_6.str
tonyplot -ttitle anelex09.in -st anelex09_*.str
#
quit
```

27.1.10. anelex10.in: Ion Milling of Aluminum

This example demonstrates ATHENA/ELITE ion milling using the RIE model. Ion milling shows a strong directional dependence. For this example, the incoming ion flux is vertical. This is modelled by using the RIE model with a directional component only. This results in a vertical penetration.

A machine called `ionmill` is defined and applied to the window structure. After each one minute etch, a structure file is saved. Following the last ion milling step, the structures are plotted sequentially.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.10: Evolution of Ion Milling through Metal

Input File athena_elite/anelex10.in:

```
go athena

#TITLE: Ion Milling of Aluminum
#
line x loc=0.0 spacing=0.1
line x loc=3.0 spacing=0.1
line y loc=0.0 spacing=0.1
line y loc=1.0 spacing=0.2
initial
#
deposit  aluminum thick=.75 div=10
#
deposit material=KTI820   thick=.5 div=8
etch material=KTI820 start x=.95  y=-1.25
etch  cont  x=1.00 y=-.75
etch  cont  x=2.00 y=-.75
etch  done  x=2.05 y=-1.25
#
structure outfile=anelex10_0.str
#
rate.etch machine=ionmill rie aluminum direct=2000 a.m
rate.etch machine=ionmill rie material=KTI820  direct=500.0 a.m
rate.etch machine=ionmill rie silicon  direct=1000.0 a.m
#
etch mach=ionmill time=1.00 minute
structure outfile=anelex10_1.str
#
etch mach=ionmill time=1.00 minute
structure outfile=anelex10_2.str
#
etch mach=ionmill time=1.00 minute
structure outfile=anelex10_3.str
#
etch mach=ionmill time=1.00 minute
structure outfile=anelex10_4.str
#
etch mach=ionmill time=1.00 minute
structure outfile=anelex10_5.str
#
tonyplot -st anelex10_*.str
#
```

```
quit

```

27.1.11. anelex11.in: Two-Step Trench Etch Process

This example simulates trench etching. Etching is described as a combination of wet etching (30nm/min) and ion induced etching(100nm/min).

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.11: Trench formed though multi-step etch process in ELITE

Input File athena_elite/anelex11.in:

```
go athena

#TITLE: Trench etching with redeposition effect
# anelex11.in
#
line x loc=0.0 spacing=0.05
line x loc=2.0 spacing=0.05
line y loc=0.0 spacing=0.05
line y loc=2.0 spacing=0.05
initial
```

```
#
deposit oxide thick=0.35 div=10
#
etch oxide start x=0.7 y=-0.36
etch continue x=1.3 y=-0.36
etch continue x=1.3 y=0.01
etch done x=0.7 y=0.01
#
rate.etch machine=det2 rie oxide n.m chemical=10.0 divergence=1.0
#
rate.etch machine=det2 rie silicon n.m isotropic=5.0 chemical=75.0
#
etch mach=det2 time=10.0 minutes
#
structure outfile=anelex11_0.str
#
etch mach=det2 time=10.0 minutes
#
structure outfile=anelex11_1.str
tonyplot -st anelex11_*.str -ttitle anelex11.in
#

quit
```

27.1.12. anelex12.in: Lift-off Profile Polymer Etching

The example described in this section originated from studies accompanying the development of a lift-off resist profile for sub-micron MESFET fabrication using a tri-level resist technique. The goal of this process development is to get the lift-off profile in the bottom resist of the tri-level system without using special e-beam exposure techniques or multi-layer e-beam resists.

The resist system used instead is quite simple. The bottom layer is a hard baked standard Novolac or polymide resist covered with a silicon nitride intermediate layer. This intermediate layer is used during the etching process as a hard mask. The top layer is a PMMA resist structured by conventional electron beam lithography. The top PMMA resist is stripped and etching begun.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.12: Resist Etching though multi-step etch process in ELITE

Input File athena_elite/anelex12.in:

```
go athena

#TITLE: Resist etching using rie model
line x loc = 0.0  spac=0.05
line x loc = 2.0  spac=0.05
#
line y loc = 0    spac=1
line y loc = 1.0  spac=1
#
#calculate the mesh
init
#
deposit material=polyimide thick=1.1 div=22
#
deposit oxide thick=.1 div=2
etch  oxide start    x=0.86 y=-5
etch           cont     x=0.86 y=5
etch           cont     x=1.14 y=5
etch           done     x=1.14 y=-5
#
structure outfile=anelex12_0.str
```

```
#
rate.etch machine=det rie material=polyimide n.m isotropic=50.0 \
chemical=100.0 divergence=30
#
etch mach=det time=8.0 minutes
#dx.mult=.5
#
structure outfile=anelex12_1.str
#
rate.etch machine=det rie material=polyimide n.m isotropic=5.0 \
chemical=85.0 divergence=3
#
etch mach=det time=5.0 minutes
#
structure outfile=anelex12_2.str
tonyplot -st anelex12_*.str -ttitle anelex12.in
#
quit
```

27.1.13. anelex13.in: Interconnect Metallization

This example performs a sequence of deposition and etch steps corresponding to a double metal power IC process. The step coverage and topography of the second metal level is critical in determining the success of the process.

This example compares the structures created by two final metallization techniques. The first technique includes surface diffusion that would result from a high temperature during deposition or from increased temperature or RF bias. The second technique does not use surface diffusion. The improved step coverage, due to surface diffusion, is evident from the plots produced.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.13: Second Layer Metal Deposition onto non-planar substrate

Input File athena_elite/anelex13.in:

```
go athena

line x loc=0.00 spac=0.10
line x loc=4.0  spac=0.10
line x loc=7.0  spac=0.10
line x loc=9.0  spac=0.10
line x loc=12.0 spac=0.10
line x loc=16.5 spac=0.10
line x loc=19.5 spac=0.10
line x loc=20.0 spac=0.10
#
line y loc=0.00 spac=0.1
```

```
line y loc=1.   spac=1.0
#
init space.m=3
#
deposit oxide  thick=1.0 div=10
#
deposit photoresist  thick=2.0 div=5

etch    photo start x=4. y=-100
etch     continue x=4. y=0
etch     continue x=7. y=0
etch     done  x=7.    y=-100
#
rate.etch machine=boe photoresist a.m  rie dir=40 iso=0.0
rate.etch machine=boe oxide a.m  rie dir=800 iso=0.0
#
etch machine=boe time=15 minutes
strip
#
# remove excess surface grid
relax surface dx.surf=0.1

rate.depo machine=pe_tiw material=TiW a.m  sigma.dep=0.20 \
          hemisphe dep.rate=200 angle1=70 angle2=-70
deposit machine=pe_tiw time=5.0 minute divis=3

rate.depo machine=PE4450 aluminum u.m  sigma.dep=0.65 hemisphe \
          dep.rate=1.0 angle1=70 angle2=-70
deposit machine=PE4450 time=1.5 minute divis=8

deposit photoresist thick=1.9 div=8
#
etch    photo start x=9 y=-100
etch     continue x=9 y=0
etch     continue x=12 y=0
etch     done  x=12 y=-100

# run Aluminum etch
rate.etch machine=dq1 aluminum u.m  rie isotropic=.06 dir=.6
rate.etch machine=dq1 photoresist u.m  rie isotropic=.06 dir=.3
rate.etch machine=dq1 material=TiW u.m  rie isotropic=.005 dir=.05
rate.etch machine=dq1 oxide u.m  rie isotropic=.005 dir=.03
rate.etch machine=dq1 silicon u.m  rie isotropic=.005 dir=.001
```

```
etch machine=dq1 time=3. minute

# run TiW Etch
rate.etch machine=dq2 aluminum u.m  rie isotropic=.06 dir=.05
rate.etch machine=dq2 photoresist u.m  rie isotropic=.06 dir=.3
rate.etch machine=dq2 material=TiW u.m  rie isotropic=.045 dir=.18
rate.etch machine=dq2 oxide u.m  rie isotropic=.05 dir=.1
rate.etch machine=dq2 silicon u.m  rie isotropic=.005 dir=.001
etch machine=dq2 time=35 sec

strip photoresist

# remove excess surface grid
relax surface dx.surf=0.1

rate.depo machine=ozone_teos oxide a.m cvd dep.rate=1000 step.cov=.7
#
deposit machine=ozone_teos time=12.5 minute divis=10

deposit photoresist thick=1.9 div=10

# VIA etch
etch    photo start x=16.5 y=-100
etch     continue x=16.5 y=0
etch     continue x=19.5 y=0
etch     done  x=19.5 y=-100

rate.etch machine=DRY-384 oxide u.m  rie isotropic=.1 dir=.9
rate.etch machine=DRY-384 nitride u.m  rie isotropic=.1 dir=.9
rate.etch machine=DRY-384 silicon u.m  rie isotropic=.02 dir=.02
rate.etch machine=DRY-384 photoresist u.m  rie isotropic=.1 dir=.3

etch machine=DRY-384 time=2 minute dx.mult=0.5

strip photoresist

# remove excess surface grid
relax surface dx.surf=0.1

# save structure file for use later
structure outfile=anelex13_0.str

#second Metal with surface diffusion
```

```
rate.depo machine=varian aluminum u.m  sigma.dep=1.5 hemisphe \
          dep.rate=1.0 angle1=70 angle2=-70
deposit machine=varian time=3.0 minute divis=10

tonyplot -ttitle "anelex13.in: Double Metal w/ Surface Diffusion"
structure outfile=anelex13_1.str

#second Metal without surface diffusion
init infile=anelex13_0.str

rate.depo machine=varian aluminum u.m  sigma.dep=0.05 hemisphe \
          dep.rate=1.0 angle1=70 angle2=-70
deposit machine=varian time=3.0 minute divis=10

tonyplot -ttitle "anelex13.in: Double Metal w/ Low Surface Diffusion"
structure outfile=anelex13_2.str

quit
```

27.1.14. anelex14.in: MOS LDD Spacer Formation

This example simulates poly gate etching and spacer formation. The structure is initialized, a stack of layers is deposited, a photomask is patterned, and a directional etch of polysilicon is performed. The photoresist is removed, the oxide is deposited using CVD with a step coverage of 70%, and finally a highly directional oxide etch is performed.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.14: Process sequence for MOS Spacer formation using ELITE

Input File athena_elite/anelex14.in:

```
go athena

#TITLE: A simple Poly Gate Etching and Spacer formation
#
line x loc=0.00 spac=0.20
line x loc=1.5  spac=0.1
line x loc=2.5  spac=0.1
line x loc=4.0  spac=0.20
#
line y loc=0.00 spac=0.05
line y loc=2    spac=0.3
init
#
depo oxi thi=0.02 div=1
depo poly thi=0.35 div=10
depo photores thi=1 div=5
etch photo left  p1.x=1.5
etch photo right p1.x=2.5
#
structure outfile=anelex14_0.str
#
```

```
rate.etch mach=dr1 rie photo direct=0.02 isotropic=0.0 u.m
rate.etch mach=dr1 rie polysilicon  direct=1.0 isotropic=0.0 u.m
rate.etch mach=dr1 rie oxide direct=0.25 isotropic=0.25 u.m
#
etch mach=dr1 time=60 second dx.mult=0.5
#
structure outfile=anelex14_1.str
#
strip photores
#
structure outfile=anelex14_2.str
#
rate.depo mach=de1 cvd oxide step.cov=0.7 dep.rate=0.1 u.m
#
deposit mach=de1 time=5 minute div=15
#
structure outfile=anelex14_3.str
#
rate.etch mach=dr2 rie photo dir=0.02 iso=0.0 u.m
rate.etch mach=dr2 rie polysilicon  dir=0.1 iso=0.0 u.m
rate.etch mach=dr2 rie oxide dir=0.55 iso=0.00 u.m
#
etch mach=dr2 time=30 second dx.mult=.5
structure outfile=anelex14_4.str
etch mach=dr2 time=30 second dx.mult=.5
structure outfile=anelex14_5.str
#
tonyplot -ttitle anelex14.in -st anelex14_*.str
#
quit
```

27.1.15. anelex15.in: Metal Passivation

This example simulates metal passivation. The presence of defects (metal feet) on the sides of the metal line causes cracks in the deposited insulator that could be a source of corrosion or inter-level shorts.

By contrast, sputtering on a hot substrate can prevent crack formation. ELITE models the effect of the hot substrate by incorporating surface diffusion using the `sigma.dep` parameter.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.15: Effect of surface diffusion on dielectric deposition. The lower plot features a higher surface diffusion. This smooths out the sharp corners seen on the upper plot.

Input File athena_elite/anelex15.in:

```
go athena

#TITLE: Passivation oxide sputter on cold and hot substrate
line x loc = 0.0  spac=0.05
line x loc = 1.0  spac=0.05
#
line y loc = 0    spac=0.25
line y loc = 0.5  spac=0.25
#
init
#
# define non-planar top profile
#
deposit aluminum thick=.5 divis=8
etch aluminum start  x=0.0 y=0.0
etch          cont   x=0.5 y=0.0
etch          cont   x=0.43 y=-.07
etch          cont   x=0.45 y=-.1
etch          cont   x=0.47 y=-.1
etch          cont   x=0.52 y=-.45
etch          cont   x=0.55 y=-.5
```

```
etch          cont   x=0.55 y=-1
etch          done   x=0.0  y=-1

# mirror it!
structure mirror right
#
# Save initial profile into the file anelex15_0.str
structure outfile=anelex15_0.str
#
rate.depo  machine=depc hemis oxide u.m \
angle1=75. angle2=-75. dep.rate=0.3 sigma.dep=0.00
#
# run deposition on cold substrate
#
deposit mach=depc time=1.0 minute divis=5
#
# Save the structure for plotting
structure outfile=anelex15_1.str
#
initialize infile=anelex15_0.str
#
#
rate.depo  machine=deph hemis oxide u.m \
angle1=75. angle2=-75. dep.rate=0.3 sigma.dep=0.27
#
#run deposition on hot substrate
#
deposit mach=deph time=1.0 minute divis=5
#
# Save the structure for plotting
structure outfile=anelex15_2.str
tonyplot -st anelex15_*.str -ttitle "anelex15.in: Surface Diffusion Com-
  parison"
#
quit

```

27.1.16. anelex16.in: Intermetal Dielectric Formation

This example shows the use of two conductors (poly and aluminum) that come close together. The narrow gap between them can form voids as demonstrated in this example for atmospheric pressure CVD. The type of inter-metal dielectric material, the thickness of this dielectric, and the method of insulation can have an impact on design rules for ICs with multi-level metallization.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.16: Sequence for Metal Deposition over polysilicon gate

Input File athena_elite/anelex16.in :

```
go athena

#TITLE: Intermetal Dielectric Formation
# anelex16.in
line x loc = 0.0  spac=0.1
line x loc = 1.5  spac=0.1
line x loc = 2.5  spac=0.1
line x loc = 6.0  spac=0.1
#
line y loc = 0    spac=0.1
line y loc = 0.50 spac=0.1
#
init
#
deposit poly thick=.5 div=10
etch  poly   left  p1.x=1.5
etch  poly   right p1.x=2.5
```

```
#
rate.depo mach=d1 oxide \
hemispher angle1=90 angle2=-90 dep.rate =700.0 a.m \
sigma.dep=0.33
#
deposit mach=d1 time=10. minutes divis=5
#
structure outfile=anelex16_0.str
#
deposit alumin thick=.5 divis=10
#
structure outfile=anelex16_1.str
#
etch aluminum left  p1.x=3.2
etch aluminum right p1.x=4.2
#
structure outfile=anelex16_2.str
#
rate.depo mach=d2 oxide \
hemispher angle1=90 angle2=-90 dep.rate=700.0 a.m \
sigma.dep=0.25
#
deposit mach=d2 time=17.0 minutes divis=10
#
structure outfile=anelex16_3.str
tonyplot -st anelex16_*.str -ttitle anelex16.in
#
quit
```

27.1.17. anelex17.in: Custom Deposit

This example shows the use of ELITE for a custom deposit which can be the equivalent of many unidirectional deposits. This model is called `user.data.2.`

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.17: Deposition simulation using User-defined Deposit Model at zero degrees



Figure 27.18: Deposition simulation using User-defined Deposit Model at 45 degrees



Figure 27.19: Deposition simulation using User-defined Deposit Model at -45 degrees

Input File athena_elite/anelex17.in:

```
go athena

line x loc=0.00 spac=0.1
line x loc=2.00 spac=0.1
#
line y loc=0.00 spac=0.1
line y loc=1.0 spac=0.5
initialize

deposit oxide thick=0.8 div=16
etch oxide start x=0.5 y=-10.0
etch continue x=0.5 y=10.0
etch continue x=1.5 y=10.0
etch done x=1.5 y=-10.0

structure outf=anelex17.str
#
### Unidirectional Model at 0 degrees ###
#
rate.depo aluminum unidirec machine=uni angle1=0.0 a.s \
dep.rate=100.0 sigma.dep=0.2
```

```

deposit machine=uni time=20 sec divis=4 substeps=11

structure outf=anelex17_0.str
#
### The new CUSTOM model is called USER.DATA.2 at 0 degrees ###
#
init infile=anelex17.str

rate.depo aluminum user.data.2 machine=test infile=anelex17_0.dat a.s \
dep.rate=100.0 sigma.dep=0.2

deposit machine=test time=20 sec divis=4 substeps=11

structure outf=anelex17_1.str

tonyplot -overlay anelex17_0.str anelex17_1.str -set anelex17_1.set
#
### Unidirectional Model at 45 degrees ###
#
init infile=anelex17.str

rate.depo aluminum unidirec machine=uni angle1=45.0 a.s \
dep.rate=100.0 sigma.dep=0.2

deposit machine=uni time=20 sec divis=4 substeps=11

structure outf=anelex17_2.str
#
### The new CUSTOM model is called USER.DATA.2 at 45 degrees ###
#
init infile=anelex17.str

rate.depo aluminum user.data.2 machine=test infile=anelex17_1.dat a.s \
dep.rate=100.0 sigma.dep=0.2

deposit machine=test time=20 sec divis=4 substeps=11

structure outf=anelex17_3.str

tonyplot -overlay anelex17_2.str anelex17_3.str -set anelex17_2.set
#
### Unidirectional Model at -45 degrees ###
```

```
#
init infile=anelex17.str

rate.depo aluminum unidirec machine=uni angle1=-45.0 a.s \
dep.rate=100.0 sigma.dep=0.2

deposit machine=uni time=20 sec divis=4 substeps=11

structure outf=anelex17_4.str
#
### The new CUSTOM model is called USER.DATA.2 -45 degrees ###
#
init infile=anelex17.str

rate.depo aluminum user.data.2 machine=test infile=anelex17_2.dat a.s \
dep.rate=100.0 sigma.dep=0.2

deposit machine=test time=20 sec divis=4 substeps=11

structure outf=anelex17_5.str

tonyplot -overlay anelex17_4.str anelex17_5.str -set anelex17_3.set

quit
```

27.1.18. anelex18.in: Trench Etch

This example shows the use of ELITE to model trench etching.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.20: Trench Etch including erosion of the Oxide Mask Layer

Input File athena_elite/anelex18.in:

```
go athena

#TITLE: Trench etch using rie model
#
line x loc=0.00 spac=0.05
line x loc=3.00 spac=0.05
#
line y loc=0.00 spac=0.05
line y loc=1.00 spac=0.05
#
init orientation=100 space.m=1
#
deposit oxide thick=0.10 div=5
#
etch oxide start x=1.2 y=-10
etch       cont  x=1.2 y=10
etch       cont  x=1.8 y=10
etch       done  x=1.8 y=-10
#
```

```
rate.etch machine=det2 rie silicon n.m chemical=125 directional=100 di-
  vergence=90
#
rate.etch machine=det2 rie oxide n.m chemical=5.0 directional=5
#
etch mach=det2 time=5.0 minutes dx.mult=.5
#
tonyplot

structure outfile=anelex18_0.str

quit
```

27.1.19. anelex19.in: Contact Etch, Reflow, and Metallization

This example shows the use of ELITE to model a contact opening by using an isotropic etch followed by a directional etch. This results in an opening that has edges that may be prone to metal thinning during interconnect metal deposition. To reduce this effect, reflow is performed using the ELITE reflow model. This is accomplished by specifying:

```
material oxide visc.0=1.862e-20 gamma.reflo=1e3 reflow
method viscous
diff time=10 temp=950 reflow
```

which invokes a time temperature cycle that calculates the effects of surface tension on the structure.

Following reflow, the metallization is performed using a simple CVD model that is characterized using deposit rate and step coverage parameters.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.21: Sequence of Contact VIA Etch, Reflow of the dielectric to round the corners and metal deposition

Input File athena_elite/anelex19.in:

```
go athena

#TITLE: Trench etch, reflow, and metal deposit
#
line x loc=0.00 spac=0.1
line x loc=1 spac=0.05
line x loc=2 spac=0.05
line x loc=3.0 spac=0.1
#
line y loc=0.00 spac=0.5
line y loc=0.5 spac=0.5
initialize
#
deposit oxide      thick=1.5 div=10
deposit nitride      thick=.3
#
etch    nitride      start x=1.0 y=-10
etch                 cont  x=1.0 y=10
etch                 cont  x=2.0 y=10
```

```
etch                done  x=2.0 y=-10

rate.etch machine=m1 rie oxide iso=1.0 dir=0.0 u.m
#
etch machine=m1 time=40.0 second dx.mult=0.25

rate.etch machine=m2 rie oxide iso=0.0 dir=1.0 u.m
#
etch machine=m2 time=60.0 second dx.mult=0.5

strip nitride

structure outfile=anelex19_0.str

#perform reflow here!
material oxide visc.0=1.862e-20 gamma.reflo=1e3 reflow
diff time=10 temp=950 reflow

structure outfile=anelex19_1.str

rate.depo machine=PE4450 aluminum u.m  sigma.dep=0.65 hemisphe \
          dep.rate=0.5 angle1=70.0 angle2=-70.0
deposit machine=PE4450 time=45 second divis=8

structure outfile=anelex19_2.str
tonyplot -st anelex19_*.str

quit

```

27.1.20. anelex20.in: Sticking Coefficient for Monte Carlo Deposit

This example shows the use of ELITE to model deposit into a test structure with changing sticking coefficient. The step coverage varies as a function of the sticking coefficient in a complex manner as a function of geometry.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.22: Demonstration of the effect of sticking coefficient during depositions. A lower coefficient allows more diffusion to the underside of the nitride.

Input File athena_elite/anelex20.in:

```
go athena

#TITLE: Trench etch and monte carlo deposit
#
line x loc=0.0 spac=0.1
line x loc=1.0 spac=0.05
line x loc=2.0 spac=0.05
line x loc=3.0 spac=0.1
#
line y loc=0.0 spac=0.05
line y loc=1.0 spac=0.1
initialize
#
deposit oxide        thick=.3 divis=30
deposit nitride      thick=.1
#
etch    nitride      start x=1.0 y=-10
etch                 cont  x=1.0 y=10
etch                 cont  x=2.0 y=10
etch                 done  x=2.0 y=-10
```

```

rate.etch machine=m2 rie oxide iso=1.0 dir=0.0 u.m
#
etch machine=m2 time=50.0 second dx.mult=.2

structure outfile=anelex20_0.str

rate.depo machine=m2 monte1 tungsten dep.rate=0.5 u.m stick.c=1 sig-
  ma.dep=0.1

deposit div=1 mach=m2 time=.1 minute n.particle=10000

structure outfile=anelex20_1.str
init infile=anelex20_0.str
rate.depo machine=m2 monte1 tungsten dep.rate=0.5 u.m stick.c=.1 sig-
  ma.dep=0.1

deposit div=1 mach=m2 time=.1 minute n.particle=30000

structure outfile=anelex20_2.str
init infile=anelex20_0.str
rate.depo machine=m2 monte1 tungsten dep.rate=0.5 u.m stick.c=.01 sig-
  ma.dep=0.1

deposit div=1 mach=m2 time=.1 minute n.particle=90000

structure outfile=anelex20_3.str
tonyplot -st anelex20_*.str

quit

```

27.1.21. anelex21.in: Surface Diffusion for Monte Carlo Deposit

This example shows the use of ELITE to model deposit into a test structure with surface diffusion. The deposit gradually fills the hole in the structure so that deposit into the hole is reduced. This results in a tip structure whose sharpness is a function of the surface diffusion.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.23: Effect of surface diffusion during deposition. Deposited ions can diffuse away from the location they are first deposited.

Input File athena_elite/anelex21.in :

```
go athena
#
line x loc=-4.00 spac=0.10
line x loc=10.00 spac=0.10
line y loc=0.00 spac=0.02
line y loc=0.50 spac=0.5
#
init silicon orientation=100
#
deposit oxide thick=1.5 divisions=5
deposit nitride thick=0.5 divisions=3
#
etch nitride start x=2.0 y=-10.0
etch cont x=2.00 y=10.0
etch cont x=4.00 y=10.0
etch done x=4.00 y=-10.0
#
rate.etch machine=test1 oxide a.m  wet.etch isotropic=1000.00
etch machine=test1 time=25 minutes dx.mult=.5
```

```
rate.depo machine=Aldepo1 alumin n.m sigma.dep=0.75 monte1 dep.rate=100.0
  angle1=0.0
deposit machine=Aldepo1 time=20 minutes divis=10 n.particle=100000
#

tonyplot

structure outfile=anelex21_0.str

quit

```

27.1.22. anelex22.in: Monte Carlo Deposit into Contact Cut

This example shows the use of ELITE to model deposit into a test structure with an atomistic level surface repositioning calculation. Individual particle resting places can be seen in the final structure plots. Increasing the surface diffusion parameter, `sigma.dep,` changes the resulting density, due to atom re-positioning.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DeckBuild **run** button to execute the example.



Figure 27.24: Monte Carlo deposition into a cut. The MC algorithm displays the location of individual metal 'fragments' as dots (On-screen the dots are red)

Input File athena_elite/anelex22.in:

```
go athena
#
line x loc=-4.00 spac=0.10
line x loc=10.00 spac=0.10
line y loc=0.00 spac=0.02
line y loc=0.50 spac=0.5
#
init silicon orientation=100 space.m=1
#
deposit oxide thick=1.5 divisions=5
deposit nitride thick=0.5 divisions=3
#
etch nitride start x=2.0 y=-10.0
etch cont x=2.00 y=10.0
etch cont x=4.00 y=10.0
etch done x=4.00 y=-10.0
#
rate.etch machine=test1 oxide a.m  wet.etch isotropic=1000.00
etch machine=test1 time=25 minutes dx.mult=.5

structure outfile=anelex22_0.str

rate.depo machine=Aldepo1 alumin a.m sigma.dep=0.1 monte2 dep.rate=1000.0 angle1=0.0
deposit machine=Aldepo1 time=1 minutes divis=1 n.particle=2000 outfile=anelex22_1.str

init infile=anelex22_0.str

rate.depo machine=Aldepo1 alumin a.m sigma.dep=0.5 monte2 dep.rate=1000.0 angle1=0.0
deposit machine=Aldepo1 time=1 minutes divis=1 n.particle=2000 outfile=anelex22_2.str


tonyplot -st anelex22_*.str

quit
```

27.1.23. anelex23.in: Monte Carlo Deposit on Sloped Sidewall

This example shows the use of ELITE to model deposit onto a test structure with an atomistic level surface repositioning calculation. Individual particle resting places can be seen in the final structure plots. Increasing the surface diffusion parameter, `sigma.dep`, changes the resulting density due to atom re-positioning.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.25: Monte Carlo deposition shows metal grain formation along a sloped sidewall. A variation in surface diffusion is shown.

Input File athena_elite/anelex23.in:

```
go athena
#
line x loc=-0.5  spac=0.10
line x loc=0.5   spac=0.10
line y loc=0.00  spac=0.02
line y loc=0.75  spac=0.5
#
init silicon orientation=100 space.m=1
#
etch silicon right p1.x=0.0 p1.y=0.5 p2.x=-0.5 p2.y=0.0

structure outfile=anelex23_0.str

rate.depo machine=Aldepo1 alumin a.m sigma.dep=0.0 monte2 dep.rate=1000.0
  angle1=0.0
deposit machine=Aldepo1 time=1 minutes divis=1 n.particle=10000 out-
  file=anelex23_1.str
```

```

init infile=anelex23_0.str

rate.depo machine=Aldepo1 alumin a.m sigma.dep=0.2 monte2 dep.rate=1000.0
  angle1=0.0
deposit machine=Aldepo1 time=1 minutes divis=1 n.particle=10000 out-
  file=anelex23_2.str

tonyplot -st anelex23_*.str

quit
```

27.1.24. anelex24.in: Customizable Deposit from an ASCII File

This example shows the use of ELITE to model a customizable deposit. Input is done via an ASCII file containing angle and rate information.

The example begins by creating a test structure using the `Start Cont Done` sequence of `Etch` statements. This creates a globe on which deposition will be performed and compared. The globe is then deposited with rates that are included in a data file using the `USER.DATA.1` model. The data file specifies that the deposit rates will be similar at all angles except at a single angle where the deposit is enhanced.

A second example uses a data file that specifies rates that are set to duplicate a unidirectional deposit with vertical incidence. This results in deposit on the exposed portion of the globe test structure. Finally, the deposit is repeated using the unidirectional model.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.26: Demonstration of User-Specified Diffusion Models

Input File athena_elite/anelex24.in:

```
go athena

# Custom Deposit machine example
#
line x loc=-1.5 spac=0.025
line x loc=0.0   spac=0.025
#
line y loc=1.1  spac=0.5
line y loc=1.5  spac=0.5
#
initialize
#
deposit oxide        thick=2.1 divis=40

# Use etch sequence to create a globe test structure
etch oxide start x=-10. y=1.1
etch cont x=-.1 y=1.1
etch cont x=-.1 y=1.0
etch cont x=-.17 y=.98
```

```
etch cont x=-.34 y=.94
etch cont x=-.5 y=.86
etch cont x=-.64 y=.76
etch cont x=-.76 y=.64
etch cont x=-.86 y=.5
etch cont x=-.94 y=.34
etch cont x=-.98 y=.17
etch cont x=-1. y=0
etch cont x=-.98 y=-.17
etch cont x=-.94 y=-.34
etch cont x=-.86 y=-.5
etch cont x=-.76 y=-.64
etch cont x=-.64 y=-.76
etch cont x=-.5 y=-.86
etch cont x=-.34 y=-.94
etch cont x=-.17 y=-.98
etch cont x=0.0  y=-1.0
etch done x=-10 y=-1.0

structure mirror right

# save the structure
structure outfile=anelex24_0.str
#
rate.depo aluminum user.data.1 machine=test infile=anelex24_0.dat a.m

deposit machine=test time=10.0 minutes

structure outfile=anelex24_1.str

init infile=anelex24_0.str
#
rate.depo aluminum user.data.1 machine=test infile=anelex24_1.dat a.m

deposit machine=test time=10.0 minutes

structure outfile=anelex24_2.str

init infile=anelex24_0.str
#
rate.depo aluminum unidirec machine=test angle1=0.0 dep.rate=100 a.m

deposit machine=test time=10.0 minutes
```

```
structure outfile=anelex24_3.str

tonyplot -st anelex24_*.str

quit
```

27.1.25. anelex25.in: The Hard Polish Model

This example shows the use of ELITE to model chemical mechanical polishing using the hard polish model.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.27: Sequence showing CMP of Interconnect dielectric. The two figures on the left are initial structures and on the right the structures after CMP

Input File athena_elite/anelex25.in:

```
go athena

#  anelex25.in
line x loc=-10.00 spac=0.1
line x loc=10.00 spac=0.1
```

```
#
line y loc=0 spac=0.5
line y loc=3.0  spac=1
initialize silic

deposit aluminum thick=.5 divis=6

structure outfile=anelex25_0.str

etch aluminum start x=-1 y=-10.0
etch continue x=-1 y=10.0
etch continue x=1 y=10.0
etch done x=1 y=-10.0

etch aluminum start x=3 y=-10.0
etch continue x=3 y=10.0
etch continue x=5 y=10.0
etch done x=5 y=-10.0

etch aluminum start x=-5 y=-10.0
etch continue x=-5 y=10.0
etch continue x=-3 y=10.0
etch done x=-3 y=-10.0
#
rate.depo machine=oxide oxide u.s  smooth.win=0.20 smooth.step=1 cvd \
        dep.rate=1.0 step.cov=0.6

deposit machine=oxide time=1 seconds div=10 substeps=15

structure outfile=anelex25_1.str

rate.polish oxide machine=polish u.s max.hard=.15 min.hard=.03 isotropic=0.001

polish machine=polish time=5 seconds dx.mult=0.5

structure outfile=anelex25_2.str

init infile=anelex25_0.str

etch aluminum start x=-1 y=-10.0
etch continue x=-1 y=10.0
etch continue x=1 y=10.0
```

```
etch done x=1 y=-10.0
etch aluminum left x=-3
etch aluminum right x=3
#
deposit machine=oxide time=1 seconds div=10 substeps=15
structure outfile=anelex25_3.str
polish machine=polish time=5 seconds dx.mult=.5
structure outfile=anelex25_4.str
tonyplot -st anelex25_1.str anelex25_2.str anelex25_3.str anelex25_4.str
quit
```

27.1.26. anelex26.in: The Soft Polish Model

This example shows the use of ELITE to model chemical mechanical polishing using the soft polish model. This example calculates results for the the soft polish model so that they can be compared to the results published by J. Warnock, “A Two Dimensional Process Model for Chemimechanical Polish Planarization”, Electrochem. Soc. Vol 138, No. 8, August 1991.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.28: CMP planarization of dielectric over 20 Metal Lines. These results can be compared to published CMP Models.



Figure 27.29: CMP planarization of dielectric over Nine Metal Lines



Figure 27.30: CMP planarization of dielectric over Four Metal Lines

Input File athena_elite/anelex26.in:

```
go athena
#
#
line x loc=0.000 tag=left
line x loc=100.00 spac=1.000
line x loc=300.000 spac=1.000
line x loc=401.000 tag=right
line y loc=1.00 spac=5.00 tag=top
line y loc=5.00 spac=5.00 tag=bottom
#
init silicon
#
depo oxide thick=1.00 divis=20
#
struct outfile=anelex26.str
#
etch oxide start x=-0.1 y=-1000
etch cont x=-0.1 y=.400
etch cont x=103.5 y=.400
etch cont x=103.5 y=-1000
etch cont x=108.5 y=-1000
```

```
etch cont x=108.5 y=.400
etch cont x=113.5 y=.400
etch cont x=113.5 y=-1000
etch cont x=118.5 y=-1000
etch cont x=118.5 y=.400
etch cont x=123.5 y=.400
etch cont x=123.5 y=-1000
etch cont x=128.5 y=-1000
etch cont x=128.5 y=.400
etch cont x=133.5 y=.400
etch cont x=133.5 y=-1000
etch cont x=138.5 y=-1000
etch cont x=138.5 y=.400
etch cont x=143.5 y=.400
etch cont x=143.5 y=-1000
etch cont x=148.5 y=-1000
etch cont x=148.5 y=.400
etch cont x=153.5 y=.400
etch cont x=153.5 y=-1000
etch cont x=158.5 y=-1000
etch cont x=158.5 y=.400
etch cont x=163.5 y=.400
etch cont x=163.5 y=-1000
etch cont x=168.5 y=-1000
etch cont x=168.5 y=.400
etch cont x=173.5 y=.400
etch cont x=173.5 y=-1000
etch cont x=178.5 y=-1000
etch cont x=178.5 y=.400
etch cont x=183.5 y=.400
etch cont x=183.5 y=-1000
etch cont x=188.5 y=-1000
etch cont x=188.5 y=.400
etch cont x=193.5 y=.400
etch cont x=193.5 y=-1000
etch cont x=198.5 y=-1000
etch cont x=198.5 y=.400
etch cont x=203.5 y=.400
etch cont x=203.5 y=-1000
etch cont x=208.5 y=-1000
etch cont x=208.5 y=.400
etch cont x=213.5 y=.400
etch cont x=213.5 y=-1000
```

```
etch cont x=218.5 y=-1000
etch cont x=218.5 y=.400
etch cont x=223.5 y=.400
etch cont x=223.5 y=-1000
etch cont x=228.5 y=-1000
etch cont x=228.5 y=.400
etch cont x=233.5 y=.400
etch cont x=233.5 y=-1000
etch cont x=238.5 y=-1000
etch cont x=238.5 y=.400
etch cont x=243.5 y=.400
etch cont x=243.5 y=-1000
etch cont x=248.5 y=-1000
etch cont x=248.5 y=.400
etch cont x=253.5 y=.400
etch cont x=253.5 y=-1000
etch cont x=258.5 y=-1000
etch cont x=258.5 y=.400
etch cont x=263.5 y=.400
etch cont x=263.5 y=-1000
etch cont x=268.5 y=-1000
etch cont x=268.5 y=.400
etch cont x=273.5 y=.400
etch cont x=273.5 y=-1000
etch cont x=278.5 y=-1000
etch cont x=278.5 y=.400
etch cont x=283.5 y=.400
etch cont x=283.5 y=-1000
etch cont x=288.5 y=-1000
etch cont x=288.5 y=.400
etch cont x=293.5 y=.400
etch cont x=293.5 y=-1000
etch cont x=298.5 y=-1000
etch cont x=298.5 y=.400
etch cont x=401.1 y=.400
etch done x=401.1 y=-1000
#
struct outfile=anelex26_1.str
#
rate.polish oxide machine=polish n.m soft=25 height.fac=0.02 \
length.fac=4.2 kinetic.fac=10
#
polish machine=polish time=2 min dx.mult=3 dt.fact=0.0005
```

```
#
struct outfile=anelex26_2.str
#
polish machine=polish time=3 min dx.mult=2 dt.fact=0.0005
#
struct outfile=anelex26_3.str
#
init infile=anelex26.str
#
etch oxide start x=-0.1 y=-1000
etch cont x=-0.1 y=.400
etch cont x=116 y=.400
etch cont x=116 y=-1000
etch cont x=126 y=-1000
etch cont x=126 y=.400
etch cont x=136 y=.400
etch cont x=136 y=-1000
etch cont x=146 y=-1000
etch cont x=146 y=.400
etch cont x=156 y=.400
etch cont x=156 y=-1000
etch cont x=166 y=-1000
etch cont x=166 y=.400
etch cont x=176 y=.400
etch cont x=176 y=-1000
etch cont x=186 y=-1000
etch cont x=186 y=.400
etch cont x=196 y=.400
etch cont x=196 y=-1000
etch cont x=206 y=-1000
etch cont x=206 y=.400
etch cont x=216 y=.400
etch cont x=216 y=-1000
etch cont x=226 y=-1000
etch cont x=226 y=.400
etch cont x=236 y=.400
etch cont x=236 y=-1000
etch cont x=246 y=-1000
etch cont x=246 y=.400
etch cont x=256 y=.400
etch cont x=256 y=-1000
etch cont x=266 y=-1000
etch cont x=266 y=.400
```

```
etch cont x=276 y=.400
etch cont x=276 y=-1000
etch cont x=286 y=-1000
etch cont x=286 y=.400
etch cont x=401.1 y=.400
etch done x=401.1 y=-1000
#
struct outfile=anelex26_4.str
#
polish machine=polish time=3 min dx.mult=2 dt.fact=0.0005
#
struct outfile=anelex26_5.str
#
polish machine=polish time=3 min dx.mult=3 dt.fact=0.0005
#
struct outfile=anelex26_6.str
#
init infile=anelex26.str
#
etch oxide start x=-0.1 y=-1000
etch cont x=-0.1 y=.400
etch cont x=131 y=.400
etch cont x=131 y=-1000
etch cont x=151 y=-1000
etch cont x=151 y=.400
etch cont x=171 y=.400
etch cont x=171 y=-1000
etch cont x=191 y=-1000
etch cont x=191 y=.400
etch cont x=211 y=.400
etch cont x=211 y=-1000
etch cont x=231 y=-1000
etch cont x=231 y=.400
etch cont x=251 y=.400
etch cont x=251 y=-1000
etch cont x=271 y=-1000
etch cont x=271 y=.400
etch cont x=401.1 y=.400
etch done x=401.1 y=-1000
#
struct outfile=anelex26_7.str
#
polish machine=polish time=3 min dx.mult=2 dt.fact=0.0005
```

```
#
struct outfile=anelex26_8.str
#
polish machine=polish time=3 min dx.mult=2 dt.fact=0.0005
#
struct outfile=anelex26_9.str
#
tonyplot -overlay anelex26_1.str anelex26_2.str anelex26_3.str -set
  anelex26_1.set
#
tonyplot -overlay anelex26_4.str anelex26_5.str anelex26_6.str -set
  anelex26_2.set
#
tonyplot -overlay anelex26_7.str anelex26_8.str anelex26_9.str -set
  anelex26_3.set
#
quit
```

27.1.27. anelex27.in: Hard and Soft Pad Modeling Using the Soft Polish Model

This example shows the use of ELITE to model chemical mechanical polishing using the soft polish model. A hard polishing pad is simulated by setting `LENGTH.FAC = 30`. The soft pad uses `LENGTH.FAC = 4.2.` `LENGTH.FAC` is a measure of the flexibility of the pad.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.31: CMP over large scales. The x extent of the simulation is 1mm

Input File athena_elite/anelex27.in:

```
go athena

line x loc=0.000000 tag=left
line x loc=25.000000 spac=5.000000
line x loc=495.000000 spac=5.000000
line x loc=1020.000000 tag=right
line y loc=0.000000 spac=0.020000 tag=top
line y loc=0.100000 spac=0.050000
line y loc=0.500000 spac=0.080000
line y loc=1.000000 spac=0.100000
line y loc=5.000000 spac=3.000000 tag=bottom
init silicon orientation=100
#
deposit nitride thick=0.2

depo barrier thick=0.010000 divis=1 dy=0.005000 ydy=0.000000
etch barrier start x=-0.1 y=-1000
etch cont x=-0.1 y=1000
etch cont x=30 y=1000
```

```
etch cont x=30 y=-1000
etch cont x=230 y=-1000
etch cont x=230 y=1000
etch cont x=290 y=1000
etch cont x=290 y=-1000
etch cont x=490 y=-1000
etch cont x=490 y=1000
etch cont x=1020.1 y=1000
etch done x=1020.1 y=-1000
#
#
etch nitride dry thick=.3
etch silicon dry thick=.5
strip
#
deposit oxide thick=0.7
struct outfile=anelex27_1.str
# The hard polish
rate.polish nitride machine=anelex27 n.m soft=4 height.fac=0.02 \
length.fac=30 kinetic.fac=10
rate.polish oxide machine=anelex27 n.m soft=25 height.fac=0.02 \
length.fac=30 kinetic.fac=10
#
polish machine=anelex27 time=30 min dx.mult=1 dt.fact=0.0005
struct outfile=anelex27_2.str
init infile=anelex27_1.str
# The hard polish
rate.polish nitride machine=anelex27 n.m soft=4 height.fac=0.02 \
length.fac=4.2 kinetic.fac=10
rate.polish oxide machine=anelex27 n.m soft=25 height.fac=0.02 \
length.fac=4.2 kinetic.fac=10
polish machine=anelex27 time=30 min dx.mult=1 dt.fact=0.0005
```

```
struct outfile=anelex27_3.str
#
tonyplot anelex27_1.str anelex27_2.str anelex27_3.str
quit
```

27.1.28. anelex28.in: Multi-layer CMOS structure

This example demonstrates the use of a global shrink parameter with application to a simple multi layer interconnect structure sitting on top of some cmos transistors.

It takes around three hours to run on a sparc 20 - 75MHz cpu.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.32: Multiple ELITE process steps are used to form a complete device cross-section including all active and interconnect layers

Input File athena_elite/anelex28.in:

```
go athena
#define and use a global shrink variable...
set shrink=0.9
line x loc=0*$shrink spac=0.1
line x loc=5.6*$shrink spac=0.1
```

```

line y loc=0 spac=0.1
line y loc=1 spac=1
line y loc=2 spac=2

init silicon c.boron=1e14 spac=0.25

deposit oxide thick=0.01
deposit nitride thick=0.15 divi=2
deposit photo thick=0.5 divi=2

etch photo left p1.x=2.2*$shrink
etch nitride left p1.x=2.2*$shrink

implant boron energy=50 dose=1.2e13

strip

diffuse temp=1000 time=30 wet

strip nitride

implant phosph dose=1e13 energy=30

diffuse temp=1200 time=120 nitrogen

strip oxide

deposit oxide thick=0.01
deposit nitride thick=0.15 divi=5

etch nitride left p1.x=0.3*$shrink
etch nitride right p1.x=4.1*$shrink
etch nitride start x=1.7*$shrink y=-10
etch nitride continue x=1.7*$shrink y=10
etch nitride continue x=2.7*$shrink y=10
etch nitride  done    x=2.7*$shrink y=-10


implant boron dose=1e12 energy=30

method grid.oxide=0.05
diffuse temp=1000 time=20 wet
```

```

strip nitride

etch oxide dry thick=0.01

diffuse temp=900 time=10 dry

implant boron dose=1.4e12 boron energy=10 pear

deposit poly thick=0.2 divi=4

etch poly left p1.x=0.9*$shrink

etch poly start x=1.1*$shrink y=-10
etch continue x=1.1*$shrink y=10
etch continue x=3.3*$shrink y=10
etch done     x=3.3*$shrink y=-10

etch poly right p1.x=3.5*$shrink

structure outf=fpdetch.str

#NLDD Implant#####################################################
deposit photo thick=0.5 divi=3

etch photo left p1.x=2.2*$shrink

implant phos dose=1e13 energy=10 pear

strip

#PLDD Implant#####################################################
deposit photo thick=0.5 divi=3

etch photo right p1.x=2.2*$shrink

implant bf2 dose=1e13 energy=10 pear

strip

#Spacer Dep#####################################################

deposit oxide thick=0.1 divi=3
```

```
etch dry oxide thick=0.12
#######################################################
#N++ Implant###################################################
deposit photo thick=0.5 divi=3
etch photo left p1.x=2.2*$shrink
implant phos dose=1e15 energy=20 pear
strip
#P++ Implant###################################################
deposit photo thick=0.5 divi=3
etch photo right p1.x=2.2*$shrink
implant bf2 dose=1e15 energy=30 pear
strip
#######################################################
structure outf=trans.str
deposit nitride thick=0.02
#
rate.depo machine=bpsg1 material=bpsg a.s  cvd dep.rate=1.0 step.cov=0.80
#
deposit machine=bpsg1 time=4000 seconds divisions=5
deposit photo thick=0.1 divi=2
etch photo start x=0.5*$shrink y=-10
etch continue    x=0.5*$shrink y=10
etch continue    x=0.7*$shrink y=10
etch done        x=0.7*$shrink y=-10
etch photo start x=1.3*$shrink y=-10
etch continue    x=1.3*$shrink y=10
etch continue    x=1.5*$shrink y=10
```

```
etch done        x=1.5*$shrink y=-10

etch photo start x=2.9*$shrink y=-10
etch continue    x=2.9*$shrink y=10
etch continue    x=3.1*$shrink y=10
etch done        x=3.1*$shrink y=-10

etch photo start x=3.7*$shrink y=-10
etch continue    x=3.7*$shrink y=10
etch continue    x=3.9*$shrink y=10
etch done        x=3.9*$shrink y=-10


#
rate.etch machine=mcc1 material=bpsg a.s  rie isotropic=0.02 dir=1.00 \
        chem=0.00 div=0.01
rate.etch machine=mcc1 nitride a.s  rie isotropic=0.01 dir=1.00 \
        chem=0.00 div=0.01
rate.etch machine=mcc1 oxide a.s  rie isotropic=0.01 dir=1.00 \
        chem=0.00 div=0.01

#
etch machine=mcc1 time=5000 seconds

structure outf=mccetch.str
strip


#
rate.depo machine=met1 tung a.s  sigma.dep=0.60 unidirec dep.rate=1.0 \
        angle1=0.00
#
deposit machine=met1 time=4000 seconds divisions=5

structure outf=met1dep.str

deposit photo thick=0.1 divi=2

etch photo start x=0.9*$shrink y=10
etch continue    x=1.1*$shrink y=10
etch continue x=1.1*$shrink y=-10
etch done    x=0.9*$shrink y=-10

```

```
etch photo start x=3.3*$shrink y=-10
etch continue    x=3.3*$shrink y=10
etch continue x=3.5*$shrink y=10
etch done    x=3.5*$shrink y=-10

#
rate.etch machine=fmd1 tung a.s  rie isotropic=0.20 dir=1.00 chem=0.00 \
        div=0.01
#
etch machine=fmd1 time=4000 seconds

structure outf=fmd1.str

strip

deposit nitride thick=0.05 divi=1



rate.depo machine=bsg material=bsg a.s  cvd dep.rate=1.0 step.cov=0.80
#
deposit machine=bsg time=2000 seconds divisions=5

deposit oxide thick=0.4

etch above p1.y=-1.09

deposit material=bpsg thick=0.3 divi=3

structure outf=ml.str

deposit photo thick=0.1
etch photo start x=2. y=-10
etch continue    x=2 y=10
etch continue    x=2.4 y=10
etch done        x=2.4 y=-10

etch photo start x=4.8 y=-10
etch continue    x=5.2 y=-10
etch continue    x=5.2 y=10
etch done        x=4.8 y=10

# define metal contact cut etch..
```

```
rate.etch machine=mcc2 oxynitr a.s  rie isotropic=0.18 dir=1 chem=0.00
  div=0.01
rate.etch machine=mcc2 oxide a.s  rie isotropic=0.01 dir=1 chem=0.00
  div=0.01
rate.etch machine=mcc2 material=bsg a.s  rie isotropic=0.01 dir=1
  chem=0.00 div=0.01
rate.etch machine=mcc2 material=bpsg a.s  rie isotropic=0.05 dir=1
  chem=0.00 \
        div=0.01
rate.etch machine=mcc2 nitride a.s  rie isotropic=0.01 dir=1 chem=0.00
  div=0.01

#
etch machine=mcc2 time=6000 seconds

structure outf=mcc2.str

etch photo above p1.y=-1.2

####################################################################
rate.depo machine=tung tung a.s  cvd dep.rate=1.0 step.cov=0.60
#
deposit machine=tung time=4000 seconds divisions=5

etch tung dry thick=0.4
####################################################################
#
rate.depo machine=aldep alumin a.s  cvd dep.rate=1.0 step.cov=0.60
#
deposit machine=aldep time=4000 seconds divisions=5

deposit photo thick=0.15

etch photo start x=3*$shrink y=-10
etch continue x=3*$shrink y=10
etch continue x=3.5*$shrink y=10
etch done     x=3.3*$shrink y=-10


rate.etch machine=fmd3 alumin a.s  rie isotropic=0.40 dir=1.00 chem=0.00 \
        div=0.01
#
etch machine=fmd3 time=4200 seconds

```

```
strip photo
rate.depo machine=oxydep oxynitr a.s  cvd dep.rate=1.0 step.cov=0.80
#
deposit machine=oxydep time=6000 seconds divisions=5
structure mirror right
structure outf=anelex28.str
tonyplot
```

27.1.29. anelex29.in: Simple Plasma Etched Trench with Sidewall Implant

This example demonstrates the use of plasma etching to etch a simple trench structure.

The example etch machine uses a Monte Carlo simulated plasma model to describe the angular ion distribution incident on the substrate surface. At each point on the surface, an angular “window” of visibility is calculated. This window of visibility is used in the integration of the angular distribution to calculate the etch rate.

Each surface string is then moved and remeshed at the end of the etch step. This windowing, ray tracing technique is highlighted in this example if you change the SPR500 resist thickness.

After trench etch, the structure is implanted at an angle and the trench is oxidized. This highlights the ability of ELITE to perform physical etch alongside oxidation and implant functions.

In the first `rate.etch` command, the parameter for the plasma etcher are defined. These parameters define the characteristics of the machine and are used to calculate the ion angular and energy distributions which, in turn, are used to calculate the ion flux at each point on the surface during etching.

The first parameter, `plasma,` is used to select the plasma etching model.

The parameters for the Monte Carlo plasma etching simulation are `pressure` in mT which will effect the mean free path and therefore, the fractional angular window “seen” by the plasma by a given surface string element.

`tgas` is the gas temperature in Kelvin and and will effect the pressure and hence the mean free path calculation.

`tion` is the temperature of the ion and is typically at a higher value that that of the gas.

`vpdc` is the specified DC bias on the cathode electrode. This will control the energy and angular anisotropy of the incoming ions energy.

`vpac` is the AC component of the electrodes bias.

`lshdc` in meters is the mean debye sheath thickness.

`lshac` is the alternating component of the sheath thickness in meters.

`freq` is the AC frequency of the RF power supply in MegaHertz.

`nparticles` is the number of Monte Carlo particles used in the calculation.

mgas is the gas atomic mass in AMU's. This is used in the collision calculation and will effect the momentum transfer characterists.

mion is the atomic mass of the ion in AMU's.

elec.model selects the type of plasma electric field model to be used. Here there are four options:

constant, linear, collision, child. These four models respectively include Constant, Linear electric field models and then Collision dominated and collisionless Child-Langmuir models for the spatial dependence of the electric field.

energy.div specifies the number of energy divisions for the energy distribution.

angle.div specifies the number of angular quantizations for the angular energy distribution.

qio is the momentum transfer cross section in meters^2.

qcht is the charge exchange cross section in meters^2.

Subsequent rate.etch statements define the parameters for each of the materials etched.

When the plasma model is invoked, the angular window calculation is very important to the final profile. The Thickness of the photoresist, in this example, will determine largely the angular ion energy distribution via a resist shadowing calculation. Thus, making the resist thicker will result in less 'bowing' of the sidewall. Try making the resist thinner to say, 0.1um to see this effect.

Execution takes less than five minutes to run on a sparc 20 - 75MHz CPU.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.33: Trench simulated using the Plasma Etch Model



Figure 27.34: Sidewall implant into plasma-etched trench using ELITE and SSUPREM4

Input File athena_elite/anelex29.in:

```
#
# $Id$
#
# Example demonstrating the plasma etching module of Athena.
# This example plasma etches a simple trench structure which
# is then implanted with boron followed by a short anneal.
#

go athena

#
# Define the simulation domain with mesh spacing
# First in the x direction (along the surface)
Line X Location=0.0 Spacing=0.20
Line X Location=0.5 Spacing=0.05
Line X Location=1.0 Spacing=0.05
Line X Location=1.5 Spacing=0.05
Line X Location=2.0 Spacing=0.05
Line X Location=2.5 Spacing=0.05
```

```
Line X Location=3.0 Spacing=0.20
#
# And now the y direction (depth)
Line Y Location=0.0 Spacing=0.02
Line Y Location=2.5 Spacing=0.1

#
# Initialize the substrate
Initialize C.Boron=1e16

#
# Deposit a layer of photo-resist
Deposition Material=SPR500 Thickness=0.5 Divisions=12

#
# Mimic resist exposure and develop via geometrical etching
Etch Material=SPR500 Start X=1.1 Y=-10
Etch                 Cont  X=1.1 Y= 10
Etch                 Cont  X=1.9 Y= 10
Etch                 Done  X=1.9 Y=-10

#
# Define the parameter of the plasma etcher
# See the Athena manual for parameter details
Rate.Etch Machine=PEMach \
     Plasma \
     Pressure = 300 \
     Tgas = 300.0 \
     Tion = 3000.0 \
     Vpdc = 32.5 \
     Vpac = 32.5 \
     Lshdc = 0.005 \
     Lshac = 0.0 \
     Freq = 13.56 \
     Nparticles = 4000 \
     Mgas = 40.0 \
     Mion = 40.0 \
     Constant \
     Energy.Div = 50 \
     Angle.Div = 60 \
     Qio = 1.7e-19 \
     Qcht = 2.1e-19 \

```

```
#
# Define the plasma etch parameters for silicon
Rate.Etch Machine=PEMach \
     Plasma \
     Material=Silicon \
     k.i = 1.1

#
# Define the plasma etch parameters for the photo-resist
Rate.Etch MAchine=PEMach \
     Plasma \
     Material=SPR500 \
     k.i = 0.1

#
# Etch the trench using the plasma etcher PEMach
Etch Machine=PEMach Time=3.0 minutes Dx.Mult=0.5

#
# Save the trench structure
Structure Outfile=anelex29_0.str

#
# And plot it
tonyplot -set anelex29_0.set anelex29_0.str

#
# Implant Boron into the trench
Implant Boron Dose=1.0e14 Energy=55 Tilt=45

#
# Save the implanted trench structure
Structure Outfile=anelex29_1.str

#
# Remove the remaining photo-resist
Etch Material=SPR500 All

#
# Grow a thin oxide
Diffuse Temperature=900 Time=10 WetO2

#
```

```
# Save the oxidize structure
Structure Outfile=anelex29_2.str

#
# Plot the plasma etched structure
tonyplot -set anelex29_2.set anelex29_2.str

#
# Finished, exit Athena
Quit
```

27.1.30. anelex30.in: Dopant Concentration Enhanced Etch Rate

This example demonstrates the use of dopant enhanced etching, using the ELITE module of ATHENA. A simple trench is etched into the silicon substrate with half of the region of the trench passing through high concentration Phosphorus. A normal etch is first performed giving a symmetrical trench structure. The simulation is then re-initialized to the pre-etch state and Phosphorus enhanced etching is defined for the etch machine. The trench is etched again, but this time the final trench structure is asymmetrical because the etch rates were enhanced as the surface moved through the high concentration Phosphorus region.

The dopant enhanced etch rate is defined by the equation

```
ERenh = (1 + enh) ER
```

where ER is the normal (un-enhanced) etch rate and enh is the enhancement due to the dopant. This enhancement is calculated as a function of the dopant using the equation:

```
enh = 0.5*Enh.Max*(tanh(Enh.Scale*(S - Enh.MinC)) + 1)
```

where the parameters have the following meanings

`Enh.Max` defines the maximum enhancement factor

`Enh.MinC` gives the solution value below which enhancement decays and

`Enh.Scale` gives the spread of the enhancment over solution values, i.e., how quickly does the enhancement factor reach its maximum?

For exponentially varying solutions, `S` and `Enh.MinC` are used as a logarithm.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.

This example takes about 15 minutes to run on a Sparc 20.



Figure 27.35: Plasma Etch with doping enhancement



Figure 27.36: Comparison of etch shape with and without doping enhancement

Input File athena_elite/anelex30.in:

```
#
# $Id$
#
# Example Athena Input File demonstrating the dopant enhancement
# features of the Athena etching module.
#
# Note:
#  The process flow used in this example does not represent a
#  realistic process flow.  It is used purely to demonstrate the
#  dopant enhanced etching capabilities.
#

go athena

#
# Define the structure dimensions and the mesh spacing (density)
#   to use
# Define the mesh spacing in the x direction
Line X Location=0.0 Spacing=0.20
Line X Location=0.5 Spacing=0.05
Line X Location=1.0 Spacing=0.05
Line X Location=1.5 Spacing=0.05
Line X Location=2.0 Spacing=0.05
Line X Location=2.5 Spacing=0.05
Line X Location=3.0 Spacing=0.20
#
#  and the y direction
Line Y Location=0.0 Spacing=0.02
Line Y Location=2.5 Spacing=0.10

#
# Initialize the structure: For this example, the mesh spacing
#   defined by the Line statements above are scaled by 2 to speed
#   up the simulation
Initialize C.Boron=1E16 Space.Mult=2

#
# Deposit some nitride and etch a window in it.  The number of
#   mesh divisions used to mesh the deposited layer is 12
Deposition Nitride Thickness=0.5 Divisions=12

#
```

```
# Etch the left half of the structure
Etch Nitride Left P1.X=1.35

#
# Implant some Phosphorus.  The etch step below will be enhanced
#   by this implanted Phosphorus.
Implant Phosphorus Dose=1.0E14 Energy=120

#
# Remove the remaining Nitride
Strip Nitride

#
# Deposit a thin layer of oxide prior to the diffusion step.  Only 2
#   mesh layers are used in the deposited oxide
Deposition Oxide Thickness=0.05 Divisions=2

#
# Anneal the implanted Phosphorus for 45 minutes in an inert ambient
#   (Nitrogen)
Diffuse Time=45 Temperature=950 Nitrogen

#
# Plot the structure showing the Phosphorus concentration after
#   the anneal
tonyplot -set anelex30_1.set

#
# Strip the oxide
Strip Oxide

#
# Deposit some masking oxide
Deposition Oxide Thickness=0.25 Divisions=8

#
# Etch a window in the oxide to prepare for the deep etch of the silicon
Etch Oxide Start X=1.1 Y=-10.0
Etch       Cont  X=1.1 Y= 10.0
Etch       Cont  X=2.5 Y= 10.0
Etch       Done  X=2.5 Y=-10.0

#
```

```
# Define out etching machine.  This is a plasma etching machine, the long
#   list of parameters are needed to define the conditions of the plasma.
#   The name of the machine is PEMach
Rate.Etch Machine=PEMach \
          Plasma \
          Pressure = 50 \
          TGas = 300.0 \
          TIon = 3000.0 \
          Vpdc = 32.5 \
          Vpac = 32.5 \
          Lshdc = 0.005 \
          Lshac = 0.0 \
          Freq = 13.56 \
          NParticles = 10000 \
          MGas = 40.0 \
          MIon = 40.0 \
          Constant \
          Energy.Div = 50 \
          Angle.Div = 60 \
          Qio = 1.7e-19 \
          Qcht = 2.1e-19

#
# Define the characterization parameters for etching silicon using this
#   machine.  Only the parameters k.n, k.x and k.y need be defined for
#   material etched by this machine
Rate.Etch Machine=PEMach \
          Plasma \
          Silicon \
          k.i = 0.1

#
# Define the characterization parameters for etching oxide
Rate.Etch Machine=PEMach \
          Plasma \
          Oxide \
          k.i = 0.01

#
# Save the current structure before the silicon etch to allow a
#   comparison between doping enhanced etching and the normal
#   etching
Structure Outfile=anelex30_0.str
```

```
#
# Etch a trench in the silicon.  This etch is not dopant enhanced.
Etch Machine=PEMach Time=7.0 Minutes Dx.Mult=0.5
#
# Save the structure after etching for comparison later
Structure Outfile=anelex30_1.str
#
# Re-initialize the simulation to the point before the silicon etch
Initialize Infile=anelex30_0.str
#
# Define a doping enhancement based on the Phosphorus concentration
Rate.Dope Machine=PEMach \
          Material=Silicon \
          Impurity=Phosphorus \
          Enh.Max = 2 \
          Enh.Scale = 5.0 \
          Enh.MinC = 17
#
# Re-perform the silicon etch, this time with doping enhancement
Etch Machine=PEMach Time=7.0 Minutes Dx.Mult=0.5
#
# Save the structure after doping enhanced etch for comparison later
Structure Outfile=anelex30_2.str
#
# Plot the two etch result files to compare the results of the
#   doping enhanced etching
tonyplot -overlay anelex30_1.str anelex30_2.str -set anelex30_2.set
#
# Finished, exit Athena
Quit
```

27.1.31. anelex31.in: Oxidation Induced Stress Enhanced Etch rate

This example demonstrates the use of oxidation enhanced etching using the ELITE module of ATHENA. A simple trench is etched into the silicon substrate through a region near a field oxides bird's beak. This region contains two high S.XY stress areas. An etch machine is defined which includes oxidation enhanced etching. The two high stress areas are reflected in the final shape of the trench where the bottom of the trench is not flat.

The oxidation enhanced etch rate is defined by the same equation as the dopant enhanced etch rate:

```
ERenh = (1 + enh) ER
```

where ER is the normal (un-enhanced) etch rate and enh is the enhancement due to the dopant. This enhancment is calculated as a function of the dopant using the equation:

```
enh = 0.5*Enh.Max*(tanh(Enh.Scale*(S - Enh.MinC)) + 1)
```

where the parameters have the following meanings

`Enh.Max` defines the maximum enhancement factor

`Enh.MinC` gives the solution value below which enhancement decays and

`Enh.Scale` gives the spread of the enhancment over solution values, i.e., how quickly does the enhancement factor reach its maximum.

For exponentially varying solutions, e.g., oxidation stress, `S` and `Enh.MinC` are logged.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.

This example takes about 45 minutes to run on a Sparc 20.



Figure 27.37: Plasma Etched trench including stress enhanced etch rate



Figure 27.38: Stress from oxidation used during trench etch

```
Input File athena_elite/anelex31.in:
# $Id$
#
# Example Athena Input File demonstrating oxide induced silicon stress
# enhanced etching.
#
# Note: The process flow used in this example does not represent a
#       realistic process flow.  It is used purely to demonstrate etch
#       enhancement.
#

go athena

#
# Define the mesh spacing:  the bird's beak occurs around
# (0.0, 0.0) so the mesh is finer there:
Line Y Location= 0.0 Spacing=0.20
Line Y Location= 3.0 Spacing=0.60

Line X Location=-1.5 Spacing=0.20
```

```
Line X Location=-1.0 Spacing=0.10
Line X Location=-0.5 Spacing=0.06
Line X Location=-0.2 Spacing=0.06
Line X Location= 0.0 Spacing=0.05
Line X Location= 1.0 Spacing=0.10
Line X Location= 1.5 Spacing=0.20

#
# Initialize the structure
Initialize Orientation=100

#
# Deposit a pad Oxide
Deposition Oxide Thickness=0.04

#
# Deposit the Nitride mask
Deposition Nitride Thickness=0.16 Divisions=4

#
# Etch the Nitride to expose the oxide for field oxide growth
Etch Nitride Left P1.X= 0.0

#
# Perform the field oxidation
#  Include stress dependent oxidation coefficients
Oxide Stress.Dep=True
#  Use a viscous model for Nitride
Material Nitride Visc.0=1.8E15
#  Use a viscous model for Oxide
Method Viscous Oxide.Rel=1E-2

# Perform the wet oxidation
Diffuse Time=60 Temperature=1000 WetO2

#
# Perform an additional 0.1 minutes of oxidation with
#  Silicon stress dependence turned on
Method Viscous Oxide.Rel=1E-2 Skip.Sil=False
Diffuse Time=0.1 Temperature=1000 WetO2

#
# Strip the Nitride
```

```
Strip Nitride

# Deposit Nitride to mask for the trench etch
Deposition Nitride Thickness=0.16 Divisions=4

#
# Etch window in the Nitride
Etch Nitride Start X=0.0 Y=-10.0
Etch         Cont  X=0.0 Y= 10.0
Etch         Cont  X=0.5 Y= 10.0
Etch         Done  X=0.5 Y=-10.0

#
# Etch window in the Oxide
Etch Oxide Start X=0.0 Y=-10.0
Etch       Cont  X=0.0 Y= 10.0
Etch       Cont  X=0.5 Y= 10.0
Etch       Done  X=0.5 Y=-10.0

#
# Define the etcher machine
Rate.Etch Machine=rie.etcher \
          Silicon \
          RIE \
          Isotropic=0.01 Directional=0.1 u.m

#
# Define the stress enhancement for the "rie.etcher" machine
Rate.Dope Machine=rie.etcher \
          Material=Silicon \
          Impurity=S.xy \
          Enh.Max = 0.75 \
          Enh.Scale = 1.5 \
          Enh.MinC = 6

#
# Etch the substrate
Etch Machine=rie.etcher Time=10 Minutes

#
# Save the structure
Structure Outfile=anelex31_0.str
```

```
#
# Plot the resultant structure show regions of enhanced etching
tonyplot anelex31_0.str

#
# Quit Athena
Quit
```

27.1.32. anelex32.in: Metal Void Formation

This example demonstrates the formation of a void in a deposited metal layer. The input files defines a machine for the metal deposition into a contact. The machine is described using the hemishperical model.

To activate the formation of any voids in the metal layer, the parameter, `void`, is required on the deposit statement.

The resulting plot shows the void formation and that the subsequent deposition of a dielectric layer *does not fill the void.*

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.39: Metal Void formation during contact refill. Voids can be removed by reflow or opened up by etchback

Input File athena_elite/anelex32.in:

```
go athena
#
```

```
line x loc=0.00 spac=0.10
line x loc=2 spac=0.10
#
line y loc=0.00 spac=0.02
line y loc=0.1 spac=0.05

#
init silicon orientation=100
#
deposit oxide thick=0.50 divisions=5

etch oxide start x=0.8 y=-5
etch oxide cont x=0.8 y=5
etch oxide cont x=1.2 y=5
etch oxide done x=1.2 y=-5
#
rate.depo machine=xxx alumin a.m  sigma.dep=0.20 hemisphe dep.rate=1000 \
        angle1=90.00 angle2=-90.00
#
deposit machine=xxx time=6 minutes divisions=5 void

deposit oxide thickness=0.5 div=5

structure outf=anelex32.str

tonyplot  anelex32.str

quit
```

27.1.33. anelex33.in : Polymer Redeposition Effect

This example demonstrates basic capabilities of Monte Carlo Plasma Etching Module. It shows effect of polymer redeposition on etching of deep trench in silicon.

Oxide is used as a masking material. First, 0.4 micron opening is formed in masking oxide. Than, parameters of plasma for machine MCETCH are specified. Also, parameters of interaction of incoming ion fluxes as well as polymer flux with all materials in the structure including polymer are specified in the ETCH.RATE statement.

After that, 3 minutes etching is applied to intial structure with and without redeposition taken into account. Redeposition is controlled by the MC.REDEPO parameter.

The resulting plot compares shape of the trenches obtained with and without redeposition taken into account. It is seen that redeposition slows down etching process. Also resulting trench has positive slopes instead of a “barrel” shape.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.40: Effect of polymer redeposition on etching of deep trench. Left structure is simulation using MC plasma etch model with polymer redeposition. Right structure is the same but without polymer redeposition

Input Deck athena_elite/anelex33.in :

```

go athena

#TITLE: MC Plasma Etch Example.
#Demonstration of Polymer Redeposition Effect

# Substrate mesh definition
line y loc=0    spac=0.04
line y loc=2  spac=0.04



line x loc=-0.5   spac=0.1
line x loc=-0.2 spac=0.025
line x loc=0 spac=0.05

init orient=100

```

```
# Make 0.4 um opening in oxide mask
depo oxide thick=0.5 div=10
etch oxide right p1.x=-0.2
structure mirror right

# Save initial structure
struct outf=anelex33_0.str

#Specify parameters for MCETCH machine for all 3 participating materials

rate.etch machine=MCETCH silicon mc.plasma ion.types=2 mc.parts1=10000
  mc.norm.t1=10.0 mc.lat.t1 = 1.0 mc.ion.cu1=10 mc.etch1=1e-05
  mc.alb1=0.2 mc.plm.alb=0.9 mc.parts2=10000 mc.norm.t2=6.0 mc.lat.t2 =
  1.0 mc.ion.cu2=5 mc.etch2=1e-05 mc.alb2=0.2 mc.polympt=3000 mc.rflct-
  dif=0.5

# Plasma parameters ion.types=2 mc.parts1,  mc.norm.t1,  mc.lat.t1,
# mc.ion.cu1, mc.parts2,  mc.norm.t2,  mc.lat.t2,  mc.ion.cu2 should
# be specified only in one rate.etch statement

# Set etch rate for oxide (mask) to zero, while polymer reflection
  (mc.plm.alb)
# is set to small calue of 0.1

rate.etch machine=MCETCH oxide   mc.plasma   mc.etch1=0.0 mc.alb1=0.9 \
mc.etch2=0.0 mc.alb2=0.9 mc.plm.alb=0.1 mc.rflctdif=0.5

# Set etch rate for polymer equal to that of silicon, while reflection
# of polymer from already redeposited polymer (mc.plm.alb) much lower
# then for silicon but higher then for oxide

rate.etch machine=MCETCH material=polymer   mc.plasma   mc.etch1=1e-05 \
 mc.alb1=0.2  mc.etch2=1e-05 mc.alb2=0.2 mc.plm.alb=0.5 mc.rflctdif=0.5

# First run the machine with redeposition (mc.redepo=t)
# Time steps are selected twice as high as the default ( mc.dt.fact=2)

etch machine=MCETCH time=3 minutes mc.sm=0.001 mc.redepo=t mc.dt.fact=2

structure outf=anelex33_1.str remove.gas

# The same but without polymer redeposition

init inf=anelex33_0.str
```

```
etch machine=MCETCH time=3 minutes mc.sm=0.001 mc.redepo=f mc.dt.fact=2

structure outf=anelex33_2.str remove.gas

tonyplot anelex33_1.str anelex33_2.str

```

27.1.34. anelex34.in : Polymer Etch Rate and Reflection Effect

This example is similar to anelex33.in. It demonstrates how characteristics of redeposited polymer affect etching efficiency and shape of etched trench.

The effect of polymer redeposition could be increased either by decreasing of polymer etch rate (parameters `MC.ETCH1` and `MC.ETCH2`) or by decreasing of polymer particle reflection from polymer layer (parameter `MC.PLM.ALB`)

The resulting plot shows that in both cases trench becomes shallower and narrower.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.41: Effect of polymer characteristics on the shape of a deep trench. Left graphic is using MC plasma etch model; middle graphic has five times lower polymer etching rates; right graphic has five times lower particle reflection from the polymer material.

Input Deck athena_elite/anelex34.in :

```

go athena

#TITLE: MC Plasma Etch Example.
#Demonstration of Polymer Etch Rate Effect

# Substrate mesh definition
line y loc=0    spac=0.04
line y loc=2  spac=0.04



line x loc=-0.5   spac=0.1
line x loc=-0.2 spac=0.025
line x loc=0 spac=0.05

init orient=100

# Make 0.4 um opening in oxide mask
depo oxide thick=0.5 div=10
etch oxide right p1.x=-0.2
structure mirror right

# Save initial structure
struct outf=anelex34_1.str

#Specify parameters for MCETCH machine for all 3 participating materials

rate.etch machine=MCETCH silicon mc.plasma ion.types=2 mc.parts1=10000
  mc.norm.t1=10.0 mc.lat.t1 = 1.0 mc.ion.cu1=10 mc.etch1=1e-05
  mc.alb1=0.2 mc.plm.alb=0.9 mc.parts2=10000 mc.norm.t2=6.0 mc.lat.t2 =
  1.0 mc.ion.cu2=5 mc.etch2=1e-05 mc.alb2=0.2 mc.polympt=3000 mc.rflct-
  dif=0.5


rate.etch machine=MCETCH oxide   mc.plasma   mc.etch1=0.0 mc.alb1=0.9 \
mc.etch2=0.0 mc.alb2=0.9 mc.plm.alb=0.1 mc.rflctdif=0.5



# Set the etch rate for polymer equal to that of silicon

rate.etch machine=MCETCH material=polymer   mc.plasma   mc.etch1=1e-05 \
```

```
 mc.alb1=0.2  mc.etch2=1e-05 mc.alb2=0.2 mc.plm.alb=0.5 mc.rflctdif=0.5

etch machine=MCETCH time=3 minutes mc.sm=0.001 mc.redepo=t mc.dt.fact=2

structure outf=anelex34_2.str remove.gas


# The same as the first tun but with etch rate for polymer five times
# lower than that of silicon

init inf=anelex34_1.str
rate.etch machine=MCETCH material=polymer   mc.plasma   mc.etch1=2e-06 \
 mc.alb1=0.2  mc.etch2=2e-06 mc.alb2=0.2 mc.plm.alb=0.5 mc.rflctdif=0.5

etch machine=MCETCH time=3 minutes mc.sm=0.001 mc.redepo=t mc.dt.fact=2

structure outf=anelex34_3.str remove.gas


# The same as the first run but with five times lower reflection of polymer
# particles from already redeposited polymer material


init inf=anelex34_1.str

rate.etch machine=MCETCH material=polymer   mc.plasma   mc.etch1=1e-05 \
 mc.alb1=0.2  mc.etch2=1e-05 mc.alb2=0.2 mc.plm.alb=0.1 mc.rflctdif=0.5

etch machine=MCETCH time=3 minutes mc.sm=0.001 mc.redepo=t mc.dt.fact=2

structure outf=anelex34_4.str remove.gas
tonyplot  anelex34_2.str anelex34_3.str anelex34_4.str
```

27.1.35. anelex35.in : Effect of Plasma Characteristics

This example is similar to anelex33.in. It demonstrates how plasma characteristics affect etching efficiency and shape of etched trench. Fluxes of ions and neutrals leaving plasma sheath are represented by bimaxwell velocity distribution function. In this approximation higher relative normal plasma temperatures `MC.NORM.T1/MC.LAT.T1` and `MC.NORM.T2/MC.LAT.T2` result in particle fluxes less dispersed in the lateral direction.

In this example only one sort of particles is considered. Two normal to lateral temperature ratios are considered: first is 7:1 and second is 15:1.

The resulting plot shows that “narrower” plasma ion flux results in a narrower and deeper trench.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 27.42: Effect of plasma characteristics on the shape of a deep trench. Left graphic is using MC plasma etch model with normal to lateral plasma temperature ratio of 7:1. The right graphic uses a ratio of 15:1.

Input Deck athena_elite/anelex35.in :

```

go athena

#TITLE: MC Plasma Etch Example.
#Effect of Relative Plasma Temperature

# Substrate mesh definition
line y loc=0    spac=0.04
line y loc=2  spac=0.04



line x loc=-0.5   spac=0.1
line x loc=-0.2 spac=0.025
line x loc=0 spac=0.05

```

```
init orient=100

# Make 0.4 um opening in oxide mask
depo oxide thick=0.5 div=10
etch oxide right p1.x=-0.2
structure mirror right

# Save initial structure
struct outf=anelex35_0.str

#Specify parameters for MCETCH machine for all 3 participating materials

rate.etch machine=MCETCH silicon mc.plasma ion.types=1 mc.parts1=20000
  mc.norm.t1=14.0 mc.lat.t1 = 2.0 mc.ion.cu1=15 mc.etch1=1e-05
  mc.alb1=0.2 mc.plm.alb=0.5  mc.polympt=5000 mc.rflctdif=0.5


rate.etch machine=MCETCH oxide   mc.plasma   mc.etch1=0.0 mc.alb1=0.9 \
 mc.plm.alb=0.5 mc.rflctdif=0.5


rate.etch machine=MCETCH material=polymer   mc.plasma   mc.etch1=1e-05 \
 mc.alb1=0.2   mc.plm.alb=0.3 mc.rflctdif=0.5

etch machine=MCETCH time=3 minutes mc.sm=0.001 mc.redepo=t mc.dt.fact=2

structure outf=anelex35_1.str remove.gas


# The same but with higher normal plasma temperature

init inf=anelex35_0.str
rate.etch machine=MCETCH silicon mc.plasma ion.types=1 mc.parts1=20000
  mc.norm.t1=15.0 mc.lat.t1 = 1.0 mc.ion.cu1=15

etch machine=MCETCH time=3 minutes mc.sm=0.001 mc.redepo=t mc.dt.fact=2
structure outfile=anelex35_2.str remove.gas

tonyplot  anelex35_1.str anelex35_2.str
```

This page is intentionally left blank

28.1. ATHENA_OPTOLITH: Optical Lithography Simulation

28.1.1. anopex01.in: Aerial Image Simulation

This example demonstrates the first step in lithography simulation. A simple mask has been used: two elbows and a contact hole (cd=1 um). The stepper description corresponds to: Canon FRA-1550, g-line , NA=0.43, sigma=0.5.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.1: Contours of light intensity through an elbow mask

Input File athena_optolith/anopex01.in:

```
go athena
#
#                    OPTOLITH input file: anopex01.in
#                    --------------------------------
#
#                 Aerial image of a complex mask feature.
#
# The illumination wavelength
#
illumination g.line
```

```
#
# The shape of the illuminating source
#
illum.filter clear.fil circle sigma=0.3
#
# The projection system numerical aperture
#
projection na=.43
#
# The shape of the pupil of the projection system
#
pupil.filter clear.fil circle
#
# Define the mask : two elbows and a contact hole (cd=1.0 um)
#
layout lay.clear x.low=-2.5 z.low=-2.5 x.high=-1.5 z.high=2.5
layout x.low=-1.5 z.low=1.5 x.high=2.5 z.high=2.5
layout x.low=-.5 z.low=-2.5 x.high=.5 z.high=.5
layout x.low=.5 z.low=-.5 x.high=2.5 z.high=.5
layout x.low=1.5 z.low=-2.5 x.high=2.5 z.high=-1.5
#
# Calculation of the aerial image
#
image win.x.lo=-3 win.z.lo=-3 win.x.hi=3 win.z.hi=3 dx=.2 opaque
#
# Store the aerial image in a structure file
#
structure outfile=anopex01.str intensity
#
# Plot the aerial image during the run
#
tonyplot -st anopex01.str -set anopex01.set
#
quit
```

28.1.2. anopex02.in: Arbitrary Source Shapes

This example is a comparison of imaging using a SHRINC source, a pair of vertical, off-axis line sources, and a pair of horizontal, off-axis line sources. The source is defined and aerial images are then calculated for different defocus distances.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.2: Comparison of aerial images for three different source types

Input File athena_optolith/anopex02.in:

```
go athena
#
#                   OPTOLITH input file: anopex02.in
#                   --------------------------------
#
#           This example is a comparison between imaging using a SHRINC source,
#           a pair of vertical, off axis line sources and a pair of horizontal,
#         off axis line sources. The source is defined and aerial images are
#    then calculated for different defocus distances.
#
# _  _ _  _  ___________
#|_|      |_|| |      | ||___________|
#| |      | |
#| |      | |
#| |      | |
# _  _| |      | | ___________
#|_|      |_||_|      |_||___________|
#    (1)    (2)     (3)
#
```

```
#
# define projection system
#
projection na=.54
#
pupil.filter clear.fil circle
#
# define mask : set of three lines (0.6x4.0) um
#
layout lay.clear x.low=-0.15 z.low=-0.5 x.high=0.15 z.high=0.5
#
# define illumination system
#
illumination i.line
#
# The SHRINC type source
#
illum.filter clear.fil square sigma=0.15 radius=0.70711 angle=45
illum.filter square sigma=0.15 radius=0.70711 angle=-45
illum.filter square sigma=0.15 radius=0.70711 angle=135
illum.filter square sigma=0.15 radius=0.70711 angle=-135
#
# Run the image module without defocus.
#
image win.x.lo=-1 win.z.lo=0 win.x.hi=1 win.z.hi=0 dx=0.05 clear \
defocus=2
#
structure outfile=anopex02_1.str intensity
#
# The vertical line sources
#
illum.filter clear.fil square sigma=0.15 radius=0.5 angle=0
illum.filter square sigma=0.15 radius=0.5 angle=180
illum.filter square sigma=0.15 radius=0.70711 angle=45
illum.filter square sigma=0.15 radius=0.70711 angle=-45
illum.filter square sigma=0.15 radius=0.70711 angle=135
illum.filter square sigma=0.15 radius=0.70711 angle=-135
#
# Run the image module without defocus.
#
image win.x.lo=-1 win.z.lo=0 win.x.hi=1 win.z.hi=0 dx=0.05 clear \
defocus=2
#
```

```
structure outfile=anopex02_2.str intensity
#
# The horizontal line sources
#
illum.filter clear.fil square sigma=0.15 radius=0.5 angle=90
illum.filter square sigma=0.15 radius=0.5 angle=-90
illum.filter square sigma=0.15 radius=0.70711 angle=45
illum.filter square sigma=0.15 radius=0.70711 angle=-45
illum.filter square sigma=0.15 radius=0.70711 angle=135
illum.filter square sigma=0.15 radius=0.70711 angle=-135
#
# Run the image module without defocus.
#
image win.x.lo=-1 win.z.lo=0 win.x.hi=1 win.z.hi=0 dx=0.05 clear \
defocus=2
#
structure outfile=anopex02_3.str intensity
#
tonyplot -st anopex02_*.str
#
quit
```

28.1.3. anopex03.in: Planar Lithography with Phase Shift Masks

This example shows lithography with phase shift masks. The aerial image is calculated for a 0.3 um contact hole and a 0.3 contact hole with sub-imageable (0.1um) 180 degree phase shift outriggers. Inhibitor concentration is calculated as well as 2D resist development.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.3: Effect of phase shift mask on the image (left) and subsequent exposure and development of the resist (right)

Input File athena_optolith/anopex03.in:

```
go athena
#
#                    ATHENA Input File: anopex03.in
#                    ----------------------------
#
#      Aerial Image, Inhibitor Concentration and 2D Resist Development.
#
#      Mask layout: 0.3 um contact hole and 0.3 contact hole
#     with sub-imageable (0.1 um) 180 degree Phase Shifters (outriggers).
#
# Define the structure
#
line x loc=0 spac=0.1
line x loc=0.15 spac=0.01
line x loc=0.6 spac=0.1
#
line y loc=1.1 spac=1
line y loc=2.1 spac=1
#
```

```
init silicon
#
rate.develop name.resist= OiR32 i.line c.dill=0.026
#
deposit nitride thick=0.035
#
deposit name.resist=OiR32 thick=.8 divisions=30
#
structure outfile=anopex03.str
#
# Aerial image calculation
#
illumination i.line
#
illum.filter clear.fil circle sigma=0.38
#
projection na=.54
#
pupil.filter clear.fil circle
#
layout lay.clear x.lo=-0.15 z.lo=-0.15 x.hi=0.15 z.hi=0.15 phase=0.
  trans=1.
layout x.lo=-0.45 z.lo=0.45 x.hi=0.45 z.hi=0.55 phase=180. trans=1.
layout x.lo=-0.45 z.lo=-0.55 x.hi=0.45 z.hi=-0.45 phase=180. trans=1.
layout x.lo=-0.55 z.lo=-0.45 x.hi=-0.45 z.hi=0.45 phase=180. trans=1.
layout x.lo=0.45 z.lo=-0.45 x.hi=0.55 z.hi=0.45 phase=180. trans=1.
#
image win.x.lo=0 win.x.hi=0.6 win.z.lo=0 win.z.hi=0 dx=0.01 n.pupil=2
#
structure outfile=anopex03_1.str intensity
#
# Resist exposure
#
expose dose=230 num.refl=3
#
# Post exposure bake
#
bake time=45 temp=115
#
tonyplot
#
# Resist Development
#
```

```
develop mack time=60 steps=4 substeps=30
#
structure mirror left
#
structure outfile=anopex03_2.str
#
# Run image, exposure and development without phase shift outriggers
#
initialize infile=anopex03.str
#
layout lay.clear x.lo=-0.15 z.lo=-0.15 x.hi=0.15 z.hi=0.15 phase=0.
  trans=1.
#
image win.x.lo=0 win.x.hi=0.6 win.z.lo=0 win.z.hi=0 dx=0.01 n.pupil=2
#
structure outfile=anopex03_3.str intensity
#
# Resist exposure
#
expose dose=220 num.refl=10
#
tonyplot
#
# Post exposure bake
#
bake time=45 temp=115
#
# Resist Development
#
develop mack time=60 steps=4 substeps=30
#
structure mirror left
#
structure outfile=anopex03_4.str
#
tonyplot -st anopex03_*.str
#
quit
```

28.1.4. anopex04.in: Imaging Using an Annular Illuminator

This example compares DUV lithography using annular and circular apertures.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.4: Comparison of an ANNULAR SOURCE (left) with a CIRCULAR SOURCE (right) shows improved imaging. The graphs below are cross sections through the 2D image

Input File athena_optolith/anopex04.in:

```
go athena
#
#                    OPTOLITH input file: anopex04.in
#                    ------------------------------
#
#               Imaging using annular illumination aperture
#
# ANNULAR APERTURE
#
#
# Compensate for intensity difference with intensity=2
#
illumination intensity=2 duv.line
#
illum.filter clear.fil circle sigma=0.8
illum.filter circle in.radius=0 out.radius=0.7 transmit=0
#
```

```
projection na=.42
#
pupil.filter clear.fil circle
#
# define mask : three different cd's 0.8, 0.6, 0.4)
#
layout lay.clear x.low=-1.8 z.low=-0.4 x.high=-1. z.high=0.4
layout x.low=1 z.low=-0.2 x.high=1.4 z.high=0.2
layout x.low=-.3 z.low=-.3 x.high=.3 z.high=.3
#
image win.x.lo=-2 win.z.lo=-1 win.x.hi=2 win.z.hi=1 x.p=45 z.p=25
  n.pup=10
#
structure outfile=anopex04_0.str intensity
#
#
# CIRCULAR APERTURE
#
#
illumination intensity=1 duv.line
#
illum.filter clear.fil circle sigma=0.5
#
image win.x.lo=-2 win.z.lo=-1 win.x.hi=2 win.z.hi=1 x.p=45 z.p=25
  n.pup=10
#
structure outfile=anopex04_1.str intensity
#
tonyplot -st anopex04_0.str anopex04_1.str -set anopex04.set
#
quit
```

28.1.5. anopex05.in: Imaging Using Spatial Filtering Techniques

This example simulates DUV lithography using more complicated annular apertures.

The Input File is composed of two parts. The first part shows standard imaging, where the Aerial Image was generated using a circular aperture.

The second part of the Input File repeats the same simulation conditions, but it uses spatial filtering in the projection system with an annular aperture constructed from five zones.

Note the difference in aerial image data generated by the two parts of the Input File. The image for the filtered system shows an improved quality for smaller CDs when compared with the image for the circular aperture.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.5: Demonstration of the effect of spatial filtering on the image

Input File athena_optolith/anopex05.in:

```
go athena
#
#                    OPTOLITH input file: anopex05.in
#                    -------------------------------
#
#                Simulation of spatial filtering techniques
#
illumination duv.line
#
illum.filter clear.fil circle sigma=0.5
#
#
# WITHOUT SPATIAL FILTER
#
#
projection na=.42
#
pupil.filter clear.fil circle
#
```

```
# Define mask
#
layout lay.clear x.low=-2.0 z.low=-4.0 x.high=-1.0 z.high=4.0
layout x.low=0.0 z.low=-4.0 x.high=0.5 z.high=4.0
layout x.low=1.5 z.low=-4.0 x.high=1.8 z.high=4.0
layout x.low=2.8 z.low=-4.0 x.high=3.05 z.high=4.0
#
image win.x.lo=-3.4 win.z.lo=-.1 win.x.hi=3.4 win.z.hi=.1 x.p=100 z.p=2
#
structure outfile=anopex05_1.str intensity
#
#
# WITH SPATIAL FILTER
#
#
# Set intensity to approx. 2.0 to compensate for intensity loss
# caused by spatial filtering
#
illumination duv.line intensity=2
#
# Define the annular filters
#
pupil.filter clear.fil circle
pupil.filter circle in.rad=0.2  out.rad=0.5  trans=0.36
pupil.filter circle in.rad=0.5  out.rad=0.6  trans=0.49
pupil.filter circle in.rad=0.7  out.rad=0.8  trans=0.81
pupil.filter circle in.rad=0.0  out.rad=0.2  trans=0.25
pupil.filter circle in.rad=0.6  out.rad=0.7  trans=0.64
#
image win.x.lo=-3.4 win.z.lo=-.1 win.x.hi=3.4 win.z.hi=.1 x.p=100 z.p=2
#
structure outfile=anopex05_2.str intensity
tonyplot -st anopex05_1.str anopex05_2.str -set anopex05.set
#
quit
```

28.1.6. anopex06.in: Aberrated Aerial Image

This example simulates aerial image for a projection system that has 0.5 lambda “astigmatism”, and the position of the image patch under consideration is located at the upper right boundary of the circular image field.

The Input File is composed of two parts: simulation of the aerial image without and with aberrations (astigmatism).

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.6: Demonstration of the effect of astigmatism on the image. Astigmatism causes a loss of XY symmetry in the image

Input File athena_optolith/anopex06.in:

```
go athena
#
#
#                    OPTOLITH input file: anopex06.in
#                    -------------------------------
#
#                 Aerial Image simulation and
#                 aberrated aerial image (astigmatism)
#                 simulation.
#
illumination g.line
#
# Circular aperture
#
illum.filter clear.fil circle sigma=0.5
#
projection na=.43
```

```
#
pupil.filter clear.fil circle
#
# define mask
#
layout lay.clear x.low=-.3 z.low=-.3 x.high=.3 z.high=.3
#
# Image without astigmatism
#
image win.x.lo=-1 win.z.lo=-1 win.x.hi=1 win.z.hi=1 x.p=25 z.p=25 n.pu-
  pil=10
#
structure outfile=anopex06_1.str intensity
#
# Image with astigmatism
#
aberration astigmatism=0.5 x.field=.7 z.field=.7
#
image win.x.lo=-1 win.z.lo=-1 win.x.hi=1 win.z.hi=1 x.p=25 z.p=25 n.pu-
  pil=10
#
structure outfile=anopex06_2.str intensity
#
tonyplot -st anopex06_1.str anopex06_2.str -set anopex06.set
#
quit
```

28.1.7. anopex07.in: SHRINC from Circular Sources

This example shows the use of arbitrary source positions to make a SHRINC-type source.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.7: Demonstration of imaging from several circular sources emulating a SHRINC source

Input File athena_optolith/anopex07.in:

```
go athena
#
#                    OPTOLITH input file: anopex07.in
#                    --------------------------------
#
#               Imaging using SHRINC illumination aperture
#
# Illumination system
#
illumination lambda=.365
#
# Define the Shrinc illuminator using the arbitrary source technique
#
# Define the circular illuminator
#
illum.filter clear.fil circle sigma=0.5
#
projection na=.54
#
```

```
pupil.filter clear.fil circle
#
# Define the mask
#
layout lay.clear x.lo=-1.575 z.lo=-1.50 x.hi=-1.225 z.hi=1.50
layout x.lo=-0.875 z.lo=-1.50 x.hi=-0.525 z.hi=1.50
layout x.lo=-0.1750 z.lo=-1.50 x.hi=0.175 z.hi=1.50
layout x.lo=0.525 z.lo=-1.50 x.hi=0.875 z.hi=1.50
layout x.lo=1.225 z.lo=-1.50 x.hi=1.575 z.hi=1.50
#
# Run the imaging module
#
image win.x.lo=-0.70 win.z.lo=-0.1 win.x.hi=0.70 win.z.hi=0.1 x.p=51
  z.p=3 n.pupil=10 defocus=-1
#
structure outfile=anopex07_1.str intensity
#
# Clear the circular illuminator and define the Shrinc illuminator using
  the
# arbitrary source technique
#
illum.filter clear.fil circle sigma=0.22 radius=.47 angle=45
illum.filter circle sigma=0.22 radius=.47 angle=135
illum.filter circle sigma=0.22 radius=.47 angle=225
illum.filter circle sigma=0.22 radius=.47 angle=315
#
image win.x.lo=-0.70 win.z.lo=-0.1 win.x.hi=0.70 win.z.hi=0.1 x.p=51
  z.p=3 n.pupil=10 defocus=-1
#
structure outfile=anopex07_2.str intensity
tonyplot -st anopex07_1.str anopex07_2.str -set anopex07.set
#
quit
```

28.1.8. anopex08.in: SHRINC/QUEST Imaging Comparison

This example simulates i-line lithography using a quadrupole illuminator. The Input File generates aerial image data with and without the SHRINC illuminator to determine the difference in results between SHRINC and circular apertures.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.8: Demonstration of imaging from a true SHRINC source

Input File athena_optolith/anopex08.in:

```
go athena
#
#                    OPTOLITH input file: anopex08.in
#                    --------------------------------
#
#               Imaging using SHRINC illumination aperture
#
# Define the circular illuminator
#
illumination lambda=.365
#
illum.filter clear.fil circle sigma=0.5
#
# Projection numerical aperture
#
projection na=.54
#
# Projection aperture
#
```

```
pupil.filter clear.fil circle
#
# Define the mask
#
layout lay.clear x.lo=-1.575 z.lo=-1.50 x.hi=-1.225 z.hi=1.50
layout x.lo=-0.875 z.lo=-1.50 x.hi=-0.525 z.hi=1.50
layout x.lo=-0.1750 z.lo=-1.50 x.hi=0.175 z.hi=1.50
layout x.lo=0.525 z.lo=-1.50 x.hi=0.875 z.hi=1.50
layout x.lo=1.225 z.lo=-1.50 x.hi=1.575 z.hi=1.50
#
# Run the imaging module
#
image win.x.lo=-0.70 win.z.lo=-0.1 win.x.hi=0.70 win.z.hi=0.1 x.p=51
  z.p=3 n.pupil=10 defocus=-1
#
structure outfile=anopex08_1.str intensity
#
# Define the SHRINC illuminator
#
illum.filter
#
illum.filter clear.fil shrinc sigma=0.22 radius=.47 angle=45
#
image win.x.lo=-0.70 win.z.lo=-0.1 win.x.hi=0.70 win.z.hi=0.1 x.p=51
  z.p=3 n.pupil=10 defocus=-1
#
structure outfile=anopex08_2.str intensity
#
tonyplot -st anopex08_1.str anopex08_2.str -set anopex08.set
#
quit
```

28.1.9. anopex09.in: Defocus Looped 2D Resist Development

This example simulates looped defocus (1.5,0.0,and -1.5 um) to create 2D photoresist profiles.

Note that resist profiles simulated under the same defocus amount but with opposite sign (1.5 and -1.5um) have different shapes. In this simulation, Mack's development model has been used, and the Post-Exposure Bake (PEB) module has been run. This example demonstrates the application of user defined materials, as well as user defined photoresists.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.9: Effects of defocus on final developed resist profiles

Input File athena_optolith/anopex09.in:

```
go athena
#
#                    OPTOLITH input file: anopex09.in
#                    --------------------------------
#
#            2d  development simulation (with PEB)
#            for 3 different focus planes (1.5 , 0.0, and -1.5 um)
#            using:
#            - looping procedure
#            - Mack development model
#
# define illumination system
#
illumination h.line
#
illum.filter clear.fil circle sigma=0.5
#
# define projection system
#
projection na=.45
#
```

```
pupil.filter clear.fil circle
#
# define mask : set of three lines (0.6x4.0) um
#
layout lay.clear x.low=-0.30 z.low=-0.9 x.high=0.30 z.high=0.9
#
# Define the structure
#
line x loc=0 spac=0.1
line x loc=.3 spac=0.01
line x loc=0.9 spac=0.1
#
line y loc=1.675 spac=1.0
line y loc=2.7 spac=1.0
#
init silicon
#
# Deposit user defined materials XXX and YYY
#
deposit material=XXX  thick=0.15 div=1 min.space=0.01
deposit material=YYY  thick=0.03 div=1 min.space=0.01
#
# Define exposure parameters for the Dill exposure model and development
# rate parameters for the Mack development rate model for the photoresist
# ZZZ.
#
rate.develop name.resist=ZZZ h.line a.dill=0.83 b.dill=0.03 c.dill=0.016\
rmax.mack=0.1 rmin.mack=0.001 mth.mack=0.3 n.mack=5 Dix.0 = 7.55e-13
  Dix.E = 3.34e-2
#
# Define index of refraction of user defined materials XXX and YYY as well
  as
# for the user defined photoresist ZZZ.
#
optical material=XXX h.line refrac.real=2.05 refrac.imag=0
optical material=YYY h.line refrac.real=1.47 refrac.imag=0
optical photoresist name.resist=ZZZ h.line refrac.real=1.7
#
# Deposit user defined photoresist ZZZ
#
deposit photoresist name.resist=ZZZ thick=1.5 div=45 min.space=0.01
#
structure outfile=anopex09_1.str
#
```

```
# Set looping statement for defocus looping
# from -1.5 um (above phototoresist surface)
# to 1.5 um (beneath phototoresist surface),
# with increment -1.5 um
#
foreach JJ ( -1.5 to 1.5 step 1.5)
#
# Start of the loop
#
init infile=anopex09_1.str
#
# Run the image module with defocus.
#
image win.x.lo=0 win.z.lo=0.0 win.x.h=0.9 win.z.h=0.0 dx=0.05 clear \
 n.pupil=2 defocus=JJ one.d
#
# Run exposure module
#
expose dose=150
#
# Run post exposure bake
#
bake time=45 temp=115
#
structure outf=anopex09_2_JJ.str
#
# Run development
#
develop mack time=90 steps=6 substeps=30
#
structure mirror left
#
structure outfile=anopex09_3_JJ.str
#
# End of the loop
#
end
#
tonyplot -st anopex09_3_*.str
#
quit
```

28.1.10. anopex10.in SHRINC Source for 45 Degree Lines

This example shows the effect of a SHRINC imaging system on two sets of lines that are at 45 degree angles to each other.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.10: Imaging of 45 degree layouts using a SHRINC source. These sources produce a poorer image for the 45 degree lines.

Input File athena_optolith/anopex10.in:

```
go athena
#
#                    OPTOLITH input file: anopex10.in
#                    --------------------------------
#
#               Imaging using SHRINC illumination aperture
#        This example illustrates the effect of a SHRINC
#        type source on two sets of lines at 45 degree angles to
#        each other.
#
# Illumination system
#
illumination lambda=.365
```

```
#
illum.filter clear.fil shrinc sigma=0.22 radius=.47 angle=45
#
# Projection system
#
projection na=.54
#
pupil.filter clear.fil circle
#
# Run the imaging module with a mask input file from Maskviews
#
image inf=anopex10.sec dx=0.1 defocus=-1
#
structure outfile=anopex10.str intensity
tonyplot -st anopex10.str -set anopex10.set
#
quit
```

28.1.11. anopex11.in: Swing Curves for CD Variation with Resist Thickness

This example demonstrates CD variation as a function of resist thickness. This is due to reflection effects inside the resist. This example should be compared with the results from anopex12.in. The example generates a swing curve for resist thickness variation from 0.9 to 1.2 microns. To plot the swing curve, type: tonyplot -da anopex11.data.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.11: Image for the simple single feature mask



Figure 28.12: Image for the simple single feature mask

Input File athena_optolith/anopex11.in:

```
go athena
#
#                   OPTOLITH input file: anopex11.in
#                   --------------------------------
#
#                 Exposure using an without AquaTAR.
#
# Illumination wavelength
#
illumination lambda= 0.365
#
illum.filter clear.fil circle sigma=0.5
#
# Projection numerical aperture
#
projection na=.52
#
# Projection aperture
#
pupil.filter clear.fil circle
#
# The mask layout
#
layout lay.clear x.lo=-.2 z.lo=-3 x.hi=.2 z.hi=3
#
# Run the imaging module
#
image clear win.x.lo=0.0 win.z.lo=0.0 win.x.hi=0.5 win.z.hi=0.0 \
      dx=0.05 one.d
#
structure outfile=anopex11.str intensity
#
tonyplot -st anopex11.str
#
# Define the structure
#
line x loc=0 spac=0.05
line x loc=0.2 spac=0.01
line x loc=0.5 spac=0.05
#
line y loc=1 spac=1.0
line y loc=2 spac=1.0
```

```
#
init silicon orientation=100
#
deposit oxide thick=0.08 div=1 min.space=0.01
deposit aluminum thick=1.04 div=3 min.space=0.01
#
structure outfile=anopex11_1.str

rate.dev name.resist=OiR897i i.line c.dill=0.018

#
# CD data will be put into a file called anopex11.data
#
printf ATHENA > anopex11.data
printf 31 2 2 > anopex11.data
printf THICKNESS > anopex11.data
printf CDs > anopex11.data
#
foreach j (.9 to 1.2 step 0.01)
#
# Beginning of the loop.
#
initialize infile=anopex11_1.str
#
deposit name.resist=OiR897i thick=j div=40 min.space=0.01
#
expose dose=250 num.refl=10 na=0
#
bake time=60 temp=125
#
#structure outfile=anopex11_2_j.str
#
develop kim time=60 steps=5 substeps=24
#
structure mirror left
#
#structure outfile=anopex11_3_j.str
#
printf j ( photo|gas(-0.5) - gas|photo(-0.5) ) > anopex11.data
#
# End of the loop.
#
end
```

```
#
tonyplot -da anopex11.data
#
quit
```

28.1.12. anopex12.in: Use of AquaTAR for CD Variation with Resist Thickness

This example demonstrates the use of an AquaTAR to minimize reflection effects inside the resist. The example generates a swing curve for resist thickness variation from 0.9 to 1.2 microns. To plot the swing curve, type: `tonyplot -da anopex11.data.`

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.13: Swing Curve of CD versus resist thickness for a single feature using AquaTAR. This result can be compared to the previous example to show improvement in CD variation.

Input File athena_optolith/anopex12.in:

```
go athena
#
#                    OPTOLITH input file: anopex12.in
#                    --------------------------------
#
#                  Exposure using AquaTAR.
#
# Illumination wavelength
```

```
#
illumination lambda= 0.365
#
illum.filter clear.fil circle sigma=0.5
#
# Projection numerical aperture
#
projection na=.52
#
# Projection aperture
#
pupil.filter clear.fil circle
#
# The mask layout
#
layout lay.clear x.lo=-.2 z.lo=-3 x.hi=.2 z.hi=3
#
# Run the imaging module with a mask input file from Maskviews
#
image clear win.x.lo=0.0 win.z.lo=0 win.x.hi=0.5 win.z.hi=0 dx=0.05 one.d
#
structure outfile=anopex12.str intensity
#
tonyplot -st anopex12.str
#
# Define the structure
#
line x loc=0 spac=0.05
line x loc=0.2 spac=0.01
line x loc=0.3 spac=0.01
line x loc=0.5 spac=0.05
#
line y loc=1 spac=1.0
line y loc=2 spac=1.0
#
init silicon orientation=100
#
deposit oxide thick=0.08 div=1 min.space=0.01
deposit aluminum thick=1.04 div=3 min.space=0.01
#
# Define exposure parameters for the Dill exposure model and development
# rate parameters for the Mack development rate model for the TAR, AZ Aqua-
  TAR.
```

```
#
rate.develop name.resist=AquaTAR i.line a.dill=0.83 b.dill=0.03 c.dill=0.014\
rmax.mack=0.15 rmin.mack=0.15 mth.mack=0.0 n.mack=1.01\
r1.kim = 0.50 r2.kim = 0.50 r3.kim = 15.0 \
r4.kim = 0.0 r5.kim = 0.0 r6.kim = 0.0 r7.kim = 0.0 \
r8.kim = 0.0 r9.kim = 0.0 r10.kim = 0.0 \
        Dix.0 = 7.55e-13 Dix.E = 3.34e-2
#
# Define index of refraction of user defined TAR AZ AquaTAR which is entered
# as a photoresist.
#
optical photoresist name.resist=AquaTAR i.line refrac.real=1.41 \
refrac.imag=0.0

rate.dev name.resist=OiR897i i.line c.dill=0.018


#
structure outfile=anopex12_1.str
#
# CD data will be put into a file called anopex12.data
#
printf ATHENA > anopex12.data
printf 31 2 2 > anopex12.data
printf THICKNESS > anopex12.data
printf CDs > anopex12.data
#
foreach j (.9 to 1.2 step 0.01)
#
# Beginning of the loop.
#
initialize infile=anopex12_1.str
#
deposit name.resist=OiR897i thick=j div=40 min.space=0.01
#
deposit photoresist name.resist=AquaTAR thick=0.0647 div=1 min.space=0.01
#
expose dose=250 num.refl=6 na=0
#
bake time=60 temp=125
#
#structure outfile=anopex12_2_j.str
#
```

```
develop kim time=60 steps=5 substeps=24
#
structure mirror left
#
#structure outfile=anopex12_3_j.str
#
printf j ( photo|gas(-0.5) - gas|photo(-0.5) ) > anopex12.data
#
# End of the loop.
#
end
#
tonyplot -da anopex12.data
#
quit
```

28.1.13. anopex13.in: Demonstration of Multilayer Resists

This example demonstrates the use of multiple resists for a possible contrast enhancement layer.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.14: Exposure and Development of MultiLayer Photoresist

Input File athena_optolith/anopex13.in:

```
go athena
#
#                   OPTOLITH input file: anopex13.in
#                   --------------------------------
#
#            Example of the use of contrast enhancement layers
#
# Define the structure
#
line x loc=0.00 spac=0.2 tag=left
line x loc=0.50 spac=0.025
line x loc=2.00 spac=0.2 tag=right
#
line y loc=1. spac=0.5 tag=top
line y loc=2.0 spac=0.5 tag=bottom
#
init silicon orientation=100
#
# Define exposure parameters for the Dill exposure model and development
```

```
# rate parameters for the Kim development rate model for the photoresists
# AAA and BBB.
#
rate.develop name.resist=AAA  i.line\
    r1.kim = 0.085329 r2.kim = 0.000002 r3.kim = 11.74276 \
    r4.kim = 0.0 r5.kim = 0.0 r6.kim = 0.0 r7.kim = 0.0 \
    r8.kim = 0.0 r9.kim = 0.0 r10.kim = 0.0 \
    a.dill =  0.73 b.dill =  0.07 c.dill =  0.018 \
    Dix.0 = 7.55e-13 Dix.E = 3.34e-2
rate.develop name.resist=BBB  i.line\
    r1.kim = 0.085329 r2.kim = 0.000002 r3.kim = 11.74276 \
    r4.kim = 0.0 r5.kim = 0.0 r6.kim = 0.0 r7.kim = 0.0 \
    r8.kim = 0.0 r9.kim = 0.0 r10.kim = 0.0 \
    a.dill =  0.73 b.dill =  0.07 c.dill =  0.018 \
    Dix.0 = 7.55e-13 Dix.E = 3.34e-2
#
# Define index of refraction of user defined photoresists AAA and BBB.
#
optical photo name.resist=AAA lambda=0.365 refrac.real=1.6 refrac.imag=0.02
optical photo name.resist=BBB lambda=0.365 refrac.real=1.9 refrac.imag=0.01
#
# Deposit user defined photoresists AAA and BBB
#
deposit photoresist name.resist=AAA thick=.5 divisions=20
deposit photoresist name.resist=BBB thick=.5 divisions=20
#
structure mirror left
#
structure outfile=anopex13_0.str
#
# Run the imaging module for a 1.0 micrometer feature
#
illumination i.line
#
illum.filter clear.fil circle sigma=0.3
#
projection na=.35 flare=2
#
pupil.filter clear.fil circle
#
layout lay.clear x.lo=-0.5 z.lo=-2 x.hi=0.5 z.hi=2
#
```

```
image clear win.x.l=-3 win.z.l=0 win.x.h=3 win.z.h=0 x.p=31 z.p=3 one.d
#
structure outfile=anopex13_1.str intensity
#
# Resist exposure, post exposure bake, and development
#
expose dose=250
#
structure outfile=anopex13_2.str
#
bake time=60 temp=125
#
develop kim time=100 steps=5
#
structure outfile=anopex13_3.str
#
tonyplot -st anopex13_*.str
#
quit
```

28.1.14. anopex14.in: Imaging of a Complex Mask using MaskViews

This example shows the ability of OPTOLITH to simulate aerial image formation of a complicated mask. The mask is stored in the anopex14.lay file for use in MASKVIEWS. The mask is loaded into OPTOLITH via the file anopex14.sec.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.15: Fast imaging results for a layout supplied by GDSII file using MASKVIEWS



Figure 28.16: Layout for imaging calculation shown in MaskViews

Input File athena_optolith/anopex14.in:

1 `go athena`

```
#
#                   OPTOLITH input file: anopex14.in
#                   --------------------------------
#
#
# Illumination system
#
illumination lambda=.193
#
illum.filter clear.fil circle sigma=0.01
#
# Projection system
#
projection na=.52
#
pupil.filter clear.fil circle
#
# Run the imaging module with a mask input file from Maskviews
#
image inf=anopex14.sec dx=0.1 win.x.h=6 win.x.l=-6 win.z.h=4 win.z.l=-4
#
structure outfile=anopex14.str intensity
tonyplot -st anopex14.str
#
quit
```

28.1.15. anopex15.in: Swing Curve Generation

A swing curve is generated by varying resist thickness from 1.0 to 1.5 microns for a 0.6 micron feature.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.17: Swing curve of CD versus resist thickness for a single feature

Input File athena_optolith/anopex15.in:

```
go athena
#
#                    OPTOLITH input file: anopex15.in
#                    --------------------------------
#
#2d  development simulation for varying
#photoresist thickness demonstrates the
#looping procedure and swing curves.
#
# Define illumination system
#
illumination h.line
#
illum.filter clear.fil circle sigma=0.5
#
# Define projection system
#
projection na=.45
#
pupil.filter clear.fil circle
#
```

```
# Define mask : set of three lines (0.6x4.0) um
#
layout lay.clear x.low=-0.30 z.low=-2.00 x.high= 0.30 z.high= 2.00
#
# Run the image module.
#
image win.x.lo=0.0 win.z.lo=0 win.x.hi=2 win.z.hi=0 dx=0.05 clear one.d
#
structure outfile=anopex15_0.str intensity
#
# Define structure
#
line x loc=0   spac=0.1
line x loc=0.3   spac=0.01
line x loc=2    spac=0.5
#
line y loc=1.5     spac=1.0
line y loc=2.5     spac=1.0
#
init silicon
#
# Define exposure parameters for the Dill exposure model and development
# rate parameters for the Mack development rate model for the photoresist
# ZZZ.
#
rate.develop name.resist=ZZZ h.line a.dill=0.83 b.dill=0.03 c.dill=0.014\
rmax.mack=0.15 rmin.mack=0.001 mth.mack=0.3 n.mack=5 Dix.0 = 7.55e-13
  Dix.E = 3.34e-2
#
# Define index of refraction of user defined photoresist ZZZ.
#
optical photoresist name.resist=ZZZ h.line refrac.real=1.7 re-
  frac.imag=0.001
#
structure outfile=anopex15_1.str
#
# CD data will be put into a file called anopex15.data
#
printf ATHENA > anopex15.data
printf 21 2 2 > anopex15.data
printf THICKNESS > anopex15.data
printf CDs > anopex15.data
#
foreach j (.1 to 0.3 step 0.01)
```

```
#
# Beginning of the loop.
#
initialize infile=anopex15_1.str
#
deposit nitride thick=j div=1 min.space=0.01
#
deposit photoresist name.resist=ZZZ thick=1 div=30 min.space=0.01
#
expose dose=150 num.refl=10 na=0
#
bake time=45 temp=115
#
#structure outfile=anopex15_2_j.str
#
develop mack time=45 steps=3 substeps=15
#
structure mirror left
#
#structure outfile=anopex15_3_j.str
#
printf j ( ZZZ|gas(1.4-j) - gas|ZZZ(1.4-j) ) > anopex15.data
#
# End of the loop.
#
end
#
tonyplot -da anopex15.data
#
#tonyplot -st anopex15_3_*.str
#
quit
```

28.1.16. anopex16.in: Interference Imaging Technique

This example shows the use of OPTOLITH to model a technique that utilizes interference created by a mirror to form an interference pattern for imaging.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.18: Effect of reflections from a sloped metal sidewall on image quality within the resist

Input File athena_optolith/anopex16.in:

```
go athena
#
#                    OPTOLITH input file: anopex16.in
#                    --------------------------------
#
#    Interference Imaging Technique
#
line x loc=0.00 spac=0.01
line x loc=1.00 spac=0.01
#
line y loc=0.0  spac=.5
line y loc=0.5  spac=.5
#
init silicon
#
deposit aluminum thick=1.0 divis=50
#
# Etch the aluminum to create a simple wedge structure
#
```

```
etch aluminum right p1.x=0.0 p1.y=-1.0 p2.x=0.5 p2.y=0.0
#
deposit photo thick=.1 divis=10 min.spac=0.001
deposit photo thick=.9 divis=30 min.spac=0.001
#
etch photo start x=0.0 y=-1
etch cont        x=1.0 y=-1
etch cont        x=1.0 y=-2
etch done        x=0.0 y=-2
#
expose inf=anopex16.exp dose=200.0 na=0
#
tonyplot -ttitle anopex16.in:Final -set anopex16.set
#
structure outfile=anopex16_0.str
#
quit

```

28.1.17. anopex17.in: Smile Plot Generation

This example demonstrates how to generate smile plots. This example loops through development of resists for varying defocus at each exposure dose. The end result can be plotted as a smile plot and shows cd variation at each exposure dose as a function of defocus.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.19: Smile Plot. This shows CD variation as a function of defocus for several different exposure doses

Input File athena_optolith/anopex17.in:

```
go athena

#
#                    OPTOLITH input file: anopex17.in
#                    --------------------------------
#
#2d  development simulation for varying
#exposure and defocus demonstrates the
#looping procedure and making smile plots.
#
# define illumination system
#
illumination i.line
#
illum.filter clear.fil circle sigma=0.5
#
# define projection system
#
projection na=.52
#
pupil.filter clear.fil circle
```

```
#
# Define mask
#
layout lay.clear x.low=-0.2 z.low=-0.5 x.high= 0.2 z.high=0.5
#
# Define the structure
#
line x loc=0   spac=0.01
line x loc=.5    spac=0.01
#
line y loc=1.5     spac=1.0
line y loc=2.5     spac=1.0
#
init silicon
#
rate.develop name.resist= OiR32 i.line c.dill=0.027
#
deposit photoresist name.resist=OiR32 thick=.97 div=40 min.space=0.01
#
structure outfile=anopex17_1.str
#
# CD data will be put into a file called anopex17.data
#
printf ATHENA > anopex17.data
printf 15 3 3 > anopex17.data
printf DEFOCUS (um) > anopex17.data
printf CDs (um) > anopex17.data
printf DOSE (mJ) > anopex17.data
#
#
# Beginning of the outer loop.
#
foreach XX (180 to 220 step 20)
#
# Beginning of the inner loop.
#
foreach YY (40 to 80 step 10)
#
initialize infile=anopex17_1.str
#
# Run the image module.
#
image win.x.lo=-.4 win.z.lo=0 win.x.hi=.4 win.z.hi=0 dx=0.5 \
```

```
defocus=(-1.2 + YY*0.03) clear one.dim
#
structure outfile=anopex17_0.str intensity
#
expose dose=XX num.refl=10
#
#structure outfile=anopex17_2_XX_YY.str
#
bake diff.length=0.055
#
#structure outfile=anopex17_3_XX_YY.str
#
develop mack time=60 steps=4 substeps=30
#
structure mirror left
structure outfile=anopex17_4_XX_YY.str
#
printf (-1.2 + YY*0.03) ( OiR32|gas(1.49) - gas|OiR32(1.49) ) XX >
  anopex17.data
#
# End of the inner loop.
#
end
#
# End of the outer loop.
#
end
#
# On the following TonyPlot select the Plot Display popup and under Group
# select DOSE to get the final smile plot.
#
tonyplot -da anopex17.data
#
tonyplot -st anopex17_4_*_*.str
#
quit
```

28.1.18. anopex18.in: Demonstration of Multiple Exposures

This example demonstrates exposure for two wavelengths. The example shows how imaging and exposure must be done for each wavelength.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.20: Image for the multiple exposure example for first wavelength



Figure 28.21: Image for the multiple exposure example for second wavelength



Figure 28.22: Concentration of the photoactive compound in the photoresist after multiple exposures



Figure 28.23: Final developed resist profile from multiple exposures

Input File athena_optolith/anopex18.in:

1 `go athena`

```
#
#                    ATHENA Input File: anopex18.in
#                    ----------------------------
#
#      This example demonstrates multiple exposure of photoresist with
#      different wavelengths.
#
#
# Define the structure
#
line x loc=0 spac=0.05
line x loc=.4 spac=0.01
line x loc=0.80 spac=0.05
#
line y loc=1.0 spac=1
line y loc=2.0 spac=1
#
init silicon
#
deposit name.resist=KTI820 thick=1.0 divisions=40
#
# Mask layout
#
layout lay.clear x.lo=-.4 x.hi=.4 z.lo=-.4 z.hi=.4
#
# Aerial image calculation
#
illumination i.line
#
illum.filter clear.fil circle sigma=0.38
#
projection na=.54
#
pupil.filter clear.fil circle
#
image win.x.lo=0 win.x.hi=.8 win.z.lo=0 win.z.hi=0 dx=0.01 clear one.d
#
structure outfile=anopex18_1.str intensity
tonyplot -st anopex18_1.str -set anopex18_1.set
#
# Resist exposure
#
expose dose=140 num.refl=10 na=0
```

```
#
tonyplot
#
illumination g.line
#
image win.x.lo=0 win.x.hi=.8 win.z.lo=0 win.z.hi=0 dx=0.01 clear one.d
#
structure outfile=anopex18_2.str intensity
#
tonyplot -st anopex18_2.str -set anopex18_1.set
#
expose mult.expose dose=140 num.refl=10 na=0
#
tonyplot
#
# Post exposure bake
#
bake time=45 temp=115
#
structure outfile=anopex18_3.str
#
tonyplot -st anopex18_3.str -set anopex18_3.set
#
# Resist Development
#
develop kim time=40 steps=5 substeps=24
#
structure mirror left
#
structure outfile=anopex18_4.str
#
tonyplot -st anopex18_4.str
#
quit
```

28.1.19. anopex19.in: Non-planar Lithography

This example shows non-planar lithography. The effects of exposing photoresist over a change in topography are demonstrated. The example uses a user defined aerial image cross section. The intensity data are read via the file anopex02.exp. The photoresist is developed and effects of the light reflected off of the 45 degrees slope can be seen.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.24: Demonstration of resist exposure over a step in the substrate

Input File athena_optolith/anopex19.in:

```
go athena
#
#                    OPTOLITH input file: anopex19.in
#                    --------------------------------
#
#            This example illustrates exposure and development
#     over nonplanar topography using a user defined
#     aerial image cross section.
#
# Define the structure
#
line x loc=0.00 spac=0.25
line x loc=1.25  spac=0.1
line x loc=1.75  spac=0.1
line x loc=3.00 spac=0.25
#
line y loc=0.5  spac=.1
line y loc=1.00 spac=.1
line y loc=1.50 spac=.1
```

```
#
init silicon
#
# Etch the silicon to create a simple wedge structure
#
etch silicon right p1.x=1.25 p1.y=.5 p2.x=1.75 p2.y=1.0
#
# Deposit default photoresist
#
deposit photo thick=1.00 divisions=60
#
# Etch photoresist to create a flat top surface
#
etch photoresist start x=0.00 y=-0.5
etch cont x=0.00 y=0.00
etch cont x=3.00 y=0.00
etch done x=3.00 y=-0.5
#
# Expose using a user intensity cross section
#
expose inf=anopex19.exp dose=200.0 na=0
#
tonyplot
#
# Diffuse the PAC for the post exposure bake
#
bake time=45 temp=115
#
structure outfile=anopex19_1.str
#
tonyplot -st anopex19_1.str -set anopex19_1.set
#
# Develop the photoresist
#
develop kim time=60 steps=8
#
structure outfile=anopex19_2.str
#
tonyplot
#
quit
```

28.1.20. anopex20.in: Resist Flow Model for Post-development Bake

This example shows planar lithography. The photoresist is baked following development to show the flow that takes place.

The plot shows the resist before and after the bake.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.25: Post Development Bake allows final resist profile to be flowed. This removes some non-uniformities from the sidewalls before substrate etching

Input File athena_optolith/anopex20.in:

```
go athena
#
# Define illumination system
#
illumination h.line
#
illum.filter clear.fil circle sigma=0.5
#
# Define projection system
#
projection na=.45
#
```

```
pupil.filter clear.fil circle
#
# Define mask
#
layout lay.clear x.lo=-0.3 z.lo=-2.0 x.hi=0.3 z.hi=2.0
#
# Run the image module.
#
image win.x.lo=-1.0 win.x.hi=1.0 win.z.lo=0.0 win.z.hi=0.0 dx=0.05 clear one.d
#
# Define structure
#
line x loc=0    spac=0.05
line x loc=0.3  spac=0.02
line x loc=0.5  spac=0.1
#
line y loc=0.0  spac=1.0
line y loc=1.0  spac=1.0
#
init silicon space.m=1
#
deposit photoresist thick=.5 div=20
#
expose dose=220
#
bake time=30 temp=115
#
develop mack time=60 steps=5 substeps=24
#
structure outfile=anopex20_0.str
#
# Perform bake to smooth the resist
# First define reflow parameters
#
material photoresist gamma.reflo=2e2 reflow visc.0=1.862e-13 visc.E=1.85
#
bake time=5 temp=125 reflow
#
struct outf=anopex20_1.str
#
tonyplot -st anopex20_*.str
#
```

```
quit

```

28.1.21. anopex21.in: Superposition of Multiple Images

Imaging using a SHRINC illumination aperture where images with different defocus values are superimposed. This example demonstrates the multiple image superposition technique.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.26: Multiple images can be superimposed in OPTOLITH

Input File athena_optolith/anopex21.in:

```
go athena
#
#                    OPTOLITH input file: anopex21.in
#                    --------------------------------
#
#               Imaging using SHRINC illumination aperture with varying
#        defocussed images superimposed. This example demonstrates the
#        multiple image superposition technique.
#
# Illumination system
#
```

```
illumination lambda=.365 intensity=.2
#
# Define the Shrinc illuminator using the arbitrary source technique
#
illum.filter clear.fil circle sigma=1 radius=0 angle=0
illum.filter square sigma=0.2 radius=0 angle=0 trans=0
illum.filter square sigma=0.2 radius=.4 angle=0 trans=0
illum.filter square sigma=0.2 radius=.8 angle=0 trans=0
illum.filter square sigma=0.2 radius=.4 angle=180 trans=0
illum.filter square sigma=0.2 radius=.8 angle=180 trans=0
illum.filter square sigma=0.2 radius=.4 angle=90 trans=0
illum.filter square sigma=0.2 radius=.8 angle=90 trans=0
illum.filter square sigma=0.2 radius=.4 angle=-90 trans=0
illum.filter square sigma=0.2 radius=.8 angle=-90 trans=0
#
projection na=.54
#
pupil.filter clear.fil circle
#
# Define the mask
#
layout lay.clear x.lo=-1.575 z.lo=-1.50 x.hi=-1.225 z.hi=1.50
layout x.lo=-0.875 z.lo=-1.50 x.hi=-0.525 z.hi=1.50
layout x.lo=-0.1750 z.lo=-1.50 x.hi=0.175 z.hi=1.50
layout x.lo=0.525 z.lo=-1.50 x.hi=0.875 z.hi=1.50
layout x.lo=1.225 z.lo=-1.50 x.hi=1.575 z.hi=1.50
#
# Run the imaging module
#
image win.x.lo=-0.70 win.z.lo=-0.1 win.x.hi=0.70 win.z.hi=0.1 dx=0.01
  n.pupil=10 defocus=0.1
#
structure outfile=anopex21_1.str intensity
#
# Start looping after one image
#
foreach XX (2 to 4 step 1)
#
# Now use mult.image in the image command
#
image mult.image win.x.lo=-0.70 win.z.lo=-0.1 win.x.hi=0.70 win.z.hi=0.1
  dx=0.01 n.pupil=10 defocus=(0.1*XX)
#
structure outfile=anopex21_XX.str intensity
```

```
#
# End of the loop
#
end
#
tonyplot -st anopex21_*.str
#
quit

```

28.1.22. anopex22.in: Optimization of a Rim Shifter Phase Shift Mask

This example uses looping and functions of the looping variable to change the size of a contact hole and its rim shifters.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.27: Plot showing results from all combinations of contact size and rim phase shifters. Looping within OPTOLITH allows such plots to be created quickly for analysis of many mask options.

Input File athena_optolith/anopex22.in:

```
go athena
#
#                    OPTOLITH input file: anopex22.in
```

```
#                    --------------------------------
#
#                 Optimization of a rim shifter phase shift mask.
# This example uses looping and functions of the
# looping variable to change the size of a contact
# hole and its rim shifters
#
# The illumination wavelength
#
illumination i.line
#
# The shape of the illuminating source
#
illum.filter clear.fil circle sigma=0.01
#
# The projection system numerical aperture
#
projection na=.54
#
# The shape of the pupil of the projection system
#
pupil.filter clear.fil circle
#
# Begin the outside optimization loop
#
foreach X (0 to 2 step 1)
#
# Begin the inside optimization loop
#
foreach Y (1 to 8 step 1)
#
# Define the mask : two elbows and a contact hole (cd=1.0 um)
#
layout lay.clear x.lo= (-.125-X*.025) x.hi= (.125+X*.025) z.lo= (-.125-
  X*.025) z.hi= (.125+X*.025) phase=0. trans=1.
printf x.lo= (-.125-X*.025) x.hi= (.125+X*.025) z.lo= (-.125-X*.025)
  z.hi= (.125+X*.025) phase=0. trans=1.
layout x.lo= (-.125-X*.025-.025*Y) x.hi= (.125+X*.025+.025*Y) z.lo=
  (.125+X*.025) z.hi= (.125+X*.025+.025*Y) phase=180. trans=1.
printf x.lo= (-.125-X*.025-.025*Y) x.hi= (.125+X*.025+.025*Y) z.lo=
  (.125+X*.025) z.hi= (.125+X*.025+.025*Y) phase=180. trans=1.
layout x.lo= (-.125-X*.025-.025*Y) x.hi= (.125+X*.025+.025*Y) z.lo= (-
  .125-X*.025-.025*Y) z.hi= (-.125-X*.025) phase=180. trans=1.
printf x.lo= (-.125-X*.025-.025*Y) x.hi= (.125+X*.025+.025*Y) z.lo= (-
  .125-X*.025-.025*Y) z.hi= (-.125-X*.025) phase=180. trans=1.
```

```
layout x.lo= (-.125-X*.025-.025*Y) x.hi= (-.125-X*.025) z.lo= (-.125-
  X*.025) z.hi= (.125+X*.025) phase=180. trans=1.
printf x.lo= (-.125-X*.025-.025*Y) x.hi= (-.125-X*.025) z.lo= (-.125-
  X*.025) z.hi= (.125+X*.025) phase=180. trans=1.
layout x.lo= (.125+X*.025) x.hi= (.125+X*.025+.025*Y) z.lo= (-.125-
  X*.025) z.hi= (.125+X*.025) phase=180. trans=1.
printf x.lo= (.125+X*.025) x.hi= (.125+X*.025+.025*Y) z.lo= (-.125-
  X*.025) z.hi= (.125+X*.025) phase=180. trans=1.
#
# Calculation of the aerial image
#
image win.x.lo=-.6 win.z.lo=-.6 win.x.hi=.6 win.z.hi=.6 dx=.05 n.pupil=10
#
# Store the aerial image in a structure file
#
structure outfile=anopex22_X_Y.str intensity
#
# End of the inside loop
#
end
#
# End of the outside loop
#
end
#
# Plot the aerial images at the end of the run
#
tonyplot -st anopex22_*_*.str
#
quit
```

28.1.23. anopex23.in: Critical Dimensions of an Aerial Image

This example calculates the aerial image of phase shift contacts and uses DECKBUILD's **Extract** tool to find the width of the aerial image.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.28: Plots of Aerial Images such as these are useful. However, judging CD from the image in a plot is tricky. This example demonstrates a way to automate CD extraction in batch mode.

Input File athena_optolith/anopex23.in:

```
go athena

#
#                    OPTOLITH input file: anopex23.in
#                    -------------------------------
#
#
# define illumination system
#
illumination i.line
#
illum.filter clear.fil circle sigma=0.32
#
# define projection system
#
projection na=.54
#
pupil.filter clear.fil circle
#
```

```
# Define mask
#
set CONTACT=0.4
set SPACE=0.4
set DEFOC=1
#
# Clear previous features
#
layout lay.clear
#
# Phase shifting contact holes for an isolated contact with 10 percent
  transmittance in the
# spaces
#
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=-(3*$"CONTACT"/2+$"SPACE")
  \
       x.high=-($"CONTACT"/2+$"SPACE")  z.high=-($"CONTACT"/2+$"SPACE")
  trans=0.1 phas=0
layout x.low=-($"CONTACT"/2)           z.low=-(3*$"CONTACT"/2+$"SPACE") \
       x.high=($"CONTACT"/2)            z.high=-($"CONTACT"/2+$"SPACE")
  trans=0.1 phas=0
layout x.low=($"CONTACT"/2+$"SPACE")    z.low=-(3*$"CONTACT"/2+$"SPACE")
  \
       x.high=(3*$"CONTACT"/2+$"SPACE") z.high=-($"CONTACT"/2+$"SPACE")
  trans=0.1 phas=0
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=-($"CONTACT"/2)           \
       x.high=-($"CONTACT"/2+$"SPACE")  z.high=($"CONTACT"/2)
  trans=0.1 phas=0
layout x.low=($"CONTACT"/2+$"SPACE")    z.low=-($"CONTACT"/2)           \
       x.high=(3*$"CONTACT"/2+$"SPACE") z.high=($"CONTACT"/2)
  trans=0.1 phas=0
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=($"CONTACT"/2+$"SPACE")
  \
       x.high=-($"CONTACT"/2+$"SPACE")  z.high=(3*$"CONTACT"/2+$"SPACE")
  trans=0.1 phas=0
layout x.low=-($"CONTACT"/2)           z.low=($"CONTACT"/2+$"SPACE")    \
       x.high=($"CONTACT"/2)            z.high=(3*$"CONTACT"/2+$"SPACE")
  trans=0.1 phas=0
layout x.low=($"CONTACT"/2+$"SPACE")    z.low=($"CONTACT"/2+$"SPACE")    \
       x.high=(3*$"CONTACT"/2+$"SPACE") z.high=(3*$"CONTACT"/2+$"SPACE")
  trans=0.1 phas=0
#
# Spaces
#
layout x.low=-($"CONTACT"/2+$"SPACE")   z.low=-($"CONTACT"/2)           \
       x.high=-($"CONTACT"/2)           z.high=($"CONTACT"/2)
  trans=0.1 phas=0
```

```
layout x.low=($"CONTACT"/2)            z.low=-($"CONTACT"/2)            \
       x.high=($"CONTACT"/2+$"SPACE")   z.high=($"CONTACT"/2) trans=0.1 phas=0
layout x.low=-($"CONTACT"/2+$"SPACE")  z.low=($"CONTACT"/2+$"SPACE")    \
       x.high=-($"CONTACT"/2)           z.high=(3*$"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=($"CONTACT"/2)            z.low=($"CONTACT"/2+$"SPACE")    \
       x.high=($"CONTACT"/2+$"SPACE")   z.high=(3*$"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=-($"CONTACT"/2+$"SPACE")  z.low=-(3*$"CONTACT"/2+$"SPACE") \
       x.high=-($"CONTACT"/2)           z.high=-($"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=($"CONTACT"/2)            z.low=-(3*$"CONTACT"/2+$"SPACE") \
       x.high=($"CONTACT"/2+$"SPACE")   z.high=-($"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=($"CONTACT"/2)            \
       x.high=(3*$"CONTACT"/2+$"SPACE") z.high=($"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=-($"CONTACT"/2+$"SPACE") \
       x.high=(3*$"CONTACT"/2+$"SPACE") z.high=-($"CONTACT"/2) trans=0.1 phas=0
#
# Add a defect for printability
#
set DEFECT=0.08
set XCENTER=-.1
set ZCENTER=0.0
#
# Central feature
#
layout x.low=-($"CONTACT"/2)           z.low=-($"CONTACT"/2) \
       x.high=(-$"DEFECT"/2+$"XCENTER") z.high=($"CONTACT"/2) trans=1 phas=180
layout x.low=(-$"DEFECT"/2+$"XCENTER") z.low=($"DEFECT"/2+$"ZCENTER")\
       x.high=($"DEFECT"/2+$"XCENTER")  z.high=($"CONTACT"/2) trans=1 phas=180
layout x.low=(-$"DEFECT"/2+$"XCENTER")  z.low=-($"CONTACT"/2) \
       x.high=($"DEFECT"/2+$"XCENTER")  z.high=(-$"DEFECT"/2+$"ZCENTER") trans=1 phas=180
layout x.low=($"DEFECT"/2+$"XCENTER")   z.low=-($"CONTACT"/2) \
       x.high=($"CONTACT"/2)            z.high=($"CONTACT"/2) trans=1 phas=180
#
# The defect
#
```

```
layout x.low=(-$"DEFECT"/2+$"XCENTER") z.low=(-$"DEFECT"/2+$"ZCENTER") \
       x.high=($"DEFECT"/2+$"XCENTER") z.high=($"DEFECT"/2+$"ZCENTER") trans=1 phas=180
#
# Calculate the image
#
image win.x.lo=-.35 win.z.lo=0 win.x.hi=.35 win.z.hi=0 dx=0.025 opaque defocus=$DEFOC n.pupil=5
structure outfile=anopex23_1_$"DEFOC".int int
#
#extract init infile="anopex23_1_$"DEFOC".int"
extract name="xint1" x.val from curve(depth,impurity="Norm Intensity" material="All" y.val=0) where y.val=0.15 and val.occno =1

extract name="xint2" x.val from curve(depth,impurity="Norm Intensity" material="All" y.val=0) where y.val=0.15 and val.occno =2

extract name="cd1"  ($xint2 - $xint1)

#
# Clear previous features
#
layout lay.clear
#
# Phase shifting contact holes for dense contacts with 10 percent transmittance in the
# spaces
#
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=-(3*$"CONTACT"/2+$"SPACE") \
       x.high=-($"CONTACT"/2+$"SPACE")  z.high=-($"CONTACT"/2+$"SPACE") trans=1 phas=180
layout x.low=-($"CONTACT"/2)           z.low=-(3*$"CONTACT"/2+$"SPACE") \
       x.high=($"CONTACT"/2)           z.high=-($"CONTACT"/2+$"SPACE") trans=1 phas=180
layout x.low=($"CONTACT"/2+$"SPACE")    z.low=-(3*$"CONTACT"/2+$"SPACE") \
       x.high=(3*$"CONTACT"/2+$"SPACE") z.high=-($"CONTACT"/2+$"SPACE") trans=1 phas=180
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=-($"CONTACT"/2)           \
       x.high=-($"CONTACT"/2+$"SPACE")  z.high=($"CONTACT"/2) trans=1 phas=180
layout x.low=($"CONTACT"/2+$"SPACE")    z.low=-($"CONTACT"/2)           \
       x.high=(3*$"CONTACT"/2+$"SPACE") z.high=($"CONTACT"/2) trans=1 phas=180
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=($"CONTACT"/2+$"SPACE") \
```

```
        x.high=-($"CONTACT"/2+$"SPACE")  z.high=(3*$"CONTACT"/2+$"SPACE") trans=1 phas=180
layout x.low=-($"CONTACT"/2)            z.low=($"CONTACT"/2+$"SPACE")    \
        x.high=($"CONTACT"/2)            z.high=(3*$"CONTACT"/2+$"SPACE") trans=1 phas=180
layout x.low=($"CONTACT"/2+$"SPACE")    z.low=($"CONTACT"/2+$"SPACE")    \
        x.high=(3*$"CONTACT"/2+$"SPACE") z.high=(3*$"CONTACT"/2+$"SPACE") trans=1 phas=180
#
# Spaces
#
layout x.low=-($"CONTACT"/2+$"SPACE")   z.low=-($"CONTACT"/2)            \
        x.high=-($"CONTACT"/2)           z.high=($"CONTACT"/2) trans=0.1 phas=0
layout x.low=($"CONTACT"/2)             z.low=-($"CONTACT"/2)            \
        x.high=($"CONTACT"/2+$"SPACE")   z.high=($"CONTACT"/2) trans=0.1 phas=0
layout x.low=-($"CONTACT"/2+$"SPACE")   z.low=($"CONTACT"/2+$"SPACE")    \
        x.high=-($"CONTACT"/2)           z.high=(3*$"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=($"CONTACT"/2)             z.low=($"CONTACT"/2+$"SPACE")    \
        x.high=($"CONTACT"/2+$"SPACE")   z.high=(3*$"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=-($"CONTACT"/2+$"SPACE")   z.low=-(3*$"CONTACT"/2+$"SPACE") \
        x.high=-($"CONTACT"/2)           z.high=-($"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=($"CONTACT"/2)             z.low=-(3*$"CONTACT"/2+$"SPACE") \
        x.high=($"CONTACT"/2+$"SPACE")   z.high=-($"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=($"CONTACT"/2)            \
        x.high=(3*$"CONTACT"/2+$"SPACE") z.high=($"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=-($"CONTACT"/2+$"SPACE") \
        x.high=(3*$"CONTACT"/2+$"SPACE") z.high=-($"CONTACT"/2) trans=0.1 phas=0
#
# Add a defect for printability
#
# Central feature
#
layout x.low=-($"CONTACT"/2)            z.low=-($"CONTACT"/2) \
        x.high=(-$"DEFECT"/2+$"XCENTER") z.high=($"CONTACT"/2) trans=1 phas=180
layout x.low=(-$"DEFECT"/2+$"XCENTER") z.low=($"DEFECT"/2+$"ZCENTER")\
```

```
       x.high=($"DEFECT"/2+$"XCENTER")  z.high=($"CONTACT"/2)
   trans=1 phas=180
layout x.low=(-$"DEFECT"/2+$"XCENTER")  z.low=-($"CONTACT"/2) \
       x.high=($"DEFECT"/2+$"XCENTER")  z.high=(-$"DEFECT"/2+$"ZCENTER")
   trans=1 phas=180
layout x.low=($"DEFECT"/2+$"XCENTER")   z.low=-($"CONTACT"/2) \
       x.high=($"CONTACT"/2)            z.high=($"CONTACT"/2)
   trans=1 phas=180
#
# The defect
#
layout x.low=(-$"DEFECT"/2+$"XCENTER") z.low=(-$"DEFECT"/2+$"ZCENTER") \
       x.high=($"DEFECT"/2+$"XCENTER") z.high=($"DEFECT"/2+$"ZCENTER")
   trans=1 phas=180
#
# Calculate the image
#
image win.x.lo=-.35 win.z.lo=0 win.x.hi=.35 win.z.hi=0 dx=0.025 opaque
   defocus=$DEFOC n.pupil=5
structure outfile=anopex23_2_$"DEFOC".int int
#
#extract init infile="anopex23_2_$"DEFOC".int"
extract name="xint1" x.val from curve(depth,impurity="Norm Intensity" ma-
   terial="All" y.val=0) where y.val=0.15 and val.occno =1
extract name="xint2" x.val from curve(depth,impurity="Norm Intensity" ma-
   terial="All" y.val=0) where y.val=0.15 and val.occno =2
extract name="cd2"  ($xint2 - $xint1)
set SPACE=0.6
#
# Clear previous features
#
layout lay.clear
#
# Phase shifting contact holes for dense contacts with 10 percent trans-
   mittance in the
# spaces
#
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=-(3*$"CONTACT"/2+$"SPACE")
   \
       x.high=-($"CONTACT"/2+$"SPACE")  z.high=-($"CONTACT"/2+$"SPACE")
   trans=1 phas=180
layout x.low=-($"CONTACT"/2)          z.low=-(3*$"CONTACT"/2+$"SPACE") \
```

```
       x.high=($"CONTACT"/2)           z.high=-($"CONTACT"/2+$"SPACE") trans=1 phas=180
layout x.low=($"CONTACT"/2+$"SPACE")    z.low=-(3*$"CONTACT"/2+$"SPACE") \
       x.high=(3*$"CONTACT"/2+$"SPACE") z.high=-($"CONTACT"/2+$"SPACE") trans=1 phas=180
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=-($"CONTACT"/2)           \
       x.high=-($"CONTACT"/2+$"SPACE")  z.high=($"CONTACT"/2) trans=1 phas=180
layout x.low=($"CONTACT"/2+$"SPACE")    z.low=-($"CONTACT"/2)           \
       x.high=(3*$"CONTACT"/2+$"SPACE") z.high=($"CONTACT"/2) trans=1 phas=180
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=($"CONTACT"/2+$"SPACE") \
       x.high=-($"CONTACT"/2+$"SPACE")  z.high=(3*$"CONTACT"/2+$"SPACE") trans=1 phas=180
layout x.low=-($"CONTACT"/2)           z.low=($"CONTACT"/2+$"SPACE")    \
       x.high=($"CONTACT"/2)           z.high=(3*$"CONTACT"/2+$"SPACE") trans=1 phas=180
layout x.low=($"CONTACT"/2+$"SPACE")    z.low=($"CONTACT"/2+$"SPACE")    \
       x.high=(3*$"CONTACT"/2+$"SPACE") z.high=(3*$"CONTACT"/2+$"SPACE") trans=1 phas=180
#
# Spaces
#
layout x.low=-($"CONTACT"/2+$"SPACE")   z.low=-($"CONTACT"/2)           \
       x.high=-($"CONTACT"/2)           z.high=($"CONTACT"/2) trans=0.1 phas=0
layout x.low=($"CONTACT"/2)            z.low=-($"CONTACT"/2)           \
       x.high=($"CONTACT"/2+$"SPACE")   z.high=($"CONTACT"/2) trans=0.1 phas=0
layout x.low=-($"CONTACT"/2+$"SPACE")   z.low=($"CONTACT"/2+$"SPACE")    \
       x.high=-($"CONTACT"/2)           z.high=(3*$"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=($"CONTACT"/2)            z.low=($"CONTACT"/2+$"SPACE")    \
       x.high=($"CONTACT"/2+$"SPACE")   z.high=(3*$"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=-($"CONTACT"/2+$"SPACE")   z.low=-(3*$"CONTACT"/2+$"SPACE") \
       x.high=-($"CONTACT"/2)           z.high=-($"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=($"CONTACT"/2)            z.low=-(3*$"CONTACT"/2+$"SPACE") \
       x.high=($"CONTACT"/2+$"SPACE")   z.high=-($"CONTACT"/2+$"SPACE") trans=0.1 phas=0
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=($"CONTACT"/2)           \
       x.high=(3*$"CONTACT"/2+$"SPACE") z.high=($"CONTACT"/2+$"SPACE") trans=0.1 phas=0
```

```
layout x.low=-(3*$"CONTACT"/2+$"SPACE") z.low=-($"CONTACT"/2+$"SPACE")
  \
      x.high=(3*$"CONTACT"/2+$"SPACE") z.high=-($"CONTACT"/2)
  trans=0.1 phas=0
#
# Add a defect for printability
#
# Central feature
#
layout x.low=-($"CONTACT"/2)           z.low=-($"CONTACT"/2) \
      x.high=(-$"DEFECT"/2+$"XCENTER") z.high=($"CONTACT"/2)
  trans=1 phas=180
layout x.low=(-$"DEFECT"/2+$"XCENTER") z.low=($"DEFECT"/2+$"ZCENTER") \
      x.high=($"DEFECT"/2+$"XCENTER")  z.high=($"CONTACT"/2)
  trans=1 phas=180
layout x.low=(-$"DEFECT"/2+$"XCENTER")  z.low=-($"CONTACT"/2) \
      x.high=($"DEFECT"/2+$"XCENTER")  z.high=(-$"DEFECT"/2+$"ZCENTER")
  trans=1 phas=180
layout x.low=($"DEFECT"/2+$"XCENTER")   z.low=-($"CONTACT"/2) \
      x.high=($"CONTACT"/2)            z.high=($"CONTACT"/2)
  trans=1 phas=180
#
# The defect
#
layout x.low=(-$"DEFECT"/2+$"XCENTER") z.low=(-$"DEFECT"/2+$"ZCENTER") \
      x.high=($"DEFECT"/2+$"XCENTER") z.high=($"DEFECT"/2+$"ZCENTER")
  trans=1 phas=180
#
# Calculate the image
#
image win.x.lo=-.35 win.z.lo=0 win.x.hi=.35 win.z.hi=0 dx=0.025 opaque
  defocus=$DEFOC n.pupil=5
structure outfile=anopex23_3_$"DEFOC".int int
#
extract name="xint1" x.val from curve(depth,impurity="Norm Intensity" ma-
  terial="All" y.val=0) where y.val=0.15 and val.occno =1

extract name="xint2" x.val from curve(depth,impurity="Norm Intensity" ma-
  terial="All" y.val=0) where y.val=0.15 and val.occno =2


extract name="cd3"  ($xint2 - $xint1)

printf $DEFOC $cd1 > anopex23_1.data

printf $DEFOC $cd2 > anopex23_2.data
```

```

printf $DEFOC $cd3 > anopex23_3.data

#tonyplot -overlay anopex23_*.int -set anopex23.set
#
quit
```

28.1.24. anopex24.in: CMOS Example Using MaskViews and OPTOLITH

This example is a modification of 27.18anex18.in: Simple CMOS Example Using MaskViews. The purpose is to show the interconnection between OPTOLITH and the mask layout generated by the cross-sectioning in MaskViews. In particular, the poly mask is selected for the demonstration.

In the original example, the following statement is included,

```
mask name="POLY"
```

When this statement is executed in DECKBUILD, a barrier layer will be deposited and it will be automatically etched according to the mask layout. In order to make the process more general so that photolithographic procedures can be employed, an optional OPTOLITH flag is added, that is,

```
mask name="POLY" optolith
```

During execution, the layout statements for the mask will be listed instead. By default, the mask is assumed opaque, that is, the background transmittance is zero. If `reverse` is specified, the openings will have zero transmittance.

As opposed to the original example, the processing steps for the poly gate are interrupted in a sense that a typical photolithographic procedure is used to deposit the poly gate. For the photolithographic procedure, it includes the following steps,

- specification of the light source and the optical system
- photoresist deposition
- exposure
- post bake
- photoresist development

In addition, statements are included to extract the critical dimension and the height for the final photoresist profile.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.29: MOSFET formation using OPTOLITH models for polysilicon gate definition

Input File athena_optolith/anopex24.in:

```
go athena cutline=anopex24_0.sec
#
#
init c.phos=1.0e14 orientation=100

# Initial Oxide
method fermi compress
diffus time=30 temp=1000 dryo2 press=1.00 hcl.pc=3
# Initial Nitride
deposit nitride thick=0.15 divisions=2

# Pwell photo
mask name="WELL"
# Pwell definition etch
etch nitride dry thick=0.15
# Oxide overetch
etch oxide dry thick=0.015
# pwell implant
implant boron dose=1.0e13 energy=100
```

```
strip
struct outfile=anopex24_0.str

# Nwell maox
method fermi compress
diffus time=45 temp=950 weto2 press=1.00 hcl.pc=3
struct outfile=anopex24_1.str

# strip init nit
etch nitride all
# oxide overetch
etch oxide dry thick=0.015
# nwell implant
implant phosphor dose=1.0e13 energy=60

# planarising oxide
method fermi compress
diffus  time=45 temp=800 t.final=1000 dryo2 press=0.05 hcl.pc=3
diffus time=45 temp=1000 weto2 press=0.05 hcl.pc=3
# well drive
method fermi compress
diffus time=170 temp=1200 nitro press=0.05 hcl.pc=0
structure outfile=anopex24_2.str
# wellox strip
etch oxide all

# Stress relief Oxide
deposit oxide thick=0.035 divisions=1
# Masking Nitride
deposit nitride thick=0.15 divisions=1
# Active Area Photo
mask name="AAD"
# AA Definition
etch nitride dry thick=0.30
# AA Definition
etch oxide dry thick=0.10
# AA Definition
etch silicon dry thick=0.15
struct outfile=anopex24_3.str
strip

# Field Oxide growth
method fermi compress
```

```
diffus time=70 temp=1000 weto2 press=1.00 hcl.pc=3
struct outfile=anopex24_4.str
#
extract name="pwell_fieldox" thickness oxide mat.occno=1 region="pwell_fld"
extract name="nwell_fieldox" thickness oxide mat.occno=1 region="nwell_fld"
# AA stack strip
etch nitride dry thick=0.25
# AA stack strip
etch oxide dry thick=0.07
# AA stack strip
etch nitride all
#
struct outfile=anopex24_5.str
# Sacox growth
diffus time=25 temp=950 dryo2 press=1.00 hcl.pc=3
# Sacox strip
etch oxide dry thick=0.06
#Gate oxide recipe for example msex03.in
# push in boat, timed from after boat reaches the cold zone....
diffuse temp=800  time=15 nitrogen
#ramp up to temp in dilute o2 and tca
diffuse temp=850 t.final=950 dryo2 hcl.pc=3 time=15 press=0.25
#increase pp of o2 and stabilise
diffuse temp=950 time=5 dryo2 hcl.pc=3 press=0.8
#main growth step
diffuse temp=950 dryo2 hcl.pc=3 time=15 press=0.8
#anneal step
diffuse temp=950 time=35 nitrogen
# Ramp down
diffuse temp=950 t.final=800 time=30 nitrogen
# Pull out
diffuse temp=800 time=15 nitrogen
struct outfile=anopex24_6.str
# Extract gate ox thickness. This is used as optimisation target and as a
# VWF design variables....
extract thickness oxide mat.occno=1 region="ngate" name="ngateox"
extract thickness oxide mat.occno=1 region="pgate" name="pgateox"
# ################################################
```

```
# The following section is modified for OPTOLITH
# to deposit the photoresist for the poly gate.
# ################################################
# Vt adjust Implant
implant boron dose=1.5e12 energy=25
# Poly Dep
deposit poly thick=0.30 divisions=5
# The illumination wavelength
illumination g.line
# The shape of the illuminating source
illum.filter clear.fil circle sigma=0.3
# The projection system numerical aperture
projection na=.43
# The shape of the pupil of the projection system
pupil.filter clear.fil circle
# mask for poly defined by CUT in MaskView
mask name="POLY" optolith
# Calculation of the aerial image
image win.x.lo=0 win.z.lo=-1 win.x.hi=3 win.z.hi=1 clear
# Photoresist processing
deposit photo thick=1.00 divisions=30
expose dose=200.0 NA=0 num.refl=10
bake diff.leng=0.05
develop kim time=45 steps=8
struct outfile=anopex24_7.str

# CD extraction at y = -1.0, -0.6 and -0.3
printf (gas|photo(-1.0) - photo|gas(-1.0))
printf (gas|photo(-0.6) - photo|gas(-0.6))
printf (gas|photo(-0.3) - photo|gas(-0.3))
# extract the height of the photoresist
extract name="photoresist" thickness material="Photoresist" mat.occno=1
  x.val=1.3
# ###################
# end of modification
# ###################

# Poly etch
etch poly dry thick=0.40
#photostrip
strip
# Poly is bare here, so name the gate electrodes.....
autoelectrode
```

```
# Poly Reox
method fermi compress
diffus time=23 temp=900 dryo2 press=1.00 hcl.pc=3
struct outfile=anopex24_8.str
# Spacer Nitride Dep
deposit nitride thick=0.20 divisions=4
# Spacer etch
etch nitride dry thick=0.20
# SDN++
mask name="WELL"
# N++ Implant
implant arsenic dose=4.5e15 energy=50
# Nitride spacer removal for the N’s
etch nitride dry thick=1
# N-LDD Implant
implant phosphor dose=3.5e13 energy=30
struct outfile=anopex24_9.str
strip
# P++ photo
mask name="WELL" reverse
# P++ Implant
implant boron dose=1.0e15 energy=25
# Nitride spacer removal for the P’s
etch nitride dry thick=1
# P-LDD Implant
implant bf2 dose=1.0e14 energy=40
struct outfile=anopex24_10.str
strip
# Capping Nitride
deposit nitride thick=0.05 divisions=1

# BPSG Dense
method fermi compress
diffus time=20 temp=900 nitro press=1.00 hcl.pc=0
struct outfile=anopex24_11.str

# Extract the two sd junction depths....
extract name="p++ xj" xj silicon mat.occno=1 region="pmcc" junc.occno=1
extract name="n++ xj" xj silicon mat.occno=1 region="nmcc" junc.occno=1

# MCC
mask name="VIA2"
# MCC etch
```

```
etch nitride dry thick=0.10
# MCC etch
etch oxide dry thick=0.10
strip
# Metal Depo
deposit alumin thick=0.05 divisions=2
# First Metal Defn
mask name="MET1"
# Metal1 etch
etch aluminum dry thick=0.50
strip
# Aluminium is bare here, so place the electrodes.....
autoelectrode
electrode backside name=substrate
structure outf=anopex24_12.str

tonyplot

quit
```

28.1.25. anopex25.in: Optical Proximity Correction on Mask

The purpose of this example is to domonstrate the ability of OPTOLITH and MASKVIEWS to perform optical proximity correction on the layout. The image quality is shown by two figures of merit. One is the percentage of unfilled area inside the mask features. The other one is the percentage of image area lying outside of the mask features.

- Preparation:

Two files need to be prepared by using MASKVIEWS**,**

1. **anopex25.sec -** unbiased original mask
2. **anopex25_opc.sec -** OPC biased mask

- Output:

Four files will be created,

1. anopex25_1.str - image of the unbiased mask
2. anopex25_1_img.sec - automatically generated image file of the unbiased mask, to be used by MASKVIEWS
3. anopex25_2.str - image of the biased mask
4. anopex25_2_img.sec - automatically generated image file of the biased mask, to be used by MASKVIEWS

- Two sets of figures of merit will be printed.

The normalized intensity threshold level is defined by **opc=???** in the `structure` command. For this example, the threshold level is assumed to be 0.5.

```
structure outfile=anopex25_1.str infile=anopex25.sec opc=0.5
```

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 28.30: Original Uncorrected Layout for OPC example



Figure 28.31: Image from corrected mask



Figure 28.32: Image from corrected mask



Figure 28.33: Corrected Mask in MaskViews

Input File athena_optolith/anopex25.in:

1
```
go athena
```

```
#
#                   OPTOLITH input file: anopex25.in
#                   --------------------------------
#
# Purpose:
# Demonstrate the advantage of using the OPC biased mask over
# the unbiased mask.
#
# Preparation:
#Two files need to be prepared by using MASKVIEWS,
#anopex25.sec - unbiased original mask
#anopex25_opc.sec - OPC biased mask
# Output:
#1. Four files will be created,
#anopex25_1.str   - image of the unbiased mask
#anopex25_1_img.sec - image of the unbiased mask, to be used
#     by MASKVIEWS
#anopex25_2.str   - image of the biased mask
#anopex25_2_img.sec - image of the biased mask, to be used
#     by MASKVIEWS
#2. Two sets of error will be printed.
#
# The illumination wavelength
#
illumination i.line
#
# The shape of the illuminating source
#
illum.filter clear.fil circle sigma=0.3
#
# The projection system numerical aperture
#
projection na=.43
#
# The shape of the pupil of the projection system
#
pupil.filter clear.fil circle
#
########################################################################
  #
#
# Calculation of the aerial image of the original mask
#
```

```
##########################################################################
#
image inf=anopex25.sec win.x.lo=-2 win.z.lo=-2 win.x.hi=2 win.z.hi=2 dx=.05
#
# Store the aerial image in a structure file and a file called
# anopex25_1_img.sec will also be created by ATHENA automatically.
# This .sec file contains the aerial image which can be viewed in
# MASKVIEWS.
#
structure outfile=anopex25_1.str infile=anopex25.sec opc=0.5
#
# Plot the aerial image during the run
#
tonyplot -st anopex25.str -set anopex25.set
#
##########################################################################
#
# Calculation of the aerial image of the OPC biased mask
#
########################################################################
#
image inf=anopex25_opc.sec win.x.lo=-2 win.z.lo=-2 win.x.hi=2 win.z.hi=2 dx=.05
#
# Store the aerial image in a structure file and a file called
# anopex25_2_img.sec will also be created by ATHENA automatically.
# This .sec file contains the aerial image which can be viewed in
# MASKVIEWS.
#
structure outfile=anopex25_2.str infile=anopex25.sec opc=0.5
#
# Plot the aerial image during the run
#
tonyplot -st anopex25_2.str -set anopex25.set
#
quit
```

28.1.26. anopex26.in: Aerial Image Calculation for Mask Pattern with Non-Manhattan Geometry

Calculate the aerial image of a mask with non-Manhattan geometry. This example includes a 45 degrees metal joint and a circular contact opening.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DeckBuild **run** button to execute the example.



Figure 28.34: Imaging of Circular and 45-degree masks

Input File athena_optolith/anopex26.in:

```
go athena
#
#                    OPTOLITH input file: anopex26.in
#                    --------------------------------
#
# Calculate the aerial image of a mask with non-Manhattan geometry. This example
# includes a 45 degree metal joint and a circular contact opening.
#
# The illumination wavelength
#
illumination h.line
#
# The shape of the illuminating source
#
illum.filter clear.fil circle sigma=0.3
#
# The projection system numerical aperture
```

```
#
projection na=0.43
#
# The shape of the pupil of the projection system
#
pupil.filter clear.fil circle
#
# Define the mask : a circular contact hole (cd=1.0 um) and a 45 degree
#                   metal joint
#
layout lay.clear
layout x.circle=0z.circle=0 radius=0.25
layoutx.lo=1z.lo=-0.75x.hi=1.3z.hi=0.25
layoutx.tri=1.3 z.tri=0.25 width=0.3height=0.3rot.angle=90
layoutx.lo=0.562868z.lo=0.2446699x.hi=0.987132z.hi=1.30533rot.angle=45
layoutx.tri=0.25 z.tri=1.3 width=0.3height=0.3rot.angle=-90
layout x.lo=-0.75z.lo=1x.hi=0.25z.hi=1.3
#
# Calculation of the aerial image
#
image win.x.lo=-1.5 win.z.lo=-1.5 win.x.hi=1.5 win.z.hi=1.5 dx=.02 opaque
#
# Store the aerial image in a structure file
#
structure outfile=anopex26_1.str intensity
#
# Plot the aerial image during the run
#
tonyplot -st anopex26_1.str
#
quit
```

This page is intentionally left blank

29.1. ATHENA_FLASH: Compound Semiconductor Process Simulation

29.1.1. anflex01.in: Ion Implant Tilt Angle Dependence

This example calculates 10 keV Be implant profiles in GaAs for zero and five degrees implant angles. The crystalline Monte Carlo implant model is used. The results show that ion channeling influences not only the tails of the profiles, but also the position of the profile peak.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 29.1: Tilt Dependence of a Be Implant into GaAs

Input File athena_flash/anflex01.in:

```
go athena

#TITLE: Tilt angle dependence for MC implantation in GaAs

line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing = 0.01
line y loc = 0.20  spacing = 0.01
init gaas

```

```
structure outfile=anflex01.str

implant beryllium energy=10 dose=1e14 monte tilt=0 n.ion=3000
structure outfile=anflex01_1.str

init infile=anflex01.str
implant beryllium energy=10 dose=1e14 monte tilt=5 n.ion=3000
structure outfile=anflex01_2.str


tonyplot -overlay -st anflex01_*.str -set anflex01.set

quit
```

29.1.2. anflex02.in: Ion Implant Rotation Angle Dependence

This example calculates 10 keV Be implant profiles in GaAs for 10 degrees implant angle. Results for four different rotation angles (0, 5, 10, and 10 degrees) are compared using the Overlay capability of TONYPLOT. Note that the channeling effect is more pronounced when plane of implantation is parallel to a major crystallographic plane (zero degrees rotation).

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 29.2: Rotation Dependence of a Be Implant into GaAs

Input File athena_flash/anflex02.in:

```
go athena

#TITLE: Rotation angle dependence for MC implantation in GaAs

line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing = 0.01
line y loc = 0.20  spacing = 0.02
init gaas

implant beryllium energy=10 dose=1e14 monte tilt=10 rotation=0.0
structure outfile=anflex02_1.str

line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing = 0.01
line y loc = 0.20  spacing = 0.02
init gaas

implant beryllium energy=10 dose=1e14 monte tilt=10 rotation=5.0
structure outfile=anflex02_2.str

line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing = 0.01
line y loc = 0.20  spacing = 0.02
init gaas

implant beryllium energy=10 dose=1e14 monte tilt=10 rotation=10.0
structure outfile=anflex02_3.str

line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing = 0.01
line y loc = 0.20  spacing = 0.02
init gaas

implant beryllium energy=10 dose=1e14 monte tilt=10 rotation=15.0
structure outfile=anflex02_4.str

```

```
tonyplot -overlay -st anflex02_*.str -set anflex02.set

quit
```

29.1.3. anflex03.in: Ion Implant Dose Dependence

This example calculated 10 keV Si implant profiles in GaAs for 3 implant doses: 1e14, 1e15, and 1e16 ions/cm**3. Calculation of point defect generation during cascade of atomic collision is also included by specifying `damage` and `rec.frac` parameters. This allows the user to predict the crystal amorphization effect which results in shortening of channeling tails with dose increase.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 29.3: Dose Dependence of Si Implant into GaAs. The tail is shortened at high doses due to damage in the substrate

Input File athena_flash/anflex03.in:

```
go athena

#TITLE: Dose dependence for MC implantation in GaAs

line x loc = 0.0    spacing=0.25
line x loc = 0.25   spacing=0.25
line y loc = 0      spacing = 0.01
line y loc = 0.20   spacing = 0.01
init gaas

```

```
material gaas max.dam=1e22 dam.thresh=25
structure outfile=anflex03.str

implant silicon energy=10 dose=1e14 monte tilt=0 damage rec.frac=1
  n.ion=8000
structure outfile=anflex03_0.str

implant silicon energy=10 dose=1e15 monte tilt=0 damage rec.frac=1
  n.ion=8000
structure outfile=anflex03_1.str

implant silicon energy=10 dose=1e16 monte tilt=0 damage rec.frac=1
  n.ion=8000
structure outfile=anflex03_2.str

tonyplot -overlay -st anflex03_*.str -set anflex03.set

quit
```

29.1.4. anflex04.in: Ion Implant Species Dependence

This example compares results of the Monte Carlo and analytic implant models for Be, Mg, Si-28, Zn and Se implants into GaAs material.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 29.4: Comparison of Be implant using Analytic and MC Implants. The analytic profile does not account for channeling



Figure 29.5: Comparison of Mg implant using Analytic and MC Implants. The analytic profile does not account for channeling



Figure 29.6: Comparison of Si implant using Analytic and MC Implants. The analytic profile does not account for channeling



Figure 29.7: Comparison of Zn implant using Analytic and MC Implants. The analytic profile does not account for chan-

neling.



Figure 29.8: Comparison of Se implant using Analytic and MC Implants. The analytic profile does not account for channeling, though the effect is reduced for the large Se ion compared with the earlier Be example

Input File athena_flash/anflex04.in:

```
go athena

#TITLE: Species dependence for implant in GaAs with MC and Analytic models

# implant species penetration should be less with increasing mass:
#   Species      Atomic Mass
#   Beryllium       9.012
#   Magnesium      24.305
#   Silicon        28.086
#   Zinc           65.38
#   Selenium       78.96
line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing = 0.005
line y loc = 0.50  spacing = 0.1
init gaas

implant beryllium energy=15 dose=1e13 monte n.ion=5000
implant magnesium energy=15 dose=1e13 monte n.ion=5000
implant silicon   energy=15 dose=1e13 monte n.ion=5000
```

```
implant zinc      energy=15 dose=1e13 monte n.ion=5000
implant selenium  energy=15 dose=1e13 monte n.ion=5000
structure outfile=anflex04_0.str

line x loc = 0.0   spacing=0.25
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing = 0.005
line y loc = 0.50  spacing = 0.1
init gaas

implant beryllium energy=15 dose=1e13
implant magnesium energy=15 dose=1e13
implant silicon   energy=15 dose=1e13
implant zinc      energy=15 dose=1e13
implant selenium  energy=15 dose=1e13
structure outfile=anflex04_1.str

tonyplot -overlay anflex04_*.str -set anflex04_0.set
tonyplot -overlay anflex04_*.str -set anflex04_1.set
```

```
tonyplot -overlay anflex04_*.str -set anflex04_2.set
tonyplot -overlay anflex04_*.str -set anflex04_3.set
tonyplot -overlay anflex04_*.str -set anflex04_4.set

quit
```

This page is intentionally left blank

30.1. ATHENA_CALIBRATION: Process Simulator Calibration

30.1.1. ancaex01.in: OED Tuning Using TWO.DIM for DRYO2

Requires: SSUPREM4

This example compares the `two.dim` and `Fermi` methods for simulation of diffusion in an oxidizing ambient. The two.dim method shows the Oxidation Enhanced Diffusion (OED) effect. The next example, `ancaex02.in`, demonstrates how OED profiles could be calibrated.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 30.1: Importance of including Oxidation Enhanced Diffusion via the TWO.DIM model for diffusions in oxidizing ambients

Input File athena_calibration/ancaex01.in:

```
go athena

#TITLE: OED using two.dim for dryo2
line x loc = 0.0
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing=0.02
line y loc = 1.50  spacing=0.05
line y loc = 5.0   spacing=1.0
line y loc = 100.0
init silicon c.boron=1.0e14
```

```
#
#perform uniform boron implant
implant boron dose=3e14 energy=70
#
structure outfile=ancaex01_0.str
method fermi compress
diffuse time=30 temperature=1000 dryo2
#
structure outfile=ancaex01_1.str

init infile=ancaex01_0.str
#perform diffusion with point defects
method two.dim compress
diffuse time=30 temperature=1000 dryo2
structure outfile=ancaex01_2.str

tonyplot -overlay -st ancaex01_*.str -set ancaex01.set

quit
```

30.1.2. ancaex02.in: OED Calibration Using THETA.0

Requires: SSUPREM4

This example shows how to use the interstitial injection coefficient `theta.0` for calibration of Oxidation Enhanced Diffusion. Only prefactor theta.0 is used in this example. If data on temperature dependence of OED was available, `theta.E` could be calibrated as well.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 30.2: Effect of THETA.0 Parameter on Diffusion Profile in oxidizing ambient

Input File athena_calibration/ancaex02.in:

```
go athena

#TITLE: OED calibration using theta.0
# original
#inter silic /oxide theta.0=3.67e-5 theta.E=-0.902 Gpow.0=0 Gpow.E=0
  vmole=5e22

line x loc = 0.0
line x loc = 0.25  spacing=0.25
line y loc = 0     spacing=0.02
line y loc = 1.50  spacing=0.05
line y loc = 5.0   spacing=1.0
line y loc = 40.0
init silicon c.boron=1.0e14
#
#perform uniform boron implant
implant boron dose=3e14 energy=70
#
structure outfile=ancaex02_0.str

#perform diffusion with point defects
```

```
method two.dim compress
diffuse time=30 temperature=1000 dry
structure outfile=ancaex02_1.str

init infile=ancaex02_0.str
inter silic /oxide theta.0=2e-5 theta.E=-0.902 Gpow.0=0 Gpow.E=0
  vmole=5e22
diffuse time=30 temperature=1000 dry
structure outfile=ancaex02_2.str

init infile=ancaex02_0.str
inter silic /oxide theta.0=1e-5 theta.E=-0.902 Gpow.0=0 Gpow.E=0
  vmole=5e22
diffuse time=30 temperature=1000 dry
structure outfile=ancaex02_3.str

init infile=ancaex02_0.str
inter silic /oxide theta.0=5e-6 theta.E=-0.902 Gpow.0=0 Gpow.E=0
  vmole=5e22
diffuse time=30 temperature=1000 dry
structure outfile=ancaex02_4.str

tonyplot -overlay -st ancaex02_*.str -set ancaex02.set


quit

```

30.1.3. ancaex03.in: Calibration of Thin WetO2 Oxidation

Requires: SSUPREM4/OPTIMIZER

This example performs calibration of thin oxide model parameters using OPTIMIZER. These parameters are quite accurately characterized for dryo2 ambient, but virtually no information exists for wetO2 ambient. Historically, wet oxidation has been predominantly used for thick oxide growth. However, low temperature, low pressure, short time, wet oxidation is getting more popular now. Unfortunately, to the best of our knowledge, extensive experimental results do not exist for such oxidations. Therefore, calibrations based on just a few experimental points may need to be done for wet thin oxidations.

The default value for thin oxidation term for wetO2 is zero. This usually underestimates thin oxide thickness. The default values result in 71 A and 173 A oxide thicknesses for 30 minutes and 90 minutes wetO2 oxidation at 820°C and pressure of 0.2 atmospheres. Typical experimental values could be, for example, 100 A and 230 A. These values are selected as targets for calibration procedures. Only by using OPTIMIZER can the parameters of thin oxidation models be found.

To load and execute this example, display this text and select the **Load example** button. Once loaded, select **Optimizer...** from the **Main Control** menu. The **Deckbuild: Optimizer** screen will appear in momentarily. Select **Load File...** from the **File** menu and then load the special optimizer file **ancaex03.opt** from the scrolling **Optimizer Load** menu.

Preset parameters and targets are included in this file. So when you select **Parameters** from the **Mode** menu, you will find that `thinox.0` and `thinox.l` are chosen as parameters for optimization. The default values for dryo2 oxidation are selected as an initial guess.

Oxide thicknesses `tox1` and `tox2` after 30 and 90 minutes of oxidation respectively are selected as targets.

You can modify any of these targets and/or parameters. Please refer to the **Optimizer** chapter in the VWF INTERACTIVE TOOLS MANUAL.

If you want to run the preset optimization task, it is better to slightly improve the default stop criteria as follows:

Select **Setup** from the **Mode** menu. Then decrease the maximum error down to 2, and increase the number of Iterations to 10.

After this, select the **Optimize** button on the **Optimizer** menu.

30.1.4. ancaex04.in: Tuning LOCOS Shapes using the COMPRESS Model

Requires: SSUPREM4

This example shows a simple way of tuning LOCOS bird's beak shape. It is known that the length and thickness of a bird's beak during LOCOS oxidation decrease with increase of nitride thickness. This happens because thicker nitride produces higher stresses, which decreases oxidation rates. Viscous stress-dependent oxidation model should be used to accurately predict this effect. However, this model is very time consuming, therefore one can use the `compress` model with higher Young modulus for nitride in order to decrease bird's beak length. This example calculates LOCOS for two values of nitride Young modulus coefficient 1e14 (default) and 1e15.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.

30.1.5. ancaex05.in: Tuning LOCOS Shapes using the VISCOUS Model

Requires: SSUPREM4

This example shows calibration of LOCOS bird's beak shape using `viscous` stress-dependent oxidation model. It is known that the length and thickness of bird's beak during LOCOS oxidation decreases with an increase of nitride thickness. This happens because thicker nitride produces higher stresses which decreases oxidation rates. Viscous stress-dependent oxidation model should be used to accurately predict this effect. One of the key calibration parameters for this model is nitride viscosity. This example calculates LOCOS for two values of nitride viscosity prefactor visc.0=1.8e15 (default) and 1.8e14. If the temperature dependence of bird's beak shape is to be calibrated, the exponential factor visc.E could be used.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.

30.1.6. ancaex06.in: Tuning Threshold Voltage using Dopant Segregation

Requires: SSUPREM4

This example demonstrates how segregation can be used for tuning of threshold voltage of a MOS structure. A simplified 1D simulation of an NMOS channel formation is considered. The coefficient **Seg.0**, which is the prefactor of boron segregation coefficient on the silicon/oxide interface, is selected as a tuning parameter. Simulation is repeated for seven different values of Seg.0. For each case the threshold voltage is extracted using **QuickMOS 1dvt.** The results are stored in the **ancaex06.dat** file for subsequent plotting of Vt vs Seg.0 curve. Such a curve could be used for selecting of appropriate segregation coefficient. Boron profiles near the interface are plotted separately in TONYPLOT. One can see that boron surface concentration (and therefore threshold) increases with Seg.0.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.

30.1.7. ancaex07.in Tuning Directional and Isotropic Etch for Spacer Formation

Requires: ELITE

This example demonstrates how to use etch rate parameters to tune a MOS spacer width and shape. In the first simulation, the etch rate is completely directional. This corresponds to simple dry (anisotropic) etching of SSUPREM4. The dry etch model usually overestimates spacer width. In the second simulation, an isotropic component is included which results in a narrower spacer.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.

30.1.8. ancaex08.in: Multiple Implant Profile Optimization

Requires: SSUPREM4/OPTIMIZER

This example performs optimization of implant doses and energies, as well as diffusion temperature in order to obtain a specified boron doping profile as well as pn-junction depth. The target boron profile is a constant doping with concentration of 2e16 all the way from the silicon surface until x=1.0. The target for the pn-junction depth is 1.5 microns.

To load and execute this example, display this text and select the **Load example** button. Once loaded, select **Optimizer...** from the **Main Control** menu. The **Deckbuild: Optimizer** screen will appear in momentarily. Select **Load File...** from the **File** menu and then load the special optimizer file **ancaex08.in.opt** from the scrolling **Optimizer Load** menu.

Preset parameters and targets are included in this file. So when you select **Parameters** from the **Mode** menu you will find that doses and energies for all four implants, as well as diffusion temperature, are chosen as parameters for optimization.

Boron profile **bcurve** is chosen as an optimization target. Constant concentrations of 2e16 are set as target values for depths from 0.02 to 1 micron. Then a concentration of 1e16 is set for 1.1 microns. Weights for all these points are set to 1.0. A junction depth of 1.5 microns is an additional point target with a weight of 6.0.

Of course, now you can modify any of these targets and/or parameters. Please refer to the **Optimizer** chapter in the VWF INTERACTIVE TOOLS MANUAL.

If you want to run the preset optimization task, it is better to slightly relax the default stop criteria as follows:

Select **Setup** from the **Mode** menu and increase the maximum error to 20.

After this, select the **Optimize** button.

After optimization is completed, you may execute TONYPLOT which will plot final optimized profiles.

30.1.9. ancaex09.in: Extraction of Implant Moments from Measured Data

Requires: SSUPREM4/OPTIMIZER/SPDB (Optional)

This example shows how the general purpose OPTIMIZER can be used for extraction of dual-Pearson implant parameters from an experimental profile. This technique could be useful when ATHENA's implant look-up tables do not include moments for a specific implant (e.g. 0 degrees angle, very high or low energy, "exotic" ion-material combination, etc.) or when it appears that a simulated profile does not match an experimental one. This example can also be used as a tutorial for setting up an optimization procedure with a so-called curved target.

An experimental profile for 1e13 ion/cm**2, 15 keV, O degrees Boron implant into <100> crystalline silicon is used in this example. This SIMS profile published by H. Kinoshita of the Microelectronic Research Center (Austin, TX) is stored in the SILVACO Profile Data Base (SPDB) under ID number 4256. For those users who have not purchased this invaluable source of extremely useful technological information, the profile is provided with the example in two formats: User Data Format **ancaex09_4256.str** and simple XY format which can be read into the OPTIMIZER **ancaex09_4256.opt.**

To load and execute this example, display this text and select the **Load example** button while this text is displayed. The input file and several support files including **ancaex09_4256.dat** and **ancaex09_4256.opt** will be copied to your current working directory at this time.

Selection of initial values for optimization is usually very important in setting optimization tasks. In case of implantation, one can use parameters available for similar implant. To demonstrate flexibility of this approach parameters corresponding to the same implant but at seven degreees are used. It can be done by using default SVDP model and `print.mom` parameter in the `implant` statement.

If you want to compare this initial approximation with the experimental one, just select the **run** button to execute the example. In the end, the two profiles will be overlaid in TONYPLOT, so you can see that the experimental profile has much bigger and deeper channeling peak than the default seven degrees profile.

The goal of the example is to find a set of dual-Pearson parameters, `range, std.dev, gamma,` etc. that will give a profile close to the experimental one. In order to achieve this, select **Optimizer...** from the **Main Control** menu. The **Deckbuild: Optimizer** screen will appear in momentarily. Unlike the first example **ancaex08.in**, this example does not use a special optimizer file with preset parameters and targets, so you should repeat the following steps in order to set the optimization task.

First, select **Parameters** from the **Mode** menu to display the Parameter worksheet. Select (highlight) the MOMENTS statement and choose **Add** from the **Edit** pulldown menu. The **Parameter, define** popup will appear. Check all boxes, excluding the first two which correspond to dose and energy, then press **Apply.** Nine new rows will be inserted into the Parameter worksheet. By default all min/max values for parameters are ±50%. This should be changed for some parameters. For example, we may expect a wider range for skewness (asymmetry) for Pearson distributions, therefore we set **Minimum value** to –1.0 and **Maximum value** to +3.0 for `gamma` and `sgamma` parameters. Also, the intervals for the fourth moment, `kurtosis` of both first and second Pearson functions, should be wider (2.7,10.0); `dratio` should not be higher than 1.0, therefore its maximum value should be set to 1.0. To edit any of these numeric values, position the pointer over the corresponding cell of the worksheet, click the left-hand mouse button, edit the value, and do not forget to push the return button after finishing the editing.

The next step is setting the curved target. Select **Targets** from the **Mode** menu. Highlight the `extract` line. This line extracts the Boron chemical concentration profile which corresponds to the SIMS measured profile. After that, choose **Add** from the **Edit** pulldown menu. A new row will be inserted into the Target worksheet. The target name will correspond to the name in the extract statement. Now, to insert experimental profiles as a curved target, select the only row in the worksheet by double-clicking the left-hand button while the pointer is positioned anywhere in the row, then choose **From File** from the **Insert** submenu from the **X/Y Data** menu under the **Edit** pulldown menu. As a result, the **Optimizer X/Y Target Data** popup will appear. Select **ancaex09_4256.opt** file and then press **Load.** The next step is to change the response type of the curved target from **log** to **linear, o**therwise, the estimated error will be very small because Optimizer would take the log of the result and target at each point before computing the error. Choose **Select all** from the **Edit** pulldown menu. All the rows will appear raised. Then choose **Lin/Log** from the same menu and all row's response type will be toggled.

Now, once the parameters, targets and setup information have been defined, the Optimization process can be started by pressing the **Optimize** button. During the optimization run, the **Graphics** mode shows real-time updates of parameter, target and error values. After optimization is completed, you may execute TONYPLOT, to plot the comparison between optimized and measured profiles.

31.1. ATHENA_ADAPTMESH: Process Simulation with Adaptive Meshing

31.1.1. amex01.in: Trench Isolation Structure

Requires: SSUPREM4/ELITE

This example shows the fabrication of a trench isolation structure using ELITE and SSUPREM4 to model etching and oxidation processes. We test this as a nonplanar structure application for base mesh generation.

The example begins without grid definition. The `Grid.Model` statement specifies a grid type `trench`.

Finally, the structure is plotted using TONYPLOT and the structure saved to a file.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 31.1: Adaptive Mesh in Trench Isolation. Surface refinement and Doping refinement criteria are used

Input File athena_adaptmesh/amex01.in:

```
go athena

#TITLE: Trench isolation

grid.model template=TRENCH

```

```
# We Don't need this anymore ...
#
#set trench_bottom=0.5
#line x loc=0.00 spac=0.20
#line x loc=0.35 spac=0.1
#line x loc=0.6 spac=0.1
#line x loc=1.2 spac=0.02
#line x loc=1.8 spac=0.02
#line x loc=2.4 spac=0.1
#line x loc=2.65 spac=0.1
#line x loc=3.0 spac=0.20
#
#line y loc=0.00 spac=0.02
#line y loc=$trench_bottom spac=0.02
#line y loc=2.5 spac=0.4

init silicon c.boron=1.0e15 orientation=100 spac=2 width.str=3.0
  depth.str=2.5

diffuse temp=900 time=25 dryo2 cl=3
deposit nitride thick=0.4

deposit photo thick=1. div=10
#
etch   photo   start x=1.2 y=-10
etch   photo      cont  x=1.2 y=10
etch   photo      cont  x=1.8 y=10
etch   photo      done  x=1.8 y=-10

relax y.min=0.4 dir.y=f
relax y.min=1.0 dir.y=f

etch nitride thick=0.5

strip photo

etch oxide thick=0.1

# Setup a silicon trench etching machine.....
```

```
rate.etch machine=trench_etch rie silicon iso=0.1 dir=0.9 u.m
rate.etch machine=trench_etch rie oxide iso=0.1 dir=0.9 u.m


etch machine=trench_etch time=0.5 minute dx.mult=0.5

method adapt
adapt.mesh verbose

# Oxidize the trench....
method grid.ox=0.05
diffuse temp=925 time=10 weto2

# Implant the trench
implant boron energy=35 dose=1.2e12


# Strip off all the nitride
strip nitride

# Dip off the oxide
etch oxide all

# deposit a new oxide layer to fill the trench, with CVD TEOS

# First define the TEOS machine.....
rate.depo machine=teos oxide u.m cvd dep.rate=1.0 step.cov=0.9

# Now employ it....
deposit machine=teos time=55 seconds divisions=8

etch oxide left p1.x=0.6
etch oxide right p1.x=2.4

# S/D Implant....
implant arsenic energy=85 dose=2e15

# Final anneal...
diffuse temp=925 time=25 nitrogen dump=1 dump.prefix = test

# now make electrode contacts and move it into device simulation.....
deposit alumin thick=0.1
etch alumin start x=0.35 y=-10
```

```
etch alumin cont  x=0.35 y=10
etch alumin cont  x=2.65 y=10
etch alumin done  x=2.65 y=-10

electrode name=source x=0.1
electrode name=drain x=2.9


structure outf=amex01.str

tonyplot amex01.str -set amex01_0.set
```

31.1.2. amex02.in: PBL Isolation Structure

Requires: MASKVIEWS/SSUPREM4/ELITE

This example shows the fabrication of a trench isolation structure using ELITE and SSUPREM4 to model etching and oxidation processes.

The example begins with grid definition. This is tailored to place a fine grid in the area that will be exposed during the trench etch. The `initial` statement does not specify a material type, so silicon is taken by default. A background doping of boron with a concentration of 1.0e15 is specified.

Finally, the structure is plotted using TONYPLOT and the structure saved to a file.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 31.2: Adaptive Mesh in Deep, Refilled Trench

Input File athena_adaptmesh/amex02.in:

```
go athena

grid.model template=PBL

#TITLE: Trench isolation example

# place the finest grid around the trench
#line  x    location=0.0   spacing=0.5
#line  x    location=1.50  spacing=0.5
#line  x    location=1.90  spacing=0.10
#line  x    location=1.98  spacing=0.02
#line  x    location=2.00  spacing=0.004
#line  x    location=2.02  spacing=0.007
#line  x    location=2.10  spacing=0.01
#line  x    location=2.30  spacing=0.05
#line  x    location=2.75  spacing=1.0

#line  y    location=0.0   spacing=0.01
#line  y    location=0.010 spacing=0.01
#line  y    location=0.1   spacing=0.02
#line  y    location=0.6   spacing=0.1
```

```
#line  y    location=3.4   spacing=0.8
#line  y    location=6.0   spacing=0.2
#line  y    location=6.5   spacing=0.50
#line  y    location=7.05  spacing=1.0
initial c.boron=1.0e15 no.imp space.m=2 width.str=2.8 depth.str=7.05
#initial c.boron=1.0e15 no.imp space.m=2
method adapt
deposit nitride th=.1 div=4
etch       nitride right    p1.x=2.0
# form the complete trench structure
structure  reflect
# DRYTECH QUAD RIE machine
rate.etch machine=DRYTEK silicon u.m  rie isotropic=0.0 directional=0.6
etch machine=DRYTEK time=10 minutes
structure outfile=amex02_0.str
etch nitride all
# horizontal profiles
deposit    oxide  thick=0.02 divisions=1
rate.depo machine=polydep cvd polysil u.m dep.rate=.8 step.cov=1
deposit machine=polydep time=1 minute divis=10
structure outfile=amex02_1.str
rate.depo machine=d3 hemis photoresist a.m \
 angle1=90. angle2=-90. dep.rate=1000. sigma.dep=0.2 \
 smooth.win=2.0 smooth.step=5
#
deposit mach=d3 time=15.0 minutes divis=6
structure outfile=amex02_2.str
rate.etch machine=test polysilicon a.m  wet.etch iso=1000.0
rate.etch machine=test silicon     a.m  wet.etch iso=1000.0
rate.etch machine=test nitride     a.m  wet.etch iso=1000.0
rate.etch machine=test oxide       a.m  wet.etch iso=1000.0
```

```
rate.etch machine=test photores    a.m  wet.etch iso=1000.0

# planarize
etch machine=test time=24 minutes dt.max=1
relax surface dx.surf=0.03
structure outfile=amex02_3.str

# deposit masking nitride
deposit    nitride   thick=0.03

etch nitride right p1.x=2.85

# turn off poly lifting as it is not needed
method grid.ox=0.05 lift.poly=f fermi

method adapt

# Adjust these parameters to match observed
# oxide thickness on poly and silicon
oxide silicon wet par.h.0 =2.0
oxide wet par.h.0 =10
diffuse    temp=1000 time=10 dryo2
structure outfile=amex02_4.str

diffuse   temp=1000 time=100.0 weto2 dump=1 dump.prefix = test

structure outfile=amex02_5.str
tonyplot -ttitle amex02.in:Final

quit
```

31.1.3. amex03.in: LOCOS Isolation Structure

Requires: /SSUPREM4/ELITE

This example shows the fabrication of a trench isolation structure using ELITE and SSUPREM4 to model etching and oxidation processes. We test this as a nonplanar structure application for base mesh generation.

The example begins without grid definition. The `grid.model` statement specifies a grid type `locos`.

Finally, the structure is plotted using TONYPLOT and the structure saved to a file.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 31.3: Adaptive Mesh Control for LOCOS Oxidation

Input File athena_adaptmesh/amex03.in:

```
go athena

#  Active Area Isolation Structure Punchthrough Extraction Test

grid.model template=LOCOS

method adapt

#
#line x loc=0 spac=0.1 tag=left
#line x loc=0.5 spac=0.04
#line x loc=0.7 spac=0.04
#line x loc=1.5 spac=0.1 tag=right
#
#line y loc=0.00 spac=0.01 tag=top
#line y loc=0.2 spac=0.01
#line y loc=0.35 spac=0.04
#line y loc=0.5 spac=0.04
#line y loc=3 spac=0.3 tag=bottom
```

```
#
init orientation=100 c.phos=1e12 space.m=1 width.str=1.5 depth.str=3
#pwell formation including masking off of the nwell
#
diffus time=30 temp=1000 dryo2 press=1.00 hcl=3
#
etch oxide thick=0.02
#
#P-well Implant
#
implant boron dose=1e12 energy=100 pears
#
diffus temp=950 time=100 weto2 hcl=3
#
#N-well implant not shown -
#
# welldrive starts here
diffus time=50 temp=1000 t.rate=4.000 dryo2 press=0.10 hcl=3
#
diffus time=220 temp=1200 nitro press=1
#
diffus time=90 temp=1200 t.rate=-4.444 nitro press=1
#
etch oxide all
#
#Create Active..
deposit oxide thick=0.01 divi=1
deposit nitride thick=0.3 divi=1
deposit barrier thick=.1 divi=1
#
etch barrier left p1.x=0.5
etch nitride thick=.35
etch oxide thick=.02
etch silicon dry thick=0.15
#
#Field Implant
adapt.par accur.mult=0.2
implant boron dose=1e12 energy=50
```

```
#
strip
#relax y.min=1
#relax y.min=1.5
#
#Fieldox

method grid.ox=0.07

adapt.par accur.mult=0.0100
diffuse temp=950 time=180 weto2 hcl.pc=3 dump=1 dump.prefix = test


etch nitride all
#
etch oxide thick=0.02 dry
#
#sacrificial "cleaning" oxide
diffus time=20 temp=1000 dryo2 press=1 hcl=3
#
etch oxide dry thick=0.02
#
#gate oxide grown here:-
diffus time=11 temp=925 dryo2 press=1.00 hcl=3
#
# Extract the gate oxide thickness...
extract name="gate ox" thickness oxide mat.occno=1

# Implant the VT adjust....
#vt adjust implant
adapt.par accur.mult=1
implant boron dose=1e12 energy=25 pearson
#
#LDD Spacer formation
depo oxide thick=0.120 divisions=8
#
etch oxide dry thick=0.120
#
implant arsenic dose=5.0e15 energy=50 pearson
#
method fermi compress
adapt.par accur.mult=1
diffuse time=30 temp=900 nitro press=1.0 dump=2 dump.prefix = test
```

```
#

# Make the metal contacts to be used as electrodes in Atlas...
etch oxide right p1.x=1
deposit alumin thick=0.03 divi=2
etch alumin left p1.x=0.7
structure mirror left
# Name the electrodes....
electrode name=source x=-1.2
electrode name=drain x=1.2
electrode name=substrate backside

struct outf=amex03.str

tonyplot  amex03.str -set amex03.set
```

31.1.4. amex04.in: Tilt Implant and Anneal on Irregular Mesh

Requires: SSUPREM4

This example shows the tilt implantation of a trench structure using SSUPREM4 adaptive gridding technique.

The example begins with input grid structure. The `method adapt` statement specifies an adaptive gridding option.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 31.4: Adaptive Mesh Refinement on a Trench Sidewall Implant

Input File athena_adaptmesh/amex04.in:

```
go athena
#
line x loc=0.00 spac=0.2
line x loc=0.4 spac=0.02
line x loc=0.55 spac=0.03
line x loc=0.7 spac=0.02
line x loc=1.1 spac=0.2
#
line y loc=0.00 spac=0.05
line y loc=1 spac=0.2

init

deposit oxide thick=0.011
deposit poly thick=0.07 div=4
deposit nitride thick=0.2 div=5

etch nitride start x=0.4 y=-1
etch cont x=0.4 y=1
```

```
etch cont x=0.7 y=1
etch done x=0.7 y=-1

relax y.min=0.1
oxide poly /nitride split.angle=359
method grid.ox=0.04
method visc oxide.r=3e-2
oxide stress.dep=t Vc=150 Vr=14 vd=10
material oxide visc.0=5.1 visc.E=3.48
material nitride visc.0=5.96e5 visc.E=2.5625

diffuse time=10 temp=1100 wet

deposit poly thick=0.1 div=10
rate.etch machine=m0 rie poly dir=0.1 iso=0.0 u.m
etch machine=m0 time=62.0 second dx.mult=0.5
deposit photo thick=0.5 div=5

etch photo start x=0.4 y=-1
etch cont x=0.4 y=1
etch cont x=0.7 y=1
etch done x=0.7 y=-1

rate.etch machine=m1 rie oxide dir=1.0 iso=0.00 u.m
rate.etch machine=m1 rie poly dir=0.2 iso=0.0 u.m
rate.etch machine=m1 rie photo dir=0.1 iso=0.01 u.m
etch machine=m1 time=17.0 second dx.mult=0.5

rate.etch machine=m2 rie silicon dir=1.0 iso=0.00 u.m
rate.etch machine=m2 rie poly dir=5.0 iso=5 u.m
etch machine=m2 time=20.0 second dx.mult=0.5
etch poly thick=0.15
strip
struct outf=amex04_0.str

adapt.par silicon i.arsenic conc.min = 1e14
method adapt

implant arsenic energy=30 dose=4e13 tilt=15
struct outf=amex04.str

tonyplot amex04.str -set amex04.set
```

```

quit
```

31.1.5. amex05.in: MOS Device Structure

Requires: SSUPREM4/ELITE

This example shows the fabrication of MOS structure using ELITE and SSUPREM4**.** The grid is adaptive and the **BASE.MESH** feature is automatic apply to the initial mesh generation. This is a good test for gridding on a planar structure.

The example begins without grid definition. A `MOS` grid template is used. This template is installed under the subdirectory, `var/athena.grid` below the Silvaco Installation Directory. The template can be modified and reused for other comparable type simulations. More application-specific templates will be installed to optimize grid parameters for application-specific simulation.

Finally, the structure is plotted using TONYPLOT and the structure saved to a file.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 31.5: Adaptive Mesh for 0.3 micron MOSFET

Input File athena_adaptmesh/amex05.in:

```
go athena
# Start Athena running...


grid.model template=MOS
```

```

# remember to assign width.str and depth.str

init orientation=100 c.phos=1e14 space.mul=3 width.str=0.5 depth.str=0.8

#method adapt
#adapt.par accur.mult = 1

#pwell formation including masking off of the nwell
#
diffus time=30 temp=1000 dryo2 press=1.00 hcl=3
#
etch oxide thick=0.02
#
#P-well Implant
#
implant boron dose=8e12 energy=100 pears

#
diffus temp=950 time=100 weto2 hcl=3
#
#N-well implant not shown -
#
# welldrive starts here
diffus time=50 temp=1000 t.rate=4.000 dryo2 press=0.10 hcl=3
#
diffus time=220 temp=1200 nitro press=1
#
diffus time=90 temp=1200 t.rate=-4.444 nitro press=1
#
etch oxide all
#
#sacrificial "cleaning" oxide
diffus time=20 temp=1000 dryo2 press=1 hcl=3
#
etch oxide all
#
#gate oxide grown here:-
method grid.ox=0.0035
diffus time=11 temp=925 dryo2 press=1.00 hcl=3

# Extract a design parameter.....

```

```
extract name="gateox" thickness oxide mat.occno=1 x.val=0.49
#
#vt adjust implant
adapt.par accur.mult = 5
implant boron dose=9.5e11 energy=10 pearson
#
depo poly thick=0.2 divi=10
#
#from now on the situation is 2-D
#
etch poly left p1.x=0.35
#
method adapt
method fermi compress
method grid.ox=0.035
diffuse time=3 temp=900 weto2 press=1.0 dump=1 dump.prefix = test
adapt.par accur.mult=1
#
implant phosphor dose=3.0e13 energy=20 pearson
#
depo oxide thick=0.120 divisions=8
#
etch oxide dry thick=0.120
#
implant arsenic dose=5.0e15 energy=50 pearson
#
method fermi compress
diffuse time=1 temp=900 nitro press=1.0 dump=1 dump.prefix = test
#
#
etch oxide left p1.x=0.2
deposit alumin thick=0.03 divi=2
etch alumin right p1.x=0.18
# Extract another design parameters...
# extract final S/D Xj...
extract name="nxj" xj silicon mat.occno=1 x.val=0.1 junc.occno=1
# extract the long chan Vt...
extract name="n1dvt" 1dvt ntype vb=0.0 qss=1e10 x.val=0.49
# extract a curve of conductance versus bias....
```

```
extract start material="Polysilicon" mat.occno=1 bias=0.0 bias.step=0.2
  bias.stop=2 x.val=0.45
extract done name="sheet cond v bias" curve(bias,1dn.conduct materi-
  al="Silicon" mat.occno=1  region.occno=1) outfile="extract.dat"
# extract the N++ regions sheet resistance...
extract name="n++ sheet rho" sheet.res material="Silicon" mat.occno=1
  x.val=0.05 region.occno=1
# extract the sheet rho under the spacer, of the LDD region...
extract name="ldd sheet rho" sheet.res material="Silicon" mat.occno=1
  x.val=0.3 region.occno=1
# extract the surface conc under the channel....
extract name="chan surf conc" surf.conc impurity="Net Doping" materi-
  al="Silicon" mat.occno=1 x.val=0.45

structure mirror right

electrode name=gate x=0.5 y=0.1
electrode name=source x=0.1
electrode name=drain x=0.9
electrode name=substrate backside

structure outfile=amex05_0.str

# plot the structure
tonyplot  amex05_0.str -set amex05_0.set
```

31.1.6. amex06.in: MESFET Formation

Requires: FLASH/DEVEDIT

This example demonstrates fabrication and electrical analysis of a MESFET structure using the FLASH module of ATHENA. The example uses DevEdit at various points in the process to optimize the grid. The example shows:

- MESFET fabrication using FLASH

This example starts by interfacing ATHENA and DEVEDIT to provide a GaAs MESFET structure using silicon and beryllium implants. Details of using FLASH can be found in the ATHENA_FLASH examples section.

To load and run this example, select the **Load example** button while this text is displayed. The input file and several support files will be copied to your current working directory at this time. Once loaded into DECKBUILD, select the **run** button to execute the example.



Figure 31.6: Adaptive Mesh for submicron MESFET

Input File athena_adaptmesh/amex06.in:

```
go athena

# GaAs MESFET fabrication and analysis using adaptive mesh

line x loc=-1.5 spac=0.2
line x loc=-.7  spac=0.1
line x loc=-.5  spac=0.05
line x loc=0.0  spac=0.1
line x loc=0.5  spac=0.05
line x loc=0.7  spac=0.1
line x loc=1.5  spac=0.2
#
line y loc=0.00 spac=0.02
line y loc=2.00 spac=0.5
#

grid.model template=MESFET

init gaas c.beryllium=1.0e13 orient=100 space.mult=1
```

```

method adapt
adapt.mesh verbose

implant beryllium energy=100 dose=2e11
implant silicon energy=100 dose=1e12

diffus time=10 temp=850

# deposit and pattern gate metal
deposit titanium thick=.3 divisions=10
etch titanium right p1.x=0.5
etch titanium left  p1.x=-0.5

deposit oxide thick=0.35 divisions=8

etch oxide thick=.4

structure outfile=amex06_0.str

# perform source/drain implant
implant silicon energy=50 dose=1e13
structure outfile=amex06_1.str

init inf=amex06_1.str
diff time=10 temp=850

# deposit ohmic metal
deposit aluminum thick=.2 divisions=4

etch aluminum        start x=-1 y=10
etch cont                  x=-1 y=-10
etch cont                  x=1 y=-10
etch done                  x=1 y=10
#
electrode name=source x=-1.4
electrode name=drain x=1.4
electrode name=gate x=0.0


structure outfile=amex06_2.str

```

```
tonyplot  amex06_2.str  -set amex06_2.set

```

31.1.7. amex07.in: CCD Structure Formation

Requires: SSUPREM4

In this example, a CCD structure is constructed using ATHENA process simulation. This structure can be passed to ATLAS for electrical testing.

The first stage of the input constructs the CCD geometry and doping profiles in ATHENA. The CCD device consists of a storage node on the left of the structure, controlled by a polysilicon gate. A transfer node to the right of this is also controlled by both a polysilicon gate and a drain region of heavy n+ doping with a metal contact.

The CCD structure has a n- active region at the surface above a p-type substrate. Under the transfer node an extra p-type implant is given. This implant is of sufficient magnitude to create a potential difference between the storage and transfer nodes, but is not so large as to cause a junction.

The process simulation is designed to run extremely quickly so there is a minimum of diffusion steps included. The final stages of the ATHENA input performs the electrode definition needed by ATLAS. Also some `EXTRACT` statements are used to measure the junction depth of the active region under the storage and transfer nodes.

To load and run this example, select the **Load example** button while this text is displayed. The input file and several support files will be copied to your current working directory at this time. Once loaded into DECKBUILD, select the **run** button to execute the example.

31.1.8. amex08.in: Flash EPROM Formation

This example uses **Adaptive griding** to allow fast, accurate and easy simulation over a complex topography created using SSUPREM and ELITE**.** The EPROMs are fabricated by ATHENA using the autogrid capability.

The example begins with a silicon substrate that is 10 microns wide and 10 microns deep. Two parameters on the `adapt.parm` statement specifies the error criteria and the refinement species. The parameter, `accur.mult,` specifies the multiplication factor for accurate control. This results in easy control for the dopant simulation in ATHENA.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.

31.1.9. amex09.in: Poly Emiter Bipolar Device Structure (PNP)

This example shows the fabrication of a PNP Bipolar device structure using ELITE and SSUPREM4 to model etching and oxidation processes. In this example the **Adaptive Meshing** is turned on by command `method adapt`. The grid should be adapted during ion implantation and anneal process. The initial grid setting is still using that designed by hand. However, the first time use of new grid feature `grid.model` allows the grid parameters to be optimized according to its application area, e.g. PNP Bipolar device and stored in the template `PNP`**.** At this moment, the template `PNP` only contains one line:

```
adapt.par disable oxide adapt.par disable poly
```

It says DON'T do adaptive grid in oxide poly, since the grid inside oxide and poly is not important for PNP Bipolar device.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 31.7: Adaptive Mesh for Poly Emitter Bipolar

Input File athena_adaptmesh/amex09.in:

```
go athena
#TITLE: Polysilicon Emitter Bipolar Example  (PNP)
#Silvaco International 1997

grid.model template=PNP

line x loc=0.0          spacing=0.03
line x loc=0.2          spacing=0.02
line x loc=0.24         spacing=0.01
line x loc=0.3          spacing=0.015
line x loc=0.8         spacing=0.15

line y loc=0.0          spacing=0.01
line y loc=0.1          spacing=0.01
line y loc=0.4          spacing=0.02
line y loc=0.5          spacing=0.06
line y loc=1.0          spacing=0.15

init boron conc=2e16
```

```

method adapt

# base implant
implant   phos  energy=100 dose=8e13

# base drive
diffuse time=5 temp=900

# deposit polysilicon
deposit poly thick=0.3 divisions=6 min.space=0.05

# Implant to dope polysilicon
implant bf2 dose=3e15 energy=35

# Pattern the poly
etch poly right p1.x=0.2

structure outfile=amex09_1.str

# relax the mesh in the substrate and extrinsic areas
relax y.min=.5
relax x.min=0.4

# emitter drive
method compress fermi
diffuse time=45   temp=900 nitrogen

# extrinsic emitter implant under spacer
implant phos dose=2e14 energy=70

# deposit spacer
deposit oxide thick=0.3 divisions=10  min.space=0.1

# etch the spacer back
etch oxide dry thick=0.3

structure outfile=amex09_2.str

# n+ base contact implant
implant arsenic dose=1e15 energy=50

# contact drive
```

```
diffuse time=30 temp=900   nitrogen

structure outfile=amex09_3.str

# put down Al and etch to form contacts
deposit alum thick=0.05 div=2

etch alum start x=0.16 y=-4
etch continue x=0.16 y=0.2
etch continue x=0.6 y=0.2
etch done  x=0.6 y=-4

# Name the electrodes for use with Atlas
electrode x=0.0    name=emitter
electrode x=0.7    name=base
electrode backside name=collector

# extract junction depths
extract name="EB_xj" xj material="Silicon" mat.occno=1 x.val=0.1 junc.occno=1
extract name="BC_xj" xj material="Silicon" mat.occno=1 x.val=0.1 junc.occno=2
extract name="base_width" $BC_xj - $EB_xj

#extract 1D electrical parameters
extract name="base_rho" n.sheet.res material="Silicon" mat.occno=1
  x.val=0.1 region.occno=2
extract name="poly_emitter_rho" p.sheet.res material="Polysilicon"
  mat.occno=1 x.val=0.1 region.occno=1 semi.poly

#  Save the final structure
structure outfile=amex09_4.str
tonyplot amex09_4.str
```

31.1.10. amex10.in: Large Dimension Power Device Structure

Requires: SSUPREM4/ELITE

This example shows the fabrication of a large dimension Power device structure using ELITE and SSUPREM4 to model etching and oxidation processes. In this example the **Adaptive Meshing** is turned on by the command `method adapt`. The grid should be adapted during ion implantation and anneal processes. The initial grid setting is still using that designed by hand. However, the first time use of a new grid feature `grid.model`**,** allows the grid parameters to be optimized according to its application area, e.g. POWER Bipolar device and stored in the template `POWER`. Since this device is large, the good mesh design becomes nontrivial. One might use the `relax` command numerously to tailor the original mesh, so the grid budget can be cut low. Only spend the grid on the critical area. The New Base Mesh feature will deal with issue, the user will feel more relieved when encountering this type of application. The `POWER` template contains this new feature- Base Mesh Generation.

The template reads like this:

base.par oxide grad.space= 5 ratio.box=5

base.par silicon grad.space= 3 ratio.box=1.8

base.par polysilicon grad.space= 5 ratio.box=10

base.par nitride grad.space= 5 ratio.box=10

photo grad.space= 5 ratio.box=10

and

base.mesh surf.ly=0.0 surf.dy=0.1

active.ly=1.0 active.dy=0.5 epi.ly=2.0 epi.dy=1.0

sub.ly=10.0 sub.dy=10.0 back.ly=500 back.dy=100

It specifies parameters to generate Oct-tree like mesh. This allows the placing of a fine grid at the area where the dopant is located, while the large grid spacing is assigned to the depth of device where dopant is less critical. Do remember to assign the two parameters using the initial command:

```
width.str=8 depth.str=10
```

It says the initial structure width and depth to be used for simulation with this new base mesh generation is 8 microns by 10 microns.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 31.8: Adaptive Mesh in LDMOS. A separate set of rules are used for larger dimension devices.

Input File athena_adaptmesh/amex10.in:

```
go athena
# Large  Dimension Power Device


#line x loc=0 spac=0.4
#line x loc=0.5 spac=0.1
#line x loc=0.6 spac=0.1
#line x loc=1 spac=0.08
#line x loc=2 spac=0.3

#line x loc=3 spac=0.5
#line x loc=5 spac=0.05
#line x loc=6 spac=0.3
#line x loc=7 spac=0.1
#line x loc=7.4 spac=0.1
#line x loc=7.5 spac=0.1
#line x loc=8 spac=0.4

#
```

```
#line y loc=0.00 spac=0.01
#line y loc=0.2 spac=0.015
#line y loc=0.5 spac=0.06
#line y loc=1    spac=0.12
#line y loc=10.0 spac=2.0
#

grid.model template=POWER

init orientation=100 c.boron=1e15 width.str=8 depth=10

method adapt
adapt.mesh verbose
adapt.par accur.mult = 0.2

# sacrificial oxide
diffus time=30 temp=1000 dryo2
#
etch oxide all
#
# gate oxide growth
# make sure more than one grid point is included within the gate oxide thickness
method grid.ox=0.01
diffus time=50 temp=1000 dryo2 press=1.00 hcl=3
#
extract name="gateox" thickness material="SiO~2" mat.occno=1 x.val=-10

# vt adjust implant
implant boron dose=6e11 energy=20 pearson

# Poly deposition
depo poly thick=0.35 divi=10

# Poly definition
etch poly left p1.x=1
etch poly right p1.x=5
# slightly relax grid
#relax y.min=0.4 dir.y=f
#relax y.min=0.4 dir.y=f

struct outf=amex10_1.str
tonyplot amex10_1.str
```

```

adapt.par accur.mult = 1
adapt.par silicon  i.boron i.phosphor edge.min=1.0e-5  edge.max=1.0e-4

# Light n+ implant
implant phosphor dose=2e12 energy=100 pearson
# S/D mask and implant
depo barrier thick=0.01
etch barrier left p1.x=2
etch barrier right p1.x=7
implant phos dose=3.0e15 energy=100 pearson
strip

struct outf=amex10_2.str
tonyplot amex10_2.str

# final anneal
method fermi compress
phosph poly /oxide trn.0=0.0
diffuse time=30 temp=1000 nitro press=1.0

struct outf=amex10_3.str
tonyplot amex10_3.str

# contact holes
etch oxide left p1.x=0.5
etch oxide right p1.x=7.5

# Contact metal deposition and etching
deposit alumin thick=0.1 divi=2
etch alumin start x=0.6 y=-10
etch cont x=0.6 y=10
etch cont x=7.4 y=10
etch done x=7.4 y=-10

# electrode naming
electrode name=source x=0.3
electrode name=gate x=2 y=0.0
electrode name=drain x=7.7
electrode name=substrate backside


# estimate threshold voltage
```

```
extract name="1dvt" 1dvt ntype x.val=3.0


struct outf=amex10_4.str
tonyplot amex10_4.str amex10_4.set

```

31.1.11. amex11.in: Tutorial Example for Using the Adapt/Base Meshing Feature

Requires: SSUPREM4/ELITE

This example shows the fabrication of MOS structure using ELITE and SSUPREM4**.** The grid is adaptive and the `base mesh` feature automatically applies to the initial mesh generation. This is good testing for gridding on a planar structure.

The example begins without grid definition. A MOS grid template is used. This template is installed under subdirectory **var/athena.grid,** below the Silvaco Installation Directory. The template can be modified and reused for other simulations of similar type. More application-specific templates will be installed to optimize grid parameters for application-specific simulations.

Finally, the structure is plotted using TONYPLOT and the structure saved to a file.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 31.9: Evolution of Adaptive Mesh Control

Input File athena_adaptmesh/amex11.in:

```
go athena
```

```
base.par oxide        grad.space= 3   ratio.box=5
base.par silicon      grad.space= 3   ratio.box=1.8
base.par polysilicon  grad.space= 3   ratio.box=5
base.par nitride      grad.space= 3   ratio.box=5
base.par photo        grad.space= 3   ratio.box=5

base.mesh  offset.x= -0.5 offset.y= -0.5 surf.ly=0.0 surf.dy=0.1\
           active.ly=1.0 active.dy=0.01 epi.ly=2.0 epi.dy=1.0\
           sub.ly=10.0 sub.dy=10.0 back.ly=500 back.dy=100

adapt.par silicon i.arsenic    diff.length=0.001 dose.err=.1
  dose.min=1e14
adapt.par silicon i.boron      diff.length=0.001 dose.err=.005
  dose.min=1e12
adapt.par silicon i.phosphor   diff.length=0.001 dose.err=.05
  dose.min=1e16
adapt.par silicon i.antimony   diff.length=0.001 dose.err=.05
  dose.min=1e16
adapt.par disable nitride oxide


init silicon c.boron=1.0e15 orientation=100 spac=2 width.str=3.0
  depth.str=2.5
#
deposit nitride thick=0.4

deposit photo thick=1. div=10
#
etch   photo   start x=1.2 y=-10
etch   photo      cont  x=1.2 y=10
etch   photo      cont  x=1.8 y=10
etch   photo      done  x=1.8 y=-10


etch nitride thick=0.5
method adapt
adapt.mesh verbose
implant boron dose=1e13 energy=80

diffuse time=60 temp=1100 wet

strip photo
struct outf=amex11_1.str
```

```

# test2 use offset x and y

go athena

base.par oxide        grad.space= 3   ratio.box=5
base.par silicon      grad.space= 3   ratio.box=1.8
base.par polysilicon  grad.space= 3   ratio.box=5
base.par nitride      grad.space= 3   ratio.box=5
base.par photo        grad.space= 3   ratio.box=5

base.mesh  offset.x= 0.2 offset.y= 0.2  active.ly=1.0 active.dy=0.01

adapt.par silicon i.arsenic    diff.length=0.001 dose.err=.1
  dose.min=1e14
adapt.par silicon i.boron      diff.length=0.001 dose.err=.005
  dose.min=1e12
adapt.par silicon i.phosphor   diff.length=0.001 dose.err=.05
  dose.min=1e16
adapt.par silicon i.antimony   diff.length=0.001 dose.err=.05
  dose.min=1e16
adapt.par disable nitride oxide


init silicon c.boron=1.0e15 orientation=100 spac=2 width.str=3.0
  depth.str=2.5
#
deposit nitride thick=0.4

deposit photo thick=1. div=10
#
etch   photo   start x=1.2 y=-10
etch   photo      cont  x=1.2 y=10
etch   photo      cont  x=1.8 y=10
etch   photo      done  x=1.8 y=-10


#relax y.min=0.4 dir.y=f
#relax y.min=1.0 dir.y=f

etch nitride thick=0.5
method adapt
implant boron dose=1e13 energy=80

```

```
diffuse time=60 temp=1100 wet

strip photo

struct outf=amex11_2.str

# test3 use  tensel grid as in tenselTRENCH

go athena

adapt.par silicon i.boron  diff.length=0.001 dose.err=.01 dose.min=1e15
adapt.par disable nitride oxide

grid.model template=tenselTRENCH

init silicon c.boron=1.0e15 orientation=100 spac=2

deposit nitride thick=0.4

deposit photo thick=1. div=10
#
etch   photo   start x=1.2 y=-10
etch   photo      cont  x=1.2 y=10
etch   photo      cont  x=1.8 y=10
etch   photo      done  x=1.8 y=-10


#relax y.min=0.4 dir.y=f
#relax y.min=1.0 dir.y=f

etch nitride thick=0.5

method adapt
adapt.par accur.mult=1
# 0.5 is less accurate than default 1

adapt.mesh chat
implant boron dose=1e13 energy=80

diffuse time=60 temp=1100 wet

strip photo

```

```
struct outf=amex11_3.str

tonyplot amex11_1.str amex11_2.str amex11_3.str -set amex11.set

```

31.1.12. amex12.in: CMOS Latch Up

Requires: SSUPREM4/ELITE

This example shows the fabrication of MOS structure using ELITE and SSUPREM4**.** The grid is adaptive and the `base.mesh` feature automatically applies to the initial mesh generation. This is good testing for gridding on a planar structure.

The example begins without grid definition. A MOS grid template is used. This template is installed under subdirectory **var/athena.grid,** below the Silvaco Installation Directory. The template can be modified and reused for other simulations of similar type. More application-specific templates will be installed to optimize grid parameters for application-specific simulations.

Finally, the structure is plotted using `TonyPlot` and the structure saved to a file.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 31.10: CMOS Structure formed using Adaptive Mesh

Input File athena_adaptmesh/amex12.in:

```
go athena cutfile=amex12.sec

#
```

```
# grid and mask information provided by amex12.sec


#init orientation=100 c.phos=1e14 space=1.5
#
#turn this init 1D into 2D by struct out=amex12_ini.str
#then struct in as two.d, the adaption griding on 2D can
#be tested.


grid.model template=LATCHUP

init inf= amex12_ini.str two.d

method adapt

#
diffuse temp=900 time=30 dryo2 hcl.pc=3
#
deposit nitride thick=0.2 divi=2
#
mask name="WELL"
#
etch nitride dry thick=0.25
#
implant boron dose=5e12 energy=100 pearson
#
strip

# remove some unecessary grid in the substrate
relax  y.min=2.0
relax  y.min=1.0


method compress
diffuse temp=1200 time=10 weto2

#
etch nitride all
#
implant phosphorus energy=50 dose=8e12 pearson
```

```
#
#Well Drive ...
method compress
diffuse temp=1200 time=90 nitrogen
#
etch oxide all

#
deposit oxide thick=0.03 divi=2
#
deposit nitride thick=0.2 divi=2

#
mask name="AAD"
#
etch nitride dry thick=0.4
#
mask name="PFLD"
#
implant boron energy=30 dose=1.5e13 pearson


#
strip


method compress
diffuse temp=1000 time=40 weto2
#
structure outf=amex12_0.str
#
etch nitride all
#
etch oxide dry thick=0.04
#
method compress
diffuse temp=1000 time=10 dryo2
#
etch oxide dry thick=0.025
#
method compress
diffuse temp=1000 time=12 dryo2
#
```

```
structure outf=amex12_1.str
#
implant boron energy=25 dose=1e12 pearson

#
deposit poly thick=0.25 divi=6
#
diffuse temp=900 time=10 weto2
#
mask name="POLY"
#
etch oxide thick=0.1 dry
etch poly dry thick=0.4
#
strip

method compress
diffuse temp=900 time=10 weto2
#
structure outf=amex12_2.str
#


adapt.par polysilicon \
    i.arsenic i.phosphor i.boron i.antimony i.bf2 \
    edge.min=1.0e-5 edge.max=1.0e-4

adapt.par oxide \
    i.arsenic i.phosphor i.boron i.antimony i.bf2 \
    edge.min=1.0e-5 edge.max=1.0e-4


mask name="SDN"
#
implant phosphorus dose=1e13 energy=50 pearson
#
strip
#
deposit oxide thick=0.2 divi=4
#
etch oxide dry thick=0.25
#
mask name="SDN"
```

```
#
implant arsenic dose=5e15 pearson energy=100
#
strip
#
mask name="SDP"
#
implant bf2 dose=3e15 energy=50 pearson
#
#
strip
#
structure outf=amex12_3.str
deposit nitride thick=0.04 divisions=5
deposit oxide thick=0.3 divi=3
diffuse temp=900 time=1
mask name="MCC"
#
etch oxide dry thick=0.8
etch nitride dry thick=0.2
etch oxide dry thick=0.2
#
strip
rate.depo machine=al_dep1 aluminum a.s  cvd dep.rate=100.0 \
          step.cov=0.9 smooth.win=0.1
deposit machine=al_dep1 time=30 seconds divisions=4
mask name="FMD"
#
etch alumin dry thick=0.8
#
strip
structure outf=amex12_4.str
```

```
rate.depo machine=bpsg oxynitr a.s  cvd dep.rate=1.0 \
 step.cov=0.80 smooth.win=0.5
deposit machine=bpsg time=8000 seconds divi=8
mask name="CON1"
rate.etch machine=bpe1 oxynitr a.s  rie isotropic=0.9 dir=0.05
etch machine=bpe1 time=3000 seconds dx.mult=0.2
structure outf=amex12_4_mid.str
rate.etch machine=bpe2 oxynitr  a.s  rie isotropic=0.4 dir=10
rate.etch machine=bpe2 aluminum a.s  rie isotropic=0.4 dir=2
etch machine=bpe2 time=700 seconds dx.mult=0.5
strip
structure outf=amex12_5.str
rate.depo machine=al_dep2 aluminum a.s  cvd dep.rate=100.0 step.cov=0.65
deposit machine=al_dep2 time=80 seconds divisions=8
mask name="MET2"
rate.etch machine=bpe3 aluminum a.s  rie isotropic=0.5 dir=10.0
etch machine=bpe3 time=1000 seconds dx.mult=0.9
strip
# define the electrodes
electrode name="pwell" x=0.525
electrode name="ngate" x=1.75
electrode name="vdd" x=4.25
```

```
electrode name="pgate" x=5.225
electrode name="nwell" x=6.475

structure outf=amex12_6.str
```

32.1. ATHENA: Advanced Process Examples

32.1.1. anex01.in: Non-planar Optical Lithography

This example uses OPTOLITH to show non-planar lithography over a complex topography created using SSUPREM4 and ELITE.

The example begins with a silicon substrate that is 24 microns wide and 1 micron deep. Two parameters on the `initialize` statement specifies that the no impurity (geometric) mode and the coarse grid mode be employed. The parameter, `no.imp`, specifies that the introduction of impurities will be bypassed. This results in a much faster calculation for many steps and still produces similar topographies to the calculation including impurities. For simulations that are concerned primarily with topography, the no impurity mode can typically be applied without loss of significant information.

The parameter `space.mult` specifies that all grid spacings specified by the parameter `spacing` on the `line` statement should be multiplied by the specified value. This results in a global grid coarsening by the specified factor. This speeds simulation by reducing the discretization.

The process sequence begins with deposition of a thin oxide to model the effect of gate oxidation. Next, nitride is deposited and patterned, simply by specifying the positions of a vertical plane that defines the boundary of the etch region. Following the nitride patterning, field oxidation is performed and the nitride mask is stripped.

Next, gate polysilicon is deposited and patterned using the geometric etch capability and a structure file is saved.

Following the structure file save, the structure is implanted. Since the no impurity mode is being used, the implant step will be ignored. Next, the sidewall spacers are formed by depositing oxide and etching it using the vertical etch model. The source/drain implant is performed next, but due to the use of the no impurity mode, it is bypassed. The next step is the source/drain implant anneal. This step runs much more quickly in no impurity mode because impurity diffusion, which typically limits the timestep size during diffusion causing more equations to be solved, does not need to be calculated.

The next step is to deposit spin-on-glass. The glass deposition and reflow process are approximated here by performing a deposit that includes a geometric smoothing performed during the deposit. This is accomplished by defining an ELITE machine using the `rate.depo` statement and including the parameters `smooth.win` and `smooth.step` as part of the statement. The ELITE machine is invoked by specifying the `deposit` statement with the `machine` parameter used to specify the previously defined machine.

The next sequence of process steps deposits photoresist, patterns the resist with the geometric etch capability, defines a wet etch machine and employs that machine on the structure, defines a directional etch machine and employs that machine on the structure, saves the structure and strips the photoresist.

Following this, a short wet etch is performed to remove fillets from the structure, and aluminum is deposited using the ELITE hemispherical model.

Next, photoresist is deposited using the same geometric smoothing approach that is applied to model spin-on-glass reflow. This results in a nearly planar top surface following the resist deposit. The parameter, `name.resist`, specifies the type of resist that will be used from the library of resists included in the models file. In this case, the photoresist named OiR32 is applied.

After the resist is deposited, the simulation begins the OPTOLITH portion of the process. This performs a detailed analysis of the photolithographic process of this final structure. OPTOLITH simulation begins by performing imaging. The `ILLUMINATION, ILLUM.FILTER, {bold}`

PROJECTION, and PUPIL.FILTER statements describe the illumination system. The layout statements define two mask features. The parameter, lay.clear, removes any previously defined layout information so that a new layout can be initialized. Once the illumination system and the layout have been defined, the imaging module is invoked. The image statement invokes the imaging module and specifies the window in which imaging will be performed. The parameter, DX, specifies the discretization along the x-dimension and the parameter, one.d, specifies that the imaging should consider only one dimension. The parameter, clear, specifies that the mask should be considered as a clear field with dark features defined by the layout statements.

The optical statement defines the index of refraction for BPSG material at a wavelength of 0.365 microns. The expose statement performs the resist exposure from the results of the imaging module.

Following exposure, the BAKE statement performs post-exposure bake and calculates the diffusion of the photoactive component. Finally, resist development is performed using the develop statement. The MACK model is specified along with the development time, number of steps at which regridding will be performed, and the number of non-regridded substeps that will be taken per step.

Finally the results of the simulation are plotted using TONYPLOT. The non-idealities in the photoresist that result from reflections in the underlying topography can be seen in the lines of resist that remain.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.1: Non-Planar Metal Patterning over MOS Transistor

Input File athena/anex01.in:

```
1    go athena
```

```

# Establish the grid locations and densities
line x loc=-12.0   spacing=0.3
line x loc=-5.0    spacing=0.1
line x loc=-.50    spacing=0.05
line x loc=0.50    spacing=0.05
line x loc=5.0     spacing=0.1
line x loc=7.0     spacing=0.1
line x loc=9.0     spacing=0.1
line x loc=12.0    spacing=0.3

line y loc=0.0     spacing=0.1
line y loc=1.0     spacing=0.3

# Initialize the silicon wafer, include boron background doping
init c.boron=1.0e16 space.m=4 no.imp

#deposit the gate oxide
deposit oxide thick=0.025
deposit nitride th=.25
etch nitride left  p1.x=-5.0
etch nitride right p1.x=5.0
#
method grid.ox=0.2
diff time=160 temp=965 wet

etch nitride all

#deposit the gate poly
deposit poly thick=0.500 div=5

etch poly left p1.x=-.5
etch poly right p1.x=9
etch poly start x=.5 y=10
etch      cont  x=.5 y=-10
etch      cont  x=7 y=-10
etch      done  x=7 y=10

structure outfile=anex01_0.str

# implant ldd phosphorus
implant phos dose=3e13 energy=35

```

```
#deposit the oxide spacer
deposit oxide thick=0.4 div=8

#etch the spacer back
etch oxide thick=.45

# implant source drain arsenic
implant arsenic dose=1e15 energy=50

# anneal
diff time=30 temp=920

#SOG processing
rate.depo machine=SOG material=BPSG u.m hemisphe \
          sigma.dep=2.50 dep.rate=1.0 angle1=90 angle2=-90 \
          smooth.win=5 smooth.step=3
deposit machine=SOG time=1 minutes div=5

structure outfile=anex01_1.str

deposit photoresist thick=.6 div=5

etch photoresist start x=-2.5 y=1000
etch              cont x=-2.5 y=-1000
etch              cont x=-1.5 y=-1000
etch              done x=-1.5 y=1000

etch photoresist start x=2.5 y=1000
etch              cont x=2.5 y=-1000
etch              cont x=1.5 y=-1000
etch              done x=1.5 y=1000

rate.etch machine=wet material=BPSG u.m  wet.etch isotropic=1

etch machine=wet time=.75 minutes dx.mult=.5

rate.etch machine=m1 rie material=BPSG isotropic=0.0  directional=1.0 u.m

etch machine=m1 time=1.6 minute dt.max=0.01

structure outfile=anex01_2.str

strip
```

```

rate.etch machine=wet2 material=BPSG u.m  wet.etch isotropic=1
etch machine=wet2 time=.05 minute

rate.depo machine=ALmetal aluminum u.m  sigma.dep=0.35 hemisphe \
dep.rate=1.0 angle1=70 angle2=-70

deposit machine=ALmetal time=1 minute div=6

rate.depo machine=resistspin photoresist name.res=OiR32 u.m cvd \
          dep.rate=1.0 step.cov=2.5

deposit machine=resistspin time=.4 minute divis=16

rate.depo machine=resistspin photoresist name.res=OiR32 u.m cvd \
          dep.rate=1.0 step.cov=1.0 smooth.win=1 smooth.step=1

deposit machine=resistspin time=.2 minute divis=8

#structure outfile=anex01_3.str

# OPTOLITH Module

illumination i.line

illum.filter clear.fil circle sigma=0.3

projection na=.3 flare=2

pupil.filter circle

layout x.low=-2.5 z.low=-7.5 x.high=-1.5 z.high=7.5 lay.clear
layout x.low=1.5  z.low=-7.5 x.high=2.5  z.high=7.5

image win.x.lo=-12.0 win.z.lo=-1.0 win.x.hi=12.0 win.z.hi=1.0 dx=0.25
  one.d clear

structure outfile=anex01_4.str intensity

#Input optical parameters for user materials

optical material=BPSG wavelength=0.365 refrac.real=1.78 refrac.imag=0.0

```

```
# Resist exposure
expose dose=220 na=0
#structure outfile=anex01_5.str
# Post exposure bake
bake diff.length=.08
structure outfile=anex01_6.str
# Resist Development
develop mack time=60 steps=5 substeps=24
structure outfile=anex01_7.str
tonyplot -ttitle anex01.in:Final
quit
```

32.1.2. anex02.in: Trench Isolation Example

This example shows the fabrication of a trench isolation structure using ELITE and SSUPREM4 to model etching and oxidation processes.

The example begins with grid definition. This is tailored to place a fine grid in the area that will be exposed during the trench etch. The `initial` statement does not specify a material type, so silicon is taken by default. A background doping of boron with a concentration of 1.0e15 is specified. The parameter, `no.imp`, specifies that the entire simulation should neglect all processing steps that introduce dopant into the structure. This results in a much faster calculation for diffusion and oxidation calculations but will not produce a structure suitable for electrical analysis due to the fact that dopant is left out. This is intended to provide a fast mode of calculation that is useful for initial simulations. The parameter, `space.mult`, specifies that a coarser grid be used than that specified by the `spacing` parameter on the `line` statements. This is also intended for fast initial runs of the simulator. To disable these features for a more complete calculation, simply delete the `no.imp` and `space.mult` parameters from the `initialize` statement.

Following initialization, a nitride layer is deposited and patterned by the `deposit` and `etch` statements. The structure is then reflected to create a symmetric structure for trench etching.

The trench is performed by defining an etch machine with the `rate.etch` statement. For this etch, the etch rates for all materials except silicon are assumed to be zero, so only rates for silicon are defined. If other materials are to be etched, the rates for each material should be set with a `rate.etch` statement. For this etch process, the rates are defined for the RIE model. This simple model describes the etch process as a combination of `isotropic` and `directional` components. The parameter, `U.M`, specifies that the etch rates are given in units of microns per minute.

The etch is then performed for a period of ten minutes.

Following the etch, the nitride masking layer is removed with the `ETCH NITRIDE ALL` statement.

The trench is then filled with a sandwich of materials. First, a layer of oxide is deposited with the simple conformal model. Then a layer of polysilicon is deposited with the ELITE calculation. Finally, photoresist is deposited using ELITE to model the resist flow. This is included by adding the parameters `smooth.win` and `smooth.step` to the deposit machine definition. These two parameters specify the width over which a simple averaging algorithm will be applied and the number of times it will be applied, respectively. This results in a smoothed profile that is a fast way of modeling the flowing resist during deposit.

The structure coated with resist has a surface that is nearly planar. By applying an etch process that erodes all materials at an equal rate, the planarization process can be modelled. This is performed by specifying the etch rates for an ELITE type calculation using the `wet.etch` model. Following the etch, the grid on the surface of the structure is relaxed to remove any excess grid that may have been introduced during the etch process. The parameter, `dx.surf,` on the `relax` statement specifies the minimum surface segment size in microns for the relax operation.

Next, a nitride layer is deposited and patterned to control the area that will be oxidized.

Some model parameters are set for the oxidation using the `method` and `oxide` statements. Then the diffusion is performed in two steps, first in dry oxygen, specified by using the `DRYO2` parameter, and then in water, specified by using the `WETO2` parameter.

Finally, the structure is plotted using TONYPLOT and the structure saved to a file.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.2: Trench Etch and Refill using ELITE and SSUPREM4

Input File athena/anex02.in:

```
1    go athena
2
```

```
#TITLE: Trench isolation example

# place the finest grid around the trench
line  x    location=0.0   spacing=0.5
line  x    location=1.50  spacing=0.5
line  x    location=1.90  spacing=0.10
line  x    location=1.98  spacing=0.02
line  x    location=2.00  spacing=0.004
line  x    location=2.02  spacing=0.007
line  x    location=2.10  spacing=0.01
line  x    location=2.30  spacing=0.05
line  x    location=2.75  spacing=1.0

line  y    location=0.0   spacing=0.01
line  y    location=0.010 spacing=0.01
line  y    location=0.1   spacing=0.02
line  y    location=0.6   spacing=0.1
line  y    location=3.4   spacing=0.8
line  y    location=6.0   spacing=0.2
line  y    location=6.5   spacing=0.50
line  y    location=7.05  spacing=1.0

initial c.boron=1.0e15 no.imp space.m=2

deposit nitride th=.1 div=4
etch       nitride right    p1.x=2.0

# form the complete trench structure
structure  reflect

# DRYTECH QUAD RIE machine
rate.etch machine=DRYTEK silicon u.m  rie isotropic=0.0 directional=0.6

etch machine=DRYTEK time=10 minutes
structure outfile=anex02_0.str

etch nitride all

# horizontal profiles
deposit    oxide  thick=0.02 divisions=1

rate.depo machine=polydep cvd polysil u.m dep.rate=.8 step.cov=1
deposit machine=polydep time=1 minute divis=10
```

```
structure outfile=anex02_1.str

rate.depo machine=d3 hemis photoresist a.m \
 angle1=90. angle2=-90. dep.rate=1000. sigma.dep=0.2 \
 smooth.win=2.0 smooth.step=5
#
deposit mach=d3 time=15.0 minutes divis=6
structure outfile=anex02_2.str

rate.etch machine=test polysilicon a.m  wet.etch iso=1000.0
rate.etch machine=test silicon     a.m  wet.etch iso=1000.0
rate.etch machine=test nitride     a.m  wet.etch iso=1000.0
rate.etch machine=test oxide       a.m  wet.etch iso=1000.0
rate.etch machine=test photores    a.m  wet.etch iso=1000.0

# planarize
etch machine=test time=24 minutes dt.max=1
relax surface dx.surf=0.03
structure outfile=anex02_3.str

# deposit masking nitride
deposit    nitride   thick=0.03

etch nitride right p1.x=2.85

# turn off poly lifting as it is not needed
method grid.ox=0.05 lift.poly=f fermi

# Adjust these parameters to match observed
# oxide thickness on poly and silicon
oxide silicon wet par.h.0 =2.0
oxide wet par.h.0 =10
diffuse    temp=1000 time=10 dryo2

structure outfile=anex02_4.str

diffuse   temp=1000 time=100.0 weto2

structure outfile=anex02_5.str
tonyplot -ttitle anex02.in:Final

quit

```

32.1.3. anex03.in: Comparison of Simple and Realistic Resist Surfaces

This example demonstrates optical lithography over nonplanar topography. OPTOLITH is invoked to expose and develop the photoresist.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.3: Effects on CD of Planar versus Realistic Resist Surfaces

Input File athena/anex03.in:

```
go athena


#                    ATHENA input file: anex03.in
#                    --------------------------------

#            This example illustrates exposure and development
#     over nonplanar topography using a user defined
#     aerial image cross section.

#
line x loc=0.00 spac=0.25
line x loc=1.25  spac=0.1
line x loc=1.75  spac=0.1
```

```
line x loc=3.00 spac=0.25
#
line y loc=0.5  spac=.1
line y loc=1.00 spac=.1
line y loc=1.50 spac=.1
#
init orientation=100
#
# Etch the silicon to create a simple wedge structure
etch silicon right p1.x=1.25 p1.y=.5 p2.x=1.75 p2.y=1.0
#
deposit photo thick=1.00 divisions=60
#
# Etch photoresist to create a flat top surface
etch photoresist start x=0.00 y=-0.5
etch cont x=0.00 y=0.00
etch cont x=3.00 y=0.00
etch done x=3.00 y=-0.5
#
expose inf=anex03.exp dose=200.0
tonyplot
bake diff.leng=0.05
structure outfile=anex03_1.str
tonyplot
#
develop kim time=60 steps=8
structure outfile=anex03_2.str
tonyplot
#
#
line x loc=0.00 spac=0.25
line x loc=1.25  spac=0.1
line x loc=1.75  spac=0.1
line x loc=3.00 spac=0.25
#
line y loc=0.5  spac=.1
line y loc=1.00 spac=.1
line y loc=1.50 spac=.1
#
init orientation=100
#
# Etch the silicon to create a simple wedge structure
etch silicon right p1.x=1.25 p1.y=.5 p2.x=1.75 p2.y=1.0
```

```
#
# Define a machine for realistic resist deposit
rate.depo machine=test photoresist u.m  cvd \
          dep.rate=1.0 step.cov=0.6 smooth.win=1. smooth.step=4
#
deposit machine=test time=.5 divisions=30
#
expose inf=anex03.exp dose=200.0
bake diff.leng=0.05
structure outfile=anex03_3.str
tonyplot
#
develop kim time=60 steps=8
structure outfile=anex03_4.str
tonyplot
#
quit
```

32.1.4. anex04.in: Metal Patterning Over Non-planar Topography

This example demonstrates optical lithography over non-planar topography. OPTOLITH is invoked to expose and develop the photoresist.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.4: Mask and Image for Two Stripes to be exposed over a step in topography



Figure 32.5: Images of the Two Features over a Step caused by Local Oxidation

Input File athena/anex04.in:

1 go athena

```
#TITLE: Metal Patterning with SSUPREM4, ELITE, and OPTOLITH (20 minutes)
#
line x loc=-2.0 spacing=0.1
line x loc=-1.0 spacing=0.05
line x loc=0.0  spacing=0.1
line x loc=1.0  spacing=0.05
line x loc=2.0  spacing=0.1
line y loc=0.0  spacing=0.05
line y loc=1.0  spacing=0.5
initial space.m=1
#
deposit nitride thick=.25

etch nitride right p1.x=0.0
method grid.ox=0.05
diff time=300 temp=920 weto2

etch nitride all

rate.depo machine=PE4450 aluminum u.m  sigma.dep=0.35 hemisphe
  dep.rate=.2 angle1=70 angle2=-70
deposit machine=PE4450 time=1. minutes divisions=8

tonyplot -ttitle anex04.in:Initial

rate.depo machine=PRspin photores name.res=OiR32 u.m  \
          sigma.dep=.50  hemisphe dep.rate=1.0 angle1=90 angle2=-90 \
          smooth.win=1
deposit machine=PRspin time=1 minutes div=30


illumination i.line

illum.filter circle sigma=.5

projection na=.45

pupil.filter circle

layout x.lo=-1.25 z.lo=-5.0 x.hi=-0.75 z.hi=5.0
layout x.lo=0.75  z.lo=-5.0 x.hi=1.25  z.hi=5.0

```

```
image clear dx=0.05 win.x.lo=-2.0 win.x.hi=2.0 win.z.lo=0.0 win.z.hi=0.0
  one.d

structure outfile=anex04_0.str intensity

tonyplot -st anex04_0.str -ttitle anex04.in:Intensity

expose dose=220 num.refl=1 na=0

structure outfile=anex04_1.str

tonyplot -st anex04_1.str -set anex04_1.set

bake diff.length=0.06

structure outfile=anex04_2.str

tonyplot -ttitle "anex04.in: Following Expose and Postbake" -set
  anex04_2.set

develop mack time=60 steps=5 substeps=24

structure outfile=anex04_3.str

tonyplot -ttitle "anex04.in: Following Develop"
#
rate.etch mach=test rie u.m isotropic=0.05  direction=0.0 photoresist
rate.etch mach=test rie u.m isotropic=0.5   direction=0.5 alumin
rate.etch mach=test rie u.m isotropic=0.1  direction=0.0 silicon
rate.etch mach=test rie u.m isotropic=0.1  direction=0.0 oxide
#
etch mach=test time=.40 minute dx.mult=.5
#
structure outfile=anex04_4.str
#
tonyplot -ttitle anex04.in:Final

quit
```

32.1.5. anex05.in: Multi-level Interconnect Formation

This example shows the use of ELITE to model a complete interconnect system over isolation oxide grown using SSUPREM4.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.6: Use of ELITE and SSUPREM4 to model Multi-Metal Interconnect Topology

Input File athena/anex05.in:

```
go athena

line x loc=-12.0   spacing=0.5
line x loc=-8.0   spacing=0.5
line x loc=-7.1   spacing=0.2
line x loc=-7.0   spacing=0.2
line x loc=-6.1   spacing=0.2
line x loc=-6.0   spacing=0.2
line x loc=-5.7   spacing=0.2
line x loc=-5.0   spacing=0.2
line x loc=-4.8   spacing=0.2
line x loc=-4.4   spacing=0.2
line x loc=-3.8   spacing=0.1
line x loc=-3.0   spacing=0.1
line x loc=-2.4   spacing=0.1
line x loc=-2.0   spacing=0.1
line x loc=-.50   spacing=0.1
```

```
line x loc=0.0    spacing=0.1

line y loc=0.0   spacing=0.03
line y loc=2.0   spacing=0.5

init space.m=2

#deposit the gate oxide
deposit oxide thick=0.025
deposit nitride th=.25
etch nitride left p1.x=-5.
#
diff time=160 temp=965 wet

etch nitride all

#deposit the gate poly
deposit poly thick=0.400 div=5

etch poly left p1.x=-.5

deposit oxide thick=.3 division=6
#etch the spacer back
etch dry oxide thick=0.33

#reflect the structure
structure mirror right

#deposit the BPSG intermetal dielectric
rate.depo machine=bpsg_depo material=BPSG cvd dep.rate=1.6 u.m
  step.cov=.75 smooth.win=6 smooth.step=5
deposit machine=bpsg_depo time=1 div=6
#
deposit photores thick=1. div=5
etch photores start x=-3.00 y=-10.00
etch cont x=-3 y=10.00
etch cont x=-2 y=10.00
etch done x=-2 y=-10.00
etch photores start x=3.00 y=-10.00
etch cont x=3 y=10.00
etch cont x=2 y=10.00
etch done x=2 y=-10.00

```

```
rate.etch machine=bpsg_etch material=BPSG rie directional=1.0 isotro-
  pic=.1 u.m
etch machine=bpsg_etch time=2.0 minute dx.mult=.25
#
etch photo all
rate.etch machine=bpsg_etch material=BPSG rie directional=1.0 isotro-
  pic=.1 u.m
etch machine=bpsg_etch time=0.1 minute dx.mult=0.5
#

deposit material=TiN thick=0.250 divis=1
#
deposit tungsten thick=0.5 divisions=5
rate.depo machine=tungsten_depo tungsten cvd dep.rate=.75 u.m
  step.cov=.75 smooth.win=1 smooth.step=2
deposit machine=tungsten_depo time=.5 minute divisions=3
#
deposit photores thick=.5 div=3

etch photo right p1.x=8.5

etch photo start x=6.1 y=-10
etch cont x=7.1 y=-10
etch cont x=7.1 y=10
etch done x=6.1 y=10

etch photo start x=3.8 y=-10
etch cont x=4.8 y=-10
etch cont x=4.8 y=10
etch done x=3.8 y=10

etch photo start x=-1.2 y=-10
etch cont x=1.2 y=-10
etch cont x=1.2 y=10
etch done x=-1.2 y=10
#
rate.etch machine=W_etch tungsten rie directional=1.0 isotropic=.1 u.m
etch machine=W_etch time=1.4 minute dx.mult=.3
strip

deposit material=SOG thick=0.4 div=3
#
rate.depo machine=HEMIS material=TEOS u.m  sigma.dep=0.20 smooth.win=3
  smooth.step=1 dualdirec dep.rate=1.0 angle1=45.00 angle2=-45.00
```

```
#
deposit machine=HEMIS time=0.8 minutes divis=4
#
deposit photo thick=1 divis=5
etch photo start x=-7.5 y=-10
etch cont x=-7.5 y=10
etch cont x=-6.5 y=10
etch done x=-6.5 y=-10
rate.etch machine=TEOS_etch material=TEOS rie directional=1.0 isotropic=0.1 u.m
rate.etch machine=TEOS_etch material=SOG  rie directional=1.0 isotropic=0.1 u.m
etch machine=TEOS_etch time=1.5 minutes
#
strip
structure outfile=anex05_0.str

deposit aluminum thick=.6 divis=6
rate.depo machine=aluminum_depo aluminum cvd dep.rate=1. u.m step.cov=.75 smooth.win=5 smooth.step=4
deposit machine=aluminum_depo time=.4 divisions=5
#
etch aluminum thick=1.2

deposit tungsten thick=.8 divisions=5
#
deposit alumin thick=.8 divisions=3
#
deposit tungsten thick=.1
#
deposit material=TiN thick=.2 divisions=2
deposit photores thick=1.0 divis=4
#
etch photo left p1.x=-8

etch photo start  x=-6 y=-10
etch cont x=-6 y=10
etch cont x=2.4 y=10
etch done x=2.4 y=-10

etch photo start  x=4.4 y=-10
etch cont x=4.4 y=10
etch cont x=5.7 y=10
etch done x=5.7 y=-10
```

```
etch photo right p1.x=7
rate.etch machine=met_etch rie material=TiN direct=1.0 iso=.1 u.m
rate.etch machine=met_etch rie tungsten     direct=1.0 iso=.2 u.m
rate.etch machine=met_etch rie aluminum     direct=1.0 iso=.1 u.m
etch machine=met_etch time=1.02 minute
strip
structure outfile=anex05_1.str
deposit oxide thick=.4 divisions=3
rate.depo machine=HEMIS material=TEOS u.m smooth.win=2 smooth.step=1 \
      sigma.dep=0.20 dualdirec dep.rate=1.0 angle1=45.00 angle2=-45.00
deposit machine=HEMIS time=0.8 minute divis=10
rate.depo machine=HEMIS material=SOG u.s smooth.win=5 smooth.step=1 \
     sigma.dep=0.20 dualdirec dep.rate=1.0 angle1=45.00 angle2=-45.00
#
deposit machine=HEMIS time=0.5 seconds divis=4
#
etch dry thick=0.2
#
deposit oxide thick=1 divisions=5
deposit photores thick=1. divis=3
etch photo start x=2.8 y=-10.00
etch cont x=2.8 y=10
etch cont x=4 y=10
etch done x=4 y=-10
rate.etch machine=via_etch material=TEOS rie directional=1.0 isotropic=0.0 u.m
rate.etch machine=via_etch material=SOG  rie directional=1.0 isotropic=0.0 u.m
rate.etch machine=via_etch oxide       rie directional=1.0 isotropic=0.0 u.m
etch machine=via_etch time=2. minutes
strip
structure outfile=anex05_2.str
```

```
rate.depo machine=aluminum_depo aluminum cvd dep.rate=1. u.m step.cov=.75
  smooth.win=1 smooth.step=4
deposit machine=aluminum_depo time=.2 divisions=5
deposit machine=aluminum_depo time=.8 divisions=10

etch aluminum thick=1.01

deposit material=AlCu thick=1.6 divis=10

etch material=AlCu thick=0.8

deposit tungsten thick=.1

deposit material=TiN thick=.2 divisions=2

deposit photo thick=.5 divis=3

etch photo left p1.x=-4.4
etch photo start x=-2.4 y=-20
etch cont x=-2.4 y=20
etch cont x=2.4  y=20
etch done x=2.4  y=-20
etch photo right p1.x=4.4
#
rate.etch machine=metal_etch material=TiN  rie direction=1.0 isotro-
  pic=0.1 u.m
rate.etch machine=metal_etch aluminum    rie direction=1.0 isotropic=0.1
  u.m
rate.etch machine=metal_etch material=AlCu rie direction=1.0 isotro-
  pic=0.1 u.m
rate.etch machine=metal_etch tungsten    rie direction=1.0 isotropic=0.1
  u.m
etch machine=metal_etch time=2. minutes

strip

deposit nitride thick=1.5 divisions=5

structure outfile=anex05_3.str

tonyplot

quit
```

32.1.6. anex06.in: Trench Formation and Planarization

This example shows the use of ELITE to model a trench process using a unique self-aligned process. The process begins with spacer formation over a patterned sandwich of oxide and nitride layers. Next, oxidation is performed using the spacer as a mask. This creates a bird's beak that extends under the spacer. After etching the nitride, an opening is created by etching enough oxide to remove the bird's beak. This opening is used to pattern a trench etch. The resulting trench is oddly shaped because of the simultaneous erosion of the silicon and the lateral erosion of the masking oxide.

Following the trench formation, an oxide layer is grown. The structure is then filled with oxide. Planarizing photoresist is deposited next.

Finally, the structure is planarized by eroding all materials with a similar etch rate. The sequence of structures is then plotted using TONYPLOT.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.7: Planarized Trench using ELITE

Input File athena/anex06.in:

```
go athena

line x loc=0.00 spac=0.1
line x loc=0.80 spac=0.02
line x loc=1.20 spac=0.02
line x loc=2.00 spac=0.1
#
```

```
line y loc=0.00 spac=0.02
line y loc=3.0spac=0.5

init silicon  c.boron=1.5e15 orientation=100 no.imp


method fermi compress
diffus time=20 temp=1050 dryo2 press=1.00 hcl.pc=0

deposit nitride thick=0.07 divisions=3
deposit oxide thick=0.19 divisions=5

deposit photo thick=.5
etch photo right p1.x=1

etch oxide thick=0.2
etch nitri thick=0.2
etch oxide thick=0.2

strip photo

deposit nitride thick=0.18 divisions=10

rate.etch machine=test1 nitride a.m  rie isotropic=50.00 dir=500
etch machine=test1 time=4.0 minutes dx.mult=.25

structure outfile=anex06_0.str

method fermi compress
diffus time=55 temp=900 weto2 press=1.00 hcl.pc=0

structure outfile=anex06_1.str

rate.etch machine=test2 nitride a.m  rie isotropic=50.00 dir=500
rate.etch machine=test2 oxide   a.m  rie isotropic=50.00 dir=50
rate.etch machine=test2 silicon   a.m  rie isotropic=1.00 dir=5
etch machine=test2 time=6.0 minutes dx.mult=.5

structure outfile=anex06_2.str

rate.etch machine=trench silicon a.m  rie isotropic=50 direct=250
rate.etch machine=trench oxide   a.m  rie isotropic=5 direct=10
etch machine=trench time=50 minutes dx.mult=.5
```

```
diff time=10 temp=900 dry
structure outfile=anex06_3.str
rate.depo machine=test3 cvd step.cov=1 oxide dep.rate=1 u.m
deposit machine=test3 time=.3 minute div=10
rate.depo machine=test4 cvd step.cov=1 photo dep.rate=1 u.m smooth.win=5
deposit machine=test4 time=1 minute div=10
structure outfile=anex06_4.str
rate.etch machine=planar silicon u.m  rie isotropic=0 direct=1.5
rate.etch machine=planar photo   u.m  rie isotropic=0 direct=1.5
rate.etch machine=planar oxide   u.m  rie isotropic=0 direct=1.5
rate.etch machine=planar nitride u.m  rie isotropic=0 direct=1.5
etch machine=planar time=1 minutes
structure outfile=anex06_5.str
tonyplot -st anex06_*.str
quit
```

32.1.7. anex07.in: Trench Oxidation Stress

This example shows the use of ELITE to model a trench process using two different processes. Following the trench formation, an oxide layer is grown. The stress produced during oxidation is included in the calculation.

The final structures are then plotted using TONYPLOT**.**

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.8: Trench Oxidation using ELITE and SSUPREM4

Input File athena/anex07.in:

```
go athena

#TITLE: Trench etch and subsequent oxidation example
#
line x loc=0.00 spac=0.05
line x loc=0.4  spac=0.03
line x loc=1.0  spac=0.03
line x loc=1.4  spac=0.05
#
line y loc=0.00 spac=0.04
line y loc=1.00 spac=0.1
#
init orientation=100 space.m=1
#
deposit oxide thick=0.10 div=8
#
etch oxide start x=0.4 y=-10
etch       cont  x=0.4 y=10
etch       cont  x=1.0 y=10
```

```
etch       done  x=1.0 y=-10
#
structure outfile=anex07_0.str

rate.etch machine=test1 rie silicon a.m isotropic=200. directional=100.
#
rate.etch machine=test1 rie oxide a.m isotropic=10.0 directional=0.
#
etch mach=test1 time=5.0 minutes dx.mult=.25
#
strip oxide

method viscous
diff time=30 temp=900 dry

structure outfile=anex07_1.str

init infile=anex07_0.str
#
rate.etch machine=test2 silicon u.m rie isotropic=.04 dir=.11

etch mach=test2 time=5.0 minutes dx.mult=.25
strip oxide

method viscous
diff time=30 temp=900 dry

structure outfile=anex07_2.str

tonyplot -st anex07_*.str

quit
```

32.1.8. anex08.in: DevEdit Mesh Adaptation with ELITE and SSUPREM4

This example shows the use of DEVEDIT in a mesh adaptation application. A trench is etched with an ELITE etch machine. A highly doped oxide is then deposited inside the trench for sidewall doping. Boron segregation and transport coefficients are defined. (These parameters may be used to calibrate the amount of out diffusion of boron.)

A short heat cycle segregates out some of the boron from the oxide into the silicon trench sidewall.

DEVEDIT is then invoked to regrid the structure based upon the concentration gradient of the boron and phosphorous. The weight for each impurity will determine the distance from the diffusion front that the grid is refined. These parameters should be used in relation to the estimated subsequent diffusion profile.

DEVEDIT is invoked at a point where minimal material boundary points exist to minimize the mesh density. This is the preferred, although not essential method of use.

Next, ATHENA is re-invoked and again the segregation and transport coefficients are defined in the new ATHENA session.

A second diffusion is started, and the boron diffusion front now moves within the refined mesh region.

DEVEDIT may be used in this manner to refine at any stage during a process flow.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.9: DevEdit Grid Adaption after ELITE Trench Etch and SSUPREM4 Implant

Input File athena/anex08.in:

```
go athena

#TITLE: Trench sidewall doping example using DevEdit to remesh the struc-
  ture
#
line x loc=0.00 spac=0.20
line x loc=0.4 spac=0.05
line x loc=0.6 spac=0.05
line x loc=0.9 spac=0.05
line x loc=1.2 spac=0.05
line x loc=1.4 spac=0.05
line x loc=1.8 spac=0.20
#
```

```
line y loc=0.00 spac=0.02
line y loc=1.1 spac=0.1
line y loc=2 spac=0.5
initialize c.phos=1e16
#
deposit material=SPR500      thick=1. div=5
#
etch material=SPR500 start x=0.6 y=-10
etch                 cont  x=0.6 y=10
etch                 cont  x=1.2 y=10
etch                 done  x=1.2 y=-10
#
structure outfile=anex08_0.str

rate.etch machine=m1 rie silicon iso=0.3 dir=0.8 u.m
#
etch machine=m1 time=40.0 second dx.mult=0.25

structure outfile=anex08_1.str

strip material="SPR500"

#
rate.depo machine=dep1 oxide u.m  cvd dep.rate=1.0 step.cov=0.55
#
deposit machine=dep1 time=0.1 minutes c.boron=1.0e21 divisions=3
#
tonyplot
#
# Define the segregration/transport co-efficients for calibration.....
boron silicon /oxide Seg.0=1066.0 Trn.0=1.2e-7

method fermi compress
diffus time=30 temp=900 nitro press=1.00

etch oxide all

# Save this structure for comparison...
struct outf=anex08_2.str

# Now dip off the oxide to save boundary related mesh points, and remesh on
# doping concentration gradients....
go DevEdit
```

```
base.mesh height=0.2 width=0.2
bound.cond apply=false max.ratio=300
imp.refine min.spacing=0.05
imp.refine imp="Phosphorus" sensitivity=1
imp.refine imp="Boron" sensitivity=0.5
mesh

go athena

# Note: redefine all non-default models when moving back into Athena....
boron silicon /oxide Seg.0=1066.0 Trn.0=1.2e-7

structure outf=anex08_3.str

diffuse temp=1000 time=10 nitr

structure outf=anex08_4.str

#plot the structures before and after DevEdit....
tonyplot -st anex08_*.str

quit
```

32.1.9. anex09.in: GaAs on Silicon Process

This example shows the use of ATHENA in a GaAs on Silicon example.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.10: GaAs on Silicon Process Structure using SSUPREM4 and FLASH

Input File athena/anex09.in:

```
go athena
#
# GaAs on Silicon example
#
# Define the grid
line x loc=0.0  spac=0.1
line x loc=0.1  spac=0.1
#
line y loc=0.0  spac=0.005
line y loc=0.5  spac=0.1

# initialize but include silicon as an impurity in the silicon substrate
init silicon c.silicon=5.0e22 orientation=100

# Modify and add some impurity model parameters for silicon

# original
#impurity i.silicon donor silicon Dix.0=0.0 Dix.E=0.0 Dvx.0=0.0 Dvx.E=0.0
#impurity i.silicon gaas ss.temp=700. ss.conc=1.000e+20 ss.clear
```

```
#impurity i.silicon gaas ss.temp=1200. ss.conc=1.000e+20

# modified
impurity i.silicon silicon Dix.0=100.0 Dix.E=0.0 Dvx.0=100.0 Dvx.E=0.0
impurity i.silicon silicon /gaas Seg.0=30.0 Seg.E=0.0 Trn.0=1.66e-7
  Trn.E=0.0
impurity i.silicon gaas ss.temp=700. ss.conc=1.000e+23 ss.clear
impurity i.silicon gaas ss.temp=1200. ss.conc=1.000e+23


# Deposit GaAs with background silicon doping
deposit gaas thick=.25 divis=20 c.silicon=1e16

diff time=200 temp=700

structure outfile=anex09.str

tonyplot

quit
```

32.1.10. anex10.in: Trench Etching and Lithography

This example shows the use of ATHENA in a trench etching example. OPTOLITH is used for patterning and ELITE is used for etching.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.11: Image of Via Mask



Figure 32.12: Imaging of Via (left). Subsequent Metal Etch to form Via (right)

Input File athena/anex10.in:

```
go athena

#TITLE: Trench etching using RIE model and OPTOLITH
#
line x loc=0 spac=0.05
line x loc=0.25 spac=0.05
line x loc=0.5 spac=0.20
line x loc=0.75 spac=0.20
#
line y loc=0.00 spac=1
line y loc=1 spac=1
initialize
#
deposit aluminum  thick=1. div=10
deposit name.resist=OiR32  thick=1. div=40
#
structure outfile=anex10.str

# Illumination wavelength
#
illumination lambda=0.365
illum.filter clear.fil circle sigma=0.5
#
# Projection numerical aperture
#
projection na=.52
pupil.filter clear.fil circle
#
# The mask layout
#
layout lay.clear x.lo=-.25 z.lo=-1 x.hi=.25 z.hi=1
#
# Run the imaging module with a mask input file from Maskviews
#
image opaque win.x.lo=0.0 win.z.lo=0 win.x.hi=0.5 win.z.hi=0 \
dx=0.05 one.d n.pupil=3
#
structure outfile=anex10.int intensity
#
expose dose=250 num.refl=10
#
bake time=60 temp=125
```

```
#
develop mack time=60 steps=5 substeps=24

structure mirror left

structure outfile=anex10_1.str

rate.etch machine=m1 rie aluminum iso=.05 dir=0.95 u.m
rate.etch machine=m1 rie name.resist=OiR32 iso=0.01 dir=0.09 u.m
#
etch machine=m1 time=60.0 second dx.mult=2

structure outfile=anex10_2.str

tonyplot -st anex10.int

tonyplot -st anex10_*.str
#
quit

```

32.1.11. anex11.in: C Interpreter Used for Phosphorus Segregation

This example demonstrates the use of the `C-Interpreter` to alter the form of the calculation of segregation for phosphorus.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.13: Comparison of default segregation model with a user specified model entered through the C-Interpreter

Input File athena/anex11.in:

```
go athena

# This Input File demonstrates the use of the C interpreter

line x loc=0.00 spac=0.10
line x loc=0.10 spac=0.10
line y loc=0.00 spac=0.01
line y loc=0.01 spac=0.01
line y loc=1    spac=0.2
init silicon orientation=100 c.phos=1e17

deposit oxide thick=0.20 divisions=20 c.phos=1e17

structure out=anex11_0.str

diffus time=2000 temp=1000

structure out=anex11_1.str

init infile=anex11_0.str

diffus time=2000 temp=1000 p.seg.calc=anex11.lib
```

```

structure outfile=anex11_2.str

tonyplot -overlay anex11_1.str anex11_2.str -set anex11.set -ttitle
  "Phosphorus Segregation Calculation"

quit
```

32.1.12. anex12.in: CMOS Example Using MaskViews

MaskViews is a product enabling the design of more than one device side by side. For example, if an NMOS and a PMOS pair need to be considered side by side, MaskViews is used in conjunction with a single generic input file. MaskViews sends to the process simulator the mask, grid, boolean region and electrode name information.

This example uses an IC Layout cross section, along with a generic process flow to realize a specific device. To run this example, first invoke MaskViews from the **Tools - MaskViews - Start MaskViews** menu options. An IC Layout editor will appear. Create a cutline object by selecting the **Write file** button. Then select the beginning and end of the required cross section on the layout itself. This action generates a ‘cutline object’ as a file. (If used under VWF, the cutline object will be loaded into the database). The cutline object contains four pieces of information:

- masking information – where to place the photoresist or barrier
- grid information – automatically generated from a set of rules
- boolean regions information – defining regions of interest
- electrode names – for use in a later device test

Load the cutline object into DeckBuild by using the **Tools - MaskViews - Cutlines** menu option. Once loaded into DeckBuild, press **run.**

The completed run will relate directly to your IC layout cutline selection.

Note the use of the `autoelectrode` statement to refer any available electrode names at a particular point in the process flow to a specific material region. Electrode names may be associated with poly, metal or silicide regions. The `autoelectrode` statement will search from the top of the structure until it finds an appropriate electrode material.

The lateral position of the electrode is determined from the IC layout electrode mask position. Electrodes are named automatically, half-way from the point where the cutline intercepts the layout electrode polygon to the point where the cutline leaves the electrode polygon.

The grid is derived from a set of rules relating to the edge of each mask layer. These rules are defined under the: **define - grid - x** popup window within MaskViews**.**

This example also demonstrates the use of MaskViews**,** with a view to usage under Virtual Wafer Fab. When used under VWF, split lots may include many devices, e.g., a P-channel and an N-channel MOS device.

To load and run this example, select the **Load example** button. This action will copy all associated files to your current working directory. Select the DeckBuild **run** button to execute the example.



Figure 32.14: Layout of CMOS Inverter in MaskViews



Figure 32.15: Cross-Section of Masks for NMOS device including automated mesh generation

Input File athena/anex12.in:

```
go athena cutline=anex12_0.sec
#
#
init c.phos=1.0e14 orientation=100

# Initial Oxide
method fermi compress
diffus time=30 temp=1000 dryo2 press=1.00 hcl.pc=3
# Initial Nitride
deposit nitride thick=0.15 divisions=2

# Pwell photo
mask name="WELL"
# Pwell definition etch
etch nitride dry thick=0.15
# Oxide overetch
etch oxide dry thick=0.015
# pwell implant
implant boron dose=1.0e13 energy=100
strip
struct outfile=anex12_0.str

# Nwell maox
method fermi compress
diffus time=45 temp=950 weto2 press=1.00 hcl.pc=3
struct outfile=anex12_1.str

# strip init nit
etch nitride all
# oxide overetch
etch oxide dry thick=0.015
# nwell implant
implant phosphor dose=1.0e13 energy=60

# planarising oxide
method fermi compress
diffus  time=45 temp=800 t.final=1000 dryo2 press=0.05 hcl.pc=3
diffus time=45 temp=1000 weto2 press=0.05 hcl.pc=3
# well drive
method fermi compress
diffus time=170 temp=1200 nitro press=0.05 hcl.pc=0
structure outfile=anex12_2.str
```

```
# wellox strip
etch oxide all

# Stress relief Oxide
deposit oxide thick=0.035 divisions=1
# Masking Nitride
deposit nitride thick=0.15 divisions=1
# Active Area Photo
mask name="AAD"
# AA Definition
etch nitride dry thick=0.30
# AA Definition
etch oxide dry thick=0.10
# AA Definition
etch silicon dry thick=0.15
struct outfile=anex12_3.str
strip

# Field Oxide growth
method fermi compress
diffus time=70 temp=1000 weto2 press=1.00 hcl.pc=3
struct outfile=anex12_4.str
#
extract name="pwell_fieldox" thickness oxide mat.occno=1 region="pwell_fld"
extract name="nwell_fieldox" thickness oxide mat.occno=1 region="nwell_fld"


# AA stack strip
etch nitride dry thick=0.25
# AA stack strip
etch oxide dry thick=0.07
# AA stack strip
etch nitride all
#
struct outfile=anex12_5.str
# Sacox growth
diffus time=25 temp=950 dryo2 press=1.00 hcl.pc=3
# Sacox strip
etch oxide dry thick=0.06

#Gate oxide recipe for example msex03.in
# push in boat, timed from after boat reaches the cold zone....
```

```
diffuse temp=800  time=15 nitrogen
#ramp up to temp in dilute o2 and tca
diffuse temp=850 t.final=950 dryo2 hcl.pc=3 time=15 press=0.25
#increase pp of o2 and stabilise
diffuse temp=950 time=5 dryo2 hcl.pc=3 press=0.8
#main growth step
diffuse temp=950 dryo2 hcl.pc=3 time=15 press=0.8
#anneal step
diffuse temp=950 time=35 nitrogen
# Ramp down
diffuse temp=950 t.final=800 time=30 nitrogen
# Pull out
diffuse temp=800 time=15 nitrogen
struct outfile=anex12_6.str
# Extract gate ox thickness. This is used as optimisation target and as a
# VWF design variables....
extract thickness oxide mat.occno=1 region="ngate" name="ngateox"
extract thickness oxide mat.occno=1 region="pgate" name="pgateox"


# Vt adjust Implant
implant boron dose=1.5e12 energy=25
# Poly Dep
deposit poly thick=0.30 divisions=5
# Poly Photo
mask name="POLY"
# Poly etch
etch poly dry thick=0.40
#photostrip
strip
# Poly is bare here, so name the gate electrodes.....
autoelectrode
# Poly Reox
method fermi compress
diffus time=23 temp=900 dryo2 press=1.00 hcl.pc=3
struct outfile=anex12_7.str
# Spacer Nitride Dep
deposit nitride thick=0.20 divisions=4
# Spacer etch
etch nitride dry thick=0.20
# SDN++
mask name="WELL"
# N++ Implant
```

```
implant arsenic dose=4.5e15 energy=50
# Nitride spacer removal for the N’s
etch nitride dry thick=1
# N-LDD Implant
implant phosphor dose=3.5e13 energy=30
struct outfile=anex12_8.str
strip
# P++ photo
mask name="WELL" reverse
# P++ Implant
implant boron dose=1.0e15 energy=25
# Nitride spacer removal for the P’s
etch nitride dry thick=1
# P-LDD Implant
implant bf2 dose=1.0e14 energy=40
struct outfile=anex12_9.str
strip
# Capping Nitride
deposit nitride thick=0.05 divisions=1

# BPSG Dense
method fermi compress
diffus time=20 temp=900 nitro press=1.00 hcl.pc=0
struct outfile=anex12_10.str

# Extract the two sd junction depths....
extract name="p++ xj" xj silicon mat.occno=1 region="pmcc" junc.occno=1
extract name="n++ xj" xj silicon mat.occno=1 region="nmcc" junc.occno=1

# MCC
mask name="VIA2"
# MCC etch
etch nitride dry thick=0.10
# MCC etch
etch oxide dry thick=0.10
strip
# Metal Depo
deposit alumin thick=0.05 divisions=2
# First Metal Defn
mask name="MET1"
# Metal1 etch
etch aluminum dry thick=0.50
strip
```

```
# Aluminium is bare here, so place the electrodes.....
autoelectrode
electrode backside name=substrate
structure outf=anex12_11.str

tonyplot

quit
```

32.1.13. anex13.in: Bipolar Example Using MaskViews

This example uses an IC Layout cross section, together with a generic process flow to realize a specific device.

To run this example, first invoke MASKVIEWS from the **Tools - MaskViews - Start MaskViews** options. Create a cutline object by selecting the **Write** file button, and then by selecting the beginning and end of the required cross-section. Load the cutline object into DECKBUILD by using the **Tools - MaskViews - Cutlines** menu option. Once loaded into DECKBUILD**,** press **run.**

To load and run this example, display this text and select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.16: Layout of Bipolar Transistor in MaskViews



Figure 32.17: Cross-Section of masks for active bipolar device including automated mesh generation

Input File athena/anex13.in:

```
go athena cutline=anex13.sec

#
init c.arsenic=2.0e16 orientation=100  space.mul=1

implant   boron  energy=18 dose=2.5e13

diffuse time=60 temp=920

# deposit polysilicon
deposit poly thick=0.3 divisions=6 min.space=0.05

# Implant to dope polysilicon
implant arsenic dose=7.5e15 energy=50

# Pattern the poly and name the electrode...
mask name="POLY"
etch poly dry thick=0.4
strip
autoelectrode

method compress fermi
diffuse time=25   temp=920 dryo2
```

```
diffuse time=50   temp=900 nitrogen

implant boron dose=2.5e13 energy=18

# deposit spacer
deposit oxide thick=0.4 divisions=10  min.space=0.1

# etch the spacer back
etch oxide dry thick=0.5

implant boron dose=1e15 energy=30

diffuse time=60 temp=900   nitrogen

# put down Al and etch to form contacts
deposit alum thick=0.05 div=2

mask name="FMD"
etch alumin dry thick=0.2
strip
# Autoname any bare electrodes...and the backside contact....
autoelectrode
electrode backside name="collector"

#  Save the final structure
struct outfile=anex13_0.str


# Completely remesh the structure without obtuse triangles in the semi-
  conductor
# Use the Sensitivity & Minspacing parameters to adjust the mesh densi-
  ty....
# ... the smaller the Sensitivity, the denser the mesh...
go DevEdit
BaseMesh Height=0.15 Width=0.15
BoundaryConditioning NoApply MaxRatio=200
ImpRefine MinSpacing=0.025
ImpRefine Type="net doping" Sensitivity=0.1
ConstrainMesh type=insulator maxangle=175
ConstrainMesh type=metal maxangle=180
mesh

structure outf=anex13_1.str

```

```

tonyplot


quit
```

32.1.14. anex14.in: Defining Phosphorus Models using the C Interpreter

This example demonstrates the use of the `C-Interpreter` for modifying the calculation of phosphorus diffusion coefficients, phosphorus segregation calculation, phosphorus activation calculation, and calculation of vacancy diffusion coefficients.

To load and run this example, display this text and select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 32.18: Comparison of default P diffusion model with a user specified model entered through the C-Interpreter



Figure 32.19: Comparison of default P segregation model with a user specified model entered through the C-Interpreter



Figure 32.20: Comparison of default P activation model with a user specified model entered through the C-Interpreter



Figure 32.21: Effect on doping profile of default vacancy diffusion model versus a user specified model entered through the C-Interpreter

Input File athena/anex14.in:

```
go athena
# This Input File demonstrates the use of the C interpreter
line x loc=0.00 spac=0.10
line x loc=0.10 spac=0.10
#
line y loc=0.00 spac=0.03
line y loc=2    spac=0.2
#
init silicon orientation=100
#
deposit oxide thick=0.10 divisions=5
implant phos dose=1.0e16 energy=55
structure out=anex14.str
#
# Standard phosphorus diffusion
#
```

```
diffuse time=20 temp=900

structure outfile=anex14_1.str

initialize infile=anex14.str
#
# Use the interpreter to modify the diffusion coefficient
# calculation for phosphorus
#
diffuse time=20 temp=900 p.dif.coef=anex14.lib

structure outfile=anex14_2.str

tonyplot -overlay anex14_1.str anex14_2.str -set anex14.set -ttitle
  "Phosphorus Diffusion Coefficient Calculation"
#
# Use the interpreter to modify the segregation calculation
# for phosphorus
#
initialize infile=anex14.str

diffus time=20 temp=900 p.seg.calc=anex14.lib

structure outfile=anex14_3.str

tonyplot -overlay anex14_1.str anex14_3.str -set anex14.set -ttitle
  "Phosphorus Segregation Calculation"
#
# Use the interpreter to modify the activation calculation
# for phosphorus
#
initialize infile=anex14.str

diffuse time=20 temp=900 p.act.calc=anex14.lib

structure outfile=anex14_4.str

tonyplot -overlay anex14_1.str anex14_4.str -set anex14.set -ttitle
  "Phosphorus Activation Calculation"
#
# Use the interpreter to modify the diffusion coefficient
# calculation for vacancies
#
initialize infile=anex14.str
```

```
method two.dim

diffuse time=20 temp=900

structure outfile=anex14_5.str

initialize infile=anex14.str

method two.dim

diffuse time=20 temp=900 v.dif.coef=anex14.lib

structure outfile=anex14_6.str

tonyplot -overlay anex14_5.str anex14_6.str -set anex14.set -ttitle "Va-
  cancy Diffusion Coefficient Calculation"

quit
```

This page is intentionally left blank

33.1. SSUPREM3: 1-D Process Simulation

33.1.1. s3ex01.in: NMOS Device: Gate and S/D Active Regions

Requires: SSUPREM3

This example simulates the fabrication sequence for an NMOS transistor with polysilicon gate. The example performs the initial front end processing first and then splits into back end processing for the Source/Drain region and for the channel region.

The graphics produced by this input file shows the initial structure, the structure in the Source/ Drain region and the channel structure.

Following the simulation of the Source/Drain region, several extractions are performed. Among them are **Junction Capacitance vs. Bias Curve** and **Junction Breakdown Voltage** estimated from Ionization Integral calculations.

Following the fabrication of the channel region, electrical analysis is performed using extraction capabilities built into DECKBUILD.

The electron conductivity versus gate bias curve is obtained using the extract capability. **Sheet conductance vs.** bias from the SSUPREM3 **Extract** menu is used. To obtain these extract statements, the **Sheet type** is set to **N**; the **Bias** is checked from **Define layers.** Then the bias is applied to the gate by choosing Polysilicon from the **material** menu. Then the **Bias** is set to 0 (default) and the bias is ramp from 0 V to 3 V by checking the **Ramped Bias** checkbox and choosing 0.2 V for the **Bias Step** and 3 V for the **Bias Stop.** Then click the **insert** and **write** buttons. The filename, **Nsheetcond.dat,** is chosen and the file produced by the extraction statements is plotted using the `-da` format of TONYPLOT.

The threshold voltage `vt_from_conc_curve` is extracted from the same curve using the `xintercept` and `maxslope` functions.

Finally, vt is extracted by an alternative method using **QUICKMOS 1D Vt** from the SSUPREM3 **extract** menu.

The last extraction in this example is **Sheet Resistance of the Polysilicon Layer** which is set by checking the **Poly semiconductor** box in the **sheet resistance** pop-up menu.

To load and execute this example, display this text and select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 33.1: Simulation of NMOS Channel: Layers and doping



Figure 33.2: Simulation of NMOS Channel: Doping and Layer boundaries



Figure 33.3: NMOS source/drain region with LDD



Figure 33.4: NMOS drain/bulk capacitance extraction (CJ in SPICE)



Figure 33.5: Final profile of NMOS channel showing all dopants



Figure 33.6: 1D threshold voltage extraction from SSUPREM3 doping

Input File ssuprem3/s3ex01.in:

```
go ssuprem3
```

```
TITLE:          NMOS Device: Gate and S/D Active Region
#               Active device region initial processing.

#               Initialize silicon substrate.
Initialize      <100> Silicon c.boron=1e15 Thick=4 \
                dX=.005 XdX=.02  Spaces=400

#               Grow pad oxide, 400A.
Diffusion       Temperature=1000  Time=40  DryO2

#               Deposit 800A of CVD Nitride.
Deposit         Nitride  Thickness=.0800  Spaces=15

#               Grow field oxide.
Diffusion       Temperature=1000  Time=180  WetO2

#               Etch to silicon surface.
Etch            Oxide all
Etch            Nitride all
Etch            Oxide all

#               Implant boron to shift the threshold voltage.
Implant         Boron  Dose=4e11  Energy=50

#               Grow gate oxide in dry oxygen ambient with 3% HCl.
Diffusion       Temperature=1050  Time=30  DryO2  HCL%=3

#               Deposit polysilicon
Deposit         Polysilicon  Thickness=0.5  Temperature=600

#               Heavily dope the polysilicon using POCl3
Diffusion       Temperature=1000  Time=25  dTmin=.3 \
                Phosphorus  Solidsolubility

#
#               Extract gate oxide thickness

extract name="gateox" thickness material="SiO~2" mat.occno=1

TonyPlot        -ttitle "s3ex01.in: Initial Active Region"

#               Save the structure at this point.  The simulation runs
#               are split for the gate and source/drain regions.
```

```
structure      outfile=s3ex01_1.str

#              Carry on processing the Source/drain regions.

#              Etch polysilicon and oxide over source/drain regions.
Etch           Polysilicon
Etch           Oxide

#              Implant Arsenic for source/drain regions.
Implant        Arsenic  Dose=5E15  Energy=150

#              Drive-in Arsenic and re-oxidize source/drain regions.
Diffusion      Temperature=1000  Time=30  DryO2

#              Etch contact holes to gate, source, and drain regions.
Etch           oxide

#              Deposit Phosphorus doped SiO2 using CVD.
Deposit        Oxide  Thickness=.7500  Phosphorus Concentration=1e21

#              Reflow glass to smooth surface and dope contact holes.
Diffusion      Temperature=1000  Time=30

#              Reopen contact holes.
Etch           Oxide

#              Deposit Aluminum.
Deposit        Aluminum  Thickness=1.2  Spaces=10

TonyPlot       -ttitle "s3ex01.in: Source/Drain Region"

#              Extract Integrated Active Arsenic concentration.

extract name="Active_Arsenic" 1.0e-04 * (area from curve(depth,impuri-
  ty="Active Arsenic" material="Silicon" mat.occno=1))

#              Extract and Plot Junction Capacitance.

extract start material="Silicon" mat.occno=1 bias=0.0 bias.step=0.25 bi-
  as.stop=5.0 region.occno=1
extract done name="cjcurve" curve(bias,1djunc.cap material="Silicon"
  mat.occno=1 region.occno=1 junc.occno=2) outfile="cj.dat"

Tonyplot -da cj.dat -mtitle "Junction Capacitance"
```

```
#               Extract Junction Breakdown Voltage.

extract start material="Silicon" mat.occno=1 bias=0.0 bias.step=0.25 bi-
  as.stop=30.0 region.occno=1
extract done name="jbv" x.val from curve(bias,n.ion material="Silicon"
  mat.occno=1 region.occno=1) where y.val=1.0

#               Save the structure.
structure       outfile=s3ex01_2.str

#               Process the gate region.

#               Initialize previously saved structure
Initialize      infile=s3ex01_1.str

#               Implant Arsenic for source/drain regions.
Implant         Arsenic  Dose=5E15  Energy=150

#               Drive-in Arsenic and re-oxidize source/drain regions.
Diffusion       Temperature=1000  Time=30  DryO2

#               Etch contact holes to gate, source, and drain regions.
Etch            Oxide

#               Deposit Phosphorus doped SiO2 using CVD.
Deposit         Oxide  Thickness=.75  Phosphorus Concentration=1.E21

#               Reflow glass to smooth surface and dope contact holes.
Diffusion       Temperature=1000  Time=30

#               Reopen contact holes.
Etch            Oxide

#               Deposit Aluminum.
Deposit         Aluminum  Thickness=1.2  Spaces=10

#              Extract integrated Chemical Boron concentration in Silicon

extract name="Chemical_Boron" 1.0e-04 * (area from curve(depth,impuri-
  ty="Boron" material="Silicon" mat.occno=1))

TonyPlot        -ttitle  "s3ex01.in: Gate Region"

```

```
#               Save the structure.
Structure       outfile=s3ex01_3.str

#               Extract and plot Conductance Vs. Bias Curve

extract start poly mat.occno=1 bias=0.0 bias.step=0.20 bias.stop=3.0
extract done name="nsh" curve(bias,1dn.conduct silicon mat.occno=1 re-
  gion.occno=1) outfile="Nsheetcond.dat"

tonyplot -da Nsheetcond.dat -mtitle "Sheet Conductance Curve"

#         Now  extract Vt from this curve as follows:

extract start poly mat.occno=1 bias=0.0 bias.step=0.20 bias.stop=3.0
extract done name="vt_from_cond_curve" xintercept(maxslope(curve(bi-
  as,1dn.conduct silicon mat.occno=1 region.occno=1)))

#               But there is much easier way:

extract name="1dvt" 1dvt ntype

#               Extract Sheet Resistance of Polysilicon layer

extract name="polyshro" sheet.res material="Polysilicon" mat.occno=1 re-
  gion.occno=1 semi.poly

quit
```

33.1.2. s3ex02.in: NMOS Device: Isolation Region

Requires: SSUPREM3

This example follows the same process sequence as that in s3ex01.in but applies the process sequence to model the isolation region. The example produces graphics using TONYPLOT that show the isolation region structure at two times in the process.

The **Integrated Boron Concentration in the Field Region** and **Field Threshold Voltage** are extracted. In order to properly extract the Field Threshold Voltage a **Bias Stop** of 20 Volts was set.

To load and execute this example, display this text and select the **Load example** button. This action will copy all associated file to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 33.7: NMOS Field/Isolation region: Final doping profile and layers

Input File ssuprem3/s3ex02.in:

```
go ssuprem3
TITLE:          NMOS Device: Isolation Region

#               Initialize silicon substrate.
Initialize      <100> Silicon c.Boron=1e15 Thick=4.0 dX=.01  Spaces=400

#               Grow pad oxide, 400A.
Diffusion       Temperature=1000  Time=40  DryO2

#               Implant boron to increase field region doping.
Implant         Boron  dose=1e13  energy=150

#               Grow field oxide.
Diffusion       Temperature=1000  Time=180  WetO2

#               Extract field oxide thickness
#
extract name="fieldox" thickness material="SiO~2" mat.occno=1

```

```
#               Implant boron to shift the enhancement threshold voltage.
Implant         Boron  Dose=4e11  Energy=50

#               Grow gate oxide
Diffusion       Temperature=1050  Time=30  DryO2  HCL%=3

#               Deposit polysilicon
Deposit         Polysilicon  Thickness=0.5  Temperature=600

#               Heavily dope the polysilicon using POCl3
Diffusion       Temperature=1000  Time=25  dTmin=.3 Phosphorus  Solidsol


TonyPlot        -ttitle "s3ex02.in: Initial Isolation Structure"

#               Final isolation region processing.

#               Etch polysilicon and oxide over source/drain regions.
Etch            Polysilicon
Etch            Oxide  Amount=.0700

#               Implant Arsenic for source/drain regions.
Implant         Arsenic  Dose=5E15  Energy=150

#               Drive-in Arsenic and re-oxidize source/drain regions.
Diffusion       Temperature=1000  Time=30  DryO2

#               Deposit Phosphorus doped SiO2 using CVD.
Deposit         Oxide  Thickness=.7500  Phosphorus Concentration=1e21

#               Reflow glass to smooth surface and dope contact holes.
Diffusion       Temperature=1000  Time=30

#               Deposit Aluminum.
Deposit         Aluminum  Thickness=1.2  Spaces=10

#              Extract integrated Boron concentration in the field region
extract name="Chemical_Boron" 1.0e-04 * (area from curve(depth,impuri-
  ty="Boron" material="Silicon" mat.occno=1))
#               Extract Field Threshold Voltage
extract name="Vt" 1dvt ntype bias=0.0 bias.step=0.5 bias.stop=60

TonyPlot        -ttitle "s3ex02.in: Final Isolation Structure"
```

```

#               Save the structure.
Structure       outfile=s3ex02.str
#
quit
```

33.1.3. s3ex03.in: CMOS Device: Generic Process Driven by IC Layout

Requires: SSUPREM3/MASKVIEWS

This example demonstrates how a generic process flow description and an IC layout description may be brought together to simulate a specific device.

This example, if loaded, will copy the following files into your current working directory:

s3ex03.in (Input file)

s3ex03.lay (layout file)

Select the example on the scrolling list and press load.

Once loaded, a generic process flow input file is displayed in the DECKBUILD main editing window.

Start MASKVIEWS next by selecting the: **Tools -> Maskviews -> Start Maskviews** menu option. This will invoke the file selector popup window for MASKVIEWS. Select s3ex03.lay and press “Start MaskViews”.

A few seconds later MASKVIEWS will appear with an IC layout loaded into its main screen.

Click on the **Properties...** button and then select SSUPREM3 from the **simulator** menu.

Click the left mouse button once only on the **Write File** button.

Move the pointer to the position on the IC layout you wish to simulate and click at this point.

Having completed the areal selection, MASKVIEWS will display the name of the section filename created. When this is confirmed by clicking the left mouse button, a summary of the masks that will be used will appear.

Make a note of the cutline section filename since you will need it later.

The default filename will be, **default.sec.0.** You may change this to a more meaningful name if you wish.

From DeckBuild now load the cutfile by selecting the menu option: **Tools -> MaskViews -> Cut files**

A list of cutlines will be displayed in a list. (Only one will be displayed in the first instance).

Select a cutline filename on the scrolling list so that the displayed filename becomes depressed. Press the Load button to load the cutline information into DECKBUILD.

Press **Run** on the DECKBUILD runtime control panel. Mask information will be added automatically to the runtime window.

The layout file **s3ex03.lay,** employs photoresist as the masking material. Since SSUPREM3 does not recognize the material name “barrier”, you must always be sure to use “photoresist” as masking material.



Figure 33.8: NMOS channel simulation defined by specifying simulation point in a CMOS Layout in MaskViews

Input File ssuprem3/s3ex03.in:

```

#SSUPREM3 MaskViews Example

#### This is a generic 1-D Input File describing a CMOS process flow  #####
####  Use this Input File only with deckbuild and with MaskViews      #####


go ssuprem3
#
init silicon orient=100 thick=4.00 c.phosphor=1.0e14

# Initial Oxide
diffus time=30 temp=1000 dryo2 press=1.00 hcl=3

# Initial Nitride
deposit nitride thick=0.15 space=15

# Pwell photo
```

```
mask name="WELL"

# Pwell definition etch
etch nitride  thick=0.15

# Oxide overetch
etch oxide  thick=0.015

# pwell implant
implant boron dose=1.0e13 energy=100


#photo strip
strip

#save the pwell structure for inspection later
struct outfile=s3ex03_pwell.str

# Nwell maox
diffus time=45 temp=950 weto2 press=1.00 hcl=3

# strip init nit
etch nitride all

# oxide overetch
etch oxide  thick=0.015

# nwell implant
implant phosphor dose=1.0e13 energy=60

# planarising oxide
diffus time=45 temp=950 dryo2 press=0.05 hcl=3

# planarising oxide
#
diffus time=45 temp=1000 weto2 press=0.05 hcl=3

# well drive
#
diffus time=170 temp=1200 nitro press=0.05
#
struct outfile=s3ex03_welldrive.str

```

```
# wellox strip
etch oxide all

# Stress relief Oxide
deposit oxide thick=0.035 spaces=15

# Masking Nitride
deposit nitride thick=0.15 spaces=15

# Active Area Photo
mask name="AAD"
# AA Definition
etch nitride  thick=0.30
# AA Definition
etch oxide  thick=0.10
# AA Definition
etch silicon  thick=0.15
#
struct outfile=s3ex03_aad.str
#
#photostrip
strip

# Field Oxide growth
diffus time=85 temp=1000 weto2 press=1.00 hcl=3
struct outfile=s3ex03_field.str

# AA stack strip
etch oxide thic=0.02
etch nitride  thick=0.25
etch oxide  thick=0.07
etch nitride all
#
struct outfile=s3ex03_aa.str

# Sacox growth
#
diffus time=25 temp=950 dryo2 press=1.00 hcl=3
extract thickness oxide mat.occno=1

# Sacox strip
etch oxide  thick=0.06
```

```
# Gateoxide growth
diffus time=23 temp=950 dryo2 press=1.00 hcl=3
# Gateoxide anneal
diffus time=40 temp=950 nitro press=1.00
struct outfile=s3ex03_gate.str


# Vt adjust Implant
implant boron dose=1.5e12 energy=25

# Poly Dep
deposit poly thick=0.30 spaces=30 temp=600


# Poly Photo
mask name="POLY"

# Poly etch
etch poly  thick=0.40
strip

# Poly Reox
diffus time=23 temp=900 dryo2 press=1.00 hcl.pc=3

# Spacer Nitride Dep
deposit nitride thick=0.20 spaces=20

etch nitride thick=0.20

# SDN++
mask name="WELL"

# N++ Implant
implant arsenic dose=4.5e15 energy=70

# Nitride spacer removal for the N's
etch nitride thick=1

# N++ Implant
implant phosphor dose=3.5e13 energy=40

#photostrip
strip
```

```
# P++ photo
mask name="WELL" reverse

# P++ Implant
implant boron dose=1.0e15 energy=25 pearson

# Nitride spacer removal for the P’s
etch nitride thick=1

# P-LDD Implant
implant bf2 dose=1.0e14 energy=40 pearson

# Photo-strip
strip

# Capping Nitride
deposit nitride thick=0.05 space=10

# BPSG dense
diffus time=20 temp=900 nitro press=1.00

# MCC
mask name="VIA2"

# MCC etch
etch nitride  thick=0.10

# MCC etch
etch oxide  thick=0.10

#photostrip
strip

# Metal Depo
deposit alumin thick=0.8 spaces=80

# First Metal Definition
mask name="MET1"

# Metal1 etch
etch aluminum  thick=0.90
```

```
#photostrip
strip

#
struct outfile=s3ex03_sd.str

tonyplot -ttitle "s3ex03.in: SSUPREM3 - MaskViews Example"

quit

```

33.1.4. s3ex04.in: Bipolar Poly Emitter Device: Active Region

Requires: SSUPREM3

This example forms a bipolar structure. Following Antimony Buried Layer formation, the structure is plotted using TONYPLOT. TONYPLOT is used again to plot the structure after epitaxial growth and after the final emitter drive-in step.

Following complete structure formation, several process and device parameters are extracted. These include emitter and base junction depths, base width, electron distribution at 0 V bias, electric field and a number of Gummel-Poon parameters extracted by using the built-in 1-D Bipolar solver **QuickBip.**

To load and execute this example, display this text and select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 33.9: NPN Bipolar device with poly emitter: Layer structure and Dopin

Input File ssuprem3/s3ex04.in:

```
go ssuprem3
TITLE:       Bipolar Poly Emitter Device: Active Region.

#            Initialize the silicon substrate.
Initialize   <100> Silicon  c.Boron=5e14 Thick=5. \
             dX=.01 XdX=.05 Spaces=100

#            Grow masking oxide for non-active regions.
Diffusion    Temperature=1150  Time=100  WetO2

#            Etch the oxide over the buried layer regions.
Etch         Oxide

#            Implant and drive-in the antimony buried layer.
Implant      Antimony  Dose=5E14  Energy=120
Diffusion    Temperature=1150  Time=15  DryO2
Diffusion    Temperature=1150  Time=300

TonyPlot     -ttitle "s3ex04.in: Antimony Buried Layer"

#            Etch off the oxide.
Etch         Oxide

#            Grow 1.6 micron of arsenic doped epi.
Epitaxy      Temperature=1050 Time=4 Growth.Rate=.4 \
             Arsenic Gas.Conc=5E15

#            Grow a 400A pad oxide.
Diffusion    Temperature=1060  Time=20  DryO2

#            Deposit nitride to mask the field oxidation.
Deposit      Nitride  Thickness=.08

#            Field oxide growth.  Oxidation is masked by nitride.
Diffusion    Temperature=800   Time=30         t.rate=10
Diffusion    Temperature=1000  Time=15  DryO2
Diffusion    Temperature=1100  Time=210 WetO2
Diffusion    Temperature=1100  Time=15  DryO2
Diffusion    Temperature=1100  Time=10         t.rate=-30

TonyPlot     -ttitle "s3ex04.in: After Epitaxial Growth"

```

```
#           Etch the oxide and nitride layers.
Etch        Oxide
Etch        Nitride
Etch        Oxide

#           Move the fine grid to the surface.
Grid        Layer.1  Xdx=0.

#           Implant the boron base.
Implant     Boron  Dose=1E14  Energy=50

#           Remove oxide from emitter region.
Etch        Oxide

#           Deposit arsenic doped polysilicon for emitter contacts.
Deposit     Polysilicon  Thickness=.5  Temp=620 c.Arsenic=1e20

#           Anneal to activate base and drive-in emitter.
Diffusion   Temperature=1000  Time=20  WetO2

TonyPlot    -ttitle "s3ex04.in: Final Bipolar Active Region"

#           Save the resulting active region.
Structure   outfile=s3ex04.str

######################################################################
# Now extract basic parameters using Deckbuild Extract command \
  These extract lines may be used as optimization targets, \
  or used for logging extracted parameters to the VWF....

#    Extract emitter junction
extract name="emitter_xj" xj material="Silicon" mat.occno=1 junc.occno=1

#    Extract base junction
extract name="base_xj" xj material="Silicon" mat.occno=1 junc.occno=2

#           Extract base width
extract name="base_width" $base_xj - $emitter_xj

#    Extract electron concentration
extract name="electrons" curve(depth,n.conc vg=0.0 vb=0.0 silicon mat.oc-
  cno=1) outfile="elec.dat"

```

```
#           Extract electric field
extract name="field" curve(depth,efield vg=0.0 vb=0.0 silicon mat.occno=1) outfile="field.dat"
######################################################################

############ BIP Gummel-Poone Parameters #######################

extract name="bip test bf" bf
extract name="bip test nf" nf
extract name="bip test is" gpis
extract name="bip test ne" ne
extract name="bip test ise" ise
extract name="bip test cje" cje
extract name="bip test vje" vje
extract name="bip test mje" mje
extract name="bip test rb" rb
extract name="bip test rbm" rbm
extract name="bip test irb" irb
extract name="bip test tf" tf
extract name="bip test cjc" cjc
extract name="bip test vjc" vjc
extract name="bip test mjc" mjc
extract name="bip test ikf" ikf
extract name="bip test ikr" ikr
extract name="bip test nr" nr
extract name="bip test br" br
extract name="bip test isc" isc
extract name="bip test nc" nc
extract name="bip test tr" tr



quit
```

33.1.5. s3ex05.in: Bipolar Poly Emitter Device: Isolation Region

Requires: SSUPREM3

This example simulates the isolation region formation for the bipolar example of **s3ex04.in.** TONYPLOT produces a graphical display of the structure at various points in the process.

To load and execute this example, display this text and select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 33.10: Isolation region of a bipolar device

Input File ssuprem3/s3ex05.in:

```
go ssuprem3
TITLE:          Bipolar Poly Emitter Device: Isolation Region.

#           Initialize the silicon substrate.
Initialize  <100> Silicon c.Boron=5E14 Thickness=6  dX=.03  Spaces=150

#           Grow masking oxide for the non-active regions.
Diffusion   Temperature=1150  Time=100  WetO2

#           Implant and drive in the antimony buried layer.
Implant     Antimony  Dose=1E15  Energy=80
Diffusion   Temperature=1150  Time=15   DryO2
Diffusion   Temperature=1150  Time=300


#           Etch off the oxide.
Etch        Oxide

#           Add 1.6 microns of arsenic doped epi.
```

```
Epitaxy     Temperature=1050 Time=4 Growth.R=.4 Arsenic Gas.Conc=5E15

#           Grow a 400A pad oxide.
Diffusion   Temperature=1060  Time=20  DryO2

#           Deposit a 800A layer of silicon-nitride.
Deposit     Nitride  Thickness=.08

TonyPlot    -ttitle "s3ex05.in: After Epitaxial Growth"

#           Etch the nitride and oxide layers.
Etch        Nitride
Etch        Oxide

#           Etch half the silicon epi layer.
Etch        Silicon  Amount=.8

#           Implant boron in the field region.
Implant     Boron  Dose=2E13  Energy=100

#           Grow the field oxide.
Diffusion   Temperature=800   Time=30          T.Rate=10
Diffusion   Temperature=1100  Time=15   DryO2
Diffusion   Temperature=1100  Time=210  WetO2
Diffusion   Temperature=1100  Time=15   DryO2
Diffusion   Temperature=1100  Time=10          T.Rate=-30

#           Implant the boron base.
Implant     Boron  Dose=1E14  Energy=80

#           Deposit arsenic doped polysilicon for the emitter contact.
Deposit     Polysilicon,  Thickness=.5  Temperature=620. \
            Arsenic  Concentration=1E20

#           Remove the polysilicon.
Etch        Polysilicon

#           Anneal to activate base and emitter regions.
Diffusion   Temperature=1000  Time=20  WetO2

TonyPlot    -ttitle "s3ex04.in: Final Structure"

#           Save the initial part of the isolation simulation.
```

```
Structure   outfile=s3ex05.str

#########################################################
## Now for some Deckbuild extraction, used for \
   optimization targets and VWF logging...
#
extract name="final oxide thickness" thickness oxide mat.occno=1
#
extract name="max boron conc" max.conc boron silicon mat.occno=1
#
extract name="arsen surf conc" surf.conc silicon mat.occno=1 arsen
#
#########################################################

quit
```

33.1.6. s3ex06.in: Spreading Resistance Profile (SRP) Analysis

Requires: SSUPREM3

This example illustrates SSUPREM3's capability to perform a complete Spreading Resistance Profiling (SRP) analysis. SRP profiling is known to produce an inaccurate description of impurity profiles that is especially pronounced for lightly doped junctions. The SRP analysis of SSUPREM3's capabilities gives a quick and easy way to determine how much of a discrepancy can be expected from SRP analysis. The example begins by forming a silicon substrate with some impurities present.

The SRP analysis can be easily performed using the `sprresan` command. For this example, we select to perform 120 steps in the SRP analysis with each step representing a thickness of 0.05 microns of silicon. The results of the SRP analysis shows the profile that would be predicted by SRP analysis of the impurity profile present in the SSUPREM3 calculation.

Then the same results are obtained using DeckBuild's `extract` capability. The SIMS (Chemical) and SRP net profiles are extracted, saved in the simple data format and plotted using the `-overlay` capability of TonyPlot.

To load and execute this example, display this text and select the **Load example** button. This action will copy all associated files to your current working directory. Select the DeckBuild **run** button to execute the example.



Figure 33.11: Comparison of SRP results versus the actual doping profile. Offset is due to measuring carrier concentration in SRP

Input ssuprem3/s3ex06.in:

```
go ssuprem3
TITLE:         Spreading Resistance Analysis (SRP) Analysis

#              Initialize silicon substrate.
Initialize     <100> Silicon c.phos=1e15 Thick=9.5 dX=.05 XdX=.02 Spac-
  es=450

Deposit        oxide thick=0.1

#              Produce simple structure with a junction

Implant        Boron  Dose=2e13  Energy=150 am
Diffusion      Temperature=1200  Time=120

Etch           oxide all

#              Perform SRP analysis using internal SSUPREM3 capability
Sprresan       sprstep=120 sprth=0.05

```

```
structure outfile=s3ex06.str
Tonyplot -s s3ex06_1.set -ttitle "s3ex06.in: SSuprem3 SRP Analysis"

# Now the same comparison using extract capability

extract name="sims" curve(depth,net silicon mat.occno=1) out-
  file="sims.dat"
extract name="srp" curve(depth,srp silicon mat.occno=1) outfile="srp.dat"

tonyplot -da -overlay sims.dat srp.dat -set s3ex06_2.set -ttitle
  "s3ex06.in: SRP Analysis"

quit
```

33.1.7. s3ex07.in: Implant Profile Variation with Tilt Angle

Requires: SSUPREM3

This example illustrates the implant statistics used to model tilted implantation. These are based on the work of Tasch, et al. (see manual for references) and uses the **Dual Pearson Model** to accurately represent a range of tilt angles, including zero degrees (on axis) implantation.

Simulated 35 keV boron profiles for tilts from zero to ten degrees are plotted using the **overlay** capability of TONYPLOT.

To load and execute this example, display this text and select the **Load example** button. This action will copy all associated files in your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 33.12: Variation in Boron implant range with tilt angle. Substantial channeling is seen for zero degree implants

Input File ssuprem3/s3ex07.in:

```
go ssuprem3
#
# This implant example uses data calibrated from Al Tasch's experimental
# data, generated at the University of Texas at Austin....
#
TITLE:         Implant Profile Variation with Tilt Angle

#              Initialize silicon substrate.
Initialize    <100> Silicon c.Boron=1e15 Thick=0.8 dX=.005 XdX=.02 Spac-
  es=150

Implant        Boron Dose=5e14 Energy=35 Tilt=0
struct         outfile=s3ex07_00.str

Initialize    <100> Silicon c.Boron=1e15 Thick=0.8 dX=.005 XdX=.02 Spac-
  es=150
Implant        Boron Dose=5e14 Energy=35 Twist=0 Tilt=2

struct         outfile=s3ex07_02.str

Initialize    <100> Silicon c.Boron=1e15 Thick=0.8 dX=.005 XdX=.02 Spac-
  es=150
```

```
Implant         Boron Dose=5e14 Energy=35 Twist=0 Tilt=4
struct          outfile=s3ex07_04.str
Initialize     <100> Silicon c.Boron=1e15 Thick=0.8 dX=.005 XdX=.02 Spac-
  es=150
Implant         Boron Dose=5e14 Energy=35 Twist=0 Tilt=6
struct          outfile=s3ex07_06.str
Initialize     <100> Silicon c.Boron=1e15 Thick=0.8 dX=.005 XdX=.02 Spac-
  es=150
Implant         Boron Dose=5e14 Energy=35 Twist=0 Tilt=8
struct          outfile=s3ex07_08.str
Initialize     <100> Silicon c.Boron=1e15 Thick=0.8 dX=.005 XdX=.02 Spac-
  es=150
Implant         Boron Dose=5e14 Energy=35 Twist=0 Tilt=10
struct          outfile=s3ex07_10.str
tonyplot -overlay s3ex07_*.str -set s3ex07.set  -ttitle "s3ex07.in: Tilted
  Ion Implant"
quit
```

33.1.8. s3ex08.in: Monte Carlo Implant into Multilayer Structure

Requires: SSUPREM3

The Monte Carlo ion implant capability included with SSUPREM3 is the fastest calculation of its kind in the world. This example illustrates the Monte Carlo based ion implant calculation of SSUPREM3. The final plot clearly demonstrates concentration profile discontinuity at the nitride/ silicon interface. SSUPREM3 also outputs estimated projected range, straggling, skewness, and kurtosis of this implant.

Following the calculation, in order to check the dose conservation law, the integrated boron dose is calculated using the **area from curve** capability of DECKBUILD's `Extract`.

To load and execute this example, display this text and select the **Load example** button. This action will copy all associated file to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 33.13: Boron Implant profile using the Monte Carlo technique

Input File ssuprem3/s3ex08.in:

```
go ssuprem3
TITLE:          Monte Carlo Implant into Multilayer Structure

#               Initialize silicon substrate.
Initialize      <100> Silicon c.phos=1e14 Thickness=0.4 Spaces=300

#               Deposit cap material
Deposit         nitride thick=0.1

#               Implant boron
Implant         boron  Dose=1e14  Energy=70  mc ionnum=1000

#               Check boron integrated dose
#
extract name="Boron in Nitride" area from curve(depth,impurity="Boron"
  material="Si~3N~4" mat.occno=1)
#
extract name="Boron in Silicon" area from curve(depth,impurity="Boron"
  material="Silicon" mat.occno=1)

extract name="Total Boron" $"Boron in Nitride" + $"Boron in Silicon"

```

```
structure       outfile=s3ex08.str
TonyPlot        -set s3ex08.set -ttitle "s3ex08.in: Monte Carlo Boron Im-
  plant"

quit
```

33.1.9. s3ex09.in: Monte Carlo Implant with Damage

The Monte Carlo ion implant capability of SSUPREM3 includes a damage calculation that predicts interstitial and vacancy profiles that can be important in the prediction of post-implant diffusion.

This example uses the Monte Carlo ion implant model to simulate boron implantation at 350 keV. During calculation, a status report is printed to the screen periodically. Total boron implanted dosage, as well as the integrated number of interstitials produced during this implant, are calculated using the **area from curve** capability of DECKBUILD's `Extract`. Then the average number of interstitials produced per one implanted boron ion is calculated. Finally, a plot showing the implanted boron profile and the interstitials formed by the implant is produced by TONYPLOT.

To load and execute this example, display this text and select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 33.14: Monte Carlo implant can be used to predict implant damage. SSUPREM3 differs from ATHENA in that implant damage is not used in the diffusion models.

Input File ssuprem3/s3ex09.in:

```
go ssuprem3
TITLE:          Monte Carlo Implant with Damage

```

```
#               Initialize silicon substrate.
Initialize      <100> Silicon Thickn=1.6 Space=500

#               Monte Carlo Implant

Implant         boron Dose=1e13 Energy=350 ion=3000 mc damage temp=100

#               Extract integrated number of interstitials
extract name="Interstitials" area from curve(depth,impurity="Intersti-
  tials" material="Silicon" mat.occno=1)

#                Extract implanted Boron dose
extract name="Boron" area from curve(depth,impurity="Boron" materi-
  al="Silicon" mat.occno=1)

#                Extract their ratio
extract name="Ratio"  $Interstitials/$Boron

structure       outfile=s3ex09.str
TonyPlot       -set s3ex09.set -ttitle "Monte Carlo Implant with Damage"

quit
```

33.1.10. s3ex10.in: Aluminum Impurity Implant and Diffusion

Requires: SSUPREM3

This example demonstrates the use of SSUPREM3 to simulate aluminum implantation and diffusion. Aluminum has been added to the impurity set in versions 5.5 or greater. This impurity is accessible under the name `I.aluminum` to distinguish the aluminum as an impurity from aluminum as a material. The synonym, `aluminum`, can be used in those statements which do not contain the parameter, `aluminum`, as a material notation. Aluminum is usually used as an acceptor impurity for formation of deep junctions in power devices. Typical aluminum diffusion for 24 hours at 1250 degrees is shown here.

To load and run this example, display this text and select the **Load example** button. This action will copy all associated files to your current working directory. Select the DECKBUILD **run** button to execute the example.



Figure 33.15: Demonstration of Aluminum Implant and Diffusion

Input File ssuprem3/s3ex10.in:

```
go ssuprem3

TITLE       Aluminum Impurity Implant and Diffusion

# This example illustrates the use of aluminum as a dopant

#           Initialize deep silicon substrate
Init        silicon orient=100 thick=120.0 c.phos=2.0e13 spaces=500

#           Deposit nitride cap
Deposit     nitride thick=0.1

#           Aluminum implant
Implant     alumin dose=1.0e15 energy=150

#           24 hours drive-in
Diffus      time=1320 temp=1250

#           Remove the cap
Etch        nitride all
structure outfile=s3ex10.str
```

```

tonyplot -set s3ex10.set -ttitle "Example s3ex10.in: Deep Aluminum Diffu-
  sion"

quit
```

33.1.11. s3ex11.in: Gallium Impurity Diffusion

Requires: SSUPREM3

This example demonstrates the use of SSUPREM3 to simulate gallium diffusion. Gallium has been added to the impurity set in versions 5.5 or greater.

Gallium is usually used as an acceptor impurity for formation of deep junctions in power devices. Typical gallium diffusion for 15 hours at 1200 degrees is shown here. Also, take note that gallium has quite a high diffusivity in oxide.

Since gallium does not belong to the standard set of impurities in Si, the junction depth cannot be extracted using the standard **extract xj** capability. However, as is shown in the example below, it can be done using the extract, **x.val from curve where y.val=<n>** capability.

To load and run this example, select the **Load example** button at the top of this window, and press the **run** button on DECKBUILD to execute the example.



Figure 33.16: Demonstration of Gallium Implant and Diffusion

Input File ssuprem3/s3ex11.in:

```
go ssuprem3
# This example shows the use of gallium as an impurity
```

```
init silicon orient=111 thick=40.0 c.phosphor=8.0e13 spaces=200
#
# grow some oxide
diffus time=180 temp=1175 wet

# perform Gallium surface diffusion
diffus time=900 temp=1200 c.gallium=1e18
#
extract name="Integrated Gallium" 1.0e-04 * (area from curve(depth,impu-
  rity="Gallium" material="Silicon" mat.occno=1))

etch oxide all

structure outfile=s3ex11.str


#       Extract junction depth
extract name="Junction Depth" x.val from curve(depth,(impurity="Gallium"
  material="Silicon" mat.occno=1)-(impurity="Phosphorus" material="Sili-
  con" mat.occno=1)) where y.val=0.0

tonyplot -set s3ex11.set -ttitle "Example s3ex11.in: Gallium Impurity"

quit
```

33.1.12. s3ex12.in: User-Defined Impurity Implant and Diffusion

Requires: SSUPREM3

This example demonstrates the use of SSUPREM3 to define and perform simulations that incorporate a user-defined impurity.

First, the name, atomic number, atomic weight and electronic stopping coefficients of the impurity are set. These parameters are needed to perform Monte Carlo or Boltzmann implant simulations. Next, diffusion, segregation and solubility coefficients are specified. Finally, implant and diffusion process steps are performed.

To load and run this example, select the **Load example** button at the top of this window, and press the **run** button on DECKBUILD to execute this example.



Figure 33.17: Demonstration of a user-defined impurity with implant ranges and diffusion coefficients

Input File ssuprem3/s3ex12.in:

```
go ssuprem3

# User defined impurity example

initialize silicon <100> thick=1.0

#              Specify new impurity name and type
imp.5          silicon name=indium acceptor

#              Specify parameters need for ion implantation
imp.5          silicon at.wt=114.82 at.number=49 elect.stop=30.02

#              Specify diffusion coefficients in silicon
imp.5          silicon dix.0=1e8 dix.e=3.2 dip.0=0.0 dip.e=0.0
imp.5          silicon fii.0=4.0  fii.e=0.40

#              Specify diffusion coefficients in oxide
imp.5          oxide   dix.0=0.0 dix.e=0.0 dip.0=0.0 dip.e=0.0
imp.5          oxide   fii.0=4.0  fii.e=0.40  elect.stop=30.02

#              Specify  segregation coefficients
```

```
segregation   imp.5 silicon /oxide seg.0=20.0    seg.e=0.0
segregation   imp.5 silicon /oxide trans.0=0.1 trans.e=0.0

#             Specify solid solubility
solubility    imp.5 silicon polysilicon ssol.0=1e22 ssol.e=0.0

#             Perform Monte Carlo implant
implant       imp.5 dose=1e14 energy=500 mc

#             Save structure for subsequent plotting
structure     outfile=s3ex12_0.str

#             Now perform diffusion
diffusion     temp=1050 time=100

structure      outfile=s3ex12_1.str

tonyplot -st -overlay s3ex12_*.str -set s3ex12.set -ttitle "s3ex12.in:
  User-defined Impurity"

quit
```

33.1.13. s3ex13.in: Making a User-Defined Material

Requires: SSUPREM3

This example demonstrates the use of SSUPREM3 to define and perform simulations that incorporate a user defined material.

A user-defined insulator material, Oxinitride, is specified in the example. It consists of three elements: Si, O, and N. Abundances, atomic numbers and atomic weights for all three elements are specified. Diffusion, segregation and solubility parameters of phosphorus in the new material are specified also. This should be done for all impurities involved in the simulation.

Ion implantation into the user-defined material cannot be simulated using the default Pearson method unless range parameters are specified at the implant line because the look-up tables do not exist for this material. Another alternative used in the example is the Monte Carlo implant model. The Boltzmann implant model can be used here as well.

At the end of the simulation, TONYPLOT compares phosphorus profiles before and after final diffusion.

To load and run this example, select the **Load example** button at the top of this window, and press the **run** button on DECKBUILD to execute this example.



Figure 33.18: Demonstration of a user-defined material with implant ranges and diffusion coefficients

Input File ssuprem3/s3ex13.in:

```
go ssuprem3

# User defined material example

initialize silicon <100> thick=0.5

material index=7 name=Oxinitride insulator density=2.84 \
abund.1=0.381 abund.2=0.333 abund.3=0.286  \
at.num.1=14 at.num.2=8 at.num.3=7  at.wt.1=28.086 \
at.wt.2=16 at.wt.3=14  dx.default=0.005

solubility phosphor mat.7 ssol.0=1e21 ssol.e=0.0
phosphor            mat.7 dix.0=0.0   dix.e=0.0
segreg  phosphor    sili /mat.7 seg.0=1.0 seg.e=0.0 trans.0=0.1
  trans.e=0.0

deposit  mat.7 thick=0.1

implant phos energy=80 dose=1e14 mc

```

```
structure outfile=s3ex13_0.str

diff time=20 temp=1000

structure outfile=s3ex13_1.str

tonyplot -st -overlay s3ex13_*.str -set s3ex13.set -ttitle "s3ex13.in:
  User-defined Material"

quit
```

33.1.14. s3ex14.in: Optimization of Bipolar Structure

Requires: SSUPREM3/OPTIMIZER

This example demonstrates the practical use of DeckBuild's OPTIMIZER for optimization of a bipolar process.

If one runs this process with all the parameters as specified in the input file, the final structure will have both a very thin (~0.06 microns) and very low concentration (~1e16) base. The goal of this example is to show how to use OPTIMIZER in order to improve the characteristics of this device. Base implant dose and emitter drive-in time are selected as the process optimization variable parameters. The target parameters are the base sheet resistance and base width extracted at the end of the input file.

To load and execute this example, display this text and select the **Load example** button. Once loaded, select **Optimizer...** from the **Main Control** menu. The **DECKBUILD: Optimizer** screen will appear momentarily. Select **Load File...** from the **File** menu, then load the special optimizer file, **s3ex14.in.opt**, from the scrolling **Optimizer Load** menu.

Preset parameters and targets are included into this file. When **parameters** is selected from the **Mode** menu, you will find that **Implant Dose** on the Line number 22 and **Diffusion Time** on the Line number 36 are chosen as parameters for optimization. If the optimizer file did not exist, these parameters could be added as described in the section **Adding a Parameter** in the OPTIMIZER Chapter of the VWF INTERACTIVE TOOLS Manual.

The target value for the base sheet resistance `bshro` is set to 150 Ohm/square, while the `base width` is set to 0.15 microns. This can be checked by selecting **Targets** from the **Mode** menu. The targets and their values can be changed as well (see section **Targets** in the **Optimizer** chapter of VWF INTERACTIVE TOOLS Manual).

To obtain an accuracy better than the default, the user should select **Setup** from the **Mode** menu and decrease the **Maximum Error** to 1.0%.

To start the optimization process, select the OPTIMIZE button. During the optimization process, it may be monitored by selecting either **Results** or **Graphics** from the **Mode** menu.



Figure 33.19: Final doping profile structure from the optimization

Input File ssuprem3/s3ex14.in:

```
go ssuprem3

Initialize  <100> Silicon  c.Boron=5e14 Thick=5. \
            dX=.01 XdX=.05 Spaces=100


#           Implant and drive-in the antimony buried layer.
Implant     Antimony  Dose=5E14  Energy=120
Diffusion   Temperature=1150  Time=15  DryO2
Diffusion   Temperature=1150  Time=300

Etch        Oxide all

#           Epi-layer formation
Epitaxy     Temperature=1050 Time=4 Growth.Rate=.4 \
            Arsenic Gas.Conc=5E15

#           Base screen oxide
diffuse     time=20 temp=950 wet
```

```

#           Base implant
Implant     Boron  Dose=8E13  Energy=40 am

#           Base drive
diffuse     time=30 temp=950

etch oxide all

#           Polysilicon deposition
Deposit     Polysilicon  Thickness=.5  Temp=620

#           Polyemitter implant
implant     phos energy=40 dose=5e15

#           emitter drive-in
Diffusion   Temperature=1000  Time=40  WetO2

#           Extract base sheet resistance
extract name="bshro" sheet.res material="Silicon" mat.occno=1 region.occno=1

#           Extract basewidth
extract name="Xj1" xj silicon mat.occno=1 junc.occno=1
#
extract name="Xj2" xj silicon mat.occno=1 junc.occno=2
#
extract name="basewidth" $Xj2 - $Xj1

Structure outfile=s3ex14.str

tonyplot -set s3ex14.set -ttitle “Example s3ex14.in: Optimization”
```

This page is intentionally left blank